```
> restart;
> Digits := 30: interface( displayprecision=5 ):
> dU := `&Delta;U`: dT := `&Delta;T`: dH := `&Delta;H`:
> parameters := {}:
> parameter := proc( x ) global parameters; parameters :=
  parameters union {x}; end:
```

Aufgabe

In einem Autoreifen, dessen Volumen konstant mit 0,025 m³ angenommen werden soll, befindet sich Luft mit 18 °C bei einem Überdruck von 157 kPa. Der Luft wird durch Sonnenbestrahlung die Wärme 2,55 kJ zugeführt. Der Barometerstand beträgt 99 kPa.
Die Luft soll als ideales Gas angenommen werden. Für ihre spezifische Wärmekapazität soll näherungsweise der Wert bei 0° C werden.

a) Welche Lufttemperatur stellt sich im Reifen ein?
b) Welcher Reifenüberdruck tritt auf?
c) Wie ändert sich innere Energie und Enthalpie des Reifeninhalts?
d) Welche Luftmasse ist abzulassen, damit sich der ursprüngliche Reifendruck wieder einstellt (die Temperatur bleibt hierbei unverändert)?

Rechnung

Das Volumen V des Gases ist konstant laut Aufgabenstellung.

```
> V = 0.025*Unit(m^3);
  parameter(%):
```

$$V = 0.025 \; [[m^3]] \tag{1}$$

Die Temperatur der Luft am Anfang (Zustand 1) muss in Kelvin umgerechnet werden.

```
> T[1] = 18 * Unit(Celsius);
  lhs(%) = evalf( convert( rhs(%), temperature, 'K' ) );
  parameter(%):
```

$$T_1 = 18 \; [[degC]]$$

$$T_1 = 291.15 \; [[K]] \tag{2}$$

Der Druck der Luft am Anfang ist als Überdruck über dem Barometerstand angegeben. Die beiden Werte zusammen ergeben den Druck.

```
> p[1] = 157*Unit(kPa) + 99*Unit(kPa);
  parameter(%):
```

$$p_1 = 256 \; [[kPa]] \tag{3}$$

Die Masse m der Luft wird in der folgenden Rechnung benötigt.
Annahme: Der Autoreifen ist dicht.
Dann ist die Masse m der Luft konstant.
Die Masse kann über die thermische Zustandsgleichung für das ideale Gas berechnet werden.

```
> p[1] * V = m * R[i] * T[1];
```

$$p_1 \, V = m \, R_i \, T_1 \tag{4}$$

Die spezielle Gaskonstante für trockene Luft aus [1].

```
> R[i] = 287.2*'Unit(J/(kg*K))';
  parameter(%):
```

$$R_i = 287.2 \left[\left[\frac{J}{kg\,K} \right]\right] \tag{5}$$

Zustandsgleichung auflösen nach der Masse.

```
> isolate((4),m);
```

$$m = \frac{p_1 V}{R_i T_1} \tag{6}$$

Die kalorische Zustandsgleichung des idealen Gases bei konstantem Volumen.
Dabei ist U die innere Energie des Gases.

```
> dU = m*c[v]*dT;
```

$$\Delta U = m\, c_v\, \Delta T \tag{7}$$

Die spezifische isobare Wärmekapazität von Luft bei 0 °C ist in [1] gegeben. Daraus wird die spezifische isochore Wärmekapazität berechnet-.

```
> c[v] = R[i] + c[p];
  c[p] = 1.004*'Unit(kJ/(kg*K))';
  subs(%,(5),%%): lhs(%)=convert(rhs(%),units,'J/(kg*K)');
  parameter(%):
```

$$c_v = R_i + c_p$$

$$c_p = 1.004 \left[\!\left[\frac{kJ}{kg\, K} \right]\!\right]$$

$$c_v = 1291.2 \left[\!\left[\frac{J}{kg\, K} \right]\!\right] \tag{8}$$

Dem Gas wird die Wärme Q durch Sonnenstrahlung zugeführt.

```
> Q = 2.55*Unit(kJ);
  parameter(%):
```

$$Q = 2.55\, [\![kJ]\!] \tag{9}$$

Annahmen: Das Gas gibt keine Wärme an den Reifen ab. Es gibt keine andere Wärmequelle.
Die gesamte Wärme wird in innere Energie umgewandelt.

```
> dU = Q;
```

$$\Delta U = Q \tag{10}$$

Damit kann aus der kalorischen Zustandsgleichung mit der bekannten Masse die Temperaturänderung berechnet werden.

```
> subs( (10), (7) );
  isolate( %, dT );
```

$$Q = m\, c_v\, \Delta T$$

$$\Delta T = \frac{Q}{m\, c_v} \tag{11}$$

Die Temperatur nach der Erwärmung (Zustand 2).

```
> T[2] = T[1] + dT; subs( (11), % );
```

$$T_2 = T_1 + \Delta T$$

$$T_2 = T_1 + \frac{Q}{m\, c_v} \tag{12}$$

Die Masse aus Gleichung **(6)** einsetzen.

```
> subs( (6), (12) ); collect(%,T[1]);
```

$$T_2 = T_1 + \frac{Q\, R_i\, T_1}{p_1\, V\, c_v}$$

(13)

$$T_2 = \left(1 + \frac{Q R_i}{p_1 V c_v}\right) T_1 \quad \textbf{(13)}$$

Alle Werte in der Gleichung sind bekannt. Einsetzen und ausrechnen.

```
> subs( parameters, (13) ): simplify(%);
  lhs(%)=convert(rhs(%),temperature,'degC');
```

$$T_2 = 316.95 \ [\![K]\!]$$

$$T_2 = 43.803 \ [\![degC]\!] \quad \textbf{(14)}$$

a) Eine Lufttemperatur von 44 °C stellt sich ein.

Die thermische Zustandsgleichung für den Zustand 2 aufschreiben.

```
> p[2] * V = m * R[i] * T[2];
```

$$p_2 V = m R_i T_2 \quad \textbf{(15)}$$

Die Masse aus Gleichung **(6)** einsetzen und vereinfachen.

```
> subs( (6), (15) );
  isolate( %, p[2] );
```

$$p_2 V = \frac{p_1 V T_2}{T_1}$$

$$p_2 = \frac{p_1 T_2}{T_1} \quad \textbf{(16)}$$

Jetzt kann die berechnete Temperatur T_2 als Zahlenwert eingesetzt werden.

Oder die Formel **(13)** wird eingesetzt. (Je nach Lehrer notwendig.)

```
> subs( (13), (16) );
```

$$p_2 = p_1 \left(1 + \frac{Q R_i}{p_1 V c_v}\right) \quad \textbf{(17)}$$

Alle Werte einsetzen und ausrechnen.

```
> subs( parameters, (17) ): simplify(%);
```

$$p_2 = 278690. \ [\![Pa]\!] \quad \textbf{(18)}$$

Der Überdruck soll angegeben werden. Also die Differenz zum äußeren Luftdruck, angezeigt auf dem Barometer.

```
> p[2,ü] = p[2] - 99*Unit(kPa);
  subs((18),%): simplify(%);
```

$$p_{2,ü} = p_2 - 99 \ [\![kPa]\!]$$

$$p_{2,ü} = 179690. \ [\![Pa]\!] \quad \textbf{(19)}$$

b) Es tritt ein Reifenüberdruck von 180 kPa auf.

Die Änderung der Inneren Energie wurde in **(10)** angegeben.

```
> (10); subs( parameters, % );
```

$$\Delta U = Q$$

$$\Delta U = 2.55 \ [\![kJ]\!] \quad \textbf{(20)}$$

Die Enthalpie H

```
> H = U + p*V;
```

(21)

$$H = p\,V + U \tag{21}$$

Die Änderung der inneren Energie und des Drucks führen hier zu einer Änderung der Enthalpie.

```
> dH = dU + dp*V;
  subs( (10), dp = p[2]-p[1], % );
```

$$\Delta H = dp\,V + \Delta U$$

$$\Delta H = \left(p_2 - p_1\right) V + Q \tag{22}$$

Den Druck aus Gleichung **(17)** einsetzen.

```
> subs( (17), (22) ); simplify(%);
```

$$\Delta H = \left(p_1 \left(1 + \frac{Q\,R_i}{p_1\,V\,c_v}\right) - p_1\right) V + Q$$

$$\Delta H = \frac{Q\left(R_i + c_v\right)}{c_v} \tag{23}$$

Die gegebenen Werte einsetzen und ausrechnen.

```
> subs( parameters, (23) ): simplify(%);
```

$$\Delta H = 3117.2\ [\![J]\!] \tag{24}$$

<u>c) Die innere Energie nimmt um 2,55 kJ zu. Die Enthalpie nimmt um 3,12 kJ zu.</u>

Der Endzustand: Druck p_1, wie am Anfang. Temperatur T_2, wie nach der Sonneneinstrahlung. Masse ist reduziert auf m_3.

Die thermische Zustandsgleichung für diese Situation.

```
> p[1]*V = m[3]* R[i] * T[2];
```

$$p_1\,V = m_3\,R_i\,T_2 \tag{25}$$

Temperatur aus Gleichung **(13)** einsetzen.

```
> subs( (13), (25) );
```

$$p_1\,V = m_3\,R_i \left(1 + \frac{Q\,R_i}{p_1\,V\,c_v}\right) T_1 \tag{26}$$

Auflösen nach der unbekannten Masse.

```
> isolate( (26), m[3] );
```

$$m_3 = \frac{p_1\,V}{R_i \left(1 + \frac{Q\,R_i}{p_1\,V\,c_v}\right) T_1} \tag{27}$$

Die Änderung der Masse ist gesucht.

```
> dM = m - m[3];
```

$$dM = m - m_3 \tag{28}$$

Die Anfangsmasse aus Gleichung **(6)** und die Masse nach Luftablassen aus Gleichung **(27)** einsetzen.

```
> subs( (27), (6), (28) );
  simplify(%,size);
```

$$dM = \frac{p_1 V}{R_i T_1} - \frac{p_1 V}{R_i \left(1 + \frac{Q R_i}{p_1 V c_v}\right) T_1}$$

$$dM = \frac{p_1 Q V}{\left(p_1 V c_v + Q R_i\right) T_1} \quad (29)$$

```
> subs( parameters, (29) ): simplify(%);
```

$$dM = 0.0062309 \, [[kg]] \quad (30)$$

d) Die Luftmasse von 6,23 g ist abzulassen um den Ausgangsdruck zu erreichen.

Hilfsmittel

[1] Cerbe und Wilhelms: Technische Thermodynamik, Hanser Verlag

[2] Maple 17, http://www.maplesoft.com/