```
> restart;
> Digits := trunc( evalhf(Digits) ):
> interface( displayprecision=5 ):
```

Aufgabe

Das Anlaufen eines Elektromotors mit Permanenterregung berechnen.

- Der Elektromotor wird mit einer konstanten Spannung versorgt.
- Die mechanische Reibung wird vernachlässigt.

Parameter

Maximales Drehmoment des Motors = Drehmoment des Motors beim Anlauf

```
> M[A] = 56.0*Unit(N*mm): lhs(%)=convert(rhs(%),units, 'N*m' );
```

$$M_A = 0.056000 \; [\![N\, m]\!] \qquad (1)$$

Maximale Drehzahl = Drehzahl des Motors im Leerlauf

```
> n[0] = 170.0 *Unit(1/min); simplify(%);
```

$$n_0 = 170.0 \; \left[\!\!\left[\frac{1}{\mathrm{min}}\right]\!\!\right]$$

$$n_0 = 2.8333 \; \left[\!\!\left[\frac{1}{s}\right]\!\!\right] \qquad (2)$$

Winkelgeschwindigkeit des Motors im Leerlauf

```
> omega[0] = 2*Pi * n[0]; simplify( subs( (2),% ) );
```

$$\omega_0 = 2\,\pi\, n_0$$

$$\omega_0 = 17.802 \; \left[\!\!\left[\frac{1}{s}\right]\!\!\right] \qquad (3)$$

Trägheitsmoment der Last (Trägheitsmoment des Motorrotors vernachlässigt.)

```
> J = 0.0088*Unit(kg*m^2);
```

$$J = 0.0088 \; [\![kg\, m^2]\!] \qquad (4)$$

1. Näherung, Drehmoment konstant

Das Drehmoment des Elektromotors ist von der Drehzahl (bzw. der Winkelgeschwindigkeit) abhängig.
Für eine erste Abschätzung wird die Abhängigkeit vernachlässigt. Das Anlaufdrehmoment soll konstant bleiben.

Wirkt ein Drehmoment M auf einen Körper mit Trägheitsmoment J, dann ändert sich die Winkelgeschwindigkeit ω.

```
> M = J * diff(omega(t),t);
```

$$M = J\,\dot{\omega}(t) \qquad (5)$$

```
> isolate( (5), diff(omega(t),t) );
```

$$\dot{\omega}(t) = \frac{M}{J} \qquad (6)$$

Drehmoment und Trägheitsmoment sind konstant. Die Gleichung kann einfach integriert werden.
Bei t = 0 ruht der Körper, ω = 0.

```
> int(lhs((6)),t) = int(rhs((6)),t );
```

(7)

$$\omega(t) = \frac{M\,t}{J} \tag{7}$$

Die Zeit bis zum erreichen der Leerlaufdrehzahl.

```
> subs( omega(t)=omega[0], t=t[0], M=M[A], (7) );
```

$$\omega_0 = \frac{M_A\,t_0}{J} \tag{8}$$

```
> isolate( (8), t[0] );
```

$$t_0 = \frac{\omega_0\,J}{M_A} \tag{9}$$

Die Zahlenwerte einseten.

```
> simplify( subs( (1),(3),(4), (9) ) );
```

$$t_0 = 2.7975\ [\![s]\!] \tag{10}$$

2. Drehmoment M(ω)

Das Drehmoment des Elektromotors ist beim Anlauf maximal und beim Leerlauf gleich 0. Die Abhängigkeit von der Winkelgeschwindigkeit ω ist linear.

```
> M(omega) = M[A]*( 1 - omega/omega[0] );
```

$$M(\omega) = M_A\left(1 - \frac{\omega}{\omega_0}\right) \tag{11}$$

Wirkt ein Drehmoment M wirkt auf einen Körper mit Trägheitsmoment J, dann ändert sich die Winkelgeschwindigkeit ω.

```
> M(omega) = J * diff(omega(t),t);
```

$$M(\omega) = J\,\dot{\omega}(t) \tag{12}$$

```
> M[A]*( 1 - omega(t)/omega[0] ) = J * diff(omega(t),t);
```

$$M_A\left(1 - \frac{\omega(t)}{\omega_0}\right) = J\,\dot{\omega}(t) \tag{13}$$

Die entstandende Differenzialgleichung etwas umschreiben.

```
> (13)/M[A]*omega[0]: simplify(%);
```

$$\omega_0 - \omega(t) = \frac{\omega_0\,J\,\dot{\omega}(t)}{M_A} \tag{14}$$

Darin ist eine Zeitkonstante erkennbar.

```
> Tau = coeff(rhs((14)),diff(omega(t),t));
```

$$\mathrm{T} = \frac{\omega_0\,J}{M_A} \tag{15}$$

Der Zahlenwert ist bereits aus **(10)** bekannt.

```
> simplify( subs( (1),(3),(4), (15) ) );
```

$$\mathrm{T} = 2.7975\ [\![s]\!] \tag{16}$$

Die Differenzialgleichung mit der Zeitkonstante geschrieben.

```
> algsubs( rhs((15))=lhs((15)), (14) );
```

$$\omega_0 - \omega(t) = \dot{\omega}(t)\ \mathrm{T} \tag{17}$$

Die Lösung dieser Differenzialgleichung ist bekannt.
Dabei den Startwert ω(0) = 0 verwendet.

```
> dsolve( { subs(omega[0]=K,(17)) , omega(0)=0 } ): subs(K=omega[0],%): collect(%,omega[0]);
```

$$\omega(t) = \left(1 - e^{-\frac{t}{T}}\right)\omega_0 \tag{18}$$

Damit kann auch das Drehmoment beim Anlaufen als Funktion der Zeit dargestellt werden.
Das berechnet ω(t) in Gleichung **(11)** einsetzen.

```
> M(t) = subs( omega[0]=K, omega=omega(t), (18), K=omega[0], rhs((11)) );
```

$$M(t) = M_A\, e^{-\frac{t}{T}} \tag{19}$$