


```
> restart;
> Digits:=25: interface( displayprecision= 7 );
```

## Aufgabe

Auf einer Gefällstrecke versagen bei einem Fahrzeug der Masse 1,2t die Bremsen. Es gelingt dem Fahrer, in eine steil ansteigende Seitenstraße mit der Steigung 25% einzubiegen. Die Geschwindigkeit am Anfang der Steigung beträgt 90km/h und der Fahrwiderstanskoeffizient 0,03. Der Luftwiderstand wird vernachlässigt.

- Berechnen Sie, nach welcher Zeit und nach welcher Strecke das Fahrzeug zum Stillstand kommt.
- Berechnen Sie die kinetische Energie des Fahrzeuges nach 2,25s nach dem Beginn des beschriebenen Vorgangs.

## Skizze



## Rechenweg

Das Fahrzeug ist auf einer Ebene mit 25% Steigung.  
Der Winkel  $\alpha$  zur Horizontalen

```
> tan(alpha) = 25/100;
```

$$\tan(\alpha) = \frac{1}{4} \quad (1)$$

Der Sinus des Winkels wird benötigt.

```
> sin(alpha) = tan(alpha)/sqrt(1+tan(alpha)^2); subs((1),%): evalf(%) ;
```

$$\sin(\alpha) = \frac{\tan(\alpha)}{\sqrt{1 + \tan(\alpha)^2}}$$

$$\sin(\alpha) = 0.2425356 \quad (2)$$

Der Cosinus des Winkels wird benötigt.

```
> cos(alpha) = 1/sqrt(1+tan(alpha)^2); subs((1),%): evalf(%);
```

$$\cos(\alpha) = \frac{1}{\sqrt{1 + \tan(\alpha)^2}}$$

$$\cos(\alpha) = 0.9701425 \quad (3)$$

Die Masse m des Fahrzeugs ist gegeben

```
> m = 1.2*Unit(t); simplify(%);
```

$$m = 1.2 \llbracket t \rrbracket$$

```
m = 1200.0 \llbracket kg \rrbracket
```

$$(4)$$

Das Gewicht des Fahrzeugs

```
> F[G] = m*g;
```

$$F_G = m g \quad (5)$$

Dabei ist g die Fallbeschleunigung an der Erdoberfläche

```
> g = 9.81*Unit(m/s^2);
```

$$g = 9.81 \llbracket \frac{m}{s^2} \rrbracket \quad (6)$$

Dem Fahrzeug auf der schiefen Ebene wirkt senkrecht zur Ebene die Normalenkraft entgegen.

```
> F[N] = F[G] * cos(alpha); subs((5),%);
```

$$F_N = F_G \cos(\alpha)$$

$$F_N = m g \cos(\alpha) \quad (7)$$

Der Fahrwiderstandskoeffizient ist gegeben

```
> mu = 0.03;
```

$$\mu = 0.03 \quad (8)$$

Der Fahrwiderstand ist

```
> F[R] = mu*F[N]; subs((7),%);
```

$$F_R = \mu F_N$$

$$F_R = \mu m g \cos(\alpha) \quad (9)$$

Auf das Fahrzeug wirkt die Hangabtriebskraft

```
> F[A] = F[G]*sin(alpha); subs((5),%);
```

$$F_A = F_G \sin(\alpha)$$

$$F_A = m g \sin(\alpha) \quad (10)$$

Die Summe der Kräfte, die das Fahrzeug abbremsen

```
> F = F[R] + F[A]; subs((9),(10),%);
```

$$F = F_R + F_A$$

$$F = \mu m g \cos(\alpha) + m g \sin(\alpha) \quad (11)$$

Die wirkende Kraft führt zu einer Geschwindigkeitsänderung. Die Beschleunigung nach Newton

```
> F=m*a; subs((11),%); isolate(%,a): simplify(%);
```

$$F = m a$$

$$\mu m g \cos(\alpha) + m g \sin(\alpha) = m a$$

$$a = g (\mu \cos(\alpha) + \sin(\alpha)) \quad (12)$$

Den Zahlenwert ausrechnen.

```
> subs((2),(3),(6),(8),(12));
```

$$a = 2.664787 \llbracket \frac{m}{s^2} \rrbracket \quad (13)$$

Die Geschwindigkeit des Fahrzeugs. Der Zeitpunkt  $t=0$  wird auf den Anfang des beschriebenen

Bremsvorgangs gesetzt.

```
> v(t) = v[0]-a*t;
```

$$v(t) = v_0 - a t \quad (14)$$

Die Anfangsgeschwindigkeit ist in der Aufgabe gegeben

```
> v[0] = 90*eval(Unit(km/h),1); simplify(%);
```

$$v_0 = 90 \left[ \frac{\text{km}}{\text{h}} \right]$$

$$v_0 = 25 \left[ \frac{\text{m}}{\text{s}} \right] \quad (15)$$

Das Fahrzeug kommt nach der Zeit T zum Stillstand.

```
> subs(v(t)=0, t=T, (14)); isolate(%,T);
```

$$0 = v_0 - a T$$

$$T = \frac{v_0}{a} \quad (16)$$

Den Zahlenwert ausrechnen.

```
> subs((15),(13), (16)): simplify(%);
```

$$T = 9.381611 \left[ \text{s} \right] \quad (17)$$

Der zurückgelegte Weg des Fahrzeugs. Zum Startzeitpunkt wird  $s(0)=0$  gesetzt.

```
> s(t) = int(rhs((14)),t);
```

$$s(t) = v_0 t - \frac{1}{2} a t^2 \quad (18)$$

Der zurückgelegte Weg zu Zeitpunkt T.

```
> subs(t=T,(18)); lhs(%)=subs((16),rhs(%));
```

$$s(T) = v_0 T - \frac{1}{2} a T^2$$

$$s(T) = \frac{v_0^2}{2 a} \quad (19)$$

Den Zahlenwert ausrechnen.

```
> subs((15),(13), (19)): simplify(%);
```

$$s(T) = 117.2701 \left[ \text{m} \right] \quad (20)$$

Das Fahrzeug kommt nach einer Strecke von 117 m innerhalb von 9,38 s zum Stillstand.

Gesucht ist die kinetische Energie zum Zeitpunkt

```
> t = 2.25*Unit(s);
```

$$t = 2.25 \left[ \text{s} \right] \quad (21)$$

Die kinetische Energie des Fahrzeugs bei Geschwindigkeit v.

```
> E[kin] = m*v^2/2;
```

$$E_{kin} = \frac{m v^2}{2} \quad (22)$$

Die Geschwindigkeit als Funktion der Zeit aus Gleichung (14) einsetzen.

```
> subs(v=v(t), (14), (22));
```

(23)

$$E_{kin} = \frac{m (v_0 - a t)^2}{2} \quad (23)$$

Zahlenwert ausrechnen.

```
> subs( (4),(13),(15),(21), (23) ): simplify(%); lhs(%)=convert(rhs(%),  
    'units','kJ');
```

$$E_{kin} = 216696.4 \text{ [J]}$$

$$E_{kin} = 216.6964 \text{ [kJ]} \quad (24)$$

Die kinetische Energie nach 2,25 s beträgt 217 kJ.

## ▼ Hilfsmittel

- Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch
- Bronstein et al: Taschenbuch der Mathematik, Verlag Harri Deutsch
- Maple 14, <http://www.maplesoft.com/>