

```
[> restart;
=> Digits:=30: interface( displayprecision=5 ):
=> swap := x -> rhs(x) = lhs(x):
```

▼ Aufgabe

Ein UI-Kern aus Elektroblech V360-50A soll eine magnetische Flussdichte von 800mT ($B=0,8\text{T}$) haben. Die mittlere Feldlinienlänge der Anordnung beträgt 700mm ($l_m=0,7\text{m}$). Die Spule hat einen mittleren Windungsdurchmesser von 39,5 mm.

Berechnen Sie den Drahtdurchmesser der Spule, wenn sie an einer Gleichspannung von 24V anliegt und ihre Wicklung aus Kupferdraht besteht!

Aus der Magnetisierungskurve des Elektroblechs: Die magnetische Feldstärke bei $B = 0,8\text{ T}$ beträgt $H = 180\text{ A/m}$.

▼ Rechnung

Die magnetische Feldstärke im Kern beträgt

```
> H = 180*Unit(A/m);
```

$$H = 180 \left[\frac{\text{A}}{\text{m}} \right] \quad (1)$$

Die mittlere Länge der Feldlinie durch den Kern beträgt

```
> l[m] = 0.7 * Unit(m);
```

$$l_m = 0.7 \left[\text{m} \right] \quad (2)$$

Die Durchflutung Θ durch die Öffnung des Kerns folgt daraus

```
> Theta = H * l[m];
```

$$\Theta = H l_m \quad (3)$$

Der Strom I fließt durch N Wicklungen. Die Summe der Ströme ergibt die Durchflutung.

```
> Theta = N * I; subs( (3), % );
```

$$\Theta = I N$$

$$H l_m = I N \quad (4)$$

Die Spannung U an der Wicklung ist gegeben.

```
> U = 24*Unit(V);
```

$$U = 24 \left[\text{V} \right] \quad (5)$$

Der Strom I wird über den ohmschen Widerstand R der Wicklung berechnet.

```
> R = L * rho / A;
```

$$R = \frac{L \rho}{A} \quad (6)$$

Die Länge L der Wicklung folgt aus dem mittleren Wicklungsdurchmesser d_w .

```
> d[w] = 39.5 * Unit(mm);
```

$$d_w = 39.5 \left[\text{mm} \right] \quad (7)$$

```
> L = N * Pi * d[w];
```

$$L = N \pi d_w \quad (8)$$

Die Querschnittsfläche A des Drahtes folgt aus dem Drahtdurchmesser d .

```
> A = Pi*(d/2)^2;
```

$$A = \frac{\pi d^2}{4} \quad (9)$$

Der spezifische Widerstand von Kupfer aus [1]

```
> rho = 0.0178*Unit(ohm*(mm)^2/m); lhs(%)=convert(rhs(%),units,
'ohm*mm^2/m');
```

$$\rho = 1.7800 \cdot 10^{-8} \left[\frac{m^3 kg}{s^3 A^2} \right]$$

$$\rho = 0.017800 \left[\frac{\Omega mm^2}{m} \right] \quad (10)$$

Gleichung (8) und (9) in Gleichung (6) einsetzen.

```
> subs( (8),(9), (6) );
```

$$R = \frac{4 N d_w \rho}{d^2} \quad (11)$$

Den Widerstand in das ohmsche Gesetz einsetzen.

```
> U=R*I; subs( (11), % );
```

$$U = IR$$

$$U = \frac{4 I N d_w \rho}{d^2} \quad (12)$$

Die bekannte Durchflutung aus (4) einsetzen.

```
> algsubs( swap((4)), (12) );
```

$$U = \frac{4 d_w \rho H l_m}{d^2} \quad (13)$$

Auflösen nach dem gesuchten Drahtdurchmesser d.

```
> isolate( (13), d^2 ); sqrt(lhs(%))=sqrt(rhs(%)): simplify(%)
assuming d>0;
```

$$d^2 = \frac{4 d_w \rho H l_m}{U}$$

$$d = 2 \sqrt{\frac{d_w \rho H l_m}{U}} \quad (14)$$

Die gegebenen Werte einsetzen und ausrechnen.

```
> subs( (10),(7),(5),(2),(1), (14) ): simplify(%); lhs(%)=convert(rhs(%),
units,'mm');
```

$$d = 0.00012151 \llbracket m \rrbracket$$

$$d = 0.12151 \llbracket mm \rrbracket \quad (15)$$

Der Drahtdurchmesser beträgt 0,12 mm.

▼ Hilfsmittel

[1] Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch

[2] Maple 17, <http://www.maplesoft.com/>