


```
> restart;
> Digits := 25: interface( displayprecision=3 );
```

Aufgabe

Die Innenseite der Wärmedämmung (Dicke 5 cm, Wärmeleitfähigkeit 1,5 W/(K m)) eines Trockenofens hat eine Oberflächentemperatur von 200 °C. An der Außenseite der Isolierung findet konvektiver Wärmeübergang an die Luft (Lufttemperatur 20 °C) statt.

- Wie groß muss der Wärmeübergangskoeffizient außen mindestens sein, damit die Außentemperatur der Dämmung 100 °C nicht überschreitet?
- Welche Außentemperatur stellt sich bei doppelter Dämmstärke und einem Wärmeübergangskoeffizienten von 15 W/(K m²) ein?

Skizze



An der Innseite der Dämmung Temperatur T_1 .

An der Außenseite der Dämmung Temperatur T_2 .

Lufttemperatur T_3 .

Rechnung

Vereinfachungen/Annahmen:

- Betrachtet wird ein stationärer Zustand.
- Die Temperatur der jeweiligen Fläche ist konstant über die gesamte Fläche.
- Die Dämmung hat ebene und parallele Wände. Auch hat damit die Innenseite und die Außenseite der Dämmung den gleichen Flächeninhalt.

Der Energiefluss durch die Dämmung mit Dicke d , Fläche A und Wärmeleitfähigkeit λ

```
> Phi[12] = A*lambda*(T[1]-T[2])/d;
```

$$\Phi_{12} = \frac{A \lambda (T_1 - T_2)}{d} \quad (1)$$

Der Energiefluss von der Dämmung mit Fläche A in die Luft durch Konvektion mit Wärmeübergangskoeffizient α

```
> Phi[23] = A*alpha*(T[2]-T[3]);
```

$$\Phi_{23} = A \alpha (T_2 - T_3) \quad (2)$$

Betrachtet wird ein stationärer Zustand. Daher bleibt die Energie in der Dämmung konstant. Der Energiefluss durch Dämmung muss vollständig durch die Konvektion abgeführt werden.

```
> Phi[12] = Phi[23];
```

$$\Phi_{12} = \Phi_{23} \quad (3)$$

```
> subs((1),(2),(3));
```

$$\frac{A \lambda (T_1 - T_2)}{d} = A \alpha (T_2 - T_3) \quad (4)$$

```
> (4)/A;
```

$$\frac{\lambda (T_1 - T_2)}{d} = \alpha (T_2 - T_3) \quad (5)$$

Im Teil (a) sind die Temperaturen gegeben und der Wärmeübergangskoeffizient α gesucht.
Auflösen nach α

```
> isolate((5),alpha): simplify(%);
```

$$\alpha = \frac{\lambda (T_1 - T_2)}{d (T_2 - T_3)} \quad (6)$$

Gegeben sind die Werte

```
> params := lambda = 1.5 * Unit( W/(K*m) ): lambda = convert( subs  
(params,lambda), units, W/(K*m) );
```

$$\lambda = 1.5 \left[\frac{W}{K m} \right] \quad (7)$$

```
> d = 5*Unit(cm); params:=params,%:
```

$$d = 5 \left[cm \right] \quad (8)$$

Für T_2 die maximale Temperatur einsetzen.

```
> T[1] = 200*Unit(degC), T[2] = 100*Unit(degC), T[3] = 20*Unit  
(degC);
```

$$T_1 = 200 \left[degC \right], T_2 = 100 \left[degC \right], T_3 = 20 \left[degC \right] \quad (9)$$

```
> map( x->lhs(x)=evalf(convert(rhs(x),temperature,K)),{(9)} ):  
params:=params,op(%):
```

Die Zahlenwerte in Gleichung (6) einsetzen

```
> subs(params,(6)): simplify(%): lhs(%)=convert(rhs(%),units,W/(K*  
m^2));
```

$$\alpha = 37.5 \left[\frac{W}{K m^2} \right] \quad (10)$$

Bei diesem Wärmeübergangskoeffizienten hat die Außenseite eine Temperatur von 100 °C.

```
> unassign('params');
```

Im Teil (b) ist die Temperatur T_2 der Außenseite gesucht.

Auflösen der Gleichung (5) nach der T_2

```
> isolate((5),T[2]);
```

(11)

$$T_2 = \frac{\lambda T_1 + \alpha d T_3}{\lambda + \alpha d} \quad (11)$$

Die Wärmeleitfähigkeit der Dämmung ist unverändert

```
> params := lambda = 1.5 * Unit( W/(K*m) ): lambda = convert( subs
  (params, lambda), units, W/(K*m) );
```

$$\lambda = 1.5 \left[\frac{W}{K m} \right] \quad (12)$$

Hier ist der Wärmeübergangskoeffizient gegeben

```
> alpha = 15*Unit(W/(K*m^2)): params:=params, %: lhs(%%)=convert(
  rhs(%%), units, W/(K*m^2) );
```

$$\alpha = 15 \left[\frac{W}{K m^2} \right] \quad (13)$$

Die Dicke der Dämmung ist verdoppelt auf

```
> d = 10*Unit(cm); params:=params, %:
```

$$d = 10 \left[cm \right] \quad (14)$$

Temperatur der Innenseite und der Luft sind gegeben

```
> T[1] = 200*Unit(degC), T[3] = 20*Unit(degC);
```

$$T_1 = 200 \left[degC \right], T_3 = 20 \left[degC \right] \quad (15)$$

```
> map( x->lhs(x)=evalf(convert(rhs(x), temperature, K)), {(15)} ):
  params:=params, op(%):
```

Die Zahlenwerte in Gleichung (11) einsetzen

```
> subs(params, (11)): simplify(%); lhs(%)=convert(rhs(%), temperature,
  degC);
```

$$T_2 = 383. \left[K \right]$$

$$T_2 = 110. \left[degC \right] \quad (16)$$

Bei diesen Parametern ergibt sich eine Temperatur von 110 °C an der Außenseite.

Hilfsmittel

- Cerbe und Wilhelms: Technische Thermodynamik, Hanser Verlag

- Maple 14