```
> restart;
```

Übungsaufgabe zur Optimierung mit Lagrange-Multiplikatoren:

Ein quaderförmiges Schwimmbecken mit einem Fassungsvermögen (Volumen) von V=180 m³ soll so gebaut werden, dass die Oberfläche (Boden und Seitenwände) möglichst klein wird. Wie sind die Abmessungen des Beckens zu wählen?

Übersetzen der Textaufgabe in Formeln:
Der Quader hat drei Kanten mit dem Längen a, b und c. Die Grundfläche ist a·b, die Höhe ist c.
a, b und c sind die Variablen der Optimierungsaufgabe.

Die Oberfläche soll optimiert werden. Die Oberfläche ist in dieser Aufgaben definiert als Seitenwände und Grundfläche:

```
> f(a,b,c) = a*b + 2*(b*c+a*c);
```

$$f(a, b, c) = a\, b + 2\, b\, c + 2\, a\, c \qquad \textbf{(1)}$$

Von der Funktion f(a,b,c) sind die Extremstellen zu bestimmen und davon das Minimum zu suchen.

Das Volumen des Quaders

```
> V(a,b,c) = a*b*c;
```

$$V(a, b, c) = a\, b\, c \qquad \textbf{(2)}$$

Dieses Volumen ist vorgegeben.

```
> V(a,b,c) = V[0];
```

$$V(a, b, c) = V_0 \qquad \textbf{(3)}$$

```
> subs((2),(3));
```

$$a\, b\, c = V_0 \qquad \textbf{(4)}$$

```
> V[0] = 180*Unit(m^3);
```

$$V_0 = 180\, [\![m^3]\!] \qquad \textbf{(5)}$$

Die Gleichung **(4)** ist eine Nebenbedingung der Optimierungsaufgabe.
Es gibt eine Nebenbedingung, also wird ein Lagrangescher Multiplikator λ eingeführt.
Zu lösen ist die Gleichung

```
> grad(f) + lambda * grad(V) = 0;
```

$$grad(f) + \lambda\, grad(V) = 0 \qquad \textbf{(6)}$$

Zusammen mit der Nebenbedingung **(4)**.

Die Komponenten der Vektorgleichung **(6)** getrennt aufschreiben.

```
> diff( f(a,b,c) , a ) + lambda * diff( V(a,b,c) , a ) = 0;
```

$$\frac{\partial}{\partial a} f(a, b, c) + \lambda \frac{\partial}{\partial a} V(a, b, c) = 0 \qquad \textbf{(7)}$$

```
> diff( f(a,b,c) , b ) + lambda * diff( V(a,b,c) , b ) = 0;
```

$$\frac{\partial}{\partial b} f(a, b, c) + \lambda \frac{\partial}{\partial b} V(a, b, c) = 0 \qquad \textbf{(8)}$$

```
> diff( f(a,b,c) , c ) + lambda * diff( V(a,b,c) , c ) = 0;
```

$$\frac{\partial}{\partial c} f(a, b, c) + \lambda \frac{\partial}{\partial c} V(a, b, c) = 0 \qquad \textbf{(9)}$$

Die bekannten Funktionen f(a,b,c) und V(a,b,c) einsetzen und die Ableitungen berechnen.

```
> subs((1),(2),(7)); simplify(%);
```

$$\frac{\partial}{\partial a} (a\, b + 2\, b\, c + 2\, a\, c) + \lambda \frac{\partial}{\partial a} (a\, b\, c) = 0 \qquad \textbf{(10)}$$

$$b+2c+\lambda b c=0 \quad \textbf{(10)}$$

```
> subs((1),(2),(8)); simplify(%);
```

$$\frac{\partial}{\partial b}\left(a b+2 b c+2 a c\right)+\lambda \frac{\partial}{\partial b}\left(a b c\right)=0$$

$$a+2c+\lambda a c=0 \quad \textbf{(11)}$$

```
> subs((1),(2),(9)); simplify(%);
```

$$\frac{\partial}{\partial c}\left(a b+2 b c+2 a c\right)+\lambda \frac{\partial}{\partial c}\left(a b c\right)=0$$

$$2b+2a+\lambda a b=0 \quad \textbf{(12)}$$

Die Gleichungen **(4)**, **(10)**, **(11)**, **(12)** bilden ein Gleichungssystem für die Variablen a, b, c und λ.
Das Gleichungssystem lösen.
(11) nach a auflösen und in die restlichen drei Gleichungen einsetzen.

```
> isolate((11),a);
```

$$a=-\frac{2c}{1+\lambda c} \quad \textbf{(13)}$$

```
> subs((13),(4));
```

$$-\frac{2c^2 b}{1+\lambda c}=V_0 \quad \textbf{(14)}$$

```
> subs((13),(10));
```

$$b+2c+\lambda b c=0 \quad \textbf{(15)}$$

```
> subs((13),(12));
```

$$2b-\frac{4c}{1+\lambda c}-\frac{2\lambda c b}{1+\lambda c}=0 \quad \textbf{(16)}$$

Gleichung **(15)** nach b auflösen und in die restlichen zwei Gleichungen einsetzen.

```
> isolate((15),b);
```

$$b=-\frac{2c}{1+\lambda c} \quad \textbf{(17)}$$

```
> subs((17),(14));
```

$$\frac{4c^3}{\left(1+\lambda c\right)^2}=V_0 \quad \textbf{(18)}$$

```
> subs((17),(16)); simplify(%);
```

$$-\frac{8c}{1+\lambda c}+\frac{4\lambda c^2}{\left(1+\lambda c\right)^2}=0$$

$$-\frac{4c\left(2+\lambda c\right)}{\left(1+\lambda c\right)^2}=0 \quad \textbf{(19)}$$

Gleichung **(18)** nach λ auflösen. Die quadratische Gleichung hat zwei Lösungen.

```
> solve((18),{lambda});
```

$$\left\{\lambda=\frac{-V_0+2c\sqrt{V_0 c}}{V_0 c}\right\},\ \left\{\lambda=-\frac{V_0+2c\sqrt{V_0 c}}{V_0 c}\right\} \quad \textbf{(20)}$$

Die erste Lösung in Gleichung **(19)** einsetzen.

```
> subs((20)[1],(19));
```

$$-\frac{4\,c\left(2+\frac{-V_0+2\,c\,\sqrt{V_0\,c}}{V_0}\right)}{\left(1+\frac{-V_0+2\,c\,\sqrt{V_0\,c}}{V_0}\right)^2}=0 \qquad \textbf{(21)}$$

Vereinfachen und auflösen nach c.

```
> simplify((21));
```

$$-\frac{V_0+2\,c\,\sqrt{V_0\,c}}{c^2}=0 \qquad \textbf{(22)}$$

```
> isolate((22),c);
```

$$c=\frac{2^{1/3}\,V_0^{1/3}}{2} \qquad \textbf{(23)}$$

Die zweite Lösung in Gleichung **(19)** einsetzen.

```
> subs((20)[2],(19));
```

$$-\frac{4\,c\left(2-\frac{V_0+2\,c\,\sqrt{V_0\,c}}{V_0}\right)}{\left(1-\frac{V_0+2\,c\,\sqrt{V_0\,c}}{V_0}\right)^2}=0 \qquad \textbf{(24)}$$

Vereinfachen und auflösen nach c.

```
> simplify((24));
```

$$\frac{-V_0+2\,c\,\sqrt{V_0\,c}}{c^2}=0 \qquad \textbf{(25)}$$

```
> isolate((25),c);
```

$$c=\frac{2^{1/3}\,V_0^{1/3}}{2} \qquad \textbf{(26)}$$

Auch die zweite Lösung für λ führt auf den gleichen Wert für c.
Das berechnet c in Gleichung **(20)** einsetzen um die beiden Möglichkeiten für λ zu berechnen.

```
> (20)[1][1]; subs((26),%); simplify(%) assuming V[0]>0;
```

$$\lambda=\frac{-V_0+2\,c\,\sqrt{V_0\,c}}{V_0\,c}$$

$$\lambda=\frac{\left(-V_0+\frac{2^{5/6}\,V_0^{1/3}\,\sqrt{V_0^{4/3}\,2^{1/3}}}{2}\right)2^{2/3}}{V_0^{4/3}}$$

$$\lambda=0 \qquad \textbf{(27)}$$

Die zweite Möglichkeit für λ.

```
> (20)[2][1]; subs((26),%); simplify(%) assuming V[0]>0;
```

$$\lambda = -\frac{V_0 + 2\,c\,\sqrt{V_0\,c}}{V_0\,c}$$

$$\lambda = -\frac{\left(V_0 + \frac{2^{5/6}\,V_0^{1/3}\,\sqrt{V_0^{4/3}\,2^{1/3}}}{2}\right)2^{2/3}}{V_0^{4/3}}$$

$$\lambda = -\frac{2\,2^{2/3}}{V_0^{1/3}} \tag{28}$$

Die berechneten Werte von c und λ=0 in Gleichung **(17)** einsetzen um b zu berechnen.

```
> subs((26),(27),(17));
```

$$b = -2^{1/3}\,V_0^{1/3} \tag{29}$$

Eine negative Kantenlänge ist keine Lösung. Den anderen Wert für λ aus Gleichung **(28)** einsetzen.

```
> subs((26),(28),(17));
```

$$b = 2^{1/3}\,V_0^{1/3} \tag{30}$$

Diese Lösung liefert positive Werte für die Kangenlänge.

Den Wert von a aus Gleichung **(13)** ebenfalls mit dem λ aus Gleichung **(28)** ausrechnen.

```
> subs((26),(28),(13));
```

$$a = 2^{1/3}\,V_0^{1/3} \tag{31}$$

Die Zahlenwerte für die Kantenlängen aus dem gegebenen Volumen V_0 ausrechnen.

```
> evalf( subs((5),(31)) ): simplify(%);
```

$$a = 7.113786609\ [\![m]\!] \tag{32}$$

```
> evalf( subs((5),(30)) ): simplify(%);
```

$$b = 7.113786609\ [\![m]\!] \tag{33}$$

```
> evalf( subs((5),(26)) ): simplify(%);
```

$$c = 3.556893304\ [\![m]\!] \tag{34}$$

Hilfmittel
- Klaus Jänich: Mathematik 2, Springer-Verlag
- Maple 14