```
> restart;
> with(plots):
```

Rätselaufgabe:

Zwei Bauern gehört ein kreisrundes Wiesengrundstück mit dem Durchmesser 30 m je zur Hälfte.
Ein Bauer möchte seine Kuh grasen lassen. Er rammt einen Pfosten genau auf der Grundstücksgrenze und bindet die Kuh mit einem Seil fest.
Wie lang darf das Seil sein, damit die Kuh maximal die Hälfte der Grundstücksfläche abgrasen kann?

Namen definieren:
a = Radius des Wisengrundstücks
b = Länge des Seils, Radius der von der Kuh erreichbaren Fläche.

Eine Skizze:

```
> a:=15: b:=17.3809:
> wiese := x^2+y^2=a^2:
> kuh := (x-a)^2+y^2=b^2:
> p1 := implicitplot([wiese,kuh],x=-16..+35,y=-18..+18,color=
  [green,blue],legend=["Wiese mit Radius a","Kuh mit Radius b"],
  thickness=2):
> xs := (1/2)*(2*a^2-b^2)/a:
> ys := (1/2)*sqrt(4*a^2-b^2)*b/a:
> l1 := plot( <<xs,0,xs,0,a>|<ys,0,-ys,0,0>>, color=green,
  thickness=3 ):
> l2 := plot( <<xs,a,xs>|<ys,0,-ys>>, color=blue, thickness=3 ):
> l3 := plot( <<xs,xs>|<ys,-ys>>, color=black, thickness=2 ):
> display( [p1,l1,l2,l3], scaling=constrained );
```




```
> restart;
> Digits:=30: interface(displayprecision=10):
```

Die von der Kuh nutzbare Fläche, die Nutzfläche, liegt in der Mitte. Es ist die Schnittmenge der beiden Kreise.

Mehrere Linien der Länge a in Grün und Linien der Länge b in Blau sind eingezeichnet. Es sind alles Radien der jeweiligen Kreise.

Je eine blaue Linie und zwei grüne Linien bilden zwei gleichschenklige Dreiecke. Die beiden Dreiecke sind ähnlich, werden durch Spiegelung an der x-Achse ineinander überführt.
In jedem Dreieck sind die Winkel beschriftet. Dem Winkel α liegt eine Seite mit Länge a gegenüber. Dem Winkel β liegt eine Seite mit Länge b gegenüber. Weil zwei Seiten gleiche Länge a haben, tritt auch der Winkel α in jedem Dreieck zweimal auf.

Eine schwarze Linie teilt die Nutzfläche in zwei Kreisabschnitte (Kreissegmente).
Der Abschnitt aus dem Kreis mit Radius a hat den Zentriwinkel 2β. Der Abschnitt aus dem Kreis mit Radius b hat den Zentriwinkel 2α.

Aus der Formelsammlung (Bronstein) den Flächeninhalt des Kreisabschnitts verwenden.
Flächeninhalt des Kreisabschnitts aus der Wiese:

```
> A[wiese] = a^2/2*(2*beta-sin(2*beta));
```

$$A_{wiese} = \frac{1}{2}\, a^2\, (2\,\beta - \sin(2\,\beta)) \tag{1}$$

Flächeninhalt des Kreisabschnitts aus dem Bewegungsbereichs der Kuh:

```
> A[kuh] = b^2/2*(2*alpha-sin(2*alpha));
```

$$A_{kuh} = \frac{1}{2}\, b^2\, (2\,\alpha - \sin(2\,\alpha)) \tag{2}$$

Die nutzbare Fläche hat den Inhalt

```
> A[nutz] = A[wiese] + A[kuh];
```

$$A_{nutz} = A_{wiese} + A_{kuh} \tag{3}$$

Die Winkelsumme im ebenen Dreieck ist π (=180°).
Damit kann in der Gleichung **(2)** der Winkel α eliminiert werden.

```
> Pi = 2*alpha + beta;
```

$$\pi = 2\,\alpha + \beta \tag{4}$$

```
> isolate((4),2*alpha);
```

$$2\,\alpha = \pi - \beta \tag{5}$$

```
> subs((5),(2));
```

$$A_{kuh} = \frac{1}{2}\, b^2\, (\pi - \beta - \sin(\pi - \beta)) \tag{6}$$

```
> simplify((6));
```

$$A_{kuh} = \frac{1}{2}\, b^2\, (\pi - \beta - \sin(\beta)) \tag{7}$$

Gesucht ist die Länge b. Die Rechnung wird einfacher, wenn zuerst der Winkel β berechnet wird. Ist der Winkel β bekannt, kann daraus die Länge b berechnet werden.

Ausdrücken der Länge b durch die gegebene Länge a und den gesuchten Winkel β.
Im ebenen Dreieck gilt der Sinussatz.

```
> a/sin(alpha) = b/sin(beta);
```

$$\frac{a}{\sin(\alpha)} = \frac{b}{\sin(\beta)} \tag{8}$$

```
> subs(isolate((4),alpha),(8));
```

$$\frac{a}{\sin\left(\frac{1}{2}\,\pi - \frac{1}{2}\,\beta\right)} = \frac{b}{\sin(\beta)} \tag{9}$$

```
> simplify((9));
```

$$\frac{a}{\cos\left(\frac{1}{2}\,\beta\right)} = \frac{b}{\sin(\beta)} \tag{10}$$

```
> isolate((10),b);
```

$$b = \frac{a\,\sin(\beta)}{\cos\left(\frac{1}{2}\,\beta\right)} \tag{11}$$

```
> subs((11),(7));
```

(12)

$$A_{kuh} = \frac{1}{2} \frac{a^2 \sin(\beta)^2 (\pi - \beta - \sin(\beta))}{\cos\left(\frac{1}{2}\beta\right)^2} \quad \textbf{(12)}$$

```
> simplify((12));
```

$$A_{kuh} = 2 a^2 \sin\left(\frac{1}{2}\beta\right)^2 \left(\pi - \beta - 2\sin\left(\frac{1}{2}\beta\right)\cos\left(\frac{1}{2}\beta\right)\right) \quad \textbf{(13)}$$

Die Nutzfläche ist damit

```
> subs((13),(1),(3));
```

$$A_{nutz} = \frac{1}{2} a^2 (2\beta - \sin(2\beta)) + 2 a^2 \sin\left(\frac{1}{2}\beta\right)^2 \left(\pi - \beta - 2\sin\left(\frac{1}{2}\beta\right)\cos\left(\frac{1}{2}\beta\right)\right) \quad \textbf{(14)}$$

```
> simplify((14));
```

$$A_{nutz} = -a^2 \left(2\sin\left(\frac{1}{2}\beta\right)\cos\left(\frac{1}{2}\beta\right) + \beta + 2\pi\cos\left(\frac{1}{2}\beta\right)^2 - 2\pi - 2\beta\cos\left(\frac{1}{2}\beta\right)^2\right) \quad \textbf{(15)}$$

Die Nutzfläche soll die Hälfte der Wiese sein. Die Wiese ist eine Kreisfläche mit Radius a. Also soll die Nutzfläche sein

```
> A[nutz] = Pi*a^2/2;
```

$$A_{nutz} = \frac{1}{2} a^2 \pi \quad \textbf{(16)}$$

Das liefert eine Bestimmungsgleichung für den unbekannten Winkel β.

```
> subs((16),(15));
```

$$\frac{1}{2} a^2 \pi = -a^2 \left(2\sin\left(\frac{1}{2}\beta\right)\cos\left(\frac{1}{2}\beta\right) + \beta + 2\pi\cos\left(\frac{1}{2}\beta\right)^2 - 2\pi - 2\beta\cos\left(\frac{1}{2}\beta\right)^2\right) \quad \textbf{(17)}$$

```
> simplify((17)/a^2-Pi/2);
```

$$0 = -2\sin\left(\frac{1}{2}\beta\right)\cos\left(\frac{1}{2}\beta\right) - \beta - 2\pi\cos\left(\frac{1}{2}\beta\right)^2 + \frac{3}{2}\pi + 2\beta\cos\left(\frac{1}{2}\beta\right)^2 \quad \textbf{(18)}$$

Diese transzendente Gleichung nach dem gesuchten Winkel β aufzulösen scheint nicht möglich. Maple 14 liefert:

```
> solve((18),beta) assuming 0<beta,beta<Pi/2;
```

$$RootOf\left(-4\sin\left(\frac{1}{2}\_Z\right)\cos\left(\frac{1}{2}\_Z\right) - 2\_Z - 4\pi\cos\left(\frac{1}{2}\_Z\right)^2 + 3\pi + 4\_Z\cos\left(\frac{1}{2}\_Z\right)^2\right) \quad \textbf{(19)}$$

Eine numerische Lösung der Gleichung ist möglich.

```
> beta = fsolve((18),beta);
```

$$\beta = 1.235896924279909343579976946 12 \quad \textbf{(20)}$$

Bei einer Wiese mit Radius

```
> a = 15*Unit(m);
```

$$a = 15 \,[\![m]\!] \quad \textbf{(21)}$$

folgt aus Gleichung **(11)** der Bewegungsradius der Kuh

```
> eval((11),{(20),(21)});
```

$$b = 17.3809270953 \,[\![m]\!] \quad \textbf{(22)}$$

```
>
```