```
> restart;
> swap:=x->rhs(x)=lhs(x):
```

**Schwerpunkt und Massenträgheitsmoment eines halben Hohlzylinders.**

Masse des Zylinders m.
Masse konzentriert im Zylindermantel.
Radius des Zylinders r.

Koordinatensystem: Z-Achse auf die Symmetrieachse des zu einem ganzen Zylinders ergänzten halben Zylinder.
Nullpunkt des Koordinatensystem in den Schwerpunkt dieses ergänzten Zylinders.

Länge der Linie, die die Stirnfläche des Zylinders bildet. Das ist die Länge des halben Kreisumfangs.

```
> L=Pi*r;
```

$$L = \pi r \tag{1}$$

Liniendichte des Zylinders

```
> rho = m/L;
```

$$\rho = \frac{m}{L} \tag{2}$$

```
> subs((1),(2));
```

$$\rho = \frac{m}{\pi r} \tag{3}$$

Parametrisierung der Linie über den Winkel α=0..π.

```
> x = r*cos(alpha);
```

$$x = r \cos(\alpha) \tag{4}$$

```
> y=r*sin(alpha);
```

$$y = r \sin(\alpha) \tag{5}$$

Darstellung der Stirnseite des halben Holzylinders mit Koordinatensystem.

```
> plot(subs(r=1,[rhs((4)),rhs((5)),alpha=0..Pi]),labels=[x/r,y/r],
  scaling=constrained,axes=boxed);
```




**Schwerpunkt des halben Hohlzylinders.**

Anschaulich ist, dass die Z-Koordinate des Schwerpunkts in der Mitte liegt, den gleichen Wert hat, wie beim ganzen Hohlzylinder.

```
> Z[S] = 0;
```

$$Z_S = 0 \tag{6}$$

Die X-Koordinate des Schwerpunkts muss nach dem Plot $X_S = 0$ sein. Zur Übung die X-Koordinate des Schwerpunkts über das Integral berechnen.

```
> X[S] = (1/m)*Int(x*rho,V);
```

$$X_S = \frac{\int x\,\rho\,\mathrm{d}V}{m} \tag{7}$$

Das Volumen ist auf eine Linie reduziert. Daher das Linienintegral berechnen. (Mathematisch exakter kann dieser Schritt über eine Dirac δ-Funktion gerechnet werden.)

```
> X[S] = (1/m)*int(x(alpha)*rho*sqrt(diff(x(alpha),alpha)^2+diff(y
  (alpha),alpha)^2),alpha=0..Pi);
```

**(8)**

$$X_S = \frac{\int_0^{\pi} x(\alpha)\,\rho\,\sqrt{\left(\frac{d}{d\alpha}\,x(\alpha)\right)^2 + \left(\frac{d}{d\alpha}\,y(\alpha)\right)^2}\,d\alpha}{m} \tag{8}$$

Die Ableitungen von **(4)** und **(5)**.

```
> diff(subs(x=x(alpha),(4)),alpha);
```

$$\frac{d}{d\alpha}\,x(\alpha) = -r\sin(\alpha) \tag{9}$$

```
> diff(subs(y=y(alpha),(5)),alpha);
```

$$\frac{d}{d\alpha}\,y(\alpha) = r\cos(\alpha) \tag{10}$$

Die Wurzel im Integranden berechnen.

```
> Delta=sqrt(diff(x(alpha),alpha)^2+diff(y(alpha),alpha)^2);
```

$$\Delta = \sqrt{\left(\frac{d}{d\alpha}\,x(\alpha)\right)^2 + \left(\frac{d}{d\alpha}\,y(\alpha)\right)^2} \tag{11}$$

```
> subs((9),(10),(11));
```

$$\Delta = \sqrt{r^2\sin(\alpha)^2 + r^2\cos(\alpha)^2} \tag{12}$$

```
> simplify((12)) assuming r>0;
```

$$\Delta = r \tag{13}$$

Das Integral ausrechnen.

```
> subs(subs((11),(13)),(8));
```

$$X_S = \frac{\int_0^{\pi} x(\alpha)\,\rho\,r\,d\alpha}{m} \tag{14}$$

```
> subs(subs(x=x(alpha),(4)),(14));
```

$$X_S = \frac{\int_0^{\pi} r^2\cos(\alpha)\,\rho\,d\alpha}{m} \tag{15}$$

```
> simplify((15));
```

$$X_S = 0 \tag{16}$$

Das erwartete Ergebnis.

Die Y-Koordinate des Schwerpunkts über das Integral berechnen.

```
> Y[S] = (1/m)*Int(y*rho,V);
```

$$Y_S = \frac{\int y\,\rho\,dV}{m} \tag{17}$$

Das Volumen ist auf eine Linie reduziert, das Linienintegral berechnen.

```
> Y[S] = (1/m)*int(y(alpha)*rho*sqrt(diff(x(alpha),alpha)^2+diff(y
  (alpha),alpha)^2),alpha=0..Pi);
```

$$Y_S = \frac{\int_0^{\pi} y(\alpha)\,\rho\,\sqrt{\left(\frac{d}{d\alpha}\,x(\alpha)\right)^2 + \left(\frac{d}{d\alpha}\,y(\alpha)\right)^2}\,d\alpha}{m} \tag{18}$$

```
> subs(subs((11),(13)),(18));
```

$$Y_S = \frac{\int_0^{\pi} y(\alpha)\,\rho\,r\,d\alpha}{m} \tag{19}$$

```
> subs(subs(y=y(alpha),(5)),(19));
```

$$Y_S = \frac{\int_0^{\pi} r^2 \sin(\alpha)\,\rho\,d\alpha}{m} \tag{20}$$

```
> simplify((20));
```

$$Y_S = \frac{2\,r^2\,\rho}{m} \tag{21}$$

Die Liniendichte aus **(3)** einsetzen.

```
> subs((3),(21));
```

$$Y_S = \frac{2\,r}{\pi} \tag{22}$$

Zahlenwert ausrechnen.

```
> evalf((22));
```

$$Y_S = 0.6366197722\,r \tag{23}$$

Der Schwerpunkt des halben Hohlzylinders ist

$(X_S, Y_S, Z_S) = \left(0, \frac{2\,r}{\pi}, 0\right)$.

**Das Massenträgheitsmoment für die Rotation um die Z-Achse des Koordinatensystems.**

(Nicht um eine Achse durch den Schwerpunkt.) Das Massenträgheitsmoment des ganzen Holzylinders ist im Tabellenbuch mit $m\,r^2$ angegeben. Der halbe Hohlzylinder sollte den halben Wert haben.

```
> J[Z] = Int(R^2*rho,V);
```

$$J_Z = \int R^2\,\rho\,dV \tag{24}$$

Darin ist r der Abstand zur Rotationsachse.

```
> R(x,y)=sqrt(x^2+y^2);
```

$$R(x, y) = \sqrt{x^2 + y^2} \tag{25}$$

Das Integral reduziert sich auf ein Linienintegral über die Wand des Zylinders.

```
> J[Z] = int(R(alpha)^2*rho*sqrt(diff(x(alpha),alpha)^2+diff(y
  (alpha),alpha)^2),alpha=0..Pi);
```

**(26)**

$$J_Z = \int_0^{\pi} R(\alpha)^2 \rho \sqrt{\left(\frac{d}{d\alpha} x(\alpha)\right)^2 + \left(\frac{d}{d\alpha} y(\alpha)\right)^2}\, d\alpha \tag{26}$$

Der Abstand des Mantels zur Rotationsachse ist r.

```
> subs(R(alpha)=r,(26));
```

$$J_Z = \int_0^{\pi} r^2 \rho \sqrt{\left(\frac{d}{d\alpha} x(\alpha)\right)^2 + \left(\frac{d}{d\alpha} y(\alpha)\right)^2}\, d\alpha \tag{27}$$

Die Länge des Linienelements **(13)** einsetzen.

```
> subs(subs((11),(13)),(27));
```

$$J_Z = \int_0^{\pi} r^3 \rho \, d\alpha \tag{28}$$

```
> simplify((28));
```

$$J_Z = r^3 \rho \pi \tag{29}$$

Die Liniendichte aus **(3)** einsetzen.

```
> subs((3),(29)): sort(%);
```

$$J_Z = m r^2 \tag{30}$$

Das Massenträgheitsmoment ist die Hälfte des ganzen Hohlzylinders, weil der ganze Hohlzylinder die Masse 2m hat.

**Massenträgheitsmoment um eine Achse durch den Schwerpunkt in Z-Richtung.**

Die Achse durch den Schwerpunkt ist gegenüber der Z-Achse um die Strecke 2r/π verschoben.
Mit dem Steinerschen Satz.

```
> J[S] = J[Z] - m*d^2;
```

$$J_S = J_Z - m d^2 \tag{31}$$

Wobei d der Abstand Achse - Schwerpunkt ist.

```
> subs(Y[S]=d,(22));
```

$$d = \frac{2 r}{\pi} \tag{32}$$

Massenträgheitsmoment um die Z-Achse **(30)** und die Verschiebung **(32)** in den Steinerschen Satz einsetzen.

```
> subs((30),(32),(31));
```

$$J_S = m r^2 - \frac{4 m r^2}{\pi^2} \tag{33}$$

```
> simplify((33),size);
```

$$J_S = \frac{m r^2 (\pi^2 - 4)}{\pi^2} \tag{34}$$

Der Zahlenwert.

```
> evalf((34));
```

$$J_S = 0.5947152656\, m\, r^2 \tag{35}$$

Hilfsmittel:
- Maple 14
- Hering, Martin, Stohrer: Physikalisch-Technisches Taschenbuch, VDI-Verlag