```
> restart;
> Digits:=24: interface( displayprecision=3 ):
```

Berechnet werden soll ein Wurf auf einer fahrenden Rolltreppe. Die werfende Person und das Ziel befinden sich auf der Rolltreppe und bewegen sich mit konstanter Geschwindigkeit. Wird der Luftwiderstand vernachlässigt, dann kann in das mitbewegte Koordinatensystem transformiert werden. Weil die Transformation mit einer gleichförmigen Geschwindigkeit erfolgt, treten keine Scheinkräfte auf.

Vereinfache Skizze der Situation:



Der Wurf findet in einer Ebene statt.
Koordinatensystem definieren: x-Achse nach rechts, y-Achse nach oben, Nullpunkt in den Zielpunkt.
Zeitnullpunkt gleich dem Abwurf.

Betrag der Fallbeschleunigung in -y Richtung.

```
> g = 9.81 * Unit(m/s^2); params:=%:
```

$$g = 9.810 \left[\!\left[\frac{m}{s^2} \right]\!\right] \tag{1}$$

Die Koordinaten des Zielpunkts.

```
> Z = <0,0>*Unit(m);
```

$$Z = \begin{bmatrix} 0 \\ 0 \end{bmatrix} \tag{2}$$

Die Koordinaten des Abwurfpunkts.

```
> A = <4,3>*Unit(m); params:=%,params:
```

$$A = \begin{bmatrix} 4 \left[\!\left[m \right]\!\right] \\ 3 \left[\!\left[m \right]\!\right] \end{bmatrix} \tag{3}$$

Der Winkel der Flugbahn am Zielpunkt, also der Winkel zwischen den Geschwindigkeitskomponenten.

```
> alpha = 10 * Unit(deg);
```

(4)

$$\alpha = 10 \; [\![arcdeg]\!] \tag{4}$$

```
> evalf(simplify((4))); params:=%,params:
```

$$\alpha = 0.175 \; [\![rad]\!] \tag{5}$$

Die Abwurfgeschwindigkeit.

```
> vs = <vs[1],vs[2]>;
```

$$vs = \begin{bmatrix} vs_1 \\ vs_2 \end{bmatrix} \tag{6}$$

Die Geschwindigkeit zum Zeitpunkt t.

```
> v(t) = vs + <0,-g*t>;
```

$$v(t) = vs + \begin{bmatrix} 0 \\ -g\,t \end{bmatrix} \tag{7}$$

```
> subs((6),(7)): simplify(%);
```

$$v(t) = \begin{bmatrix} vs_1 \\ vs_2 - g\,t \end{bmatrix} \tag{8}$$

Position des Balls zum Zeitpunkt t.

```
> P(t) = A + vs*t + <0,-g*t^2/2>;
```

$$P(t) = A + vs\,t + \begin{bmatrix} 0 \\ -\frac{1}{2}\,g\,t^2 \end{bmatrix} \tag{9}$$

```
> subs(A = <A[1],A[2]>,(6),(9)): simplify(%);
```

$$P(t) = \begin{bmatrix} A_1 + t\,vs_1 \\ A_2 + t\,vs_2 - \frac{1}{2}\,g\,t^2 \end{bmatrix} \tag{10}$$

Zum Zeitpunkt T wird das Ziel erreicht.

```
> Z = P(T);
```

$$Z = P(T) \tag{11}$$

```
> Z = rhs(subs(t=T,(10)));
```

$$Z = \begin{bmatrix} A_1 + T\,vs_1 \\ A_2 + T\,vs_2 - \frac{1}{2}\,g\,T^2 \end{bmatrix} \tag{12}$$

Für den Zielpunkt wurde der Koordinatenursprung gewählt. Die Zahlenwerte sofort einsetzen vereinfacht die folgende Rechnung.

```
> subs((2),(12));
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} = \begin{bmatrix} A_1 + T\,vs_1 \\ A_2 + T\,vs_2 - \frac{1}{2}\,g\,T^2 \end{bmatrix} \tag{13}$$

Die Geschwindigkeit zum Zeitpunkt T.

```
> subs(t=T,(8));
```

$$v(T) = \begin{bmatrix} vs_1 \\ vs_2 - g\,T \end{bmatrix} \tag{14}$$

Der Winkel ist gegeben. (Beide Geschwindigkeitskomponenten sind beim Treffen des Ziels negativ. Siehe Skizze).

```
> tan(alpha) = v(T)[1]/v(T)[2];
```

$$\tan(\alpha) = \frac{v(T)_1}{v(T)_2} \tag{15}$$

```
> subs((14),(15)): simplify(%);
```

$$\tan(\alpha) = -\frac{vs_1}{-vs_2 + g\,T} \tag{16}$$

Die Gleichungen **(13)** und **(16)** bilden ein Gleichungssystem für die drei Unbekannten T, vs_1 und vs_2. Alle anderen Größen in diesen Gleichungen sind bekannt. Das Gleichungssystem zusammengefasst:

```
> lhs((13))[1]=rhs((13))[1];
```

$$0 = A_1 + T\,vs_1 \tag{17}$$

```
> lhs((13))[2]=rhs((13))[2];
```

$$0 = A_2 + T\,vs_2 - \frac{1}{2}\,g\,T^2 \tag{18}$$

```
> (16);
```

$$\tan(\alpha) = -\frac{vs_1}{-vs_2 + g\,T} \tag{19}$$

Gleichung **(17)** nach T auflösen und in **(18)** und **(19)** einsetzen.

```
> isolate((17),T);
```

$$T = -\frac{A_1}{vs_1} \tag{20}$$

```
> subs((20),(18));
```

$$0 = A_2 - \frac{A_1\,vs_2}{vs_1} - \frac{1}{2}\,\frac{g\,A_1^2}{vs_1^2} \tag{21}$$

```
> subs((20),(19));
```

$$\tan(\alpha) = -\frac{vs_1}{-vs_2 - \frac{g\,A_1}{vs_1}} \tag{22}$$

```
> simplify((22));
```

$$\tan(\alpha) = \frac{vs_1^2}{vs_2\,vs_1 + g\,A_1} \tag{23}$$

Gleichung **(23)** mach vs_2 auflösen und in Gleichung **(21)** einsetzen.

```
> isolate((23),vs[2]);
```

$$vs_2 = \frac{\frac{vs_1^2}{\tan(\alpha)} - g\,A_1}{vs_1} \quad \textbf{(24)}$$

```
> subs((24),(21));
```

$$0 = A_2 - \frac{A_1\left(\frac{vs_1^2}{\tan(\alpha)} - g\,A_1\right)}{vs_1^2} - \frac{1}{2}\,\frac{g\,A_1^2}{vs_1^2} \quad \textbf{(25)}$$

Die Gleichung **(25)** enthält nur noch die Unbekannte vs_1. Nach der Unbekannten auflösen.

```
> expand((25));
```

$$0 = A_2 - \frac{A_1}{\tan(\alpha)} + \frac{1}{2}\,\frac{g\,A_1^2}{vs_1^2} \quad \textbf{(26)}$$

```
> isolate((26),vs[1]^2 );
```

$$vs_1^2 = -\frac{1}{2}\,\frac{g\,A_1^2}{A_2 - \frac{A_1}{\tan(\alpha)}} \quad \textbf{(27)}$$

Aus der Skizze ist ersichtlich, dass die x-Komponente der Geschwindigkeit vs_1 negativ sein muss.

```
> vs[1]=-sqrt(rhs((27)));
```

$$vs_1 = -\frac{1}{2}\sqrt{-\frac{2\,g\,A_1^2}{A_2 - \frac{A_1}{\tan(\alpha)}}} \quad \textbf{(28)}$$

Die Zahlenwerte in **(28)** einsetzen liefern die x-Komponente der Abwurfgeschwindigkeit.

```
> subs(params,(28)): simplify(%);
```

$$vs_1 = -1.997\,\left[\!\left[\frac{m}{s}\right]\!\right] \quad \textbf{(29)}$$

Gleichung **(24)** liefert die y-Komponente der Abwurfgeschwindigkeit.

```
> subs(params,(29),(24)): simplify(%);
```

$$vs_2 = 8.329\,\left[\!\left[\frac{m}{s}\right]\!\right] \quad \textbf{(30)}$$

Gleichung **(20)** liefert die Zeit für den Wurf.

```
> subs(params,(29),(30),(20)): simplify(%);
```

$$T = 2.003\,[\![s]\!] \quad \textbf{(31)}$$

Der Betrag der Abwurfgeschwindigkeit.

```
> vs = sqrt( vs[1]^2 + vs[2]^2 );
```

$$vs = \sqrt{vs_1^2 + vs_2^2} \quad \textbf{(32)}$$

```
> eval((32),{(29),(30)}): simplify(%);
```

$$vs = 8.565\,\left[\!\left[\frac{m}{s}\right]\!\right] \quad \textbf{(33)}$$

Der Winkel beim Abwurf zur Horizontalen, wie in der Aufgabenstellung eingezeichnet.

```
> tan(beta)=-vs[1]/vs[2];
```

$$\tan(\beta) = -\frac{vs_1}{vs_2} \tag{34}$$

```
> solve((34),{beta})[1];
```

$$\beta = -\arctan\left(\frac{vs_1}{vs_2}\right) \tag{35}$$

```
> lhs((35))=eval(rhs((35)),{(29),(30)})*Unit(rad);
```

$$\beta = 0.235 \; [\![rad]\!] \tag{36}$$

```
> lhs((36)) = convert(rhs((36)),'units','arcdeg');
```

$$\beta = 13.481 \; [\![arcdeg]\!] \tag{37}$$

Der Ball muss mit 8,6 m/s (relativ zur Rolltreppe) unter einem Winkel von 13° (zur Horizontalen) abgeworfen werden.

Die Bahnkurve **(10)** mit gefundenen Werten.

```
> subs(params,(29),(30),(10));
```

$$P(t) = \begin{bmatrix} 4 \; [\![m]\!] - 1.997 \, t \left[\!\left[\frac{m}{s} \right]\!\right] \\ 3 \; [\![m]\!] + 8.329 \, t \left[\!\left[\frac{m}{s} \right]\!\right] - 4.905 \left[\!\left[\frac{m}{s^2} \right]\!\right] t^2 \end{bmatrix} \tag{38}$$

```
> subs(t=x*Unit(s),rhs((38)))/Unit(m): simplify(%):
> toplot := [%[1],%[2],x=0..2.003]:
> plot(toplot,title='Wurfparabel',labels=['x/m','y/m']);
```



(Erstellt mit Maple 17.)