

```
> restart;
> Digits:= 25: interface(displayprecision=7):
```

Aufgabe

In einem gut isolierten Zylinder-Kolben-System wird Stickstoff ($m=1,15\text{g}$ $V_1=0,001\text{m}^3$) vom Umgebungszustand ($p_1=1,0\text{bar}$, $\vartheta_1=20^\circ\text{C}$) auf 30% des Ausgangsvolumens verdichtet. Der Druck im Zylinder steigt dabei auf das 5-fache des Ausgangsdruckes an.

Die spezifische isobare Wärmekapazität von Luft (Stickstoff) im idealen Gaszustand ist mit

Die spezifische isobare Wärmekapazität von Luft (Stickstoff) im idealen Gaszustand ist mit

```
> c[p](T)=K[0]+K[1]*T+K[2]*T^2+K[3]*T^3;
```

$$c_p(T) = K_0 + K_1 T + K_2 T^2 + K_3 T^3 \quad (1)$$

und den Koeffizienten

```
> K[0] = 1112.0*eval(Unit(J/(kg*K)),1),
> K[1] = -0.484*eval(Unit(J/(kg*K^2)),1),
> K[2] = 9.567E-4*eval(Unit(J/(kg*K^3)),1),
> K[3] = -4.169E-7*eval(Unit(J/(kg*K^4)),1);
```

$$K_0 = 1112.0 \left[\frac{\text{J}}{\text{kg K}} \right], K_1 = -0.484 \left[\frac{\text{J}}{\text{kg K}^2} \right], K_2 = 0.0009567 \left[\frac{\text{J}}{\text{kg K}^3} \right], K_3 = -4.169 \cdot 10^{-7} \left[\frac{\text{J}}{\text{kg K}^4} \right] \quad (2)$$

gegeben.

Die individuelle Gaskonstante von Stickstoff beträgt

```
> R[i] = eval( 296.8*Unit(J/(kg*K)), 1 );
```

$$R_i = 296.8 \left[\frac{\text{J}}{\text{kg K}} \right] \quad (3)$$

Berechne die Temperatur T_2 und die im Verlaufe der Verdichtung verrichtete Arbeit

Rechenweg

Weil der Zylinder gut isoliert ist, kann der Prozess vereinfacht als adiabat angenommen werden.

Wird weiter Reibung vernachlässigt, dann ist der Prozess reversibel.

Zusammen: Der Prozess ist isentrop.

Das System ist geschlossen. Die Masse des eingeschlossenen Gases ist konstant

```
> m = 1.15*Unit(g); simplify(%);
```

$$m = 1.15 \left[\text{g} \right]$$

$$m = 0.001150000 \left[\text{kg} \right] \quad (4)$$

Vom Anfangszustand (Index 1) ist Volumen V_1 , Druck p_1 und Temperatur T_1 gegeben.

```
> v[1] = 0.001*Unit(m^3);
```

$$V_1 = 0.001 \left[\text{m}^3 \right]$$

$$(5)$$

```
> p[1] = 1*Unit(bar): simplify(%);
```

$$p_1 = 100000 \left[\text{Pa} \right]$$

$$(6)$$

```
> T[1] = (273.15+20)*Unit(K);
```

$$T_1 = 293.15 \left[\text{K} \right]$$

$$(7)$$

Alle 3 Zustandsgrößen sind gegeben. Die Zustandsgleichung des idealen Gases kann zur Kontrolle der Konsistenz der Angaben verwendet werden.

> p[1]*V[1] = m*R[i]*T[1];

$$p_1 V_1 = m R_i T_1 \quad (8)$$

Einsetzen, ausrechnen, vergleichen.

> subs((3),(4),(5),(6),(7), (8)) : simplify(%);

$$100.000 \llbracket J \rrbracket = 100.0580 \llbracket J \rrbracket \quad (9)$$

Vom Endzustand (Index 2) ist Volumen V_2 und Druck p_2 gegeben.

> V[2]=V[1]*0.3; subs((5),%);

$$V_2 = 0.3 V_1$$

$$V_2 = 0.0003 \llbracket m^3 \rrbracket \quad (10)$$

> p[2]=5*p[1]; subs((6),%);

$$p_2 = 5 p_1$$

$$p_2 = 500000 \llbracket Pa \rrbracket \quad (11)$$

Die Temperatur T_2 kann über die Zustandsgleichung des idealen Gases berechnet werden.

> p[2]*V[2] = m*R[i]*T[2]; isolate(%,T[2]);

$$p_2 V_2 = m R_i T_2$$

$$T_2 = \frac{p_2 V_2}{m R_i} \quad (12)$$

Einsetzen der bekannten Größen und ausrechnen.

> subs((3),(4),(5),(10),(11), (12)) : simplify(%);

$$T_2 = 439.4703 \llbracket K \rrbracket \quad (13)$$

> theta[2]=convert(rhs((13)), 'temperature', degC);

$$\theta_2 = 166.3203 \llbracket degC \rrbracket \quad (14)$$

Die Temperatur im Endzustand beträgt 439 Kelvin oder 166 °C.

Für die Volumenänderungsarbeit des idealen Gases im geschlossenen System beim isentropen Prozess steht im Buch von Cerbe und Wilhelms die Gleichung:

> W[v] = m * c[v,T1,T2] * (T[2]-T[1]);

$$W_v = m c_{v, T1, T2} (T_2 - T_1) \quad (15)$$

Der Mittelwert der spezifischen isochoren Wärmekapazität ist zu bestimmen.

> c[v,T1,T2] = int(c[v](T), T=T[1]..T[2]) / (T[2]-T[1]);

$$c_{v, T1, T2} = \frac{\int_{T_1}^{T_2} c_v(T) dT}{T_2 - T_1} \quad (16)$$

Die Beziehung zwischen isobarer und isochorer spezifischer Wärmekapazität

> c[p](T) - c[v](T) = R[i]; isolate(%,c[v](T));

$$c_p(T) - c_v(T) = R_i$$

$$c_v(T) = -R_i + c_p(T) \quad (17)$$

Einsetzen in (16).

> subs((17),(16)); c[v,T1,T2] = -R[i] + int(c[p](T),T=T[1]..T[2])/(T[2]-T[1]);

$$c_{v, T1, T2} = \frac{\int_{T_1}^{T_2} (-R_i + c_p(T)) dT}{T_2 - T_1}$$

$$c_{v, T1, T2} = -R_i + \frac{\int_{T_1}^{T_2} c_p(T) dT}{T_2 - T_1} \quad (18)$$

Das Polynom aus der Aufgabenstellung (1) einsetzen.

> subs((1),(18));

$$c_{v, T1, T2} = -R_i + \frac{\int_{T_1}^{T_2} (K_0 + K_1 T + K_2 T^2 + K_3 T^3) dT}{T_2 - T_1} \quad (19)$$

Integrieren.

> expand((19));

$$c_{v, T1, T2} = -R_i + \frac{K_0 T_2}{T_2 - T_1} - \frac{K_0 T_1}{T_2 - T_1} + \frac{K_1 T_2^2}{2 (T_2 - T_1)} - \frac{K_1 T_1^2}{2 (T_2 - T_1)} + \frac{K_2 T_2^3}{3 (T_2 - T_1)} - \frac{K_2 T_1^3}{3 (T_2 - T_1)} + \frac{K_3 T_2^4}{4 (T_2 - T_1)} - \frac{K_3 T_1^4}{4 (T_2 - T_1)} \quad (20)$$

Zahlenwert ausrechnen durch Einsetzen der gegebenen Koeffizienten und bekannten Temperaturen.

> subs((2),(3),(7),(13),(20)): simplify(%): lhs(%)= convert(rhs(%), 'units', 'J/(kg*K)');

$$c_{v, T1, T2} = 746.6766 \left[\frac{J}{kg K} \right] \quad (21)$$

Damit sind alle Zahlenwerte für die Gleichung (15) berechnet.

Die Gleichung (15) liefert

> subs((4),(7),(13),(21),(15)): simplify(%);

$$W_v = 125.6420 [J] \quad (22)$$

Die Volumenänderungsarbeit beträgt 126 Joule.

Hilfsmittel

- Cerbe, Wilhelms: Technische Thermodynamik, Hanser-Verlag
- Maple 14, <http://www.maplesoft.com/>