

```

> restart;
> Digits := trunc(evalhf(Digits)):
> interface(displayprecision=5):

```

## ▼ Aufgabe

Die Luft in einem Behälter mit konstantem Volumen und kleiner Öffnung soll betrachtet werden. Der Behälter ist in einer Umgebung, deren Luftdruck fällt. Anfangs ist der Behälter mit Luft im Standardzustand gefüllt. Der Umgebungsdruck ist ebenfalls 1 bar. Dann wird der Umgebungsdruck mit konstant -10 mbar/s gesenkt. Nach 100 s ist der Umgebungsdruck 0 und bleibt auf 0.

Gesucht ist die Druckdifferenz zwischen dem Luftdruck im Behälter und dem Umgebungsdruck.

## ▼ Hilfsmittel

- Maple 17
- Cerbe und Wilhelms: Technisches Thermodynamik, Hanser Verlag
- Meschede: Gerthsen Physik, Springer Verlag
- Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch

## ▼ Parameter

Volumen des Behälters

```

> V=0.001*Unit(m^3); params:=%:

```

$$V = 0.001 \text{ [m}^3\text{]} \quad (1)$$

Querschnitt der Öffnung mit 1 mm Radius

```

> A = evalf(Pi*(1*Unit(mm))^2); params:=params,%:

```

$$A = 3.141592654 \text{ [mm]}^2 \quad (2)$$

Anfangszustand: Temperatur

```

> T[a]=298.15*Unit(K); params:=params,%:

```

$$T_a = 298.15 \text{ [K]} \quad (3)$$

Anfangszustand: Druck im Behälter und Umgebungsdruck

```

> P[a]=1e5*Unit(Pa); params:=params,%:

```

$$P_a = 100000. \text{ [Pa]} \quad (4)$$

Der vorgegebene Abfall des Umgebungsdrucks

```

> S=-1e5/100*Unit(Pa/s): params:=params,%: lhs(%%)=convert(rhs(%%),units,Pa/s);

```

$$S = -1000.000000 \left[ \frac{\text{Pa}}{\text{s}} \right] \quad (5)$$

Stoffeigenschaft: individuelle Gaskonstante der Luft (beim Normzustand)

```

> R[i] = 287.2*Unit(J/(kg*K)): params:=params,%: lhs(%%)=convert(rhs(%%),units,J/(kg*K));

```

$$R_i = 287.2 \left[ \frac{\text{J}}{\text{kg K}} \right] \quad (6)$$

## ▼ 1. Schritt: stationär und inkompressibel, T und P<sub>u</sub> konstant

Vereinfachungen:

- Luft über die Zustandsgleichung des idealen Gases beschreiben. Das ist bis hinab zu einem Druck

von 1 mbar eine sehr gute Beschreibung der Luft.

- Umgebungsdruck  $P_u$  ist konstant. Aber der Umgebungsdruck kann ungleich dem Anfangsdruck  $P_a$  im Behälter sein.
- Temperatur  $T=T_a$  des Gases im Behälter ist konstant, Wärme wird dem Gas im Behälter ausreichend zugeführt.
- Ausströmgeschwindigkeit  $c$  über Bernoulli-Gleichung abschätzen.

Bernoulli-Gleichung für stationäre, reibungsfreie Strömung eines inkompressiblen Fluids.

>  $p + \rho \cdot c^2 / 2 + \rho \cdot g \cdot h = \text{constant};$

$$p + \frac{1}{2} \rho c^2 + \rho g h = \text{constant} \quad (7)$$

Darin ist  $p$  der Druck,  $\rho$  die Dichte,  $c$  die Strömungsgeschwindigkeit,  $g$  die Fallbeschleunigung und  $h$  die Höhe.

Die Gleichung im Behälter und am Ende der Öffnung anwenden.

- Im Behälter die Strömungsgeschwindigkeit vernachlässigen. Es wird davon ausgegangen, dass die Luft aus dem Behälter strömt.
- Die Dichte der Luft im Behälter wird verwendet, weil das Gas aus dem Behälter kommt und das Gas als inkompressible angenommen wurde.
- Höhenunterschiede vernachlässigen.
- Die Druckänderung im Behälter im Laufe der Zeit vernachlässigen. Damit wird die Strömung stationär.

Die Druckdifferenz ist  $\Delta(t)$ . Der Innendruck ist  $P_u + \Delta$ .

>  $P[u] + \Delta(t) = P[u] + \rho \cdot c^2 / 2;$

$$P_u + \Delta(t) = P_u + \frac{\rho c^2}{2} \quad (8)$$

Auflösen nach der Ausströmgeschwindigkeit  $c$ . Bei der Wurzel wird davon ausgegangen, dass die Strömungsgeschwindigkeit positiv ist.

> `solve((8),{c}): simplify(%[1][1]) assuming rho::positive: combine(% ,radical ,symbolic) ;`

$$c = \sqrt{2} \sqrt{\frac{\Delta(t)}{\rho}} \quad (9)$$

Die Dichte ist abhängig vom Druck. Die Zustandsgleichung des idealen Gases.

>  $p \cdot V = m \cdot R_i \cdot T;$

$$p V = m R_i T \quad (10)$$

Den Druck im Behälter einsetzen.

> `subs (p=P[u]+Delta (t) ,(10)) ;`

$$(P_u + \Delta(t)) V = m R_i T \quad (11)$$

Auflösen nach der Masse.

> `isolate ((11),m) ;`

$$m = \frac{(P_u + \Delta(t)) V}{R_i T} \quad (12)$$

Die Dichte ist allgemein

>  $\rho = m/V;$

$$\rho = \frac{m}{V} \quad (13)$$

Die Masse aus (12) einsetzen.

> subs ((12), (13)) ;

$$\rho = \frac{P_u + \Delta(t)}{R_i T} \quad (14)$$

Die berechnete Dichte in die Formel (9) für die Auströmgeschwindigkeit einsetzen.

> subs ((14), (9)) ;

$$c = \sqrt{2} \sqrt{\frac{R_i T \Delta(t)}{P_u + \Delta(t)}} \quad (15)$$

Die Strömungsgeschwindigkeit ist neben dem Druck auch von der Temperatur abhängig. Für eine korrekte Berechnung ist eine korrekte Berechnung der Temperaturfunktion T(t) notwendig. In diesem Schritt wird vereinfacht die Temperatur konstant angenommen.

Den Massenstrom durch die Fläche A berechnen. Die Masse m ist die Gasmasse im Behälter. Die Masse strömt aus dem Behälter, daher das negative Vorzeichen.

> diff (m(t), t) = -c\*rho\*A;

$$\dot{m}(t) = -c \rho A \quad (16)$$

Die Abhängigkeit zwischen Masse und Druck ist aus (12) bekannt. Die Gleichung ableiten.

> subs (m=m(t), (12)) : diff(%, t) ;

$$\dot{m}(t) = \frac{\dot{\Delta}(t) V}{R_i T} \quad (17)$$

Der Umgebungsdruck  $P_u$  entfällt beim Ableiten, weil der Umgebungsdruck in diesem Schritt als konstant angenommen wird.

> isolate ((17), diff(Delta(t), t)) ;

$$\dot{\Delta}(t) = \frac{\dot{m}(t) R_i T}{V} \quad (18)$$

Den Massenstrom (16) einsetzen.

> subs ((16), (18)) ;

$$\dot{\Delta}(t) = - \frac{c \rho A R_i T}{V} \quad (19)$$

Die Dichte mit (14) durch den Druck ausdrücken.

> subs ((14), (19)) ;

$$\dot{\Delta}(t) = - \frac{c (P_u + \Delta(t)) A}{V} \quad (20)$$

Die Strömungsgeschwindigkeit (15) einsetzen.

> subs ((15), (20)) ;

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \sqrt{\frac{R_i T \Delta(t)}{P_u + \Delta(t)}} (P_u + \Delta(t)) A}{V} \quad (21)$$

Der Druck im Behälter ist immer größer 0, daher kann  $P_u + \Delta$  quadriert in die Wurzel gezogen werden.

```
> simplify((21)) assuming P[u]+Delta(t) > 0;
```

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \sqrt{P_u + \Delta(t)} \sqrt{R_i T \Delta(t)} A}{V} \quad (22)$$

Damit ist die Differenzialgleichung für die Druckdifferenz aufgestellt. Alle anderen Größen in der Gleichung sind konstant und bekannt.

```
> # dsolve({(22),Delta(0)=P[a]-P[u]});
```

### Spezialfall $P_u=0$

Wird der Umgebungsdruck konstant auf 0 gesetzt, wird die Differenzialgleichung analytisch lösbar.

```
> P[u]=0;
```

$$P_u = 0 \quad (23)$$

In die Gleichung (22) einsetzen.

```
> subs((23),(22));
```

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \sqrt{\Delta(t)} \sqrt{R_i T \Delta(t)} A}{V} \quad (24)$$

```
> simplify((24)) assuming Delta(t)::positive;
```

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \Delta(t) \sqrt{R_i T} A}{V} \quad (25)$$

Diese Differenzialgleichung ist analytisch lösbar.

```
> dsolve({(25),Delta(0)=P[a]});
```

$$\Delta(t) = P_a e^{-\frac{\sqrt{2} \sqrt{R_i T} A t}{V}} \quad (26)$$

Die Zeitkonstante des Druckabfalls ist abhängig von der Behältergeometrie (Volmen V und Öffnungsquerschnitt A) und von der Lufttemperatur T (die in dem hier verwendeten Modell konstant ist).

```
> log(rhs((26))/P[a]): simplify(%) assuming positive: combine(%,
radical,symbolic): tau=-t/%;
```

$$\tau = \frac{\sqrt{2} V}{2 \sqrt{R_i T} A} \quad (27)$$

```
> subs(T=T[a],params,(27)): simplify(%) ;
```

$$\tau = 0.76918 \llbracket s \rrbracket \quad (28)$$

Die Exponentialfunktion mit eingesetzter Zeitkonstante ist deutlich übersichtlicher.

```
> algsols(isolate((27),A),(26)): simplify(%) assuming positive;
```

$$\Delta(t) = P_a e^{-\frac{t}{\tau}} \quad (29)$$

Die Druckdifferenz  $\Delta$ , die in diesem Spezialfall gleich dem Druck im Behälter ist, fällt mit

```
> diff((29),t);
```

$$\dot{\Delta}(t) = - \frac{P_a e^{-\frac{t}{\tau}}}{\tau} \quad (30)$$

Im Zähler steht die Druckdifferenz.

```
> algsubs(rhs((29))=lhs((29)),(30));
```

$$\dot{\Delta}(t) = -\frac{\Delta(t)}{\tau} \quad (31)$$

Der maximale Druckabfall ist bei  $t=0$ , wenn die Druckdifferenz maximal ist.

```
> lhs((30))[max] = subs(t=0,rhs((30))): simplify(%);
```

$$\dot{\Delta}(t)_{\max} = -\frac{P_a}{\tau} \quad (32)$$

```
> subs(params,(28),(32)): simplify(%): lhs(%)=convert(rhs(%),
units,Pa/s);
```

$$\dot{\Delta}(t)_{\max} = -130010. \left[ \frac{Pa}{s} \right] \quad (33)$$

Die Strömungsgeschwindigkeit bei  $t=0$  aus (15) mit  $\Delta(t)=P_a$  und  $P_u=0$ .

```
> subs(c=c[max],Delta(t)=P[a],P[u]=0,T=T[a],(15));
```

$$c_{\max} = \sqrt{2} \sqrt{R_i T_a} \quad (34)$$


```
> subs(params,(34)): simplify(%);
```

$$c_{\max} = 413.83 \left[ \frac{m}{s} \right] \quad (35)$$

Die Schallgeschwindigkeit in Luft beträgt 331 m/s im Normzustand. Ist die Strömungsgeschwindigkeit unter 0,3 Schallgeschwindigkeit, dann kann die Kompressibilität vernachlässigt werden. Bei dem extremen Druckunterschied am Anfang ist die Bernoulli-Gleichung eine schlechte Näherung.

Der zeitliche Verlauf des Drucks  $p(t)$  beim Ausströmen ins Vakuum.

```
> convert(subs((28),params,t=t*Unit(s),rhs((29))),unit_free): plot
(%,t=0..10,labels=[t/Unit(s),P/Unit(Pa)]);
```



Bei diesen Modellannahmen wird der Druckausgleich auf 0 bar in rund 7 s erreicht. Der Druckausgleich durch die Bohrung ist deutlich schneller als der Druckabfall in der Umgebung. Also wird die Druckdifferenz klein gegenüber 1 bar sein. Damit ist sogar möglich, dass die inkompressible Strömung eine brauchbare Näherung ist.

### Spezialfall $P_u = 0,9$ bar

Der Fall  $P_u=0$  stellte ein Extremfall dar, der nicht im Bereich der Aufgabe liegt. Jetzt wird von einer 0,1 bar Druckdifferenz zwischen Behälter (1 bar) und Umgebung ausgegangen. Diese deutlich kleinere Druckdifferenz passt besser zu den Druckdifferenzen, die im Rahmen der Aufgabe erwartet werden.

>  $P[u] = 0.9 * P[a]$  ;

$$P_u = 0.9 P_a \quad (36)$$

>  $\text{subs}(\text{params}, (36))$  ;

$$P_u = 90000. \text{ [Pa] } \quad (37)$$

Die Druckdifferenz zum Startzeitpunkt.

>  $\Delta(0) = P[a] - P[u]$  ;

$$\Delta(0) = P_a - P_u \quad (38)$$

>  $\text{subs}((36), \text{params}, (38))$  ;

$$\Delta(0) = 10000. \text{ [Pa] } \quad (39)$$

Die Differenzialgleichung (22) ist in diesem Spezialfall nur schwer analytisch zu lösen. Das Verhalten der Lösung wird analog zum exponentiellen Abfall (29) erwartet. Das ist eine gute Ausgangssituation für eine numerische Lösung der Differenzialgleichung.

Die Gleichung (22) mit allen Zahlenwerten eingesetzt und die Einheiten entfernt. Die Zeit t ist in

s und die Druckdifferenz  $\Delta$  in Pa in die Gleichung einsetzen..

```
> subs( (37), T=T[a], params, Delta(t)=Delta(t)*Unit(Pa), (22) ):
lhs(%) / Unit(Pa) = rhs(%) / Unit(Pa/s) : simplify(%);
```

$$\dot{\Delta}(t) = -1.3001 \sqrt{90000. + \Delta(t)} \sqrt{\Delta(t)} \quad (40)$$

Der Versuch diese Gleichung numerisch mit Maple zu lösen führt zu einem Abbruch bei der Zeit  $t=0,5$ . Zu diesem Zeitpunkt ist die Druckdifferenz  $\Delta$  fast auf 0 abgefallen. Kleine numerische Fehler gibt es bei der Lösung immer, so können negative Werte von  $\Delta$  entstehen. Dieses führt sofort zu einer imaginären Wurzel und damit zum Abbrechen der Berechnung.

Die Ursache liegt in der Modellation der Strömungsgeschwindigkeit in Gleichung (9). Diese Gleichung liefert die Strömungsgeschwindigkeit nur für positive Druckdifferenzen. Das Modell kann erweitert werden mit einer negativen Strömungsgeschwindigkeit für negative Druckdifferenzen. Das Vorzeichen der Druckdifferenz  $\Delta$  muss dazu "vor die Wurzel gezogen" werden.

```
> subs( sqrt(Delta(t))=signum(Delta(t))*sqrt(abs(Delta(t))),
(40) );
```


$$\dot{\Delta}(t) = -1.3001 \sqrt{90000. + \Delta(t)} \operatorname{signum}(\Delta(t)) \sqrt{|\Delta(t)|} \quad (41)$$

Dieser Gleichung kann besser gelöst werden. Maple kann diese Gleichung numerisch lösen mit dem Runge-Kutta-Fehlberg Verfahren, aber einer absoluten Fehlerschranke von 0.4.

```
> sol := dsolve( {(41), lhs((39))=convert(rhs((39)), unit_free) },
numeric, abserr=0.4, range=0..10 );
```


Das Ergebnis der Rechnung ist bis zur Zeit  $t=0,5$ s der erwartete schnelle Druckabfall.

```
> plots[odeplot]( sol, axes=boxed );
```



Nach dem Abfall beginnt der Druck im Behälter zu oszillieren. Das ist kein physikalischer Effekt, es ist ein Effekt durch die numerische Lösung mit der Fehlerschranke von 0.4.


```
> plots[odeplot]( sol, t=0.5..2, axes=boxed );
```



Zur Erklärung ein Blick auf die Gleichung **(41)**. Diese Gleichung liefert den Druckabfall  $d\Delta/dt$  als Funktion der Druckdifferenz  $\Delta$ . (Der Abfall der Druckdifferenz ist gleich dem Abfall des Drucks im Behälter weil der Umgebungsdruck konstant ist.) Der Zusammenhang graphisch dargestellt:

```
> plot( subs(Delta(t)=Delta,rhs((41))), Delta=1e4..-100, axes=
        boxed, labels=[Delta,diff(Delta(t),t)] );
```





Für  $\Delta > 0$  ist die Aussage: je größer die Druckdifferenz, desto größer der Druckabfall. Für  $\Delta < 0$  gilt umgekehrt: je größer die Druckdifferenz, desto größer der Druckanstieg. (In dem Plot ist nur ein kleiner Teil  $< 0$  gezeigt.) Um  $\Delta = 0$  ist die Kurve sehr steil. Das bedeutet, eine kleine Druckdifferenz führt sofort zu einem vergleichsweise großen Druckabfall bzw. Druckanstieg. Die Tangente an die Kurve ist bei  $\Delta = 0$  senkrecht.

Kleine Fehler in der Druckdifferenz werden so zu großen Fehler im Druckabfall bzw. Druckanstieg. Das macht die numerische Berechnung in diesem Bereich sehr schwierig. So entstehen die im Plot oben sichtbaren Schwingungen.

Interessant ist, dass die Rechnung an einer Stelle fehlschlägt, die physikalisch nicht korrekt modelliert wurde. Die Bernoulli-Gleichung gilt für eine stationäre Strömung. Wird die Druckdifferenz 0, so muss die Strömung stoppen. Wird die Druckdifferenz ungleich 0, so muss die Strömung starten. Dieses "Bremsen" und "Beschleunigen" ist nicht in den Gleichungen enthalten. Vermutlich würde genauere physikalische Beschreibung ein besseres Ergebnis liefern. Erwartet würde eine gedämpfte Schwingung um 0.

Doch ein Blick auf die Zahlen: Die Schwankungen der Druckdifferenz sind kleiner als  $1 \text{ Pa} = 0,01 \text{ mbar}$ . Dieser winzige Rechenfehler hat für die Auswertung der Daten keine Bedeutung.

Diese numerische Schwierigkeit kann beseitigt werden, indem für die Druckdifferenz um 0 eine andere Beziehung verwendet. Eine Formel, die ähnlich dem verwendeten Modell ist, aber keine numerischen Schwierigkeiten macht.


Ein funktionierender Ansatz ist eine lineare Näherung für Druckdifferenzen unter  $10 \text{ Pa} = 0,1 \text{ mbar}$ . Dabei wird vorausgesetzt, dass in diesem Druckbereich keine exakten Daten benötigt werden. (Wären die Ansprüche an die Genauigkeit sehr hoch, dann müssten zuerst das physikalische Modell genauer erstellt werden.)

```
> subs( sqrt(Delta(t))=piecewise(Delta(t)>10,sqrt(Delta(t)),
Delta(t)/sqrt(10)),(40)): evalf(%);
```

$$\dot{\Delta}(t) = -1.3001 \sqrt{90000. + \Delta(t)} \begin{cases} \sqrt{\Delta(t)} & 10. < \Delta(t) \\ 0.31623 \Delta(t) & \text{otherwise} \end{cases} \quad (42)$$

Der Plot dieser Funktion zeigt bei 0 einen geringfügigen anderen Verlauf. Bei  $\Delta \ll 0$  wird die Kurve falsch, weil dort weiter lineare extrapoliert wird. Dort die Korrekte Funktion mit der Wurzel aufzuschreiben ist nicht notwendig, weil diese Druckdifferenzen in der Rechnung nicht vorkommen.


```
> plot( subs(Delta(t)=Delta,rhs((42))), Delta=1e4..-100, axes=
boxed, labels=[Delta,diff(Delta(t),t)] );
```



```
> sol := dsolve( {(42),lhs((39))=convert(rhs((39)),unit_free)},
numeric, range=0..10 );
```

Die numerische Lösung der Differenzialgleichung (42) zeigt keine Schwingung. Hier wieder der Bereich ab  $t=0,5$  dargestellt.

```
> plots[odeplot](sol, t=0.5..2, axes=boxed );
```



## 2. Schritt: stationär und inkompressible, T konstant

Gegenüber dem 1. Schritt wird der konstante Umgebungsdruck durch den linear abfallenden Druck ersetzt.

>  $P[u](t) = P[a] + S \cdot t$ ;

$$P_u(t) = P_a + S t \quad (43)$$

Diese Gleichung ist gültig bis der Umgebungsdruck 0 erreicht hat.

Die Strömung wird unverändert als stationär (als Näherung für eine sehr langsame Änderung der Strömung) und inkompressible angenommen.

Die Strömungsgeschwindigkeit (15) aus der Bernoulli-Gleichung kann übernommen werden.

>  $\text{subs}(P[u]=P[u](t), (15))$ ;

$$c = \sqrt{2} \sqrt{\frac{R_i T \Delta(t)}{P_u(t) + \Delta(t)}} \quad (44)$$

Die Masse der Luft im Behälter, die angepasste Gleichung (12).

>  $\text{subs}(P[u]=P[u](t), (12))$ ;

$$m = \frac{(P_u(t) + \Delta(t)) V}{R_i T} \quad (45)$$

Die Ableitung davon

>  $\text{diff}(\text{subs}(m=m(t), (45)), t)$ ;

$$\dot{m}(t) = \frac{(\dot{P}_u(t) + \dot{\Delta}(t)) V}{R_i T} \quad (46)$$

Aufgelöst nach dem Abfall der Druckdifferenz. (Vergleiche mit (18), hier muss auch die Ableitung des Umgebungsdrucks berücksichtigt werden.)

>  $\text{isolate}((46), \text{diff}(\Delta(t), t))$ ;

$$\dot{\Delta}(t) = \frac{\dot{m}(t) R_i T}{V} - \dot{P}_u(t) \quad (47)$$

Den Massenstrom aus der Strömungsgeschwindigkeit über (16) berechnen.

> subs ((16),(47)) ;

$$\dot{\Delta}(t) = - \frac{c \rho A R_i T}{V} - \dot{P}_u(t) \quad (48)$$

Die Dichte über den Druck mit Gleichung (14) berechnen.

> subs (subs (P[u]=P[u] (t) ,(14)) ,(48)) ;

$$\dot{\Delta}(t) = - \frac{c (P_u(t) + \Delta(t)) A}{V} - \dot{P}_u(t) \quad (49)$$

Die Strömungsgeschwindigkeit aus (44) einsetzen.

> subs ((44),(49)) ;

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \sqrt{\frac{R_i T \Delta(t)}{P_u(t) + \Delta(t)}} (P_u(t) + \Delta(t)) A}{V} - \dot{P}_u(t) \quad (50)$$

Der Druck im Behälter ist immer größer 0, daher kann  $P_u + \Delta$  quadriert in die Wurzel gezogen werden.

> simplify((50)) assuming P[u] (t)+Delta(t) > 0;

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \sqrt{P_u(t) + \Delta(t)} \sqrt{R_i T \Delta(t)} A + \dot{P}_u(t) V}{V} \quad (51)$$

Das ist die Differenzialgleichung für den Differenzdruck  $\Delta$ . Im Vergleich zum 1. Schritt (22) ist ein Summand mit der Ableitung des Umgebungsdrucks hinzugekommen.

Den vorgegebenen Umgebungsdruck (43) einsetzen.

> subs ((43),(51)) : simplify(%);

$$\dot{\Delta}(t) = - \frac{\sqrt{2} \sqrt{P_a + S t + \Delta(t)} \sqrt{R_i T \Delta(t)} A + S V}{V} \quad (52)$$

> # dsolve ({(52),Delta(0)=0}) ;

Der Versuch diese Gleichung mit Maple analytisch zu lösen scheitert. Die Zahlenwerte einsetzen für den numerischen Lösungsweg. Dabei die Zeit t in s und den Differenzdruck  $\Delta$  in Pa einsetzen.

> subs ( S\*t=S\*t\*Unit(s) , T=T[a] , params , Delta(t)=Delta(t)\*Unit (Pa) , rhs ((52)) ) : lhs ((52))=%/Unit(Pa/s) : simplify(%);


$$\dot{\Delta}(t) = -1.3001 \sqrt{100000. - 1000. t + \Delta(t)} \sqrt{\Delta(t)} + 1000. \quad (53)$$

Diese Gleichung kann numerisch von Maple mit dem Runge-Kutta-Fehlberg Verfahren gelöst werden. In dieser Gleichung führt der Wurzelterm zu keinen numerischen Schwierigkeiten.

> sol := dsolve( {(53),Delta(0)=0} , numeric , range=0..100 );

sol := proc(x\_rkf45) ... end proc (54)

> plots[odeplot]( sol , axes=boxed );



Der Differenzdruck zeigt ein interessantes Verhalten. Nach dem Zeitpunkt  $t=0$  gibt es einen schnellen Anstieg auf eine fast konstante Druckdifferenz für die folgenden 60 Sekunden. Danach gibt es einen starken Anstieg der Druckdifferenz. Die Druckdifferenz wird maximal, wenn der bei  $t=100$  der Umgebungsdruck 0 erreicht ist.

**> sol(100) : subs(%[1], %[2]) ;**

$$\Delta(100.) = 537.09 \quad (55)$$

Bei  $t=100$  endet die Rechnung, weil der Umgebungsdruck nicht mehr der Funktion (43) folgt. Ab diesem Zeitpunkt ist der Umgebungsdruck konstant  $P_u = 0$ . Der Verlauf der Druckdifferenz nach  $t=100$  wurde analytisch berechnet, siehe (29). Also folgt ein exponentieller Abfall der Druckdifferenz. Zu klären ist, ob der Anstieg des Differenzdrucks ein numerischer Fehler oder ein physikalisch plausibler Effekt ist.

Dazu ein Blick auf den einfach verständlichen und analytisch gelösten Spezialfall mit Außendruck konstant 0. In diesem Spezialfall ist der Druckabfall im Behälter, siehe (31).

**> (31);**

$$\dot{\Delta}(t) = - \frac{\Delta(t)}{\tau} \quad (56)$$

Die Druckdifferenz  $\Delta$  ist in dem Spezifall gleich dem Druck im Behälter  $P_i$ .

**> lhs((56)) = subs( Delta(t)=P[i], rhs((56)) );**

$$\dot{\Delta}(t) = - \frac{P_i}{\tau} \quad (57)$$

Dieser Druckabfall kann interpretiert werden als eine obere Grenze für den Druckabfall, bei einem Umgebungsdruck über 0. Das ist anschaulich: Ist der Umgebungsdruck größer als 0, dann kann der


Druckabfall nicht größer sein, als bei einem Umgebungsdruck von 0.

```
> subs(diff(Delta(t),t)=L,(57)) ;
```

$$L = -\frac{P_i}{\tau} \quad (58)$$

Die obere Grenze ist eine einfache lineare Funktion des Innendrucks.

```
> plot( subs((28),rhs((58))*Unit(s)), P[i]=0..1e5, labels=["Pi in Pa", "L in Pa/s"] );
```



Hier fällt der Außendruck mit konstant 10 mbar/s. In einem weiten Druckbereich könnte der Innendruck des Behälters schneller fallen. Doch es gibt eine Grenze, unterhalb der muss der Innendruck langsamer fallen als der Außendruck.

```
> S=rhs((58)) ;
```

$$S = -\frac{P_i}{\tau} \quad (59)$$

```
> isolate((59),P[i]) ;
```

$$P_i = -S \tau \quad (60)$$

```
> subs( params,(28),(60)) : simplify(%);
```

$$P_i = 769.18 \text{ [Pa]} \quad (61)$$

Ist der Innendruck unter 770 Pa, dann muss die Druckdifferenz steigen, weil die obere Grenze für den möglichen Druckabfall unter den 10 mbar/s liegt.

Also ist der Druckanstieg plausibel, der Anstieg ist kein numerischer Effekt.

```
> diff(Delta(t),t) = 0;
```

$$\dot{\Delta}(t) = 0 \quad (62)$$

```
> subs((62),(51)) ;
```

$$0 = -\frac{\sqrt{2} \sqrt{P_u(t) + \Delta(t)} \sqrt{R_i T \Delta(t)} A + \dot{P}_u(t) V}{V} \quad (63)$$

```
> subs(P[u](t)+Delta(t)=P[i], diff(P[u](t),t)=S, (63)) ;
```

$$0 = -\frac{\sqrt{2} \sqrt{P_i} \sqrt{R_i T \Delta(t)} A + S V}{V} \quad (64)$$

```
> solve((64),{Delta(t)})[1];
```

$$\Delta(t) = \frac{S^2 V^2}{2 R_i T P_i A^2}$$

(65)

```
> plot( {simplify(subs(T=T[a],params,P[i]=p*Unit(Pa),rhs((65))
/Unit(Pa))),p}, p= 1e5..100, y=0..1000 );
```

