```
> restart;
> Digits := 30:
> interface( displayprecision = 8 ):
> deltap := `&Delta;p`:
```

Aufgabe

Wasser fließt durch eine horizontal liegende Düse (also eine einfache Rohrverengung) ins Freie. Der Durchmesser am Anfang der Düse ist d_1 und am Ende der Düse d_2.

Mit welcher Geschwindigkeit verlässt das Wasser den Austrittsquerschnitt, wenn am Eintritt ein Überdruck von Δp herrscht?

```
> d[1] = 40.*Unit(mm), d[2] = 20.*Unit(mm), deltap = 1.65*Unit(bar);
```

$$d_1 = 40.\ [\![mm]\!],\ d_2 = 20.\ [\![mm]\!],\ \Delta p = 1.65\ [\![bar]\!] \quad (1)$$

Bearbeitung

Die Parameter in SI-Einheiten umgerechnet:

```
> op(simplify([(1)]));
```

$$d_1 = 0.040000000\ [\![m]\!],\ d_2 = 0.020000000\ [\![m]\!],\ \Delta p = 165000.00\ [\![Pa]\!] \quad (2)$$

Die Dichte von Wasser bei 20°C [1]

```
> rho = 998*Unit(kg/m^3);
```

$$\rho = 998\ \left[\!\!\left[\frac{kg}{m^3}\right]\!\!\right] \quad (3)$$

Skizze:



Vereinfachungen:
- Die Strömung ist laminar.
- Die Strömung ist stationär.
- Reibungsverluste (innere und äußere) bei der Strömung werden vernachlässigt.
- Das Wasser wird als inkompressible Flüssigkeit behandelt.

Die Querschnitte A_i sind Kreisflächen mit Durchmesser d_i. [2]

```
> A[i] = Pi*d[i]^2/4;
```

$$A_i = \frac{\pi d_i^2}{4} \tag{4}$$

Kontinuitätsgleichung für die Strömung einer inkompressiblen Flüssigkeit. Strömung mit Geschwindigkeit v_i durch Querschnitt A_i. [1]

```
> A[1]*v[1] = A[2]*v[2];
```

$$A_1 v_1 = A_2 v_2 \tag{5}$$

Einsetzen der Kreisflächen **(4)**;

```
> subs( subs(i=1,(4)) , subs(i=2,(4)) , (5) );
```

$$\frac{\pi d_1^2 v_1}{4} = \frac{\pi d_2^2 v_2}{4} \tag{6}$$

Bernoulli-Gleichung für die reibungsfreie stationäre Strömung. [1]
Mit statischem Druck p, Dichte der Flüssigkeit rho, Strömungsgeschwindigkeit v, Fallbeschleunigug g, Höhe h.

```
> p + rho*v^2/2 + rho*g*h = const;
```

$$p + \frac{1}{2}\rho v^2 + \rho g h = const \tag{7}$$

Die Bernoulli-Gleichungen für die Stellen 1 und 2 aufschreiben:

```
> subs(p=p[1],v=v[1],lhs((7))) = subs(p=p[2],v=v[2],lhs((7)));
```

$$p_1 + \frac{1}{2}\rho v_1^2 + \rho g h = p_2 + \frac{1}{2}\rho v_2^2 + \rho g h \tag{8}$$

```
> (8) - rho*g*h;
```

$$p_1 + \frac{\rho v_1^2}{2} = p_2 + \frac{\rho v_2^2}{2} \tag{9}$$

Gleichungen **(6)** und **(9)** bilden ein Gleichungssystem für die unbekannten Strömungsgeschwindigkeit.
Auflösen der Gleichungen nach der gesuchten Austrittsgeschwindigkeit v_2.

Gleichung **(6)** auflösen nach v_1.

```
> isolate( (6), v[1] );
```

$$v_1 = \frac{d_2^2 v_2}{d_1^2} \tag{10}$$

Einsetzen in **(9)**.

```
> subs( (10), (9) );
```

$$p_1 + \frac{\rho d_2^4 v_2^2}{2 d_1^4} = p_2 + \frac{\rho v_2^2}{2} \tag{11}$$

Auflösen nach v_2^2.

```
> isolate( (11), v[2]^2 );
```

(12)

$$v_2^2 = \frac{-p_1 + p_2}{\frac{\rho\, d_2^4}{2\, d_1^4} - \frac{\rho}{2}} \tag{12}$$

Die Druckdifferenz Δp ist gegeben.

```
> deltap = p[1] - p[2];
```

$$\Delta p = p_1 - p_2 \tag{13}$$

Einsetzen.

```
> algsubs( rhs((13))=lhs((13)) , (12) ); simplify(%,size);
```

$$v_2^2 = -\frac{\Delta p}{\frac{\rho\, d_2^4}{2\, d_1^4} - \frac{\rho}{2}}$$

$$v_2^2 = \frac{2\, d_1^4\, \Delta p}{\rho\left(-d_2^4 + d_1^4\right)} \tag{14}$$

Die Geschwindigkeit ist positiv. Die Wurzel ziehen.

```
> sqrt(lhs((14)))=sqrt(rhs((14))): simplify(%) assuming v[2] >0;
  combine(%,radical,symbolic);
```

$$v_2 = \sqrt{2}\,\sqrt{\frac{d_1^4\, \Delta p}{\rho\left(-d_2^4 + d_1^4\right)}}$$

$$v_2 = \sqrt{2}\, d_1^2 \sqrt{\frac{\Delta p}{\rho\left(-d_2^4 + d_1^4\right)}} \tag{15}$$

Die Zahlenwerte **(2)** und **(3)** einsetzen. Ausrechnen.

```
> subs((2),(3),(15) ): simplify(%);
```

$$v_2 = 18.780453\, \left[\!\left[\frac{m}{s} \right]\!\right] \tag{16}$$

Die Strömungsgeschwindigkeit des ausströmenden Wassers beträgt 19 m/s.

Hilfsmittel:

[1] Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch
[2] Bronstein et al: Taschenbuch der Mathematik, Verlag Harri Deutsch
[3] Maple 14