```
> restart;
> Digits:=24: interface( displayprecision=3 ):
> Vo := V[`o`]: vo := v[`o`]:
```

Einleitung

Laut Aufgabe darf die Bewegung eines Massenpunkts betrachtet werden. Der Berech Aerodynamik des fliegenden Autos kann vernachlässigt werden. Auf den Massenpunkt wirkt nur die Fallbeschleunigung und die Massenträgheit.
Also ist für diese Aufgabe eine Wurfparabel zu betrachten. Von diesem Wurf ist der Startpunkt und der Zielpunkt gegeben.
Der Winkel α ist der Winkel zwischen den Geschwindigkeitskomponenten beim Startpunkt.
Reduzierte Skizze mit Koordinatensystem:



Während der Bewegung wirkt die Fallbeschleunigung

```
> g = evalf( ScientificConstants[Constant](g,units) );
```

$$g = 9.807 \left[\!\left[\frac{m}{s^2} \right]\!\right] \tag{1}$$

Der Startpunkt ist der Nullpunkt des Koordinatensystems

```
> A = <0,0>;
```

$$A = \begin{bmatrix} 0 \\ 0 \end{bmatrix} \tag{2}$$

Der Zielpunkt ist gegeben (Bei der y-Koordinate die Höhendifferenz einsetzen).

```
> Z = <35,6>*Unit(m);
```

$$Z = \begin{bmatrix} 35 \,[\![m]\!] \\ 6 \,[\![m]\!] \end{bmatrix} \tag{3}$$

Teil A

Gesucht ist der minimale Winkel α, bei dem der Zielpunkt erreicht werden kann.
Je größer die Startgeschwindigkeit, desto geringer die Zeit bis zum erreichen des Ziels. Je geringer die Flugzeit, desto geringer die Geschwindigkeitsänderung in y-Richtung durch die Fallbeschleunigung.
Bei sehr großer Startgeschwindigkeit kann die Wurfparabel durch eine Gerade mit konstanter Geschwindigkeit angenähert werden.
Der minimale Winkel α ist der Winkel der Geraden zwischen Startpunkt A und Zielpunkt Z.

```
> tan(alpha)=Z[2]/Z[1];
```

$$\tan(\alpha) = \frac{Z_2}{Z_1} \tag{4}$$

```
> isolate((4),alpha);
```

$$\alpha = \arctan\left(\frac{Z_2}{Z_1}\right) \tag{5}$$

```
> subs((3),(5)): lhs(%)=evalf(rhs(%)*Unit(rad));
```

$$\alpha = 0.170 \,[\![rad]\!] \tag{6}$$

```
> lhs((6))=convert(rhs((6)),'units','arcdeg');
```

$$\alpha = 9.728 \,[\![arcdeg]\!] \tag{7}$$

Der Winkel der Rampe muss mindestens 10° betragen.

Teil B

Für diesen Teil muss die Bahnkurve berechnet werden.
Die Startgeschwindigkeit hat den Betrag v_o und den Vektor V_o.

```
> Vo = <vo*cos(alpha),vo*sin(alpha)>;
```

$$V_o = \begin{bmatrix} v_o \cos(\alpha) \\ v_o \sin(\alpha) \end{bmatrix} \tag{8}$$

Die Geschwindigkeit des Massenpunkts als Funktion der Zeit. In -y Richtung wirkt die Fallbeschleunigung.

```
> V(t) = Vo + <0,-g*t>;
```

$$V(t) = V_o + \begin{bmatrix} 0 \\ -g\,t \end{bmatrix} \tag{9}$$

```
> subs((8),(9)): simplify(%);
```

$$V(t) = \begin{bmatrix} v_o \cos(\alpha) \\ v_o \sin(\alpha) - g\,t \end{bmatrix} \tag{10}$$

Der Ort des Massenpunkts als Funktion der Zeit ergibt sich aus dem Integral über die Geschwindigkeit.

```
> P(t) = A + <seq(int(rhs((10))[i],t=0..t),i=1..2)>;
```

$$P(t) = A + \begin{bmatrix} v_o \cos(\alpha)\, t \\ v_o \sin(\alpha)\, t - \frac{1}{2}\, g\, t^2 \end{bmatrix} \tag{11}$$

```
> subs((2),(11)): expand(%);
```

$$P(t) = \begin{bmatrix} v_o \cos(\alpha)\, t \\ v_o \sin(\alpha)\, t - \frac{1}{2}\, g\, t^2 \end{bmatrix} \tag{12}$$

Zum Zeitpunkt t_z soll der Zielpunkt erreicht werden.

```
> Z = P(t[z]);
```

$$Z = P(t_z) \tag{13}$$

```
> subs(Z=<Z[1],Z[2]>,subs(t=t[z],(12)),(13));
```

$$\begin{bmatrix} Z_1 \\ Z_2 \end{bmatrix} = \begin{bmatrix} v_o \cos(\alpha)\, t_z \\ v_o \sin(\alpha)\, t_z - \frac{1}{2}\, g\, t_z^2 \end{bmatrix} \tag{14}$$

Die Zeit t_z kann aus der x-Komponente bestimmt werden. Die y-Komponente liefert eine Gleichung für die Startgeschwindigkeit v_o als Funktion des Winkels α.

```
> lhs((14))[1] = rhs((14))[1];
```

$$Z_1 = v_o \cos(\alpha)\, t_z \tag{15}$$

```
> isolate((15),t[z]);
```

$$t_z = \frac{Z_1}{v_o \cos(\alpha)} \tag{16}$$

```
> lhs((14))[2] = rhs((14))[2];
```

$$Z_2 = v_o \sin(\alpha)\, t_z - \frac{1}{2}\, g\, t_z^2 \tag{17}$$

```
> subs((16),(17));
```

$$Z_2 = \frac{\sin(\alpha)\, Z_1}{\cos(\alpha)} - \frac{1}{2}\, \frac{g\, Z_1^2}{v_o^2 \cos(\alpha)^2} \tag{18}$$

```
> isolate((18),vo^2);
```

$$v_o^2 = \frac{1}{2}\, \frac{g\, Z_1^2}{\left(-Z_2 + \frac{\sin(\alpha)\, Z_1}{\cos(\alpha)}\right) \cos(\alpha)^2} \tag{19}$$

Da v_o der Betrag der Startgeschwindigkeit ist, muss der positve Zweig der Wurzelfunktion gewählt werden.

```
> vo = sqrt(rhs((19)));
```

$$v_o = \frac{1}{2}\, \sqrt{2}\, \sqrt{\frac{g\, Z_1^2}{\left(-Z_2 + \frac{\sin(\alpha)\, Z_1}{\cos(\alpha)}\right) \cos(\alpha)^2}} \tag{20}$$

```
> simplify((20)): sort(%);
```

(21)

$$v_o = \frac{1}{2}\sqrt{2}\sqrt{\frac{g Z_1^2}{(Z_1 \sin(\alpha) - Z_2 \cos(\alpha)) \cos(\alpha)}} \quad \textbf{(21)}$$

Der Scheitelpunkt S der Bahnkurve wird erreicht, wenn die y-Komponente der Geschwindigkeit 0 ist.
Der Zeitpunkt beim Erreichen des Punkts S wird t_s genannt.

```
> V(t[s])[2] = 0;
```

$$V(t_s)_2 = 0 \quad \textbf{(22)}$$

```
> subs(subs(t=t[s],(10)),(22)): simplify(%);
```

$$v_o \sin(\alpha) - g\, t_s = 0 \quad \textbf{(23)}$$

```
> isolate((23),t[s]);
```

$$t_s = \frac{v_o \sin(\alpha)}{g} \quad \textbf{(24)}$$

Die maximale Höhe der Wurfparabel die y-Komponente des Scheitelpunkt plus 1 Meter. (Siehe Skizze in der Aufgabe.)

```
> h = P(t[s])[2];
```

$$h = P(t_s)_2 \quad \textbf{(25)}$$

```
> subs(subs(t=t[s],(12)),(25)): simplify(%);
```

$$h = \frac{1}{2}\, t_s\, (2\, v_o \sin(\alpha) - g\, t_s) \quad \textbf{(26)}$$

```
> subs((24),(26));
```

$$h = \frac{1}{2}\,\frac{v_o^2 \sin(\alpha)^2}{g} \quad \textbf{(27)}$$

Die Startgeschwindigkeit vo durch **(19)** ausdrücken.

```
> subs((19),(27));
```

$$h = \frac{1}{4}\,\frac{Z_1^2 \sin(\alpha)^2}{\left(-Z_2 + \frac{\sin(\alpha)\, Z_1}{\cos(\alpha)}\right) \cos(\alpha)^2} \quad \textbf{(28)}$$

```
> simplify((28));
```

$$h = \frac{1}{4}\,\frac{Z_1^2 \sin(\alpha)^2}{\cos(\alpha)\, (Z_1 \sin(\alpha) - Z_2 \cos(\alpha))} \quad \textbf{(29)}$$

Damit ist h als Funktion des Winkels α aufgeschrieben.

Die Funktion $v_o(\alpha)$ **(21)** mit den Zahlenwerten.

```
> subs((1),(3),(21)): simplify(%);
```

$$v_o = 77.502 \sqrt{\frac{1}{(35.000 \sin(\alpha) - 6.000 \cos(\alpha)) \cos(\alpha)}} \left[\!\left[\frac{m}{s} \right]\!\right] \quad \textbf{(30)}$$

```
> plot(subs(alpha=alpha*Pi/180,rhs((30))),alpha=15..45,title=
  "Startgeschwindigkeit Vo");
```



Die Funktion der Scheitelhöhe h(α) **(29)** mit den Zahlenwerten.

```
> subs((1),(3),(29)): evalf(simplify(%));
```

$$h = \frac{306.250 \sin(\alpha)^2 [[m]]}{\cos(\alpha)\ (35.000 \sin(\alpha) - 6.000 \cos(\alpha))} \quad \textbf{(31)}$$

In der Aufgabe ist die Höhe H verlangt. Diese ist 1 m größer, siehe die beiden Skizzen.

```
> H = h + 1.0*Unit(m);
```

$$H = h + 1.000\ [[m]] \quad \textbf{(32)}$$

```
> subs((31),(32));
```

$$H = \frac{306.250 \sin(\alpha)^2 [[m]]}{\cos(\alpha)\ (35.000 \sin(\alpha) - 6.000 \cos(\alpha))} + 1.000\ [[m]] \quad \textbf{(33)}$$

```
> plot( subs(alpha=alpha*Pi/180,rhs((33))), alpha=12..45, title=
  "Scheitelhöhe H" );
```



Erstellt mit Maple 17.