```
> restart;
> Digits:=25: interface(displayprecision=3):
> with( Physics[Vectors] ):
```

Aufgabe

Zwei Punktladungen Q_1 und Q_2 befinden sich in den Punkten $P_1(x_1,y_1)$ und $P_2(x_2,y_2)$ des kartesischen Koordinatensystems.
Berechnen Sie den Vektor der elektrischen Feldstärke im Punkt $P(x_0,y_0)$, wenn der Feldraum mit Luft gefüllt ist.

```
> Q[1]= 10.0^(-7)*Unit(A*s), X[1]=-3.0*Unit(cm), Y[1]= 0.0*Unit(cm)
  ;
```

$$Q_1 = 1.00\;10^{-7}\;[\![C]\!],\; X_1 = -3.0\;[\![cm]\!],\; Y_1 = 0. \tag{1}$$

```
> Q[2]=-2.*10^(-7)*Unit(A*s), X[2]= 4.0*Unit(cm), Y[2]=0.0*Unit(cm)
  ;
```

$$Q_2 = -2.00\;10^{-7}\;[\![C]\!],\; X_2 = 4.0\;[\![cm]\!],\; Y_2 = 0. \tag{2}$$

```
> X[0]=3.0*Unit(cm), Y[0]=4.0*Unit(cm);
```

$$X_0 = 3.0\;[\![cm]\!],\; Y_0 = 4.0\;[\![cm]\!] \tag{3}$$

Rechnung

Das elektrische Feld E einer Punktladung der Größe Q im Vakuum am Ort r von der Punktladung aus.
(Formel aus dem Physik Buch.)

```
> E_ = Q/(4*Pi*epsilon[0]*r^2)*e_[r];
```

$$\vec{E} = \frac{Q\,\vec{e}_r}{4\,\pi\,\varepsilon_0\,r^2} \tag{4}$$

Die elektrische Feldkonstante des Vakuums gilt in guter Näherung auch für Luft. (Die Möglichkeit der Ionisierung vernachlässigt.)

```
> epsilon[0] = evalf( ScientificConstants[Constant](epsilon[0],
  units) );
```

$$\varepsilon_0 = 8.85\;10^{-12}\;\left[\!\!\left[\frac{A^2\,s^4}{kg\,m^3}\right]\!\!\right] \tag{5}$$

Der Einheitsvektor wird berechnet durch

```
> e_[r] = r_/r;
```

$$\vec{e}_r = \frac{\vec{r}}{r} \tag{6}$$

Eingesetzt in **(4)**.

```
> subs((6),(4));
```

$$\vec{E} = \frac{Q\,\vec{r}}{4\,\pi\,\varepsilon_0\,r^3} \tag{7}$$

Das Feld der ersten Punktladung.

```
> subs( E_=E_[1],Q=Q[1],r_=r_[1],r=r[1],(7));
```

$$\vec{E}_1 = \frac{Q_1\,\vec{r}_1}{4\,\pi\,\varepsilon_0\,r_1^3} \tag{8}$$

Vektor von der Punktladung 1 zum Messpunkt.

```
> r_[1]=(X[0]-X[1])*e_[x] + (Y[0]-Y[1])*e_[y]: simplify(%,size);
```
$$\vec{r}_1 = (X_0 - X_1)\,\vec{e}_x + (Y_0 - Y_1)\,\vec{e}_y \tag{9}$$

```
> subs((1),(3),(9));
```
$$\vec{r}_1 = 6.0\,\vec{e}_x\,[\![cm]\!] + 4.0\,\vec{e}_y\,[\![cm]\!] \tag{10}$$

```
> r[1] = subs((10),e_[x]=_i,e_[y]=_j, Norm(r_[1]) ): simplify(%);
```
$$r_1 = 0.0721\ [\![m]\!] \tag{11}$$

Alle Zahlenwerte in Gleichung **(8)** einsetzen.

```
> subs((1),(5),(10),(11),(8)): simplify(%): expand(%): lhs(%) = map(
  z->convert( z, units, kV/m ), rhs(%) );
```
$$\vec{E}_1 = 144.\,\vec{e}_x\left[\!\!\left[\frac{kV}{m}\right]\!\!\right] + 95.9\,\vec{e}_y\left[\!\!\left[\frac{kV}{m}\right]\!\!\right] \tag{12}$$

Das Feld der zweiten Punktladung.

```
> subs( E_=E_[2],Q=Q[2],r_=r_[2],r=r[2],(7));
```
$$\vec{E}_2 = \frac{Q_2\,\vec{r}_2}{4\,\pi\,\varepsilon_0\,r_2^3} \tag{13}$$

Vektor von der Punktladung 2 zum Messpunkt.

```
> r_[2]=(X[0]-X[2])*e_[x] + (Y[0]-Y[2])*e_[y]: simplify(%,size);
```
$$\vec{r}_2 = (X_0 - X_2)\,\vec{e}_x + (Y_0 - Y_2)\,\vec{e}_y \tag{14}$$

```
> subs((2),(3),(14));
```
$$\vec{r}_2 = -\vec{e}_x\,[\![cm]\!] + 4.0\,\vec{e}_y\,[\![cm]\!] \tag{15}$$

```
> r[2] = subs((15),e_[x]=_i,e_[y]=_j, Norm(r_[2]) ): simplify(%);
```
$$r_2 = 0.0412\ [\![m]\!] \tag{16}$$

Alle Zahlenwerte in Gleichung **(13)** einsetzen.

```
> subs((2),(5),(15),(16),(13)): simplify(%): expand(%): lhs(%) = map(
  z->convert( z, units, kV/m ), rhs(%) );
```
$$\vec{E}_2 = 256.\,\vec{e}_x\left[\!\!\left[\frac{kV}{m}\right]\!\!\right] - 1030.\,\vec{e}_y\left[\!\!\left[\frac{kV}{m}\right]\!\!\right] \tag{17}$$

Die beiden berechneten Felder addieren.

```
> E_ = E_[1]+E_[2];
```
$$\vec{E} = \vec{E}_1 + \vec{E}_2 \tag{18}$$

```
> subs((12),(17),(18)): lhs(%) = map( z->convert( z, units, kV/m ), rhs
  (%) );
```
$$\vec{E} = 400.\,\vec{e}_x\left[\!\!\left[\frac{kV}{m}\right]\!\!\right] - 930.\,\vec{e}_y\left[\!\!\left[\frac{kV}{m}\right]\!\!\right] \tag{19}$$

Die gesuchte elektrische Feldstärke.

Hilfsmittel:
- D. Meschede: Gerthsen Physik, Springer-Verlag
- Maple 17