```
> restart;
> swap := a -> rhs(a) = lhs(a):
```

Aufgabe

Biegelinie eines eingespannten Balkens mit Dreieckslast auf einem Teil des Balkens.



Rechnung

Annahmen:
- Verformung des Balkens ist im linear-Elastischen Bereich.
- Höhe und Breite des Balkens ist klein gegen die Länge des Balkens.
- Kinematische Hypothese von Bernoulli gilt (Querschnitte bleiben senkrecht zur verformten Balkenachse).

Koordinatensystem einführen:
- x-Achse nach rechts, 0 bei der Einspannung.
- y-Achse nach oben, 0 bei der Balkenachse.
- Vorzeichenwahl für alle Kräfte: +, wenn in Achsrichtung.
- Vorzeichenwahl für alle Momente: +, wenn mathematisch positiv (gegen Uhrzeigersinn).

Last beginnt bei x=A. Balkenende bei x=B.
Maximale Linienlast q(A)=Q.
Biegesteifigkeit des Balkens EI konstant über die gesamte Länge. (oder Elastizitätsmodul E, Deviationsmoment I)

Funktion für die Linienlast auf dem Balken konstruieren.
Im ersten Teil 0 und im zweiten Teil linear fallend Q bei x=A bis 0 bei x=B.

```
> q(x) = piecewise( x <= A, 0, A<x, (x-B)/(A-B)*Q );
```

$$q(x) = \begin{cases} 0 & x \leq A \\ \frac{(x-B)\,Q}{A-B} & A < x \end{cases} \tag{1}$$

Das rechte Ende des Balkens ist frei, damit entfällt dort das Freischneiden.
Querkraft auf den Balken an der Schnittstelle x durch Integration der Linienlast rechts von der Schnittstelle x.
(Die Querkraft wird zur Kontrolle berechnet. Danaben wird die

```
> F(x) = int( q(zeta), zeta=x..B );
  subs( eval((1),x=zeta), % );
```

$$F(x) = \int_x^B q(\zeta)\, \mathrm{d}\zeta$$

$$F(x) = \int_x^B \begin{cases} 0 & \zeta \le A \\ \frac{(\zeta - B)\, Q}{A - B} & A < \zeta \end{cases} \mathrm{d}\zeta \tag{2}$$

Das Integral für x > A ausrechnen.
Dort ist die Linienlast eine lineare Funktion, das Integral ergibt eine Quadratische Funktion.

```
> simplify((2)) assuming 0<A,A<B,A<x,x<B;
  simplify(%,size);
```

$$F(x) = -\frac{Q\left(B^2 - 2\, B\, x + x^2\right)}{2\, (A - B)}$$

$$F(x) = -\frac{Q\, (B - x)^2}{2\, A - 2\, B} \tag{3}$$

Bei x=A wirkt die gesamte Last.

```
> F[g] = F(A); subs( eval((3),x=A) , % ); simplify(%);
```

$$F_g = F(A)$$

$$F_g = -\frac{Q\, (B - A)^2}{2\, A - 2\, B}$$

$$F_g = -\frac{(A - B)\, Q}{2} \tag{4}$$

Mit der Gesamtkraft Fg kann Gleichung **(3)** umgeschrieben werden.

```
> subs( isolate((4),Q) , (3) ): simplify(%,size);
```

$$F(x) = \frac{F_g\, (B - x)^2}{(A - B)^2} \tag{5}$$

Für 0 < x < A beibt die Querlast konstant.

```
> # Zur Kontrolle wird hier das Integral ausgerechnet.
  simplify((2)) assuming 0<A,A<B,0<x,x<A: subs( swap((4)) , % );
```

$$F(x) = F_g \tag{6}$$

Die Querkraft zusammengesetzt aus **(5)** und **(6)**.

```
> F(x) = piecewise( x <= A, rhs((6)) , A<x, rhs((5)) );
```

$$F(x) = \begin{cases} F_g & x \le A \\ \frac{F_g\, (B - x)^2}{(A - B)^2} & A < x \end{cases} \tag{7}$$

Biegemoment auf den Balken an der Schnittstelle x durch Integration der Kraft rechts von der Schnittstelle x.

```
> M(x) = int( (zeta-x) * q(zeta), zeta=x..B );
  subs( eval((1),x=zeta) , % );
```

$$M(x) = \int_x^B (\zeta - x)\, q(\zeta)\, d\zeta$$

$$M(x) = \int_x^B (\zeta - x) \begin{cases} 0 & \zeta \le A \\ \frac{(\zeta - B)\, Q}{A - B} & A < \zeta \end{cases} d\zeta \tag{8}$$

Das Integral für x > A ausrechnen.
Dort ist die Linienlast eine lineare Funktion, das Integral ergibt eine kubische Funktion.

```
> simplify((8)) assuming 0<A,A<B,A<x,x<B;
  simplify(%,size);
  subs( isolate((4),Q), % ): simplify(%,size);
```

$$M(x) = -\frac{Q\,(B^3 - 3 B^2 x + 3 B x^2 - x^3)}{6\,(A - B)}$$

$$M(x) = -\frac{Q\,(B - x)^3}{6 A - 6 B}$$

$$M(x) = \frac{F_g\,(B - x)^3}{3\,(A - B)^2} \tag{9}$$

Das Integral für x < A ausrechnen.
Dort ist die Linienlast 0, das Integral ergibt eine lineare Funktion.

```
> simplify((8)) assuming 0<A,A<B,0<x,x<A;
  simplify(%,size);
  subs( isolate((4),Q), % ): simplify(%);
```

$$M(x) = -\frac{(2 A^2 - A B - 3 A x - B^2 + 3 B x)\, Q}{6}$$

$$M(x) = -\frac{(A - B)\, Q \left(A + \frac{B}{2} - \frac{3 x}{2}\right)}{3}$$

$$M(x) = \frac{(2 A + B - 3 x)\, F_g}{3} \tag{10}$$

Zusammenfassen des Biegemoments aus Gleichungen **(9)** und **(10)**.

```
> M(x) = piecewise( x<=A, rhs((10)), A<x, rhs((9)) );
```

$$M(x) = \begin{cases} \frac{(2 A + B - 3 x)\, F_g}{3} & x \le A \\ \frac{F_g\,(B - x)^3}{3\,(A - B)^2} & A < x \end{cases} \tag{11}$$

Kontrolle: Es muss gelten M'(x) = - F(x).

```
> diff( (11), x );
```

(12)

$$M'(x) = \begin{cases} -F_g & x \le A \\ -\frac{F_g (B-x)^2}{(A-B)^2} & A < x \end{cases} \tag{12}$$

Das Ergbnis passt zur Gleichung **(7)** für die Querkraft auf den Balken.

Die Biegelinie v(x) ergibt sich aus dem Biegemoment.

```
> diff(v(x),x,x) = M(x)/EI;
```

$$v''(x) = \frac{M(x)}{EI} \tag{13}$$

Für die beiden Abschnitte des Balkens getrennte Funktionen ansetzen:

```
> v(x) = piecewise( x <= A, v[1](x), A < x, v[2](x) );
```

$$v(x) = \begin{cases} v_1(x) & x \le A \\ v_2(x) & A < x \end{cases} \tag{14}$$

Randbedingungen für die beiden Funktionen:
Bei x = 0 ist der Balken fest eingespannt. Der Balken darf nicht unterbrochen und nicht geknickt sein.

```
> eval(v[1](x),x=0) = 0;
```

$$v_1(0) = 0 \tag{15}$$

```
> eval( diff( v[1](x), x ), x=0 ) = 0;
```

$$v_1'(x)\Big|_{\{x=0\}} = 0 \tag{16}$$

Bei x = A müssen bei beiden angesetzten Funktionen zusammen passen. Auch dort darf der Balken nicht unterbrochen und nicht geknickt sein.

```
> eval(v[1](x),x=A) = eval(v[2](x),x=A);
```

$$v_1(A) = v_2(A) \tag{17}$$

```
> Eval( diff( v[1](x), x ), x=A ) = Eval( diff( v[2](x), x ), x=A
  );
```

$$v_1'(x)\Big|_{x=A} = v_2'(x)\Big|_{x=A} \tag{18}$$

Die Differentialgleichung für den ersten Teil. In Gleichung **(13)** die passende Teile aus Gleichung **(11)** und **(10)** einsetzen.

```
> subs( (14), (11), (13) ): simplify(%) assuming x < A;
```

$$v_1''(x) = \frac{(2A + B - 3x) F_g}{3\,EI} \tag{19}$$

Bei der zweifachen Integration kommen zwei Integrationskonstanten in die Gleichung.

```
> dsolve( (19), v[1](x) ): subs( _C1=C[1], _C2=C[2], % );
```

$$v_1(x) = \frac{F_g \left(A x^2 + \frac{1}{2} B x^2 - \frac{1}{2} x^3\right)}{3\,EI} + C_1 x + C_2 \tag{20}$$

Die Differentialgleichung für den zweiten Teil. In Gleichung **(13)** die passende Teile aus Gleichung **(11)** und **(10)** einsetzen.

```
> subs( (14), (11), (13) ): simplify(%) assuming x > A;
```

$$v_2''(x) = \frac{F_g (B-x)^3}{3 (A-B)^2 EI} \tag{21}$$

Bei der zweifachen Integration kommen zwei weitere Integrationskonstanten in die Gleichung.

```
> dsolve( (21), v[2](x) ): subs( _C1=C[3], _C2=C[4], % );
```

$$v_2(x) = \frac{F_g (B-x)^5}{60 (A-B)^2 EI} + C_3 x + C_4 \tag{22}$$

Die vier Integrationskonstanten werden aus den vier Randbedingungen **(15)**, **(16)**, **(17)** und **(18)** bestimmt.
Aus **(15)** folgt:

```
> subs( eval((20),x=0), (15) );
```

$$C_2 = 0 \tag{23}$$

Aus **(16)** folgt:

```
> subs( (20), (16) ): simplify(%);
```

$$C_1 = 0 \tag{24}$$

Damit kann der erste Teil der Biegelinie aufgeschrieben werden.

```
> subs( (23), (24), (20) );
  collect(%,x);
```

$$v_1(x) = \frac{F_g \left(A x^2 + \frac{1}{2} B x^2 - \frac{1}{2} x^3\right)}{3 EI}$$

$$v_1(x) = -\frac{F_g x^3}{6 EI} + \frac{F_g \left(A + \frac{B}{2}\right) x^2}{3 EI} \tag{25}$$

Aus **(18)** folgt:

```
> subs( (25), (22), (18) ): value(%);
  simplify(%,size);
  isolate(%,C[3]);
  simplify(%,size);
```

$$-\frac{F_g A^2}{2 EI} + \frac{2 F_g \left(A + \frac{B}{2}\right) A}{3 EI} = -\frac{F_g (B-A)^4}{12 (A-B)^2 EI} + C_3$$

$$\frac{F_g A (A+2B)}{6 EI} = \frac{-F_g (A-B)^2 + 12 C_3 EI}{12 EI}$$

$$C_3 = \frac{2 F_g A (A+2B) + F_g (A-B)^2}{12 EI}$$

$$C_3 = \frac{F_g (3A^2 + 2AB + B^2)}{12 EI} \tag{26}$$

Aus **(17)** folgt:

```
> subs( eval((25),x=A), eval((22),x=A), (17) ); simplify(%,size);
```

$$-\frac{F_g A^3}{6 EI} + \frac{F_g \left(A + \frac{B}{2}\right) A^2}{3 EI} = \frac{F_g (B-A)^5}{60 (A-B)^2 EI} + C_3 A + C_4$$

$$\frac{F_g A^2 (A+B)}{6\,EI} = \frac{-F_g (A-B)^3 + 60\,EI\,(C_3 A + C_4)}{60\,EI} \tag{27}$$

Gleichung **(26)** einsetzen.

```
> subs( (26), (27) );
  simplify(%,size);
  isolate(%,C[4]);
  simplify(%,size);
```

$$\frac{F_g A^2 (A+B)}{6\,EI} = \frac{-F_g (A-B)^3 + 60\,EI\left(\frac{F_g (3A^2 + 2AB + B^2) A}{12\,EI} + C_4\right)}{60\,EI}$$

$$\frac{F_g A^2 (A+B)}{6\,EI} = \frac{(14A^3 + 13A^2 B + 2AB^2 + B^3) F_g + 60\,C_4\,EI}{60\,EI}$$

$$C_4 = \frac{10\,F_g A^2 (A+B) - (14A^3 + 13A^2 B + 2AB^2 + B^3) F_g}{60\,EI}$$

$$C_4 = -\frac{F_g (4A^3 + 3A^2 B + 2AB^2 + B^3)}{60\,EI} \tag{28}$$

Mit den ausgerechnen Integrationskonstanten kann der zweite Teil der Biegelinie aufgeschrieben werden.

```
> subs( (26), (28), (22) );
```

$$v_2(x) = \frac{F_g (B-x)^5}{60\,(A-B)^2\,EI} + \frac{F_g (3A^2 + 2AB + B^2)\,x}{12\,EI} - \frac{F_g (4A^3 + 3A^2 B + 2AB^2 + B^3)}{60\,EI} \tag{29}$$

Die Biegelinie ist:

```
> subs( (29), (25), (14) );
```

$$v(x) = \begin{cases} -\frac{F_g x^3}{6\,EI} + \frac{F_g \left(A + \frac{B}{2}\right) x^2}{3\,EI} & x \leq A \\ \frac{F_g (B-x)^5}{60\,(A-B)^2\,EI} + \frac{F_g (3A^2 + 2AB + B^2)\,x}{12\,EI} - \frac{F_g (4A^3 + 3A^2 B + 2AB^2 + B^3)}{60\,EI} & A < x \end{cases}$$

Hilfsmittel

- Balke: Einführung in die Technische Mechanik - Festigkeitslehre, Springer-Verlag
- Maple 17, http://www.maplesoft.com/