```
> restart;
> Digits := 24:
> interface( displayprecision=3 ):
```

Geschätzt werden soll die maximale Temperatur eines luftgefüllten Glasballons mit einer schwarz lackierten Seite im Sonnenschein.

[1] gibt die Temperatur einer schwarzen Herdplatte in strahlender Junimittagssonne mit rund 60 °C. Eine ähnliche Temperatur wird hier erwartet.

Vereinfachungen:
- Betrachtet wird eine statische Situation. Damit müssen keine Wärmekapazitäten und Zeitverläufe berechnet werden.
- Lack, Luftfüllung und Glaswand haben eine Temperatur. Dabei wird der Temperaturabfall in der Glaswand vernachlässigt.

Der Glasballon hat den Radius

```
> r = 10 * Unit(cm);
```

$$r = 10 \, [\![cm]\!] \tag{1}$$

Die Intensität der Sonnenstrahlung in Erdnähe, die extraterristische Solarkonstante [1]:

```
> I[e] = 1400 *Unit(W)/Unit(m^2);
```

$$I_e = \frac{1400 \, [\![W]\!]}{[\![m^2]\!]} \tag{2}$$

Die Sonnenstrahlung wird durch die Erdatmosphäre geschwächt. [1] gibt für den Bereich Deutschland 0,7 I_e auf der Erdoberfläche an.

Im sichtbaren Spektralbereich ist Glas transparent. Der größte Teil (bezogen auf die Leistung) des Sonnenlichts ist im transparenten Spektralbereich des Glases.
Die Strahlung wird an den Glasflächen aufgrund des Brechungsindexsprungs teilweise reflektiert. Die Reflexion ist winkelabhängig (Fresnel-Formel). Vereinfacht geschätzt: 0,9 Transmission. Die beiden Transmissionsfaktoren zusammengefasst:

```
> t = 0.7 * 0.9;
```

$$t = 0.63 \tag{3}$$

Der Emissionskoeffizient eines mattschwarzen Lacks ist [2]

```
> epsilon[S] = 0.970;
```

$$\varepsilon_S = 0.970 \tag{4}$$

Nach der Kirchhoffschen-Regel ist der Emissionskoeffizent eines Körpers gleich dem Absorptionskoeffizient.
Die Leistung durch Absorption des Sonnenlichts. Die Fläche ist dabei die Kreisscheibe durch die die Sonnenstrahlung auf die mattschwarz lackierte Kugelhälfte trifft.

```
> P[S] = \Pi * r^2 * t * epsilon[S] * I[e];
```

$$P_S = \pi r^2 t \varepsilon_S I_e \tag{5}$$

```
> simplify( subs((1),(2),(3),(4),(5)) );
```

$$P_S = 26.878 \, [\![W]\!] \tag{6}$$

Ein Kühlungsmechanismus ist die Wärmestrahlung (Eine elektromagnetische Strahlung aufgrund der Temperatur des Körpers, die Wärme zwischen Körpern transportiert, Stefan-Boltzmann Strahlungsgesetz.) des Glasballons. Bei erwarteten 60 °C liegt das Strahlungsmaximum im fernen Infrarot. In diesem Bereich absorbiert Glas die Strahlung. Der Glasballon wirkt für die Infrarotstrahlung bei einer Wellenlänge über rund 5 µm (dort liegt das Strahlungsmaximum bei einer Temperatur von rund 300 °C.) wie eine dunkle Kugel.

```
> P[G] = S[G] * epsilon[G] * sigma * T[G]^4;
```

$$P_G = S_G\,\varepsilon_G\,\sigma\,T_G^4 \tag{7}$$

Dabei ist die Stefan-Boltzmann-Konstante [1]

```
> sigma = 5.67040*10^(-8) * Unit(W/(m^2*K^4));
```

$$\sigma = 5.670\;10^{-8}\left[\!\left[\frac{kg}{s^3\,K^4}\right]\!\right] \tag{8}$$

Auf der Internetseite [3] wird ein Emissionskoeffizient von 0,85 für Glasflächen angegeben.

```
> epsilon[G] = 0.85;
```

$$\varepsilon_G = 0.85 \tag{9}$$

Die Oberfläche des kugelförmigen Glasballons

```
> S[G] = 4*Pi* r^2;
```

$$S_G = 4\,\pi\,r^2 \tag{10}$$

```
> simplify(evalf(eval((10),(1))));
```

$$S_G = 0.126\,[\![m^2]\!] \tag{11}$$

Die Wärmestrahlung der Umgebung wird vom Glasballon absorbiert

```
> P[U] =  S[G] * epsilon[G] * sigma *T[U]^4;
```

$$P_U = S_G\,\varepsilon_G\,\sigma\,T_U^4 \tag{12}$$

Eine typische Umgebungstemperatur im Sommer ist 30 °C

```
> T[U] = ( 30 + 273.15 ) * Unit(K);
```

$$T_U = 303.150\,[\![K]\!] \tag{13}$$

Damit

```
> simplify(subs((13),(10),(9),(8),(1),(12)));
```

$$P_U = 51.153\,[\![W]\!] \tag{14}$$

Die aufgenommene Leistung P_U aus der Umgebung ist größer als die Leistung P_S aufgenommen aus dem Sonnenlicht. Die Leistungsdichte der Umgebungsstrahlung σT_U^4 ist kleiner als die Leistungsdichte des Sonnenlichts, aber die gesamte Kugelfläche nimmt diese Strahlung auf. Die Sonnenstrahlung fällt nur senkrecht durch den Kugelquerschnitt (die Kreisfläche) auf die schwarze Hälfte der Kugel.

Zwischenstufe: Die Temperatur des Glasballons nur mit dem Strahlungsgleichgewicht. (Ein im Vakuum schwebender Glasballon.)

```
> P[S] + P[U] = P[G];
```

$$P_S + P_U = P_G \tag{15}$$

```
> subs((12),(7),(5),(10),(15));
```

$$\pi\,r^2\,t\,\varepsilon_S\,\mathrm{I}_e + 4\,\pi\,r^2\,\varepsilon_G\,\sigma\,T_U^4 = 4\,\pi\,r^2\,\varepsilon_G\,\sigma\,T_G^4 \tag{16}$$

```
> isolate((16),T[G]^4): simplify(%);
```

$$T_G^4 = \frac{1}{4}\,\frac{t\,\varepsilon_S\,\mathrm{I}_e + 4\,\varepsilon_G\,\sigma\,T_U^4}{\varepsilon_G\,\sigma} \tag{17}$$

```
> T[G]=rhs((17))^(1/4): evalf(%);
```

$$T_G = 0.707\left(\frac{t\,\varepsilon_S\,\mathrm{I}_e + 4.000\,\varepsilon_G\,\sigma\,T_U^4}{\varepsilon_G\,\sigma}\right)^{1/4} \tag{18}$$

```
> subs((13),(9),(8),(4),(3),(2),(18)): simplify(%);
```

$$T_G = 336.904\ [\![K]\!] \quad (19)$$

```
> theta[G] = ( rhs((19))/Unit(K) - 273.15 ) * Unit(Celsius);
```

$$\theta_G = 63.754\ [\![degC]\!] \quad (20)$$

Das ist die Temperatur des Glasballon nur basierend auf der Strahlungsbilanz. Der Glasballon befindet sich in der Umgebungsluft mit Temperatur T_U. Die Luft hat eine geringere Temperatur, daher transportiert die Luft Wärme vom Glasballon weg, die Temperatur des Glasballons wird kleiner.
Bei Windstille gibt es nur die freie Konvektion. Dann wird die maximale Temperatur erreicht.

Der Wärmeübergang durch die Fläche S_G mit einem Wärmeübergangskoeffizient α.

```
> P[K] = alpha * S[G] * ( T[G] - T[U] );
```

$$P_K = \alpha S_G (T_G - T_U) \quad (21)$$

Ein typischer Wärmeübergangskoeffizient für die freie Konvektion bei Gasen [2]

```
> alpha = 12 * Unit(W)/Unit(m^2*K);
```

$$\alpha = \frac{12\ [\![W]\!]}{[\![m^2\ K]\!]} \quad (22)$$

Die Leistungsbilanz für den stationären Zustand mit Kühlung durch Strahlung und frei Konvektion

```
> P[S] + P[U] = P[G] + P[K];
```

$$P_S + P_U = P_G + P_K \quad (23)$$

```
> subs((21),(7),(23));
```

$$P_S + P_U = S_G \varepsilon_G \sigma T_G^4 + \alpha S_G (T_G - T_U) \quad (24)$$

Die Zahlenwerte einsetzen ergibt eine Gleichung in der die Temperatur T_G in Kelvin die Unbekannte x ist.

```
> subs((14),(11),(13),(9),(8),(6),(22),(24)): subs(T[G]=x*Unit(K),%)/Unit(W):
  simplify(%);
```

$$78.031 = 6.057\ 10^{-9}\ x^4 + 1.508\ x - 457.139 \quad (25)$$

```
> solve((25),{x},useassumptions) assuming x>0;
```

$$\{x = 315.2332995900197300726 43\} \quad (26)$$

```
> T[G]=rhs((26)[1])*Unit(K);
```

$$T_G = 315.233\ [\![K]\!] \quad (27)$$

```
> theta[G] = ( rhs((27))/Unit(K) - 273.15 ) * Unit(Celsius);
```

$$\theta_G = 42.083\ [\![degC]\!] \quad (28)$$

In der Abschätzung fehlt noch die Wärmeleitung durch die Befestigung der Kugel.

Quellen:

[1] D. Meschede: Gerthsen Physik, Springer Verlag
[2] G. Cerbe, G. Wilhelms: Technische Thermodynamik, Hanser Verlag
[3] http://www.raytek.com/Raytek/de-r0/IREducation/EmissivityNonMetals.htm

Berechnung mit Maple