```
> restart;
> Digits := 24: interface( displayprecision = 5 ):
```

Aufgabe

Durch einen Schlauch mit einem Durchmesser von 60 mm der Wasser mit 6 bar Überdruck (statisch) führt. Die Mündungsöffnung am Ende der des Schlauches misst 20 mm und der barometrische Druck beträgt 1050hPa.
Gesucht: Mittlere Stömungsgeschwindigkeit im Schlauch und in der Mündung, Volumenstrom durch den Schlauch.

Rechenweg

Skizze



Die Durchmesser in beiden Bereich 1 und 2 sind gegeben.

```
> d[1] = 60*Unit(mm), d[2] = 20*Unit(mm); simplify([%])[]: evalf(%);
```

$$d_1 = 60 \, [\![mm]\!], d_2 = 20 \, [\![mm]\!]$$

$$d_1 = 0.060000 \, [\![m]\!], d_2 = 0.020000 \, [\![m]\!] \quad (1)$$

Der statische Druck im Bereich 2 ist der äußere Luftdruck

```
> p[2] = p[0];
```

$$p_2 = p_0 \quad (2)$$

Der äußere Luftdruck ist gegeben

```
> p[0] = 1050*Unit(hPa); simplify(%);
```

$$p_0 = 1050 \, [\![hPa]\!]$$

$$p_0 = 105000 \, [\![Pa]\!] \quad (3)$$

Der statische Druck im Bereich 1 hat eine Überdruck

```
> p[1] = p[0] + p[ü];
```

$$p_1 = p_0 + p_{ü} \quad (4)$$

Der Überdruck ist gegeben

```
> p[ü] = 6*Unit(bar); simplify(%);
```

$$p_{ü} = 6 \, [\![bar]\!]$$

$$p_{ü} = 600000 \, [\![Pa]\!] \quad (5)$$

Die Dichte von Wasser bei 20 °C [1]

```
> rho = 998*Unit(kg/m^3);
```

$$\rho = 998 \left[\!\left[\frac{kg}{m^3} \right]\!\right] \quad (6)$$

Annahmen: Bei der Strömung kann Reibung (äußere und innere Reibung) vernachlässigt werden.
Das Wasser kann als inkompressible betrachtet werden.
Es gilt das Gesetz von Bernoulli [1]

```
> p+rho*c^2/2 + rho*g*h = const;
```

$$p + \frac{1}{2}\rho c^2 + \rho g h = const \quad (7)$$

Dabei ist c die mittlere Strömungsgeschwindigkeit, g die Fallbeschleunigung, h die Höhe.
Annahme: Der Schlauch liegt waagerecht, damit ist h = const.
Die Gleichung **(7)** aufgeschrieben für die beiden Seiten.

```
> p[1] + rho*c[1]^2/2 = p[2] + rho*c[2]^2/2;
```

$$p_1 + \frac{\rho c_1^2}{2} = p_2 + \frac{\rho c_2^2}{2} \quad (8)$$

Die Drücke aus **(2)** und **(4)** einsetzen

```
> subs( (2), (4), (8) ); %-p[0];
```

$$p_0 + p_{ü} + \frac{\rho c_1^2}{2} = p_0 + \frac{\rho c_2^2}{2}$$

$$p_{ü} + \frac{\rho c_1^2}{2} = \frac{\rho c_2^2}{2} \quad (9)$$

Kontinuitätsgleichung für inkompressible Flüssigkeiten verwenden.

```
> c[1] * d[1]**2 * Pi /4 = c[2] * d[2]**2 * Pi/4;
```

$$\frac{c_1 d_1^2 \pi}{4} = \frac{c_2 d_2^2 \pi}{4} \quad (10)$$

Nach der mittleren Strömunggeschwindigkeit c_2 auflösen.

```
> isolate( (10), c[2] );
```

$$c_2 = \frac{c_1 d_1^2}{d_2^2} \quad (11)$$

Einsetzen in **(9)**

```
> subs( (11), (9) );
```

$$p_{ü} + \frac{\rho c_1^2}{2} = \frac{\rho c_1^2 d_1^4}{2 d_2^4} \quad (12)$$

Auflösen nach der Geschwindigkeit c_1

```
> isolate((12),c[1]^2): simplify(%,size): sort(%);
  sqrt(lhs(%))=sqrt(rhs(%)): simplify(%) assuming c[1]::positive;
```

$$c_1^2 = \frac{2 d_2^4 p_{ü}}{(d_1^4 - d_2^4)\rho}$$

$$c_1 = \sqrt{2}\sqrt{\frac{d_2^4 p_{ü}}{(d_1^4 - d_2^4)\rho}} \quad (13)$$

Die Zahlenwerte einsetzen und ausrechnen.

```
> subs( (1),(5),(6), (13) ): simplify(%);
```

$$c_1 = 3.8769 \left[\!\left[\frac{m}{s} \right]\!\right] \qquad (14)$$

Die mittlere Strömungsgeschwindigkeit im Schlauch beträgt 3,9 m/s.

Die Geschwindigkeit c_2 aus **(11)**

```
> (11);
```

$$c_2 = \frac{c_1 d_1^2}{d_2^2} \qquad (15)$$

Zahlenwerte einsetzen und ausrechnen.

```
> subs( (1),(5),(14), (15) );
```

$$c_2 = 34.892 \left[\!\left[\frac{m}{s} \right]\!\right] \qquad (16)$$

Die mittlere Strömungsgeschwindigkeit in der Mündung beträgt 35 m/s.

Der Volumenstrom aus Strömungsgeschwindigkeit und Querschnittsfläche.

```
> Q = c[1] * d[1]^2*Pi/4;
```

$$Q = \frac{c_1 d_1^2 \pi}{4} \qquad (17)$$

Die Zahlenwerte einsetzen und ausrechnen.

```
> subs( (1),(14), (17) ): simplify(%); lhs(%)=convert(rhs(%),'units',
  'L/s');
```

$$Q = 0.010962 \left[\!\left[\frac{m^3}{s} \right]\!\right]$$

$$Q = 10.962 \left[\!\left[\frac{L}{s} \right]\!\right] \qquad (18)$$

Der Volumenstrom beträgt 11 L/s.

Hilfsmittel

[1] Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch
[2] Bronstein et al: Taschenbuch der Mathematik, Verlag Harri Deutsch
[3] Maple 14, http://www.maplesoft.com/