

```
> restart;
> Digits := 25;
```

Aufgabe

Berechne die Punkt mit maximalem oder minimalem Abstand zum Ursprung auf der entarteten Hyperbel

```
> x^2+2*x*y+y^2=5;
```

$$x^2 + 2xy + y^2 = 5 \quad (1)$$

Rechnung

mit der Methode der Lagrange-Multiplikatoren.

Die Extremstellen des Abstands sind gesucht, also die Extremstellen von

```
> f(x,y) = sqrt( x^2+y^2 );
```

$$f(x, y) = \sqrt{x^2 + y^2} \quad (2)$$

Da die Wurzelfunktion auf den positiven Zahlen streng monoton steigend ist, ist es ausreichend die Extremstellen des Arguments zu suchen.

```
> lhs((2)) = rhs((2))^2;
```

$$f(x, y) = x^2 + y^2 \quad (3)$$

Die Nebenbedingung in Nullstellen einer Funktion umformen.

```
> (1)-rhs((1));
```

$$x^2 + 2xy + y^2 - 5 = 0 \quad (4)$$

```
> psi(x,y) = lhs((4));
```

$$\psi(x, y) = x^2 + 2xy + y^2 - 5 \quad (5)$$

```
> psi(x,y)=0;
```

$$\psi(x, y) = 0 \quad (6)$$

Es gibt eine Nebenbedingung, also einen Lagrange-Multiplikator einführen: λ .

Die Lagrange-Funktion bilden.

```
> L(x,y) = f(x,y) + lambda*psi(x,y);
```

$$L(x, y) = f(x, y) + \lambda \psi(x, y) \quad (7)$$

```
> subs((3),(5),(7));
```

$$L(x, y) = x^2 + y^2 + \lambda (x^2 + 2xy + y^2 - 5) \quad (8)$$

Das Lagrange Gleichungssystem aufstellen.

```
> psi(x,y)=0;
```

$$\psi(x, y) = 0 \quad (9)$$

```
> diff(L(x,y),x) = 0;
```

$$\frac{\partial}{\partial x} L(x, y) = 0 \quad (10)$$

```
> diff(L(x,y),y) = 0;
```

$$\frac{\partial}{\partial y} L(x, y) = 0 \quad (11)$$

Die drei Gleichungen (9), (10) und (11) bilden ein Gleichungssystem für die drei Unbekannte x , y und λ .

```
> subs((5),(9));
```

$$x^2 + 2xy + y^2 - 5 = 0 \quad (12)$$

```
> subs((8),(10)); simplify(%);
```

$$\frac{\partial}{\partial x} (x^2 + y^2 + \lambda (x^2 + 2xy + y^2 - 5)) = 0$$

$$2x + 2\lambda x + 2\lambda y = 0 \quad (13)$$

```
> subs((8),(11)); simplify(%);
```

$$\frac{\partial}{\partial y} (x^2 + y^2 + \lambda (x^2 + 2xy + y^2 - 5)) = 0$$

$$2y + 2\lambda x + 2\lambda y = 0 \quad (14)$$

Die Lösung berechnet mit Maples solve Funktion:

```
> solve({(12),(13),(14)}, {x,y,lambda});
```

$$\left\{ x = \frac{\text{RootOf}(Z^2 - 5)}{2}, y = \frac{\text{RootOf}(Z^2 - 5)}{2}, \lambda = -\frac{1}{2} \right\} \quad (15)$$

Die Lösung in "Handarbeit".

```
> (13)-(14); isolate(% , x);
```

$$2x - 2y = 0$$

$$x = y \quad (16)$$

```
> subs((16),(12)); solve(% , {y}); evalf(%);
```

$$4y^2 - 5 = 0$$

$$\left\{ y = \frac{\sqrt{5}}{2} \right\}, \left\{ y = -\frac{\sqrt{5}}{2} \right\}$$

$$\{y = 1.118033988749894848204587\}, \{y = -1.118033988749894848204587\} \quad (17)$$

Der Suchbereich und die gefundenen Extremstellen darstellen.

```
> r := plots[implicitplot]( (12), x=-3..+3, y=-3..+3 );
```

```
> c := map( z->rhs(z[1]), [(17)] );
```

```
> s := plots[pointplot]( {[c[1],c[1]],[c[2],c[2]]}, symbol=circle,
symbolsize=15 );
```

```
> plots[display]( r, s );
```

