

```
> restart;
> Digits:=30: interface( displayprecision = 7 );
```

## Aufgabe


Ein Ball wird von einer Rampe die sich in 35 Meter Höhe befindet nach oben geworfen.  
Die Zeit vom Abwurf bis zum Aufschlag beträgt 4,5 Sekunden.

## Ansatz

Koordinatensystem wählen:

- Die Höhe muss berücksichtigt werden: Koordinate x, positiv nach oben, Nullpunkt auf Bodenhöhe.
- Die Zeit t mit Nullpunkt beim Abwurf.

Skizze nach der Aufgabenbeschreibung



Der Ball wird ab Position  $x_0$  nach oben geworfen. Der Ball erreicht die Höhe  $x_H$ . Danach fällt der Ball bis auf die Höhe 0.

## Rechnung

Vereinfachung:

- keine Reibung des Balls mit der Luft.
- Ballgröße vernachlässigt. Als Massepunkt beschrieben.

Auf den Ball wirkt nach dem Wurf nur die Schwerkraft, die zur Fallbeschleunigung  $g$  führt. Die Fallbeschleunigung ist nach unten gerichtet.

```
> g = -9.81*Unit(m/s^2);
```

$$g = -9.81 \left[ \frac{m}{s^2} \right] \quad (1)$$

Der Ball ist zum Startzeitpunkt  $t = 0$  auf der Höhe

```
> x[0] = 35*Unit(m);
```

$$x_0 = 35 \text{ [m]} \quad (2)$$

Der Ball hat zum Startzeitpunkt die Geschwindigkeit

```
> v[0];
```

(3)

$$v_0 \quad (3)$$

Die Bewegungsgleichung des Balls.

```
> x = x[0] + v[0] * t + g * t^2/2;
```

$$x = x_0 + v_0 t + \frac{1}{2} g t^2 \quad (4)$$

Zum Zeitpunkt T ist der Ball aufgeschlagen.

```
> T = 4.5*Unit(s);
```

$$T = 4.5 \text{ [s]} \quad (5)$$

```
> subs( t=T, alsubs(x=0,(4)) );
```

$$0 = x_0 + v_0 T + \frac{1}{2} g T^2 \quad (6)$$

Auflösen nach der unbekannten Anfangsgeschwindigkeit.

```
> isolate( (6), v[0] ); collect(%,T);
```

$$v_0 = - \frac{x_0 + \frac{g T^2}{2}}{T}$$

$$v_0 = - \frac{g T}{2} - \frac{x_0}{T} \quad (7)$$

Die Geschwindigkeit des Balls als Funktion der Zeit

```
> v = v[0] + g * t;
```

$$v = g t + v_0 \quad (8)$$

## ▼ Einschub: Skizze der Bewegung

Die Skizze der Bewegung ist nicht notwendig für die Lösung der Aufgabe. Doch trägt die Skizze zum Verständnis bei

Die Startgeschwindigkeit auf (7) ausrechnen.

```
> subs( (1), (2), (5), (7) ): simplify(%);
```

$$v_0 = 14.29472 \left[ \frac{m}{s} \right] \quad (9)$$

Mit der Startgeschwindigkeit können alle Zahlenwerte in die Bewegungsgleichung (4) eingesetzt werden.

```
> subs( (1),(2),(9), (4) );
```

$$x = 35 \text{ [m]} + 14.29472 t \left[ \frac{m}{s} \right] - 4.905000 t^2 \left[ \frac{m}{s^2} \right] \quad (10)$$

```
> plot( subs(t=t*Unit(s),rhs((10))), t=0..rhs((5))/Unit(s), axes=boxed, gridlines=true, labels=["Zeit t/s","Höhe x/m"] );
```




Auch in die Gleichung (8) für die Geschwindigkeit können alle Wert eingesetzt werden.

```
> subs( (1),(9), (8) );
```

$$v = -9.81 t \left[ \frac{m}{s^2} \right] + 14.29472 \left[ \frac{m}{s} \right] \quad (11)$$

```
> plot( subs(t=t*Unit(s),rhs((11))), t=0..rhs((5))/Unit(s), axes=
boxed, gridlines=true, labels=["Zeit t/s","Geschwindigkeit v/
m/s"] );
```



Der Ball steigt auf bis die Geschwindigkeit 0 ist. Nach diesem Zeitpunkt beginnt der Fall aus dieser Ruhelage.

```
> subs( t=t[H], algs subs( v=0, (8) ) );
```

$$0 = g t_H + v_0 \quad (12)$$

```
> isolate( (12), t[H] );
```

$$t_H = -\frac{v_0}{g} \quad (13)$$

Die Fallzeit ist die Zeit vom Scheitelpunkt bis T.

```
> t[F] = T - t[H];
```

$$t_F = T - t_H \quad (14)$$

Gleichung (13) einsetzen.

```
> subs ( (13),(14) );
```

$$t_F = T + \frac{v_0}{g} \quad (15)$$

Gleichung (7) einsetzen.

```
> subs ( (7), (15) ); collect(%,T);
```

$$t_F = T + \frac{-\frac{gT}{2} - \frac{x_0}{T}}{g}$$

$$t_F = \frac{T}{2} - \frac{x_0}{gT} \quad (16)$$

Zahlenwerte einsetzen und ausrechnen.

```
> subs ( (1),(2),(5), (16) ): simplify(%);
```

$$t_F = 3.042842 \text{ [s]} \quad (17)$$

Die Position des Scheitelpunkts folgt aus dem Einsetzen der Zeit  $t_H$  in die Gleichung (4).

```
> subs ( t=t[H] , (4) ): algsubs(x=x[H],% );
```

$$x_H = x_0 + v_0 t_H + \frac{1}{2} g t_H^2 \quad (18)$$

Die Gleichung (15) einsetzen.

```
> subs ( (13), (18) );
```

$$x_H = x_0 - \frac{v_0^2}{2g} \quad (19)$$

Die Gleichung (7) einsetzen.

```
> subs ( (7), (19) ); collect(%,T);
```

$$x_H = x_0 - \frac{\left(-\frac{gT}{2} - \frac{x_0}{T}\right)^2}{2g}$$

$$x_H = -\frac{gT^2}{8} + \frac{x_0}{2} - \frac{x_0^2}{2gT^2} \quad (20)$$

Zahlenwerte einsetzen und ausrechnen.

```
> subs ( (1),(2),(5), (20) ): simplify(%);
```

$$x_H = 45.41484 \text{ [m]} \quad (21)$$

Der Ball fällt vom Scheitelpunkt eine Streckt von 45,4 Meter in 3,04 Sekunden.

## ▼ Hilfsmittel

[Maple 17, <http://www.maplesoft.com/>]