```
> restart;
> Digits := 30: interface( displayprecision=7 ):
```

Aufgabe

Gegeben ist eine Messung der Frequenz f als Funktion der Kraft F. Berechnet werden soll eine Ausgleichsgerade $f = c_0 + c_1 \sqrt{F}$. Zur Steigung c_1 soll die Standardabweichung s_{c1} berechnet werden.

Anzahl der Wertepaare

```
> n = 5;
```

$$n = 5 \quad (1)$$

Gemessene Frequenzen

```
> f = [127.1 , 155.5 , 169.6 , 182.6 , 209.0] *~ Unit(Hz);
```

$$f = [127.1 ⟦Hz⟧, 155.5 ⟦Hz⟧, 169.6 ⟦Hz⟧, 182.6 ⟦Hz⟧, 209.0 ⟦Hz⟧] \quad (2)$$

Frequenzen gemessen bei den Kräften

```
> F = [ 10.0 , 15.0 , 17.5 , 20.0 , 25.0 ] *~ Unit(N);
```

$$F = [10.0 ⟦N⟧, 15.0 ⟦N⟧, 17.5 ⟦N⟧, 20.0 ⟦N⟧, 25.0 ⟦N⟧] \quad (3)$$

Rechenweg

Die Gerade ist eine Funktion der $\sqrt{F}$, also die Wurzeln ausrechnen.

```
> x = sqrt(F);
```

$$x = \sqrt{F} \quad (4)$$

```
> x = sqrt~( rhs((3)) );
```

$$x = \left[3.162278 \sqrt{⟦N⟧}, 3.872983 \sqrt{⟦N⟧}, 4.183300 \sqrt{⟦N⟧}, 4.472136 \sqrt{⟦N⟧}, 5.000000 \sqrt{⟦N⟧}\right] \quad (5)$$

Formeln für die Berechnung der Ausgleichsgeraden gibt es in verschiedenen Schreibweisen. Hier die Formeln aus dem "Physikalisch-Technisches Taschenbuch von Hering, Martin, Stohrer". Alternativ könnten die internen Funktionen von Mathematikprogrammen, Statistikprogrammen oder Taschenrechner verwendet werden.

Zwischenwerte A bis F.

```
> A = n;
```

$$A = n \quad (6)$$

```
> B = sum(x[i],i=1..n);
```

$$B = \sum_{i=1}^{n} x_i \quad (7)$$

```
> C = sum(x[i]^2,i=1..n);
```

$$C = \sum_{i=1}^{n} x_i^2 \quad (8)$$

```
> D = sum(y[i],i=1..n);
```

$$D = \sum_{i=1}^{n} y_i \quad (9)$$

```
> E = sum(y[i]*x[i],i=1..n);
```

(10)

$$E = \sum_{i=1}^{n} y_i x_i \tag{10}$$

```
> F = sum(y[i]^2,i=1..n);
```

$$F = \sum_{i=1}^{n} y_i^2 \tag{11}$$

Parameter der Ausgleichsgeraden y = $c_0 + c_1$ x.

```
> c[0] = (C*D-B*E)/(A*C-B^2);
```

$$c_0 = \frac{C\,\mathrm{D} - B\,E}{A\,C - B^2} \tag{12}$$

```
> c[1] = (A*E-B*D)/(A*C-B^2);
```

$$c_1 = \frac{A\,E - B\,\mathrm{D}}{A\,C - B^2} \tag{13}$$

Standardabweichung des jeweiligen Parameters.

```
> s[c0] = sqrt( (F-2*c[0]*D-2*c[1]*E+2*c[0]*c[1]*B+c[0]^2*A+c[1]^2*
  C)*C/((n-2)*(A*C-B^2)) );
```

$$s_{c0} = \sqrt{\frac{\left(F - 2\,c_0\,\mathrm{D} - 2\,c_1\,E + 2\,c_0\,c_1\,B + c_0^2\,A + c_1^2\,C\right)\,C}{(n-2)\,\left(A\,C - B^2\right)}} \tag{14}$$

```
> s[c1] = sqrt( (F-2*c[0]*D-2*c[1]*E+2*c[0]*c[1]*B+c[0]^2*A+c[1]^2*
  C)*A/((n-2)*(A*C-B^2)) );
```

$$s_{c1} = \sqrt{\frac{\left(F - 2\,c_0\,\mathrm{D} - 2\,c_1\,E + 2\,c_0\,c_1\,B + c_0^2\,A + c_1^2\,C\right)\,A}{(n-2)\,\left(A\,C - B^2\right)}} \tag{15}$$

Soweit die Formelsammlung.

Einsetzen der Werte n, x, y=f.

```
> subs( y=f,(1),(2),(5), (6) );
```

$$A = 5 \tag{16}$$

```
> subs( y=f,(1),(2),(5), (7) ): simplify(%);
```

$$B = 20.69070 \left[\!\left[\frac{\sqrt{m}\,\sqrt{kg}}{s} \right]\!\right] \tag{17}$$

```
> subs( y=f,(1),(2),(5), (8) ): simplify(%);
```

$$C = 87.50000 \left[\!\left[N \right]\!\right] \tag{18}$$

```
> subs( y=f,(1),(2),(5), (9) ): simplify(%);
```

$$\mathrm{D} = 843.8 \left[\!\left[\frac{1}{s} \right]\!\right] \tag{19}$$

```
> subs( y=f,(1),(2),(5), (10) ): simplify(%);
```

$$E = 3575.274 \left[\!\left[\frac{\sqrt{m}\,\sqrt{kg}}{s^2} \right]\!\right] \tag{20}$$

```
> subs( y=f,(1),(2),(5), (11) ): simplify(%);
```

$$F = 146122.6 \left[\!\left[\frac{1}{s^2} \right]\!\right] \tag{21}$$

```
> subs( (16),(17),(18),(19),(20),(21), (12) ): simplify(%);
```

$$c_0 = -15.15840 \left[\!\left[\frac{1}{s} \right]\!\right] \quad (22)$$

```
> subs( (16),(17),(18),(19),(20),(21), (13) ): simplify(%);
```

$$c_1 = 44.44471 \left[\!\left[\frac{1}{\sqrt{m}\,\sqrt{kg}} \right]\!\right] \quad (23)$$

```
> subs( (1),(16),(17),(18),(19),(20),(21),(22),(23), (14) ): simplify(%);
```

$$s_{c0} = 5.902281 \left[\!\left[\frac{1}{s} \right]\!\right] \quad (24)$$

```
> subs( (1),(16),(17),(18),(19),(20),(21),(22),(23), (15) ): simplify(%);
```

$$s_{c1} = 1.410915 \left[\!\left[\frac{1}{\sqrt{m}\,\sqrt{kg}} \right]\!\right] \quad (25)$$

Die Ausgleichsgeraden ist

```
> f = c[0] + c[1]*sqrt(F);
```

$$f = c_0 + c_1 \sqrt{F} \quad (26)$$

```
> subs( (22),(23), (26) );
```

$$f = -15.15840 \left[\!\left[\frac{1}{s} \right]\!\right] + 44.44471 \sqrt{F} \left[\!\left[\frac{1}{\sqrt{m}\,\sqrt{kg}} \right]\!\right] \quad (27)$$

Die Steigung der Ausgleichsgeraden ist (44,4 ± 1,4) $\frac{1}{\sqrt{kg\,m}}$.

Plot der Messwerte ($x = \sqrt{F}$, y) und der berechneten Ausgleichsgeraden.

```
> d1 := plot(subs(sqrt(F)=x,subsindets(rhs((27)),has_unit,t->convert
  (t,unit_free))),x=0.0..6.0):
> d2 := plots[pointplot]( rhs((5))/~sqrt(Unit(N)),rhs((2))/~Unit(Hz),
  symbol=diagonalcross,symbolsize=15 ):
> plots[display](d1,d2,axes=boxed,labels=[x=sqrt(F/Unit(N)),y/Unit
  (Hz)]);
```



Kurzer Lösungsweg mit der eingebauten Fitfunktion von Maple.

```
> Statistics[LinearFit]( [1,x], rhs((5))/~sqrt(Unit(N)), rhs((2))
  /~Unit(Hz), x, output=[leastsquaresfunction,parametervector,
  standarderrors] );
```

$$\left[-15.15840 + 44.44471\, x, \begin{bmatrix} -15.1584048 \\ 44.4447096 \end{bmatrix}, \begin{bmatrix} 5.9022811 & 1.4109150 \end{bmatrix}\right] \quad \textbf{(28)}$$

Eine Ursprungsgerade fitten. (Die Theorie zur Aufgabe ergibt eine Ursprungsgerade.)

```
> Statistics[LinearFit]( [x], rhs((5))/~sqrt(Unit(N)), rhs((2))/~Unit
  (Hz), x, output=[leastsquaresfunction,parametervector,
  standarderrors] );
```

$$\left[40.86028\, x, \begin{bmatrix} 40.8602758 \end{bmatrix}, \begin{bmatrix} 0.3202373 \end{bmatrix}\right] \quad \textbf{(29)}$$