

```

> restart;
> parameters := {};
> addparameter := proc( x ) global parameters; parameters :=
  parameters union {x}; end proc;
> Digits:=30: interface(displayprecision=7):

```

## Aufgabe

Ein Gleichstrom-Nebenschlussmotor arbeitet bei Nennerregung mit den Daten:

Spannung am Motor

```

> U[N] = 440*Unit(V); addparameter(%):

```

$$U_N = 440 \text{ [V]} \quad (1)$$

Strom im Ankerkreis bei Nenndrehzahl

```

> I[N] = 120*Unit(A); addparameter(%):

```

$$I_N = 120 \text{ [A]} \quad (2)$$

Nenndrehzahl

```

> n[N] = 947*Unit(1/min); simplify(evalf(%)); addparameter(%):

```

$$n_N = 947 \left[ \frac{1}{\text{min}} \right]$$

$$n_N = 15.7833333 \left[ \frac{1}{s} \right] \quad (3)$$

Leerlaufdrehzahl

```

> n[0]=1028*Unit(1/min); simplify(evalf(%)); addparameter(%):

```

$$n_0 = 1028 \left[ \frac{1}{\text{min}} \right]$$

$$n_0 = 17.1333333 \left[ \frac{1}{s} \right] \quad (4)$$

1. Wie groß ist die induzierte Spannung im Nennbetrieb?
2. Welche Nennleistung liefert die Maschine?
3. Wie hoch ist das Nennmoment?
4. Wie groß ist der Ankerkreiswiderstand?
5. Welchen Wert muss ein Vorwiderstand haben, damit der Einschaltstrom den 1,5fachen Wert des Nennstromes nicht überschreitet?
6. Um welchen Faktor unterscheiden sie die Anlaufmomente mit und ohne Anlasswiderstand ?

## Rechnung

Näherungen/Vereinfachungen

- Mechanische Reibung im Motor wird vernachlässigt.
- Eisenverlust im Anker wird vernachlässigt.
- Spannungsabfall durch die Bürsten wird vernachlässigt.

Die induzierte Spannung im Anker ist proportional zur Drehzahl. Die Proportionalitätskonstante ist abhängig vom Motor und von der Erregung. Hier ist die Erregung konstant.

Im Leerlauf ist die induzierte Spannung gleich der angelegten Spannung.

```

> U[N] = k * n[0];

```

$$U_N = k n_0 \quad (5)$$

```

> isolate((5),k);

```

$$k = \frac{U_N}{n_0} \quad (6)$$

Im Nennbetrieb läuft die Maschine mit Nenndrehzahl. Die induzierte Spannung ist dann

> **U[i] = k\*n[N] ;**

$$U_i = k n_N \quad (7)$$

Gleichung (6) einsetzen.

> **subs ((6),(7)) ;**

$$U_i = \frac{U_N n_N}{n_0} \quad (8)$$

Zahlenwerte einsetzen und ausrechnen.

> **subs ( parameters , (8) ) ;**

$$U_i = 405.3307393 \text{ [ V ]} \quad (9)$$

### Ergebnis 1

Die induzierte Spannung im Nennbetrieb beträgt 405 V.

Die abgegebene Leistung  $P_N$  des Motors im Nennbetrieb.

> **P[N] = U[i] \* I[N] ;**

$$P_N = U_i I_N \quad (10)$$

Gleichung (8) einsetzen. (Je nach Lehrer muss eine Formel hergeleitet werden, die nur aus den gegebenen Werten besteht oder es dürfen die schon bekannten Zwischenergebnisse verwendet werden. Damit der Rechenweg in jedem Fall vollständig ist, hier die Formel aus den gegebenen Werten aufgeschrieben.)

> **subs ((8),(10)) ;**

$$P_N = \frac{U_N n_N I_N}{n_0} \quad (11)$$

Zahlenwerte einsetzen und ausrechnen.

> **subs (parameters ,(11)) : simplify (%) ;**

$$P_N = 48639.6887160 \text{ [ W ]} \quad (12)$$

### Ergebnis 2

Im Nennbetrieb gibt die Maschine 48,6 kW Leistung ab.

Die Leistung  $P$  ist proportional zur Drehzahl  $n$  und zum Moment  $M$ . Im Nennbetrieb:

> **P[N] = 2\*Pi\*M[N]\*n[N] ;**

$$P_N = 2 \pi M_N n_N \quad (13)$$

Das Moment ist gesucht.

> **isolate ((13),M[N]) ;**

$$M_N = \frac{1}{2} \frac{P_N}{\pi n_N} \quad (14)$$

Gleichung (11) einsetzen.

> **subs ( (11), (14) ) ;**

$$M_N = \frac{1}{2} \frac{U_N I_N}{n_0 \pi} \quad (15)$$

Zahlenwerte einsetzen und ausrechnen.

```
> subs( parameters, (15) ): simplify(%): lhs(%)=convert(rhs(%),  
units,N*m);
```

$$M_N = 490.4697079 \text{ [Nm]} \quad (16)$$

### Ergebnis 3

Im Nennbetrieb hat die Maschine ein Drehmoment von 490 Nm.

Der Ankerkreis hat einen ohmschen Widerstand  $R_A$  (ohmscher Widerstand der Ankerwicklung) und eine Spannung  $U_i$  wird in den Kreis induziert.

```
> U[N] = R[A]*I[N] + U[i];
```

$$U_N = I_N R_A + U_i \quad (17)$$

Auflösen nach dem gesuchten Widerstand.

```
> isolate((17),R[A]);
```

$$R_A = - \frac{-U_N + U_i}{I_N} \quad (18)$$

Gleichung (8) einsetzen.

```
> subs((8),(18)); simplify(% ,size);
```

$$R_A = - \frac{-U_N + \frac{U_N n_N}{n_0}}{I_N}$$
$$R_A = \frac{U_N (n_0 - n_N)}{n_0 I_N} \quad (19)$$

Zahlenwerte einsetzen und ausrechnen.

```
> subs(parameters,(19)): simplify(%);
```

$$R_A = 0.2889105 \text{ [}\Omega\text{]} \quad (20)$$

### Ergebnis 4

Der Ankerkreis hat einen Widerstand von 0,289  $\Omega$ .

Beim Einschalten steht der Anker still, damit wird keine Spannung induziert. Der Strom folgt alleine aus dem Ankerwiderstand  $R_A$  und dem Vorwiderstand  $R_{vor}$ .

Der Strom soll das f-fache des Nennstroms sein. Die angelegte Spannung ist immer die Nennspannung.

```
> f * I[N] = U[N] / ( R[A] + R[vor] );
```

$$f I_N = \frac{U_N}{R_A + R_{vor}} \quad (21)$$

Auflösen nach dem gesuchten Vorwiderstand.

```
> isolate( (21), R[vor] );
```

$$R_{vor} = \frac{U_N}{f I_N} - R_A \quad (22)$$

Gleichung (19) einsetzen.

```
> subs( R[vor]=Rvor, (19), Rvor = R[vor], (22) ); simplify(%);
```

$$R_{vor} = \frac{U_N}{f I_N} - \frac{U_N (n_0 - n_N)}{n_0 I_N}$$

$$R_{vor} = - \frac{U_N (f n_0 - f n_N - n_0)}{f I_N n_0} \quad (23)$$

Das 1,5-fache des Nennstroms ist in der Aufgabe vorgegeben.

> **f = 1.5;**

$$f = 1.5 \quad (24)$$

Der Wert für f und die Zahlenwerte aus der Aufgabe einsetzen und ausrechnen.

> **subs( (24), parameters, (23) ): simplify(%);**

$$R_{vor} = 2.1555339 \llbracket \Omega \rrbracket \quad (25)$$

### Ergebnis 5

Bei einem Vorwiderstand von 2,15 Ω wird der 1,5-fache Nennstrom beim Einschalten erreicht.

Das Drehmoment ist proportional zum Ankerstrom I bei konstantem Erregerfeld.

> **M = k\*I/(2\*Pi);**

$$M = \frac{1}{2} \frac{I k}{\pi} \quad (26)$$

Mit Vorwiderstand ist der Ankerstrom ist das f-fache des Nennstroms. Entsprechend ist das Moment beim Anlauf.

> **subs( M=M[vor], I=f\*I[N], (26) );**

$$M_{vor} = \frac{1}{2} \frac{f I_N k}{\pi} \quad (27)$$

Ohne Vorwiderstand ist der Ankerstrom beim Anlauf

> **I = U[N]/R[A];**

$$I = \frac{U_N}{R_A} \quad (28)$$

Ohne Vorwiderstand ist damit das Drehmoment

> **subs( M=M[ohne], (28), (26) );**

$$M_{ohne} = \frac{1}{2} \frac{U_N k}{R_A \pi} \quad (29)$$

Das Verhältnis v der beiden Moment.

> **v = M[ohne] / M [vor];**

$$v = \frac{M_{ohne}}{M_{vor}} \quad (30)$$

Gleichungen (27) und (29) eingesetzt. Die meisten Werte kürzen sich.

> **subs( (27), (29), (30) );**

$$v = \frac{U_N}{R_A f I_N} \quad (31)$$

Der Ankerwiderstand aus (19) einsetzen-

> **subs( (19), (31) );**

$$v = \frac{n_0}{(n_0 - n_N) f} \quad (32)$$

Die Zahlenwerte einsetzen und ausrechnen.

**> subs( (24), parameters, (32) );**

$$v = 8.46090534979423868312757201647 \quad (33)$$

### Ergebnis 6

Das Anlaufdrehmoment ohne Vorwiderstand ist um den Faktor 8,46 größer als mit Vorwiderstand.

## ▼ Hilfsmittel

- Fuest, Döring: Elektrische Maschinen und Antriebe, Vieweg
- Maple 17, <http://www.maplesoft.com/>