Aufgabe

Reibung im Flaschenzug.



Die Reibung wird hier nicht vernachlässigt. Die Gleitreibung beträgt μ=0,25 zwischen den Umlenkrollen und den Bolzen. Die Haftung beträgt 125% der Gleitreibung.

Die "Feder" ist ein mechanisches System und zieht mit einer konstanten Kraft von 100 N.

Welche Kraft muss am oben austretenden Seil wirken, damit die Flaschen gegen die Federkraft zusammengezogen wird?
Wie groß ist die Kraft an dem oben austretenden Seil, wenn die Feder die Flaschen auseinanderzieht?
Welche Kraft muss die Feder haben um die Flaschen aus der Ruhe gegen die Haftung auseinander zu ziehen?

Vorüberlegung

Zu beschreiben ist ein Flaschenzug mit Reibung. Eine fertige Formel für diesen Aufbau habe ich nicht gefunden. Also muss die Formel selbst zusammengebaut werden.

Annahme: Das Seil hat eine geringe Masse, die Rollschlitten bewegen sich mit konstanter Geschwindigkeit. Dann können Trägheitskräfte vernachlässigt werden.

Annahme: Kräfte auf den Flaschenzug direkt durch die Schwerkraft kann vernachlässigt werden. (Eine indirekte Wirkung durch die Masse einer angehängten Last darf vorhanden sein. Dann würde diese Kraft durch die Feder aufbracht werden.)

Annahme: Die Reibung zwischen Seil und Rolle (Umschlingungsreibung) ist immer so groß, dass die Seile nicht rutschen. Jede Rolle bewegt sich genau wie das Seil auf der Rolle.

Die Reibungskraft ist immer der Bewegung entgegengerichtet. Die beiden Bewegungsrichtungen (Flaschen auseinander, Flaschen zusammen) werden getrennt betrachtet.

```
> restart;
> Digits := trunc( evalhf(Digits) ): interface(displayprecision=
  5):
```

Eine Rolle mit Reibung

Ziel ist den Wirkungsgrad der Kraftübertragung einer Umlenkrolle zu bestimmen.

Um die Rolle läuft ein Seil. Auf einer Seite zieht die Kraft F_1, auf der anderen Seite die Kraft F_2. Das Seil läuft in die Richtung von Kraft F_1. Die Gleitreibung wirkt der Bewegung entgegen. Daher muss die Kraft F_1 größer als die Kraft F_2 sein.



Zwischen Rolle und Achse tritt eine Gleitreibung auf, der Reibwert ist μ. Die Achse hat einen Außenradius, der der gering kleiner ist als der Innenradius der Rolle. Durch die Kräfte wird die Rolle an einer Stelle (eine Linie, in der hier dargestellten Projektion ein Punkt) gegen die Achse gedrückt.

Die Rolle freigeschnitten und ein Koordinatensystem im Mittelpunkt der Achse festgelegt.



Der Radius R_a ist der Radius der Achse (Bolzen). Der Radius R_u ist der Radius mit dem das Seil die Rolle umschlingt. Beim Radius R_a greift die Reibungskraft an. Beim Radius R_u greifen die beiden Seilkräfte an.

Die Kraft F_a wirkt von der Achse auf die Rolle. Die Kraft steht senkrecht auf dem Innenkreis der Rolle. Richtung der Kraft und Ansatzpunkt der Kraft hängen zusammen.
Die Reibungskraft F_r wirkt tangential an der Rolle. Die Richtung ist abhängig von der Drehrichtung. Die Rolle dreht in Richtung der Kraft F_1.

Die Reibungskraft folgt aus der Normalkraft nach dem Coulombschen Reibungsgesetz.

```
> F[r] = mu * F[a];
```

$$F_r = \mu F_a \tag{3.1}$$

Die Komponenten der Kräfte. Der Winkel zwischen der Kraft F_a und der negativen x-Achse bekommt den Namen α. (Die negative x-Achse, damit der Winkel α < 90° ist und die Winkelfunktionen nur im ersten Quadranten ausgewertet werden müssen.)

```
> F[a,x] = -F[a]*cos(alpha);
```

$$F_{a,x} = -F_a \cos(\alpha) \tag{3.2}$$

```
> F[a,y] = F[a]*sin(alpha);
```

$$F_{a,y} = F_a \sin(\alpha) \tag{3.3}$$

Die Kraft F_r ist um 90° zur Kraft F_a gedreht.

```
> F[r,x] = -F[r]*sin(alpha);
```

$$F_{r,x} = -F_r \sin(\alpha) \tag{3.4}$$

```
> F[r,y] = -F[r]*cos(alpha);
```

$$F_{r,y} = -F_r \cos(\alpha) \tag{3.5}$$

Die Kräfte F_1 und F_2 sind in x-Richtung orientiert, brauchen daher nicht zerlegt zu werden.
Die Kräfte sind im Gleichgewicht (d'Alembertsches Prinzip).
Die Summe der x-Komponenten ist 0.

```
> F[a,x]+F[r,x]+F[1]+F[2] = 0;
```

$$F_{a,x} + F_{r,x} + F_1 + F_2 = 0 \tag{3.6}$$

Die Summe der y-Komponenten ist 0.

```
> F[a,y]+F[r,y]=0;
```

$$F_{a,y} + F_{r,y} = 0 \tag{3.7}$$

Das bekannte **(3.1)** bis **(3.5)** einsetzen.

```
> subs( (3.2),(3.4),(3.1), (3.6) );
```

$$-F_a \cos(\alpha) - \mu F_a \sin(\alpha) + F_1 + F_2 = 0 \tag{3.8}$$

```
> subs( (3.3),(3.5),(3.1), (3.7) );
```

$$F_a \sin(\alpha) - \mu F_a \cos(\alpha) = 0 \tag{3.9}$$

Aus **(3.9)** kann der Winkel α berechnet werden.

```
> (3.9)/(F[a]*cos(alpha)); expand(%): convert(%,tan): isolate(%,tan
  (alpha));
```

$$\frac{F_a \sin(\alpha) - \mu F_a \cos(\alpha)}{F_a \cos(\alpha)} = 0$$

$$\tan(\alpha) = \mu \tag{3.10}$$

Der Winkel α wird Reibungswinkel genannt, er tritt zum Beispiel auch beim Reibungskegel auf.
Über die Zusammenhänge zwischen den Trigonometrischen Funktionen sin(α) und cos(α) aufschreiben.

```
> sin(alpha) = sin( solve((3.10),alpha) );
```

$$\sin(\alpha) = \frac{\mu}{\sqrt{\mu^2 + 1}} \tag{3.11}$$

```
> cos(alpha) = cos( solve((3.10),alpha) );
```

(3.12)

$$\cos(\alpha) = \frac{1}{\sqrt{\mu^2+1}} \tag{3.12}$$

Die bekannten Winkelfunktionen in Gleichung **(3.8)** eingesetzen.

```
> subs( (3.11),(3.12), (3.8) );
```

$$-\frac{F_a}{\sqrt{\mu^2+1}} - \frac{\mu^2 F_a}{\sqrt{\mu^2+1}} + F_1 + F_2 = 0 \tag{3.13}$$

Auflösen nach der Kraft F_a.

```
> isolate( (3.13), F[a] ): simplify(%);
```

$$F_a = \frac{F_1 + F_2}{\sqrt{\mu^2+1}} \tag{3.14}$$

Damit ist das Kräftegleichgewicht ausgeschöpfte. Weiter kann das Drehmomentengleichgewicht ausgenutzt werden.

Die Drehmomente um den Nullpunkt aufstellen. Die Kraft F_s wirkt radial. Alle anderen Kräfte wirken tangential. Das Kreuzprodukt zwischen Kraft und Radiusvektor wird daher zu 0 (bei F_a) oder zum Produkt der Beträge.

```
> F[r]*R[a]+F[2]*R[u]-F[1]*R[u] = 0;
```

$$F_r R_a + F_2 R_u - F_1 R_u = 0 \tag{3.15}$$

Coulomb-Reibung **(3.1)** und Achskraft F_a **(3.14)** einsetzen.

```
> subs( (3.1),(3.14), (3.15) );
```

$$\frac{\mu (F_1 + F_2) R_a}{\sqrt{\mu^2+1}} + F_2 R_u - F_1 R_u = 0 \tag{3.16}$$

Die Gleichung nach F_2 auflösen.

```
> isolate((3.16), F[2] ): simplify(%,size): sort(%);
```

$$F_2 = \frac{\left(-\mu R_a + \sqrt{\mu^2+1}\; R_u\right) F_1}{\mu R_a + \sqrt{\mu^2+1}\; R_u} \tag{3.17}$$

Die Kräfte an den beiden Seilenden sind proportional. Der Koeffizient η ist der Wirkungsgrad der Kraftübertragung über die Umlenkrolle.

```
> F[2] = eta*F[1];
```

$$F_2 = \eta F_1 \tag{3.18}$$

Der Wirkungsgrad η der Kraftübertragung ist abhängig vom Gleitreibungskoeffizienten μ und dem Verhältnis $\rho = R_u/R_a$ der beiden Radien. (Es ist immer $\rho > 1$, $\mu > 0$ und $\eta < 1$.)

```
> eta = coeff( rhs((3.17)) , F[1] );
  algsubs(R[u]=rho*R[a],%): simplify(%,size): sort(%);
```

$$\eta = \frac{-\mu R_a + \sqrt{\mu^2+1}\; R_u}{\mu R_a + \sqrt{\mu^2+1}\; R_u}$$

(3.19)

$$\eta = \frac{-\mu + \sqrt{\mu^2 + 1}\ \rho}{\mu + \sqrt{\mu^2 + 1}\ \rho} \qquad (3.19)$$

Flaschen mit Seil gegen Federkraft zusammenziehen

Die Kräfte der Seile in der Skizze des Flaschenzugs eintragen.



Die Flaschen bewegen sich aufeinanderzu. Das obere Seil wird aus dem Flaschenzug hinausgezogen. Die Reibung wirkt dieser Bewegung entgegen. Daher ist die Kraft F_2 kleiner als die Kraft F_1 und so weiter für die anderen Rollen. Also ist die Kraft F_{k+1} kleiner als die Kraft F_k.

Den oben berechneten Wirkungsgrad der Kraftübertragung durch eine Umlenkrolle verwenden.

```
> F[k+1] = eta * F[k];
```

$$F_{k+1} = \eta\, F_k \qquad (4.1)$$

Diesen Zusammenhang auf die Rolle 1 und 6 anwenden.

```
> subs(k=1,(4.1));
```

$$F_2 = \eta\, F_1 \qquad (4.2)$$

```
> subs(k=2,(4.1));
```

$$F_3 = \eta\, F_2 \qquad (4.3)$$

Die Gleichung **(4.2)** in die Gleichung **(4.3)** einsetzen.

```
> subs((4.2),(4.3));
```

$$F_3 = \eta^2\, F_1 \qquad (4.4)$$

Das Rechenschema ist zu sehen. Die weiteren Kräfte aufschreiben.

```
> subs((4.4),subs(k=3,(4.1)));
```

$$F_4 = \eta^3\, F_1 \qquad (4.5)$$

```
> subs((4.5),subs(k=4,(4.1)));
```

$$F_5 = \eta^4\, F_1 \qquad (4.6)$$

```
> subs((4.6),subs(k=5,(4.1)));
```

$$F_6 = \eta^5\, F_1 \qquad (4.7)$$

Allgemein geschrieben: Im Seil hinter der (k-1)-ten Rolle wirkt die Kraft F_k. Die Wirkungsgrade aller verwendeten Rollen multiplizieren sich, wie in den Gleichungen **(4.4)** bis **(4.7)** gezeigt.

```
> F[k] = eta^(k-1) * F[1];
```

$$F_k = \eta^{k-1} F_1 \tag{4.8}$$

Auf den linken Rollschlitten wirkt die Kraft der Feder F_s auf einer Seite und die Kräfte F_1 bis F_6 der Seile auf der anderen Seite. Die Kräfte stehen im Gleichgewicht (Prinzip von d'Alembert). Die Anzahl der Seilstränge ist n=6.

```
> F[s] = sum(F[k],k=1..n);
```

$$F_s = \sum_{k=1}^{n} F_k \tag{4.9}$$

```
> subs((4.8),(4.9));
```

$$F_s = \sum_{k=1}^{n} \eta^{k-1} F_1 \tag{4.10}$$

Die Summenformel aus der Formelsammlung.

```
> Sum(a^k,k=0..m)=sum(a^k,k=0..m): simplify(%);
```

$$\sum_{k=0}^{m} a^k = \frac{a^{m+1}-1}{a-1} \tag{4.11}$$

Mit der Summenformel die Kraft ausgerechnet.

```
> simplify((4.10));
```

$$F_s = \frac{F_1\left(\eta^n - 1\right)}{\eta - 1} \tag{4.12}$$

Gefragt war nach der Kraft F_1 bei gegebener Kraft F_s. Also die Gleichung umstellen.

```
> isolate((4.12),F[1]);
```

$$F_1 = \frac{F_s\left(\eta - 1\right)}{\eta^n - 1} \tag{4.13}$$

Die gegeben Zahlenwerte: Gleitreibungskoeffizient 0,25, Kraft der Feder 100 N, 6 Seilstränge an der Feder.
Hier für das unbekannte Radiusverhältnis ρ = 2, also Radius außen = 2 * Radius Bolzen eingesetzt!

```
> mu = 0.25, F[s] = 100*Unit(N), n = 6, rho = 2;
```

$$\mu = 0.25,\ F_s = 100\ [[N]],\ n = 6,\ \rho = 2 \tag{4.14}$$

Den Kraftübertragungswirkungsgrad einer Rolle ausrechnen mit Gleichung **(3.19)**.

```
> eval((3.19), [(4.14)]);
```

$$\eta = 0.78370 \tag{4.15}$$

Die Kraft am Seil 1 ausrechnen mit Gleichung **(4.13)**.

```
> eval((4.13), [(4.15),(4.14)]);
```

$$F_1 = 28.153\ [[N]] \tag{4.16}$$

Zum Vergleich alle Kräfte mit Gleichung **(4.8)** ausrechnen. Die Kraft wird von jeder Rolle mit dem Wirkungsgrad η übertragen, entsprechend werden die Kräfte immer kleiner.

```
> for k from 1 to 6 do
    lhs((4.8)) = eval(rhs((4.8)), [(4.15),(4.16)]);
  end do;
  unassign('k');
```

$$F_1 = 28.153 \; [\![N]\!]$$
$$F_2 = 22.063 \; [\![N]\!]$$
$$F_3 = 17.291 \; [\![N]\!]$$
$$F_4 = 13.551 \; [\![N]\!]$$
$$F_5 = 10.620 \; [\![N]\!]$$
$$F_6 = 8.3226 \; [\![N]\!] \quad \textbf{(4.17)}$$

Die Kraft an dem nach oben austretenden Seil beträgt 28 N.

Flaschen von der Feder gegen Seilkraft auseinanderziehen

Die Namen der Kräfte werden aus dem Abschnitt von oben übernommen. Auch die Rechenmethode wird übernommen.

Die Flaschen bewegen sich auseinander. Das obere Seil wird in den Flaschenzug hineingezogen. Das Seil wird vom festen Ende mit der Kraft F_6 gezogen. Die Reibung wirkt dieser Bewegung entgegen. Daher ist die Kraft F_5 kleiner als die Kraft F_6 und so weiter für die anderen Rollen. Also ist die Kraft F_k kleiner als die Kraft F_{k+1}. Die Indices der Kräfte ist gegenüber dem Abschnitt oben genau vertauscht.

Den oben berechneten Wirkungsgrad der Kraftübertragung durch eine Umlenkrolle verwenden.

```
> F[k] = eta * F[k+1];
```

$$F_k = \eta \, F_{k+1} \quad \textbf{(5.1)}$$

Diesen Zusammenhang auf die Rolle 1 und 6 anwenden.

```
> subs(k=1,(5.1));
```

$$F_1 = \eta \, F_2 \quad \textbf{(5.2)}$$

```
> subs(k=2,(5.1));
```

$$F_2 = \eta \, F_3 \quad \textbf{(5.3)}$$

Die Gleichung **(5.3)** in die Gleichung **(5.2)** einsetzen.

```
> subs((5.3),(5.2));
```

$$F_1 = \eta^2 \, F_3 \quad \textbf{(5.4)}$$

Das Rechenschema ist zu sehen. Die weiteren Kräfte aufschreiben.

```
> subs(subs(k=3,(5.1)),(5.4));
```

$$F_1 = \eta^3 \, F_4 \quad \textbf{(5.5)}$$

```
> subs(subs(k=4,(5.1)),(5.5));
```

$$F_1 = \eta^4 \, F_5 \quad \textbf{(5.6)}$$

```
> subs(subs(k=5,(5.1)),(5.6));
```

$$F_1 = \eta^5 \, F_6 \quad \textbf{(5.7)}$$

Aus den Gleichungen **(5.4)** bis **(5.7)** die allgemeine Darstellung bilden. Gegen dem Abschnitt oben steht hier $1/\eta$ anstelle von η in der Gleichung **(4.8)**.

```
> F[k] = (1/eta)^(k-1) * F[1];
```

(5.8)

$$F_k = \left(\frac{1}{\eta}\right)^{k-1} F_1 \tag{5.8}$$

Auf den linken Rollschlitten wirkt die Kraft der Feder F_s auf einer Seite und die Kräfte F_1 bis F_6 der Seile auf der anderen Seite. Die Kräfte stehen im Gleichgewicht (Prinzip von d'Alembert). Die Anzahl der Seile ist n=6.

```
> F[s] = sum(F[k],k=1..n);
```

$$F_s = \sum_{k=1}^{n} F_k \tag{5.9}$$

```
> subs((5.8),(5.9));
```

$$F_s = \sum_{k=1}^{n} \left(\frac{1}{\eta}\right)^{k-1} F_1 \tag{5.10}$$

Die Summenformel aus der Formelsammlung.

```
> Sum(a^k,k=0..m)=sum(a^k,k=0..m): simplify(%);
```

$$\sum_{k=0}^{m} a^k = \frac{a^{m+1}-1}{a-1} \tag{5.11}$$

Mit der Summenformel die Kraft ausgerechnet.

```
> simplify((5.10)): subs(eta-1=-DD,%): subs(DD=1-eta,%): sort(%);
```

$$F_s = \frac{\left(\left(\frac{1}{\eta}\right)^n - 1\right)\eta F_1}{-\eta+1} \tag{5.12}$$

Gefragt war nach der Kraft F_1 bei gegebener Kraft F_s. Also die Gleichung umstellen.

```
> solve((5.12),{F[1]})[1]: subs(eta-1=-DD,%): subs(DD=1-eta,%);
```

$$F_1 = \frac{F_s\,(-\eta+1)}{\eta\left(\left(\frac{1}{\eta}\right)^n - 1\right)} \tag{5.13}$$

Die gegeben Zahlenwerte.

```
> (4.14);
```

$$\mu = 0.25,\ F_s = 100\ [\![N]\!],\ n = 6,\ \rho = 2 \tag{5.14}$$

Den Wirkungsgrad einer Rolle ausrechnen mit Gleichung **(3.19)**.

```
> eval((3.19),[(5.14)]);
```

$$\eta = 0.78370 \tag{5.15}$$

Die Kraft am Seil 1 ausrechnen mit Gleichung **(5.13)**.

```
> eval((5.13),[(5.15),(5.14)]);
```

$$F_1 = 8.3226\ [\![N]\!] \tag{5.16}$$

Zum Vergleich alle Kräfte mit Gleichung **(5.8)** ausrechnen. Mit der Kraft F_6 am festen Ende wird das Seil durch den Flaschenzug gezogen. Mit jeder Umlenkung wird die Kraft durch die Reibung an der Umlenkrolle kleiner gemäß dem Wirkungsgrad η der Rolle. Gegenüber dem Ergebnis **(4.17)** sind alle Indices vertauscht.

```
> for k from 1 to 6 do
     lhs((5.8)) = eval(rhs((5.8)),[(5.15),(5.16)]);
  end do;
```

```
unassign('k');
```

$$F_1 = 8.3226 \ [\![N]\!]$$
$$F_2 = 10.620 \ [\![N]\!]$$
$$F_3 = 13.551 \ [\![N]\!]$$
$$F_4 = 17.291 \ [\![N]\!]$$
$$F_5 = 22.063 \ [\![N]\!]$$
$$F_6 = 28.153 \ [\![N]\!] \quad \textbf{(5.17)}$$

Die Kraft an dem nach oben austretenden Seil beträgt 8 N.

Hilfsmittel:
- Müller, Ferber: Technische Mechanik für Ingenieure, Fachbuchverlag Leipzig
- Maple 14.