```
> restart;
> Digits := 25: interface( displayprecision=10 ):
```

Aufgabe

Eine 5m lange Anlegeleiter soll auf der Kante eines 1x1m großen Absatzes aufliegen und an der Wand anliegen! In welcher Höhe liegt die Leiter an der Wand an?

Skizze



Die Leiter ist als Gerade eingezeichnet. Ein Koordinatensystem an Boden und Wand ist durch die Skizze definiert.

Text in Gleichungen übersetzen

Die Leiter ist eine Gerade.

```
> y=a*x+b;
```

$$y = a\,x + b \quad (2.1)$$

Die Leiter liegt auf dem Absatz.
Der Punkt (s_1,s_2) liegt auf der Geraden.

```
> subs(x=s[1],y=s[2],(2.1));
```

$$s_2 = a\,s_1 + b \quad (2.2)$$

Die Leiter hat eine Länge von L.
Der Abstand zwischen den Punkt $(x_0,0)$ und $(0,y_0)$ der Geraden beträgt L.
Schnittpunkt $(x_0,0)$ der Geraden mit der x-Achse.

```
> subs(x=x[0],y=0,(2.1)); isolate(%,x[0]);
```

$$0 = a\,x_0 + b$$

$$x_0 = -\frac{b}{a} \tag{2.3}$$

Schnittpunkt $(0,y_0)$ der Geraden mit der y-Achse.

```
> subs(x=0,y=y[0],(2.1));
```

$$y_0 = b \tag{2.4}$$

Abstand der Punkte muss L sein.

```
> L^2 = x[0]^2+y[0]^2;
```

$$L^2 = x_0^2 + y_0^2 \tag{2.5}$$

Gleichungen lösen

Die Gleichungen **(2.2)**, **(2.3)**, **(2.4)** und **(2.5)** bilden ein Gleichungsystem für Unbekannten a, b, x_0 und y_0.

Gesucht ist y_0, die Höhe der Leiter an der Wand.
Mit Gleichung **(2.4)**, die Unbekannte b eliminieren.

```
> isolate((2.4),b);
```

$$b = y_0 \tag{3.1}$$

Die verbleibenden drei Gleichungen.

```
> subs((3.1),(2.2));
```

$$s_2 = a\,s_1 + y_0 \tag{3.2}$$

```
> subs((3.1),(2.3));
```

$$x_0 = -\frac{y_0}{a} \tag{3.3}$$

```
> subs((3.1),(2.5));
```

$$L^2 = x_0^2 + y_0^2 \tag{3.4}$$

Gleichung **(3.3)** nach a auflösen und damit Unbekannte a aus dem Gleichungssystem eliminieren.

```
> isolate((3.3),a);
```

$$a = -\frac{y_0}{x_0} \tag{3.5}$$

Die beiden verbleibenden Gleichungen.

```
> subs((3.5),(3.2));
```

$$s_2 = -\frac{y_0\,s_1}{x_0} + y_0 \tag{3.6}$$

```
> subs((3.5),(3.4));
```

$$L^2 = x_0^2 + y_0^2 \tag{3.7}$$

Gleichung **(3.6)** nach x_0 auflösen und damit Unbekannte x_0 aus dem Gleichungsystem eliminieren.

```
> isolate((3.6),x[0]);
```

(3.8)

$$x_0 = -\frac{y_0\, s_1}{s_2 - y_0} \tag{3.8}$$

Die verbleibende Gleichung.

```
> subs((3.8),(3.7));
```

$$L^2 = \frac{y_0^2\, s_1^2}{\left(s_2 - y_0\right)^2} + y_0^2 \tag{3.9}$$

```
> convert((3.9),std)*(s[2]-y[0])^2: simplify(%): collect(%,y[0]);
```

$$-y_0^4 + 2\, s_2\, y_0^3 + \left(L^2 - s_1^2 - s_2^2\right) y_0^2 - 2\, L^2\, s_2\, y_0 + L^2\, s_2^2 = 0 \tag{3.10}$$

Die zu lösende Gleichung ist kubisch.
Daher den analytischen Lösungsweg verlassen und auf eine numerische Lösung hinarbeiten.
Die gegebeben Parameter:

```
> L = 5, s[1] = 1, s[2] = 1;
```

$$L = 5,\ s_1 = 1,\ s_2 = 1 \tag{3.11}$$

Die Parameter in die Gleichung einsetzen.

```
> subs((3.11),(3.10));
```

$$-y_0^4 + 2\, y_0^3 + 23\, y_0^2 - 50\, y_0 + 25 = 0 \tag{3.12}$$

Für eine Übersicht den linken Term der Gleichung plotten.

```
> plot( lhs((3.12)), y[0]=0..5.5 );
```



Die Lösungen numerisch ermitteln.

```
> Y0:=[fsolve((3.12))];
```

$$Y0 := [-4.9303967308,\ 0.8313772172,\ 1.2605183529,\ 4.8385011607] \tag{3.13}$$

Die zugehörigen Schnittpunkt x_0 mit der Y-Achse über Gleichung **(2.5)** ausrechnen.

```
> X0:=map( p->sqrt(5^2-p^2),Y0);
```

$$X0 := [0.8313772172,\ 4.9303967308,\ 4.8385011607,\ 1.2605183529] \tag{3.14}$$

```
> for i from 1 to 4 do p[i]:=plot( [0,X0[i]],[Y0[i],0] ) end do:
```

Die Geraden ("Leitern") plotten.

```
> plots[display](seq(p[i],i=1..4),axes=boxed,gridlines=true);
```

Nur die Werte y_0 =

```
> Y0[3],Y0[4];
```

$$1.2605183529, 4.8385011607 \tag{3.15}$$

sind brauchbare Lösungen für die Höhe der Leiter.
Die Lösung mit negativem Wert von y_0 läuft erst in der Verlängerung der Geraden durch den Punkt (1,1). Der y_0 Wert von 0,8 liefert eine Gerade die in der Verlängerung den Punkt (-1,1) durchläuft.

```
> unassign( ´X0´,´Y0´ );
```