```
> restart;
> Digits := 30: interface( displayprecision=5 ):
> # List of parameters given in the exercise
  Parameter := []:
  addParameter := proc ( newOne )
    global Parameter;
    Parameter := [op(Parameter),newOne];
  end proc:
```

Aufgabe

Zylinder A enthält Luft, die anfänglich das Volumen $V_1 = 5\ dm^3$ einnimmt. Der reibungsfrei bewegliche Kolben übt auf die Luft den konstanten Druck p = 135 kPa aus. Der unten liegende Druckluftbehälter B hat das konstante Volumen $V_B = 10\ dm^3$. Er ist ebenfalls mit Luft gefüllt, die unter dem Druck $p_B = 650$ kPa steht. Das Volumen der Leitung zwischen Zylinder und Behälter wird vernachlässigt. Das ganze System hat die Anfangstemperatur $t_1 = 15$ °C.
Nach dem Öffnen des Ventils strömt Luft aus dem Behälter langsam in den Zylinder über. Der Kolben hebt sich, bis der Druck im ganzen System denselben Wert p erreicht. Dieser Vorgang 1→2 soll adiabat ablaufen, wobei die Temperatur im Druckluftbehälter B immer gleich der Temperatur im Zylinder A ist (diatherme Wand zwischen A und B).
Anschließend wird in einem Prozess 2→3 Wärme zwischen der Luft und ihrer Umgebung ausgetauscht, bis das System wieder die Temperatur $t_3 = t_1 = 15$ °C erreicht.

Luft sei ein ideales Gas mit der speziellen Gaskonstanten R = 0, 2871 kJ/(kg K) und der spezifischen isochoren Wärmekapazität c_v = 0, 717 kJ/(kgK).

a) Wie groß ist die Gesamtmasse m der Luft im Behälter und im Zylinder?
b) Prozess 1→2: Berechnen Sie die Temperatur t_2 sowie das Volumen V_2 der Luft im Zylinder. Für die weiteren Teilaufgaben können Sie t_2 und V_2 als bekannt voraussetzen.
c) Welches Volumen V_3 nimmt die Luft im Zustand 3 im Zylinder ein?
d) Prozess 2→3: Wie groß ist die übertragene Wärme Q_{23}?

Teil a

Die Luft kann mit dem idealen Gasgesetz beschrieben werden. Für die Luftmasse m_z im Zylinder

```
> p*V[1]=m[z]*R*T[1]; isolate(%,m[z]);
```

$$p\, V_1 = m_z\, R\, T_1$$

$$m_z = \frac{p\, V_1}{R\, T_1} \qquad (1)$$

Die Luftmassse m_B im Behälter.

```
> p[B]*V[B]=m[B]*R*T[1]; isolate(%,m[B]);
```

$$p_B\, V_B = m_B\, R\, T_1$$

$$m_B = \frac{p_B\, V_B}{R\, T_1} \qquad (2)$$

Die Gesamtmasse m der Luft.

```
> m = m[z]+m[B]; subs((1),(2),%); simplify(%,size);
```

$$m = m_z + m_B$$

$$m = \frac{p\,V_1}{R\,T_1} + \frac{p_B\,V_B}{R\,T_1}$$

$$m = \frac{p\,V_1 + p_B\,V_B}{R\,T_1} \quad \textbf{(3)}$$

Alle Werte auf der rechten Seite von **(3)** sind in der Aufgabe gegeben.
Die Druckangaben werden als absolute Drücke interpretiert. Die Temperaturangabe von °C in Kelvin umgerechnet.

```
> T[1] = (15+273.15)*Unit(K); addParameter(%):
  V[1] = 5*Unit(dm^3); addParameter(%):
  V[B] = 10*Unit(dm^3); addParameter(%):
  p = 135*Unit(kPa); addParameter(%):
  p[B] = 650*Unit(kPa); addParameter(%):
  R = 0.2871*Unit(kJ/(kg*K)): lhs(%)=convert(rhs(%),units,kJ/(kg*K)
  ); addParameter(%):
```

$$T_1 = 288.15\ [\![K]\!]$$

$$V_1 = 5\ [\![dm^3]\!]$$

$$V_B = 10\ [\![dm^3]\!]$$

$$p = 135\ [\![kPa]\!]$$

$$p_B = 650\ [\![kPa]\!]$$

$$R = 0.28710\ \left[\!\!\left[\frac{kJ}{kg\,K}\right]\!\!\right] \quad \textbf{(4)}$$

Die Werte in **(3)** einsetzen und die Masse ausrechnen.

```
> subs( Parameter, (3) ): simplify(%); addParameter(%):
```

$$m = 0.086730\ [\![kg]\!] \quad \textbf{(5)}$$

Teil b

Erster Hauptsatz der Thermodynamik für geschlossene Systeme (Energieerhaltung). ΔU ist die Änderung der inneren Energie der Luft (in Behälter und Zylinder), Q_{12} ist die zugeführte Wärme, W_{12} ist die zugeführte Arbeit.

```
> deltaU := `&Delta;U`:
  deltaU = Q[12] + W[12];
```

$$\Delta U = Q_{12} + W_{12} \quad \textbf{(6)}$$

Der Kolben im Zylinder wird durch die expandierende Luft gegen den Druck p nach außen gedrückt. Die Luft leistet Verdrängungsarbeit bei der Expansion.

```
> W[12] = p*( V[1]-V[2] );
```

$$W_{12} = p\,(V_1 - V_2) \quad \textbf{(7)}$$

Das betrachtete System ist laut Aufgabe adiabatisch

```
> Q[12] = 0;
```

$$Q_{12} = 0 \quad \textbf{(8)}$$

Die innere Energie des idealen Gases wird von der Temperatur bestimmt. (Kalorische Zustandsgleichung des idealen Gases.)

```
> deltaU = m*c[v]*( T[2]-T[1] );
```

$$\Delta U = m\,c_v\,(T_2 - T_1) \quad \textbf{(9)}$$

Die isochoren Wärmekapazität ist gegeben

```
> c[v] = 0.717*Unit(kJ/(kg*K)): lhs(%)=convert(rhs(%),units,kJ/(kg*K)); addParameter(%):
```

$$c_v = 0.71700 \left[\!\left[\frac{kJ}{kg\,K} \right]\!\right] \tag{10}$$

Einsetzen in **(6)**.

```
> subs( (7), (8), (9), (6) );
```

$$m\,c_v\,(T_2 - T_1) = p\,(V_1 - V_2) \tag{11}$$

In der Gleichung sind T_2 und V_2 unbekannt. Eine weitere Gleichung wird benötigt. Die Zustandsgleichung für den Zustand 2:

```
> p*(V[2]+V[B]) = m*R*T[2];
```

$$p\,(V_2 + V_B) = m\,R\,T_2 \tag{12}$$

Auflösen nach der unbekannten Temperatur T_2.

```
> isolate((12),T[2]);
```

$$T_2 = \frac{p\,(V_2 + V_B)}{m\,R} \tag{13}$$

Einsetzen in **(11)**.

```
> subs((13),(11));
```

$$m\,c_v\left(\frac{p\,(V_2 + V_B)}{m\,R} - T_1\right) = p\,(V_1 - V_2) \tag{14}$$

Auflösen nach dem gesuchten Zylindervolumen V_2 im Zustand 2.

```
> isolate((14),V[2]); simplify(%,size): sort(%);
```

$$V_2 = \frac{-c_v p\,V_B + c_v\,T_1\,m\,R + p\,R\,V_1}{c_v p + p\,R}$$

$$V_2 = \frac{R\,m\,T_1\,c_v + (R\,V_1 - V_B\,c_v)\,p}{(R + c_v)\,p} \tag{15}$$

Die Temperatur ist ebenfalls gesucht. Einsetzen des Volumens V_2 in **(13)** liefert die Temperatur T_2.

```
> subs((15),(13)); simplify(%,size);
```

$$T_2 = \frac{p\left(\frac{R\,m\,T_1\,c_v + (R\,V_1 - V_B\,c_v)\,p}{(R + c_v)\,p} + V_B\right)}{m\,R}$$

$$T_2 = \frac{c_v\,T_1\,m + p\,(V_1 + V_B)}{(R + c_v)\,m} \tag{16}$$

Die Zahlenwerte einsetzen und ausrechnen.

```
> subs( Parameter, (15) ): simplify(%); addParameter(%):
  V[2]=convert(subs(Parameter,V[2]),units,dm^3);
```

$$V_2 = 0.032241 \left[\!\left[m^3 \right]\!\right]$$

$$V_2 = 32.241 \left[\!\left[dm^3 \right]\!\right] \tag{17}$$

```
> subs( Parameter, (16) ): simplify(%); addParameter(%):
  t[2] = convert(subs(Parameter,T[2]),temperature,degC);
```

$$T_2 = 229.01 \left[\!\left[K \right]\!\right]$$

$$t_2 = -44.137 \, [[degC]] \tag{18}$$

Teil c

Die Temperatur der Luft im Zustand 3 ist gegeben.

```
> T[3] = subs(Parameter,T[1]); addParameter(%):
```

$$T_3 = 288.15 \, [[K]] \tag{19}$$

Das gesuchte Volumen kann über die Zustandsgleichung des idealen Gases berechnet werden.

```
> p*(V[B]+V[3])=m*R*T[3]; isolate(%,V[3]);
```

$$p \left(V_B + V_3\right) = m\, R\, T_3$$

$$V_3 = \frac{m\, R\, T_3}{p} - V_B \tag{20}$$

```
> subs(Parameter,(20)): simplify(%); addParameter(%):
  V[3]=convert(subs(Parameter,V[3]),units,dm^3);
```

$$V_3 = 0.043148 \, [[m^3]]$$

$$V_3 = 43.148 \, [[dm^3]] \tag{21}$$

Teil d

Die aufgenommene Wärme kann über Formeln zur isobaren Erwärmung oder direkt über den ersten Hauptsatz berechnet werden.

Über den ersten Hauptsatz der Thermodynamik für geschlossene Systeme (Energieerhaltung):

```
> deltaU = Q[23] + W[23];
```

$$\Delta U = Q_{23} + W_{23} \tag{22}$$

Der Kolben im Zylinder wird durch die expandierende Luft gegen den Druck p nach außen gedrückt. Die Luft leistet Verdrängungsarbeit bei der Expansion.

```
> W[23] = p*( V[2]-V[3] );
```

$$W_{23} = p \left(V_2 - V_3\right) \tag{23}$$

Kalorische Zustandsgleichung des idealen Gases.

```
> deltaU = m*c[v]*( T[3]-T[2] );
```

$$\Delta U = m\, c_v \left(T_3 - T_2\right) \tag{24}$$

Einsetzen und nach der gesuchten Wärme auflösen.

```
> subs( (23), (24), (22) ); isolate(%,Q[23]);
```

$$m\, c_v \left(T_3 - T_2\right) = Q_{23} + p \left(V_2 - V_3\right)$$

$$Q_{23} = m\, c_v \left(T_3 - T_2\right) - p \left(V_2 - V_3\right) \tag{25}$$

Die Zahlenwerte einsetzen und ausrechnen.

```
> subs( Parameter, ?? ): simplify(%);
```

$$Q_{23} = 5150.0 \, [[J]] \tag{26}$$

Das Vorzeichen ist positiv, die Wärme wird aufgenommen.