

```


> restart:
> y0:=Y[0]: y1:=Y[1]: y2:=Y[2]: y3:=Y[3]: y4:=Y[4]: y5:=Y[5]: y6:=Y
[6]:
> g1:=G[1]: g2:=G[2]: g3:=G[3]: g4:=G[4]:

```

Gegeben ist ein Signalfluss mit 4 linearen Gliedern.

Die Übertragungsfunktionen sind gegeben mit  $G_1(s)$ ,  $G_2(s)$ ,  $G_3(s)$  und  $G_4(s)$ .

Im Blockschaltbild sind an 7 Stellen die Signale  $Y_0(s)$  bis  $Y_6(s)$  markiert.



Alle Glieder sind linear. Es gibt also.

```
> y2 = g1*y1;
```

$$Y_2 = G_1 Y_1 \quad (1)$$

```
> y6=g2*y3;
```

$$Y_6 = G_2 Y_3 \quad (2)$$

```
> y5=g3*y4;
```

$$Y_5 = G_3 Y_4 \quad (3)$$

```
> y4=g4*y3;
```

$$Y_4 = G_4 Y_3 \quad (4)$$

An zwei Stellen werden Signale addiert.

```
> y1=y0+y6;
```

$$Y_1 = Y_0 + Y_6 \quad (5)$$

```
> y3=y2+y5;
```

$$Y_3 = Y_2 + Y_5 \quad (6)$$

Das Gleichungssystem (1) bis (6) ist in die Form  $Y_4 = G_Z Y_0$  zu bringen.

In (1) bis (4) die Signale  $Y_1$  und  $Y_3$  durch Einsetzen von (5) und (6) eliminieren.

```
> subs ((5),(1)) ;
```

$$Y_2 = G_1 (Y_0 + Y_6) \quad (7)$$

```
> subs ((6),(2)) ;
```

$$Y_6 = G_2 (Y_2 + Y_5) \quad (8)$$

```
> (3);
```

$$Y_5 = G_3 Y_4 \quad (9)$$

```
> subs ((6),(4)) ;
```

$$Y_4 = G_4 (Y_2 + Y_5) \quad (10)$$

In (8) bis (9) das Signal Y2 eliminieren durch Einsetzen von (7).

```
> subs ((7),(8)) ;
```

$$Y_6 = G_2 (G_1 (Y_0 + Y_6) + Y_5) \quad (11)$$

```
> (9);
```

$$Y_5 = G_3 Y_4 \quad (12)$$

```
> subs ((7),(10)) ;
```

$$Y_4 = G_4 (G_1 (Y_0 + Y_6) + Y_5) \quad (13)$$

(11) auflösen nach dem Signal  $Y_6$ .

```
> isolate ((11),y6) ;
```

$$Y_6 = \frac{G_2 G_1 Y_0 + G_2 Y_5}{1 - G_2 G_1} \quad (14)$$

In (12) und (13) das Signal  $Y_6$  durch einsetzen von (14) eliminieren.

```
> subs ((14),(12)) ;
```

$$Y_5 = G_3 Y_4 \quad (15)$$

```
> subs ((14),(13)) ;
```

$$Y_4 = G_4 \left( G_1 \left( Y_0 + \frac{G_2 G_1 Y_0 + G_2 Y_5}{1 - G_2 G_1} \right) + Y_5 \right) \quad (16)$$

In (16) das Signal Y5 durch Einsetzen von (15) eliminieren.

```
> subs ((15),(16)) ;
```

$$Y_4 = G_4 \left( G_1 \left( Y_0 + \frac{G_2 G_1 Y_0 + G_2 G_3 Y_4}{1 - G_2 G_1} \right) + G_3 Y_4 \right) \quad (17)$$

(17) auflösen nach  $Y_4$ .

```
> isolate ((17),y4) ;
```

$$Y_4 = - \frac{G_4 G_1 Y_0}{-1 + G_2 G_1 + G_4 G_3} \quad (18)$$

Mit der Übertragungsfunktion

```
> G[z] = coeff(rhs ((18)) ,y0) ;
```

$$G_z = - \frac{G_4 G_1}{-1 + G_2 G_1 + G_4 G_3} \quad (19)$$

ist das System in ein einzelnes Glied zusammengefasst.

```
> y4=G[z]*y0 ;
```

$$Y_4 = G_z Y_0 \quad (20)$$

```
>
```