```
> restart;
> Digits := 24: interface( displayprecision = 5 ):
```

Aufgabe

An einer Querschnittsverengung wurden folgende Größen gemessen:
Sektor 1, Durchmesser d_1 = 100 mm
Sektor 2, Durchmesser d_2 = 60 mm
Druckdifferenz Δp = 0,55 bar
Verlustkoeffizient bezogen auf Geschwindigkeit in Sektor 1 ζ = 0,07
Dichte des Öls ρ = 860 kg/m³



Berechne den Volumenstrom.

Rechenweg

Die gegeben Größen in SI Einheiten umrechnen.

```
> d[1] = 100*Unit(mm), d[2]=60*Unit(mm); evalf(simplify([%]))[];
```

$$d_1 = 100 \; [\![mm]\!], \; d_2 = 60 \; [\![mm]\!]$$

$$d_1 = 0.10000 \; [\![m]\!], \; d_2 = 0.060000 \; [\![m]\!] \quad (1)$$

```
> dp := `&Delta;p`:
  dp = 0.55*Unit(bar); simplify(%);
```

$$\Delta p = 0.55 \; [\![bar]\!]$$

$$\Delta p = 55000. \; [\![Pa]\!] \quad (2)$$

```
> rho = 860 *Unit(kg/m^3);
```

$$\rho = 860 \left[\!\!\left[\frac{kg}{m^3} \right]\!\!\right] \quad (3)$$

```
> zeta = 0.07;
```

$$\zeta = 0.07 \quad (4)$$

Ansatz: Bernoulli-Gleichung erweitert mit dem örtlichen Energieverlust im Konvusor.
Reibungsverluste in den Rohrstücken werden nicht berücksichtigt.
Die mittleren Strömungsgeschwindigkeiten werden c_1 und c_2 genannt, die statischen Drücke p_1 und

p_2.

```
> p[1] + rho*c[1]^2/2 = p[2] + rho*c[2]^2/2 + zeta*rho*c[1]^2/2;
```

$$p_1 + \frac{\rho c_1^2}{2} = p_2 + \frac{1}{2} \rho c_2^2 + \frac{1}{2} \zeta \rho c_1^2 \tag{5}$$

Das Öl wird als nicht kompressible Flüssigkeit betrachtet. Die Kontinuitätsgleichung für den Volumenstrom:

```
> c[1]*d[1]^2*Pi/4 = c[2]*d[2]^2*Pi/4;
```

$$\frac{c_1 d_1^2 \pi}{4} = \frac{c_2 d_2^2 \pi}{4} \tag{6}$$

Auflösen nach der mittleren Strömungsgeschwindigkeit c_2.

```
> isolate( (6), c[2] );
```

$$c_2 = \frac{c_1 d_1^2}{d_2^2} \tag{7}$$

Einsetzen in Gleichung **(5)**.

```
> subs( (7), (5) );
```

$$p_1 + \frac{\rho c_1^2}{2} = p_2 + \frac{\rho c_1^2 d_1^4}{2 d_2^4} + \frac{\zeta \rho c_1^2}{2} \tag{8}$$

Umformen und Druckdifferenz einsetzen.

```
> (8) - p[2]: sort(%); algsubs( p[1]-p[2]=dp, % );
```

$$\frac{\rho c_1^2}{2} + p_1 - p_2 = \frac{\rho \zeta c_1^2}{2} + \frac{\rho c_1^2 d_1^4}{2 d_2^4}$$

$$\frac{\rho c_1^2}{2} + \Delta p = \frac{\rho \zeta c_1^2}{2} + \frac{\rho c_1^2 d_1^4}{2 d_2^4} \tag{9}$$

Auflösen nach der Geschwindigkeit c_1.

```
> (9)*2*d[2]^4: expand(%): sort(%);
  lhs(%)=simplify(rhs(%),size): sort(%);
  isolate( %, c[1]^2 ): simplify(%,size): sort(%);
```

$$\rho c_1^2 d_2^4 + 2 \Delta p d_2^4 = \rho \zeta c_1^2 d_2^4 + \rho c_1^2 d_1^4$$

$$\rho c_1^2 d_2^4 + 2 \Delta p d_2^4 = \left(\zeta d_2^4 + d_1^4\right) \rho c_1^2$$

$$c_1^2 = \frac{2 \Delta p d_2^4}{\left(d_1^4 + (\zeta - 1) d_2^4\right) \rho} \tag{10}$$

```
> sqrt(lhs((10))) = sqrt(rhs((10))):
  simplify(%) assuming c[1]::positive, d[2]::positive: simplify
  (%,size);
```

$$c_1 = \sqrt{2} d_2^2 \sqrt{\frac{\Delta p}{\left(d_1^4 + (\zeta - 1) d_2^4\right) \rho}} \tag{11}$$

Gesucht ist der Volumenstrom.

```
> Q = c[1] * d[1]^2*Pi/4;
```

$$Q = \frac{c_1\, d_1^2\, \pi}{4} \tag{12}$$

```
> subs((11),(12));
  simplify(%, symbolic): simplify(%,size): sort(%);
```

$$Q = \frac{\sqrt{2}\,\sqrt{\frac{\Delta p}{\left(d_1^4 + (\zeta - 1)\, d_2^4\right)\rho}}\ \pi\, d_1^2\, d_2^2}{4}$$

$$Q = \frac{\sqrt{2}\ \pi\,\sqrt{\Delta p}\ d_1^2\, d_2^2}{4\,\sqrt{d_1^4 + (\zeta - 1)\, d_2^4}\ \sqrt{\rho}} \tag{13}$$

Einsetzen und ausrechnen.

```
> subs( (1),(2),(3),(4), (13) ): simplify(%);
  lhs(%) = convert(rhs(%),'units','L/s');
```

$$Q = 0.034098\ \left[\left[\frac{m^3}{s}\right]\right]$$

$$Q = 34.098\ \left[\left[\frac{L}{s}\right]\right] \tag{14}$$

Der Volumenstrom beträgt 34 Liter/s.

Hilfsmittel:
- Skript Hydomechanik realer Strömungen, Uni Kassel, Prof. Dr. Koch
- Maple 14, http://www.maplesoft.com/