```
> restart;
> Digits:=24: interface( displayprecision=5 ):
> with(Physics[Vectors]):
> cross2D := (a::Vector,b::Vector) -> a[1]*b[2]-a[2]*b[1]:
```

Aufgabe

Kräfteermittlung im zweidimensionalen Fachwerk

Das Fachwerk (schwarze Linien) mit den von angreifenden Kräften (grüne Pfeile):



Ein Koordinatensystem ist eingezeichnet (rote Pfeile). Die Länge eines Pfeils ist die Längeneinheit a.
Die Stäbe sind maßstabsgetreu eingezeichnet. Die Pfeile der Kräfte geben nur die Richtung der Kräfte an.

Die Kräfte $\vec{F}_1$ und $\vec{F}_2$ sind bekannt.
(Der Index x oder y bezeichnet die entsprechende Komponente des Kraftvektors. Der große Buchstabe F steht für den Vektor, der kleine Buchstabe f steht für den "vorläufigen Betrag" des Vektors.):

```
> F_[1] = < -f[1], 0 >;
```

$$\vec{F}_1 = \begin{bmatrix} -f_1 \\ 0 \end{bmatrix} \tag{1}$$

mit dem Zahlenwert des Betrags

```
> f[1] = 200*Unit(N);
```

$$f_1 = 200 \; [\![N]\!] \tag{2}$$

```
> F_[2] = < 0, -f[2] >;
```

$$\vec{F}_2 = \begin{bmatrix} 0 \\ -f_2 \end{bmatrix} \tag{3}$$

mit dem Zahlenwert des Betrags

```
> f[2] = 100*Unit(N);
```

$$f_2 = 100 \; [\![N]\!] \tag{4}$$

a) Die Lagerkräfte

$\vec{F}_A$, $\vec{F}_B$ und $\vec{F}_C$ sind zu bestimmen.

b) Die Kräfte in den Stäbe S_1, S_2 und S_3 (siehe Skizze) sind zu bestimmen.

Lagerkräfte berechnen

Ansatz:
- Alle angreifenden Kräfte sind in der Aufgabe genannt. Insbesondere werden Gewichte vernachlässigt.
- Längenänderungen der Stäbe durch die Kräfte vernachlässigen. Das Fachwerk als einen starren Körper beschreiben.
- Die Lagerkräfte können alle Kräfte ausgleichen. Das Fachwerk ruht.
- Das Fachwerk und alle angreifenden Kräfte sind in der XY-Ebene.

Damit:
- Zweidimensionale Vektoren verwenden.
- Summe der angreifenden Käfte ist 0.
- Summe der angreifenden Drehmomente ist 0.

Durch die Lager sind die Richtungen der Kräfte $\vec{F}_A$, $\vec{F}_B$ und $\vec{F}_C$ gegeben.

Die Reichtung bestimmt die Aufteilung des Betrags auf die Komponenten der Kräfte.

```
> F_[A] = < 0, f[A] >;
```

$$\vec{F}_A = \begin{bmatrix} 0 \\ f_A \end{bmatrix} \tag{5}$$

```
> F_[B] = < -1, 1 > * eval( f[B]/sqrt(2), 1 );
```

$$\vec{F}_B = \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} \tag{6}$$

```
> F_[C] = < 0, f[C] >;
```

$$\vec{F}_C = \begin{bmatrix} 0 \\ f_C \end{bmatrix} \tag{7}$$

Die Angriffspunkte $\vec{P}$ der Kräfte an der Skizze ablesen.

```
> P_[A] = < 0, 0 >;
```

$$\vec{P}_A = \begin{bmatrix} 0 \\ 0 \end{bmatrix} \tag{8}$$

```
> P_[B] = < 4*a, 0 >;
```

$$\vec{P}_B = \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \tag{9}$$

```
> P_[C] = < 6*a, 0 >;
```

(10)

$$\vec{P}_C = \begin{bmatrix} 6\,a \\ 0 \end{bmatrix} \quad \textbf{(10)}$$

```
> P_[1] = < 1*a, 1*a >;
```

$$\vec{P}_1 = \begin{bmatrix} a \\ a \end{bmatrix} \quad \textbf{(11)}$$

```
> P_[2] = < 2*a, 0 >;
```

$$\vec{P}_2 = \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \quad \textbf{(12)}$$

Die Summe der angreifenden Kräfte ist 0.

```
> F_[A] + F_[B] + F_[C] + F_[1] + F_[2] = 0;
```

$$\vec{F}_A + \vec{F}_B + \vec{F}_C + \vec{F}_1 + \vec{F}_2 = 0 \quad \textbf{(13)}$$

Die Richtungen der Lagerkräfte **(5)**, **(6)** und **(7)** einsetzen.

```
> subs((5),(6),(7),  (13));
```

$$\begin{bmatrix} 0 \\ f_A \end{bmatrix} + \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \vec{F}_1 + \vec{F}_2 = 0 \quad \textbf{(14)}$$

Die gegebenen Kräfte aus **(1)** und **(3)** einsetzen.

```
> subs((1),(3),  (14) );
```

$$\begin{bmatrix} -f_1 \\ 0 \end{bmatrix} + \begin{bmatrix} 0 \\ -f_2 \end{bmatrix} + \begin{bmatrix} -\frac{f_B\sqrt{2}}{2} \\ f_A + \frac{f_B\sqrt{2}}{2} + f_C \end{bmatrix} = 0 \quad \textbf{(15)}$$

Addieren der Vektoren.

```
> expand( (15) );
```

$$\begin{bmatrix} -f_1 - \frac{f_B\sqrt{2}}{2} \\ -f_2 + f_A + \frac{f_B\sqrt{2}}{2} + f_C \end{bmatrix} = 0 \quad \textbf{(16)}$$

Summe der angreifenden Drehmomente ist 0.

```
> P_[A] &x F_[A] + P_[B] &x F_[B] + P_[C] &x F_[C] + P_[1] &x F_
  [1] + P_[2] &x F_[2] = 0;
```

$$\vec{P}_A \times \vec{F}_A + \vec{P}_B \times \vec{F}_B + \vec{P}_C \times \vec{F}_C + \vec{P}_1 \times \vec{F}_1 + \vec{P}_2 \times \vec{F}_2 = 0 \quad \textbf{(17)}$$

Die Richtungen der Lagerkräfte aus **(5)**, **(6)** und **(7)** einsetzen.

```
> subs((5),(6),(7),  (17));
```

$$\vec{P}_A \times \begin{bmatrix} 0 \\ f_A \end{bmatrix} + \vec{P}_B \times \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \vec{P}_C \times \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \vec{P}_1 \times \vec{F}_1 + \vec{P}_2 \times \vec{F}_2 = 0 \quad \textbf{(18)}$$

Die Angriffspunkte aus Gleichungen **(8)** - **(12)** einsetzen.

```
> subs((8),(9),(10),(11),(12), (18) );
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_A \end{bmatrix} + \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 6\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \begin{bmatrix} a \\ a \end{bmatrix} \times \vec{F}_1 + \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \times \vec{F}_2 = 0 \quad \textbf{(19)}$$

Die gegebenen Kräfte aus **(1)** und **(3)** einsetzen.

```
> subs((1),(3), (19) );
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_A \end{bmatrix} + \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 6\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \begin{bmatrix} a \\ a \end{bmatrix} \times \begin{bmatrix} -f_1 \\ 0 \end{bmatrix} + \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ -f_2 \end{bmatrix} = 0 \quad \textbf{(20)}$$

Ausrechnen der Kreuzprodukte.

```
> subsindets( (20),
     specfunc(Vector,op(0,<1> &x <1>)),
     g -> cross2D( op(g) ) );
```

$$2\,a\,f_B\sqrt{2} + 6\,a\,f_C + a\,f_1 - 2\,a\,f_2 = 0 \quad \textbf{(21)}$$

Die Kompenenten der Vektorgleichungen **(16)** und die Gleichung **(21)** ergeben ein Gleichungssystem für die Lagerkräfte.

```
> lhs((16)) [1] = rhs((16)) [1];
```

$$-f_1 - \frac{f_B\sqrt{2}}{2} = 0 \quad \textbf{(22)}$$

```
> lhs((16)) [2] = rhs((16)) [2];
```

$$-f_2 + f_A + \frac{f_B\sqrt{2}}{2} + f_C = 0 \quad \textbf{(23)}$$

```
> (21);
```

$$2\,a\,f_B\sqrt{2} + 6\,a\,f_C + a\,f_1 - 2\,a\,f_2 = 0 \quad \textbf{(24)}$$

Das lineare Gleichungssystem **(22)**, **(23)** und **(24)** nach den Lagerkräften f_A, f_B und f_C auflösen.

```
> Loesung1 := simplify( solve( [(22),(23),(24)], [f[A],f[B],f[C]] )
  [1] ): Loesung1[];
```

$$f_A = \frac{2 f_2}{3} + \frac{f_1}{2}, f_B = -\sqrt{2}\, f_1, f_C = \frac{f_1}{2} + \frac{f_2}{3} \quad \textbf{(25)}$$

Die Zahlenwert für die Kräfte ausrechnen. Die Werte aus **(2)** und **(4)** einsetzen.

```
> evalf( eval(Loesung1,[(2),(4)]) )[];
```

$$f_A = 166.67 \ [\![N]\!], f_B = -282.84 \ [\![N]\!], f_C = 133.33 \ [\![N]\!] \quad \textbf{(26)}$$

Der "Betrag" f_B ist negativ. Also hat der Vektor in der Skizze die "falsche" Richtung,

Stabkräfte berechnen

Freischneiden des linken Fachwerkteils an den Stäben 1, 2 und 3. (Ritterschnitt)



Ansatz analog der Lagerkräftebrechnung.
Die Richtung der Kräfte $\vec{F}_{S1}$, $\vec{F}_{S2}$ und $\vec{F}_{S3}$ sind durch die Richtungen der Stäbe bestimmt.

```
> F_[S1] = < -f[S1], 0 >;
```

$$\vec{F}_{S1} = \begin{bmatrix} -f_{S1} \\ 0 \end{bmatrix} \quad \textbf{(27)}$$

```
> F_[S2] = <-1,1> * eval( f[S2]/sqrt(2), 1 );
```

$$\vec{F}_{S2} = \begin{bmatrix} -\frac{f_{S2}}{\sqrt{2}} \\ \frac{f_{S2}}{\sqrt{2}} \end{bmatrix} \quad \textbf{(28)}$$

```
> F_[S3] = <-f[S3], 0 >;
```

$$\vec{F}_{S3} = \begin{bmatrix} -f_{S3} \\ 0 \end{bmatrix} \quad \textbf{(29)}$$

Die Koordinaten der drei Angriffspunkte in der Skizze ablesen.

```
> P_[S1] = <3*a,a>;
```

$$\vec{P}_{S1} = \begin{bmatrix} 3\,a \\ a \end{bmatrix} \tag{30}$$

```
> P_[S2] = <3*a,a>;
```

$$\vec{P}_{S2} = \begin{bmatrix} 3\,a \\ a \end{bmatrix} \tag{31}$$

```
> P_[S3] = <2*a,0>;
```

$$\vec{P}_{S3} = \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \tag{32}$$

Die Summe der angreifenden Kräfte ist 0.

```
> F_[A] + F_[1] + F_[2] + F_[S1] + F_[S2] + F_[S3] = 0;
```

$$\vec{F}_A + \vec{F}_1 + \vec{F}_2 + \vec{F}_{S1} + \vec{F}_{S2} + \vec{F}_{S3} = 0 \tag{33}$$

Einsetzen der Richtungen der Lagerkräfte und der Stabkräfte aus Gleichungen **(5)**, **(27)**, **(28)** und **(29)**.

```
> subs((5),(27),(28),(29), (33));
```

$$\begin{bmatrix} 0 \\ f_A \end{bmatrix} + \vec{F}_1 + \vec{F}_2 + \begin{bmatrix} -f_{S1} \\ 0 \end{bmatrix} + \begin{bmatrix} -\dfrac{f_{S2}}{\sqrt{2}} \\ \dfrac{f_{S2}}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} -f_{S3} \\ 0 \end{bmatrix} = 0 \tag{34}$$

Die gegebenen Kräfte aus Gleichungen **(1)** und **(3)** einsetzen.

```
> subs((1),(3), (34) );
```

$$\begin{bmatrix} -f_1 \\ 0 \end{bmatrix} + \begin{bmatrix} 0 \\ -f_2 \end{bmatrix} + \begin{bmatrix} -f_{S1} - \dfrac{f_{S2}\sqrt{2}}{2} - f_{S3} \\ f_A + \dfrac{f_{S2}\sqrt{2}}{2} \end{bmatrix} = 0 \tag{35}$$

Die Vektoren addieren.

```
> expand( (35) );
```

$$\begin{bmatrix} -f_1 - f_{S1} - \dfrac{f_{S2}\sqrt{2}}{2} - f_{S3} \\ -f_2 + f_A + \dfrac{f_{S2}\sqrt{2}}{2} \end{bmatrix} = 0 \tag{36}$$

Die Summe der angreifenden Drehmomente ist 0.

```
> P_[A] &x F_[A] + P_[1] &x F_[1] + P_[2] &x F_[2] + P_[S1] &x F_
  [S1] + P_[S2] &x F_[S2] + P_[S3] &x F_[S3] = 0;
```

$$\vec{P}_A \times \vec{F}_A + \vec{P}_1 \times \vec{F}_1 + \vec{P}_2 \times \vec{F}_2 + \vec{P}_{S1} \times \vec{F}_{S1} + \vec{P}_{S2} \times \vec{F}_{S2} + \vec{P}_{S3} \times \vec{F}_{S3} = 0 \tag{37}$$

Einsetzen der Richtungen der Lagerkräfte und der Stabkräfte aus Gleichungen **(5)**, **(27)**, **(28)** und **(29)**.

```
> subs((5),(27),(28),(29), (37));
```

$$\vec{P}_A \times \begin{bmatrix} 0 \\ f_A \end{bmatrix} + \vec{P}_1 \times \vec{F}_1 + \vec{P}_2 \times \vec{F}_2 + \vec{P}_{S1} \times \begin{bmatrix} -f_{S1} \\ 0 \end{bmatrix} + \vec{P}_{S2} \times \begin{bmatrix} -\frac{f_{S2}}{\sqrt{2}} \\ \frac{f_{S2}}{\sqrt{2}} \end{bmatrix} + \vec{P}_{S3} \times \begin{bmatrix} -f_{S3} \\ 0 \end{bmatrix} = 0 \qquad \textbf{(38)}$$

Einsetzen der Angriffspunkte aus den Gleichungen **(8)**, **(11)**, **(12)**, **(30)**, **(31)**, **(32)**.

```
> subs((8),(11),(12),(30),(31),(32),  (38) );
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_A \end{bmatrix} + \begin{bmatrix} a \\ a \end{bmatrix} \times \vec{F}_1 + \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \times \vec{F}_2 + \begin{bmatrix} 3\,a \\ a \end{bmatrix} \times \begin{bmatrix} -f_{S1} \\ 0 \end{bmatrix} + \begin{bmatrix} 3\,a \\ a \end{bmatrix} \times \begin{bmatrix} -\frac{f_{S2}}{\sqrt{2}} \\ \frac{f_{S2}}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} -f_{S3} \\ 0 \end{bmatrix} = 0 \qquad \textbf{(39)}$$

Die gegebenen Kräfte **(1)** und **(3)** einsetzen.

```
> subs((1),(3),  (39) );
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_A \end{bmatrix} + \begin{bmatrix} a \\ a \end{bmatrix} \times \begin{bmatrix} -f_1 \\ 0 \end{bmatrix} + \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ -f_2 \end{bmatrix} + \begin{bmatrix} 3\,a \\ a \end{bmatrix} \times \begin{bmatrix} -f_{S1} \\ 0 \end{bmatrix} + \begin{bmatrix} 3\,a \\ a \end{bmatrix} \times \begin{bmatrix} -\frac{f_{S2}}{\sqrt{2}} \\ \frac{f_{S2}}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 2\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} -f_{S3} \\ 0 \end{bmatrix} = 0 \qquad \textbf{(40)}$$

Ausrechnen der Kreuzprodukte.

```
> subsindets( (40),
     specfunc(Vector,op(0,<1> &x <1>)),
     g -> cross2D( op(g) ) );
```

$$a\,f_1 - 2\,a\,f_2 + a\,f_{S1} + 2\,a\,f_{S2}\sqrt{2} = 0 \qquad \textbf{(41)}$$

Die Komponenten der Vektorgleichung **(36)** und die Gleichung **(41)** bilden ein Gleichungssystem für die Stabkräfte.

```
> lhs((36)) [1]  = rhs((36)) [1];
```

$$-f_1 - f_{S1} - \frac{f_{S2}\sqrt{2}}{2} - f_{S3} = 0 \qquad \textbf{(42)}$$

```
> lhs((36)) [2]  = rhs((36)) [2];
```

$$-f_2 + f_A + \frac{f_{S2}\sqrt{2}}{2} = 0 \qquad \textbf{(43)}$$

```
> (41);
```

$$a\,f_1 - 2\,a\,f_2 + a\,f_{S1} + 2\,a\,f_{S2}\sqrt{2} = 0 \qquad \textbf{(44)}$$

Das lineare Gleichungssystem nach den Stabkräften f_{S1}, f_{S2} und f_{S3} auflösen.

```
> solve( [(42),(43),(44)], [ f[S1],f[S2],f[S3] ] )[1][];
```

$$f_{S1} = -f_1 - 2f_2 + 4f_A, f_{S2} = \sqrt{2}\,(f_2 - f_A), f_{S3} = f_2 - 3f_A \tag{45}$$

Die Lagerkraft f_A aus Gleichung **(25)** einsetzen.

```
> Loesung2 := simplify( subs(Loesung1, [(45)] ) ): Loesung2[];
```

$$f_{S1} = f_1 + \frac{2f_2}{3}, f_{S2} = -\frac{\sqrt{2}\,(-2f_2 + 3f_1)}{6}, f_{S3} = -f_2 - \frac{3f_1}{2} \tag{46}$$

Die Zahlenwert für die Kräfte ausrechnen. Die gegebenen Werte für die äußeren Kräfte f_1 und f_2 aus **(2)** und **(4)** einsetzen.

```
> evalf( eval( Loesung2, [(2),(4)] ) )[];
```

$$f_{S1} = 266.67\,[\![N]\!], f_{S2} = -94.281\,[\![N]\!], f_{S3} = -400.\,[\![N]\!] \tag{47}$$

Kontrolle der Stabkräfte

Zur Berechnung der Stabkräfte wurde nur der linke Teil des Fachwerks verwendet. Die analoge berechnung über den reten Teil des Fachwerks muss die gleichen Kräfte ergeben.
Freischneiden des rechten Fachwerkteils an den Stäben 1, 2 und 3.



Damit es zu keinen Vertauschungen kommt, werden die Kräfte und Punkte in diesem Abschnitt mit den Indices R1, R2 und R3 bezeichnet.

Die Richtung der Kräfte $\vec{F}_{R1}$, $\vec{F}_{R2}$ und $\vec{F}_{R3}$ sind durch die Richtungen der Stäbe bestimmt.

```
> F_[R1] = <f[R1], 0 >;
```

$$\vec{F}_{R1} = \begin{bmatrix} f_{R1} \\ 0 \end{bmatrix} \tag{48}$$

```
> F_[R2] = <1,-1>* eval( f[R2]/sqrt(2), 1 );
```

$$\vec{F}_{R2} = \begin{bmatrix} \frac{f_{R2}}{\sqrt{2}} \\ -\frac{f_{R2}}{\sqrt{2}} \end{bmatrix} \tag{49}$$

```
> F_[R3] = < f[R3], 0 >;
```

$$\vec{F}_{R3} = \begin{bmatrix} f_{R3} \\ 0 \end{bmatrix} \quad (50)$$

Die Koordinaten der drei Angriffspunkte in der Skizze ablesen.

```
> P_[R1] = < 5*a, a >;
```

$$\vec{P}_{R1} = \begin{bmatrix} 5\,a \\ a \end{bmatrix} \quad (51)$$

```
> P_[R2] = < 4*a, 0 >;
```

$$\vec{P}_{R2} = \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \quad (52)$$

```
> P_[R3] = < 4*a, 0 >;
```

$$\vec{P}_{R3} = \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \quad (53)$$

Die Summe der angreifenden Kräfte ist 0.

```
> F_[B] + F_[C] + F_[R1] + F_[R2] + F_[R3] = 0;
```

$$\vec{F}_B + \vec{F}_C + \vec{F}_{R1} + \vec{F}_{R2} + \vec{F}_{R3} = 0 \quad (54)$$

Einsetzen der Richtungen von Lagerkräften und Stabkräften aus Gleichungen **(6)**, **(7)**, **(48)**, **(49)** und **(50)**.

```
> subs((6),(7),(48),(49),(50), (54));
```

$$\begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \begin{bmatrix} f_{R1} \\ 0 \end{bmatrix} + \begin{bmatrix} \frac{f_{R2}}{\sqrt{2}} \\ -\frac{f_{R2}}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} f_{R3} \\ 0 \end{bmatrix} = 0 \quad (55)$$

Die Vektoren addieren.

```
> (55);
```

$$\begin{bmatrix} -\frac{f_B\sqrt{2}}{2} + f_{R1} + \frac{f_{R2}\sqrt{2}}{2} + f_{R3} \\ \frac{f_B\sqrt{2}}{2} + f_C - \frac{f_{R2}\sqrt{2}}{2} \end{bmatrix} = 0 \quad (56)$$

Die Summe der angreifenden Drehmomente ist 0.

```
> P_[B] &x F_[B] + P_[C] &x F_[C] + P_[R1] &x F_[R1] + P_[R2] &x
  F_[R2] + P_[R3] &x F_[R3] = 0;
```

$$\vec{P}_B \times \vec{F}_B + \vec{P}_C \times \vec{F}_C + \vec{P}_{R1} \times \vec{F}_{R1} + \vec{P}_{R2} \times \vec{F}_{R2} + \vec{P}_{R3} \times \vec{F}_{R3} = 0 \quad (57)$$

Einsetzen der Richtungen von LAgerkräften und Stabskräften aus Gleichungen **(6)**, **(7)**, **(48)**, **(49)** und **(50)**.

```
> subs((6),(7),(48),(49),(50), (57));
```

$$\vec{P}_B \times \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \vec{P}_C \times \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \vec{P}_{R1} \times \begin{bmatrix} f_{R1} \\ 0 \end{bmatrix} + \vec{P}_{R2} \times \begin{bmatrix} \frac{f_{R2}}{\sqrt{2}} \\ -\frac{f_{R2}}{\sqrt{2}} \end{bmatrix} + \vec{P}_{R3} \times \begin{bmatrix} f_{R3} \\ 0 \end{bmatrix} = 0 \qquad \textbf{(58)}$$

Einsetzen der Angriffspunkte aus den Gleichungen **(9)**, **(10)**, **(51)**, **(52)**, **(53)**.

```
> subs((9),(10),(51),(52),(53), (58) );
```

$$\begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} -\frac{f_B}{\sqrt{2}} \\ \frac{f_B}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 6\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} 0 \\ f_C \end{bmatrix} + \begin{bmatrix} 5\,a \\ a \end{bmatrix} \times \begin{bmatrix} f_{R1} \\ 0 \end{bmatrix} + \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} \frac{f_{R2}}{\sqrt{2}} \\ -\frac{f_{R2}}{\sqrt{2}} \end{bmatrix} + \begin{bmatrix} 4\,a \\ 0 \end{bmatrix} \times \begin{bmatrix} f_{R3} \\ 0 \end{bmatrix} = 0 \qquad \textbf{(59)}$$

Ausrechnen der Kreuzprodukte.

```
> subsindets( (59),
    specfunc(Vector,op(0,<1> &x <1>)),
    g -> cross2D( op(g) ) );
```

$$2\,a\,f_B\sqrt{2} + 6\,a\,f_C - a\,f_{R1} - 2\,a\,f_{R2}\sqrt{2} = 0 \qquad \textbf{(60)}$$

Die Komponenten der Vektorgleichung **(56)** und die Gleichung **(60)** bilden ein Gleichungssystem für die Stabkräfte.

```
> lhs((56))[1] = rhs((56))[1];
```

$$-\frac{f_B\sqrt{2}}{2} + f_{R1} + \frac{f_{R2}\sqrt{2}}{2} + f_{R3} = 0 \qquad \textbf{(61)}$$

```
> lhs((56))[2] = rhs((56))[2];
```

$$\frac{f_B\sqrt{2}}{2} + f_C - \frac{f_{R2}\sqrt{2}}{2} = 0 \qquad \textbf{(62)}$$

```
> (60);
```

$$2\,a\,f_B\sqrt{2} + 6\,a\,f_C - a\,f_{R1} - 2\,a\,f_{R2}\sqrt{2} = 0 \qquad \textbf{(63)}$$

Das lineare Gleichungssystem **(61)**, **(62)** und **(63)** nach den Stabkräften f_{R1}, f_{R2} und f_{R3} auflösen.

```
> solve( [(61),(62),(63)], [ f[R1],f[R2],f[R3] ] )[1][];
```

$$f_{R1} = 2\,f_C, f_{R2} = \frac{\sqrt{2}\,\left(f_B\sqrt{2} + 2\,f_C\right)}{2}, f_{R3} = -3\,f_C \qquad \textbf{(64)}$$

Die Lagerkräfte f_B und f_C aus **(25)** einsetzen.

```
> Loesung3 := simplify( subs(Loesung1, [(64)] ) ): Loesung3[];
```

$$f_{R1} = f_1 + \frac{2\,f_2}{3}, f_{R2} = -\frac{\sqrt{2}\,\left(-2\,f_2 + 3\,f_1\right)}{6}, f_{R3} = -f_2 - \frac{3\,f_1}{2} \qquad \textbf{(65)}$$

Die Zahlenwert für die Kräfte ausrechnen. Die Werte für die äußeren Kräfte aus **(2)** und **(4)** einsetzen.

```
> evalf( eval( Loesung3, [(2),(4)] ) )[];
```

$$f_{R1} = 266.67 \, [\![N]\!], f_{R2} = -94.281 \, [\![N]\!], f_{R3} = -400. \, [\![N]\!] \qquad \textbf{(66)}$$

Die Kontrolle liefert die gleichen Ergebnisse, wie Gleichungen **(46)** und **(47)**.

Hilfsmittel:
- Korrekturen von Derfnam
- Maple 14, http://www.maplesoft.com/
- Hering, Martin, Stohrer: Physikalisch-Technisches Taschenbuch, VDI-Verlag