```
> restart;
```

Betrachtet wird ein Gasspeicher aus dem Gas strömt. Gegeben ist der Startzustand und ein konstanter Massenstrom. Berechnet werden soll Druck p(t) und Temperatur T(t) des Gases im Speicher.

Skizze:



Der Behälter ist wärmeisoliert (adiabatisch) und hat ein konstantes Volumen V.
Das Gas hat die individuelle Gaskonstante R_i und den Isentropenexponent $\varkappa$.
Zum Startzeitpunkt t=0 hat das Gas im Behälter den Druck p_0 und die Temperatur T_0.

Vereinfachungen:
- Das Gas wird als mehratomiges ideales Gas beschrieben. (Insbesondere kein Joule-Thomson-Effekt.)
- Der Isentropenexponent $\varkappa$ wird als konstant angenommen. (Unabhängig von der Temperatur)
- Die Strömung ist so langsam, dass zu jedem Zeitpunkt ein thermisches Gleichgewicht für das Gas im Behälter gilt.
- Die Höhe des Behälters ist so klein, dass potentielle Energie des Gases im Schwerefeld vernachlässigt werden kann.

Der Massenstrom m_s führt aus dem Behälter, das wird mit einem negativen Vorzeichen ausgedrückt.

Die Masse m(t) im Behälter berechnen.
Thermische Zustandsgleichung Startzustand liefert die Masse m_0 zum Startzeitpunkt.

```
> p[0]*V=m[0]*R[i]*T[0];
```

$$p_0 V = m_0 R_i T_0 \quad (1)$$

```
> isolate((1),m[0]);
```

$$m_0 = \frac{p_0 V}{R_i T_0} \quad (2)$$

Die Masse als Funktion der Zeit t.

```
> m(t) = m[0] + m[s] * t;
```

$$m(t) = m_0 + m_s t \quad \textbf{(3)}$$

```
> diff((3),t);
```

$$\frac{\mathrm{d}}{\mathrm{d}t} m(t) = m_s \quad \textbf{(4)}$$

Die spezifische innere Energie u des idealen Gases ist nur von der Temperatur T abhängig.

```
> u(t) = c[v] * T(t);
```

$$u(t) = c_v T(t) \quad \textbf{(5)}$$

Dabei ist die spezifische Wärmekapazität cv bei konstantem Volumen.

```
> c[v] = R[i]/(kappa-1);
```

$$c_v = \frac{R_i}{\kappa - 1} \quad \textbf{(6)}$$

```
> subs((6),(5));
```

$$u(t) = \frac{R_i T(t)}{\kappa - 1} \quad \textbf{(7)}$$

Der 1. Hauptsatz der Thermodynanik für offene Systeme liefert eine Differentialgleichung für das System.

```
> dA[t] + dQ[a] - dm[s]*(h[s]+e[s])=dU+dE;
```

$$dA_t + dQ_a - dm_s (h_s + e_s) = dU + dE \quad \textbf{(8)}$$

Es wird keine technische Arbeit geleistet.

```
> dA[t]=0;
```

$$dA_t = 0 \quad \textbf{(9)}$$

Der Behälter ist adiabatische. Es gibt keinen Wärmestrom zwischem dem Gas im Behälter und der Umgebung.

```
> dQ[a] = 0;
```

$$dQ_a = 0 \quad \textbf{(10)}$$

Die spezifische Enthalpie h_s des strömendes Gases ist gleich der spezifischen Enthalpie h des Gases im Behälter. Das Gas im Behälter wurde nur bis zu einer Grenzschicht definiert. Dort ist das strömende Gas noch so langsam, die Enthalpie unverändert ist und die Strömungsgeschwindigkeit nahezu 0 ist. Im Bereich der Leitung aus dem Behälter hinaus wird die Strömungsgeschwindigkeit steigen und die Enthalpie fallen. Doch dieser Bereich wird hier nicht beachtet, weil es nur um die Beschreibung des Gases im Behälter geht und nicht um die Beschreibung der Strömung aus dem Behälter durch die Rohleitung hinaus.

```
> h[s] = h;
```

$$h_s = h \quad \textbf{(11)}$$

Die spezifische äußere Energie des strömenden Gases ist nahezu 0, siehe oben.

```
> e[s] = 0;
```

$$e_s = 0 \quad \textbf{(12)}$$

Das Gas im Behälter ruht, die Behälterhöhe wird vernachlässigt. Die äußere Energie des Gases im Behälter ändert sich nicht.

```
> dE = 0;
```

$$dE = 0 \quad \textbf{(13)}$$

Zusammengefasst:

```
> subs((9),(10),(11),(12),(13),(8));
```

$$-dm_s\, h = dU \tag{14}$$

Ableitungen nach der Zeit betrachten.

```
> diff(U(t),t) = m[s]*h(t);
```

$$\frac{d}{dt}\, U(t) = m_s\, h(t) \tag{15}$$

(Das negative Vorzeichen ist in der Konstante m_s enthalten.)

Die innere Energie U durch die spezifische innere Energie u ausdrücken.

```
> U(t) = m(t)*u(t);
```

$$U(t) = m(t)\, u(t) \tag{16}$$

```
> subs((16), (15));
```

$$\frac{d}{dt}\, (m(t)\, u(t)) = m_s\, h(t) \tag{17}$$

```
> expand((17));
```

$$\left(\frac{d}{dt}\, m(t)\right) u(t) + m(t) \left(\frac{d}{dt}\, u(t)\right) = m_s\, h(t) \tag{18}$$

```
> subs((4),(18));
```

$$m_s\, u(t) + m(t) \left(\frac{d}{dt}\, u(t)\right) = m_s\, h(t) \tag{19}$$

Die spezifische Enthalpie h durch die spezifische innere Energie u ausdrücken.

```
> h(t) = u(t) + p(t) * v(t);
```

$$h(t) = u(t) + p(t)\, v(t) \tag{20}$$

Dabei ist v das spezifische Volumen.

```
> subs((20),(19));
```

$$m_s\, u(t) + m(t) \left(\frac{d}{dt}\, u(t)\right) = m_s\, (u(t) + p(t)\, v(t)) \tag{21}$$

```
> expand((21));
```

$$m_s\, u(t) + m(t) \left(\frac{d}{dt}\, u(t)\right) = m_s\, u(t) + m_s\, p(t)\, v(t) \tag{22}$$

```
> (22)-m[s]*u(t);
```

$$m(t) \left(\frac{d}{dt}\, u(t)\right) = m_s\, p(t)\, v(t) \tag{23}$$

Die thermische Zustandsgleichung ist

```
> p(t)*v(t) = R[i] * T(t);
```

$$p(t)\, v(t) = R_i\, T(t) \tag{24}$$

Einsetzen.

```
> algsubs((24),(23));
```

$$m(t) \left(\frac{d}{dt}\, u(t)\right) = m_s\, R_i\, T(t) \tag{25}$$

Die spezifische innere Energie u durch die Temperatur mit Gleichung **(7)** ausdrücken.

```
> subs((7),(25));
```

$$m(t) \left(\frac{\partial}{\partial t} \left(\frac{R_i\, T(t)}{\kappa - 1}\right)\right) = m_s\, R_i\, T(t) \tag{26}$$

```
> simplify((26));
```

$$\frac{m(t)\,R_i\left(\frac{d}{dt}T(t)\right)}{\kappa-1}=m_s\,R_i\,T(t) \tag{27}$$

Die bekannte Masse m, Gleichung **(3)**, einsetzen.

```
> subs((3),(27));
```

$$\frac{\left(m_0+m_s\,t\right)R_i\left(\frac{d}{dt}T(t)\right)}{\kappa-1}=m_s\,R_i\,T(t) \tag{28}$$

In dieser Gleichung ist nur noch Temperatur T eine Unbekannte. Es ist eine Differenzialgleichung für T (t).

Die Anfangsbedingung ist T(0) = T_0, die gegebene Gastemperatur am Anfang.

Das System aus der Differenzialgleichung mit der Anfangsbedingung wird von Maple sofort gelöst:

```
> dsolve({(28),T(0)=T[0]});
```

$$T(t)=\frac{\left(m_0+m_s\,t\right)^{\kappa-1}T_0}{m_0^{(\kappa-1)}} \tag{29}$$

```
> combine((29));
```

$$T(t)=\left(m_0+m_s\,t\right)^{\kappa-1}T_0\,m_0^{(-\kappa+1)} \tag{30}$$

Damit ist die Temperatur T(t) bestimmt.

Der Druck p(t) folgt aus der thermischen Zustandsgleichung zum Zeitpunkt t.

```
> p(t)*V=m(t)*R[i]*T(t);
```

$$p(t)\,V=m(t)\,R_i\,T(t) \tag{31}$$

Die Masse m, Gleichung **(3)**, einsetzen.

```
> subs((3),(31));
```

$$p(t)\,V=\left(m_0+m_s\,t\right)R_i\,T(t) \tag{32}$$

Die Temperatur T, Gleichung **(30)**, einsetzen.

```
> subs((30),(32));
```

$$p(t)\,V=\left(m_0+m_s\,t\right)R_i\left(m_0+m_s\,t\right)^{\kappa-1}T_0\,m_0^{(-\kappa+1)} \tag{33}$$

```
> simplify((33));
```

$$p(t)\,V=\left(m_0+m_s\,t\right)^{\kappa}R_i\,T_0\,m_0^{(-\kappa+1)} \tag{34}$$

```
> isolate((34),p(t));
```

$$p(t)=\frac{\left(m_0+m_s\,t\right)^{\kappa}R_i\,T_0\,m_0^{(-\kappa+1)}}{V} \tag{35}$$

Geschriebe mit dem Druck p(t)=p_0.

```
> isolate( p[0]*V=m[0]*R[i]*T[0], p[0] );
```

$$p_0=\frac{m_0\,R_i\,T_0}{V} \tag{36}$$

```
> lhs((35))/lhs((36))=rhs((35))/rhs((36));
```

$$\frac{p(t)}{p_0} = \frac{(m_0 + m_s\, t)^{\kappa}\, m_0^{(-\kappa+1)}}{m_0} \tag{37}$$

```
> combine((37));
```

$$\frac{p(t)}{p_0} = (m_0 + m_s\, t)^{\kappa}\, m_0^{(-\kappa)} \tag{38}$$

```
> isolate((38),p(t));
```

$$p(t) = (m_0 + m_s\, t)^{\kappa}\, m_0^{(-\kappa)}\, p_0 \tag{39}$$

Damit ist der Druck p(t) bestimmt.