```
> restart;
> Digits:=25: interface(displayprecision=5):
```
> $mx := \overline{x}$: $my := \overline{y}$:

Aufgabe

Der elektrischer Widerstand R eines Heißleiters hängt von der absoluten Temperatur T nach der Gleichung $R = A \cdot e^{\frac{B}{T}}$ ab.
Die Parameter A und B sollen nach der Methoden der Ausgleichsrechnung bestimmt werden.

Die Messwerte (Temperaturen in Kelvin, Widerstände in Ohm, Einheiten vergessen - es ist Mathe):

```
> T = [293.15, 313.15, 333.15, 353.15, 373.15];
```

$$T = [293.15, 313.15, 333.15, 353.15, 373.15] \tag{1}$$

```
> R = [510, 290, 178, 120, 80];
```

$$R = [510, 290, 178, 120, 80] \tag{2}$$

Rechnung

Die gegebene Abhängigkeit

```
> R=A*exp(B/T);
```

$$R = A\, e^{\frac{B}{T}} \tag{3}$$

logarithmiert

```
> ln(lhs((3)))=ln(rhs((3)));
```

$$\ln(R) = \ln\left(A\, e^{\frac{B}{T}}\right) \tag{4}$$

(In der Mathmematik kann sofort logarithmiert werden. In der Physik müsste zuerst die Gleichung dimensionslos gemacht werden durch Division mit einem Referenzwiderstand von z.B. 1Ω.)

```
> simplify((4)) assuming real: expand(%);
```

$$\ln(R) = \frac{B}{T} + \ln(A) \tag{5}$$

Wird

```
> ln(R) = y;
```

$$\ln(R) = y \tag{6}$$

als Funktionswert angesehen und

```
> 1/T = x;
```

$$\frac{1}{T} = x \tag{7}$$

als unabhängige Variable. Dann ist die Gleichung **(5)**

```
> subs((6),(7),(5));
```

$$y = B\, x + \ln(A) \tag{8}$$

eine Geradengleichung für y als Funktion von x. Die Steigung der Funktion ist

```
> B = b;
```

$$B = b \tag{9}$$

Der y-Achsenabschnitt ist

```
> ln(A) = a;
```

$$\ln(A) = a \tag{10}$$

Die Gerade **(8)** ist damit in die Standardform

```
> subs((9),(10),(8));
```

$$y = b\,x + a \tag{11}$$

gebracht.

Die gemessenen Widerstandswerte R_i nach **(6)** in y_i Werte umrechnen.

```
> y = ln~(rhs((2))): evalf(%);
```

$$y = [6.2344, 5.6699, 5.1818, 4.7875, 4.3820] \tag{12}$$

Die gemessenen Temperaturen T_i nach **(7)** in x_i Werte umrechnen.

```
> x = 1/~rhs((1));
```

$$x = [0.0034112, 0.0031934, 0.0030017, 0.0028317, 0.0026799] \tag{13}$$

Die Formeln für die Berechnung der Ausgleichsgeraden (auch lineare Regression genannt) aus der Formelsammlung

```
b=Sum((x[i]-mx)*(y[i]-my),i=1..n)/Sum((x[i]-mx)^2,i=1..n);
```

$$b = \frac{\sum_{i=1}^{n} \left(x_i - \overline{x}\right) \left(y_i - \overline{y}\right)}{\sum_{i=1}^{n} \left(x_i - \overline{x}\right)^2} \tag{14}$$

```
> a=my-b*mx;
```

$$a = \overline{y} - b\,\overline{x} \tag{15}$$

In den Formeln werden die gemittelten Messwerte verwendet.

```
> mx=sum(x[i],i=1..n)/n;
```

$$\overline{x} = \frac{\sum_{i=1}^{n} x_i}{n} \tag{16}$$

```
> my=sum(y[i],i=1..n)/n;
```

$$\overline{y} = \frac{\sum_{i=1}^{n} y_i}{n} \tag{17}$$

Ausrechnen mit Mittelwerte von dem umgerechneten Messwerten aus **(12)** und **(13)**.

```
> lhs((16))=subs((13),n=nops(rhs((13))),rhs((16))): evalf(%);
```

$$\overline{x} = 0.0030236 \tag{18}$$

```
> lhs((17))=subs((12),n=nops(rhs((12))),rhs((17))): evalf(%);
```

$$\overline{y} = 5.2511 \tag{19}$$

Jetzt die Parameter b und a aus dem Formeln **(14)** und **(15)** berechnen.

```
> subs((18),(19),(12),(13),n=nops(rhs((13))),(14)): evalf(%);
```

$$b = 2515.4 \tag{20}$$

```
> subs((18),(19),(20),(15));
```

$$a = -2.3542 \tag{21}$$

Aus den Parametern a und b der Geraden können die gesuchten Parameter A und B der Widerstandsfunktion R(T) berechnet werden.
Gleichung **(9)** liefert sofort B.

```
> subs((20),(9));
```

$$B = 2515.4 \tag{22}$$

Gleichung **(10)** liefert A.

```
> isolate((10),A);
```

$$A = e^{a} \tag{23}$$

```
> subs((21),(23)): evalf(%);
```

$$A = 0.094970 \tag{24}$$

Die Ergebnisse der Rechnung dargestellt.

Die Regressionsgerade und die umgerechneten Messpunkte.

```
> subs((20),(21),(11));
```

$$y = 2515.4\, x - 2.3542 \tag{25}$$

```
> d1 := plot(rhs((25)),x=0.0025..0.0035):
> d2 := plots[pointplot]( rhs((13)),rhs((12)), symbol=diagonalcross,
  symbolsize=15 ):
> plots[display](d1,d2,axes=boxed,labels=[x,y]);
```



Die Exponentialfunktion und die Messpunkte.

```
> subs((22),(24),(3));
```

$$R = 0.094970\, e^{\frac{2515.4}{T}} \tag{26}$$

```
> d1 := plot(rhs((26)),T=280..380):
> d2 := plots[pointplot]( rhs((1)),rhs((2)), symbol=diagonalcross,
  symbolsize=15 ):
> plots[display](d1,d2,axes=boxed,labels=[R/Unit(ohm),T/Unit(K)]);
```



Hilfsmittel
- Bronstein: Taschenbuch der Mathematik, Verlag Harri Deutsch
- Maple 17