```
> restart;
> with(ScientificConstants):
> Digits:=30: interface(displayprecision=5):
```

Die Adwoodsche Fallmaschine ist über die Energieerhaltung zu berechnen.

Skizze:



Zwei Massen $M_1 = M_2$ hängen über ein Seil an einer Rolle.

Zwischen Seil und Rolle gibt es keinen Schlupf. Alle anderen Reibungskräfte werden vernachlässigt.
Die Masse des Seils kann gegenüber den Massen M_1, M_2 vernachlässigt werden.

Auf der rechten Seite ist eine Zusatz M_3 am Seil.

Die Masse der Rolle ist M_4. Die Rolle ist ein homogener Zylinder mit Radius r.

Gegeben sind die Werte:

```
> M[1] = 1.0*Unit(kg);
```

$$M_1 = 1.00000 \,[\![kg]\!] \tag{1}$$

```
> M[2] = M[1];
```

$$M_2 = M_1 \tag{2}$$

```
> M[3] = 0.1*Unit(kg);
```

$$M_3 = 0.10000 \,[\![kg]\!] \tag{3}$$

Aus dem Tabellenbuch die Fallbeschleungigung an der Erdoberfläche.

```
> g = evalf(Constant(standard_acceleration_of_gravity,units));
```

$$g = 9.80665 \left[\left[\frac{m}{s^2} \right]\right] \quad (4)$$

```
> parameter := ((1),(2),(3),(4)):
```

Bewegen sich die Masse M_2 und M_3 um die Strecke x nach unten, dann bewegt sich die Masse M_1 um die Strecke x nach oben, weil alle Masse durch das feste Seil verbunden sind.
Definition Nullpunkt: die potenzielle Energie ist 0 bei x=0.
Die Potentielle Energie der Massen:

```
> E[pot] = M[1]*g*x - (M[2]+M[3])*g*x;
```

$$E_{pot} = M_1\, g\, x - (M_2 + M_3)\, g\, x \quad (5)$$

```
> collect((5),[x,g]);
```

$$E_{pot} = (M_1 - M_2 - M_3)\, g\, x \quad (6)$$

Die kinetische Energie der Massen.

```
> E[kin] = 1/2 * sum(M[i],i=1..3)* v^2;
```

$$E_{kin} = \frac{1}{2}\,(M_1 + M_2 + M_3)\, v^2 \quad (7)$$

Es geht keine Energie durch Reibung verloren. Die Energie bleibt erhalten.
Die Energie zum Startzeitpunkt ist 0, weil die Massen in Ruhe waren und die potentielle Energie dort nullgesetzt wurde.

```
> 0 = E[kin] + E[pot];
```

$$0 = E_{kin} + E_{pot} \quad (8)$$

```
> subs((6),(7),(8));
```

$$0 = \frac{1}{2}\,(M_1 + M_2 + M_3)\, v^2 + (M_1 - M_2 - M_3)\, g\, x \quad (9)$$

Auflösen nach der Geschwindigkeit v. Dabei nur den positiven Zweig der Wurzel berücksichtigen, weil die x-Richtung so gewählt ist, dass v>0 ist.

```
> solve((9),v)[1]: v=simplify(sqrt(%**2));
```

$$v = \sqrt{2}\,\sqrt{-\frac{(M_1 - M_2 - M_3)\, g\, x}{M_1 + M_2 + M_3}} \quad (10)$$

Mit M_1=M_2.

```
> simplify(subs(M[2]=M[1],(10)));
```

$$v = \sqrt{2}\,\sqrt{\frac{M_3\, g\, x}{2\, M_1 + M_3}} \quad (11)$$

Der Geschwindigkeit v als Funktion der Fallstrecke x beim freie Fall zum Vergleich.

```
> v = sqrt(2*g*x);
```

$$v = \sqrt{2}\,\sqrt{g\, x} \quad (12)$$

Die Beschleunigung a folgt aus dem Vergleich.

```
> a = simplify((rhs((11))^2)/(2*x)) assuming x>0;
```

$$a = \frac{M_3\, g}{2\, M_1 + M_3} \quad (13)$$

Den Zahlenwert aus den gegebenen Massen ausrechnen.

```
> simplify(subs(parameter,(13)));
```

(14)

$$a = 0.46698 \left[\left[\frac{m}{s^2} \right]\right] \tag{14}$$

Die Rolle dreht sich mit der Winkelgeschwindigkeit ω. Das Seil hat keinen Schlupf, also muss die Umfangsgeschwindigkeit der Rolle gleich der Geschwindigkeit v des Seils sein.

```
> omega = v/r;
```

$$\omega = \frac{v}{r} \tag{15}$$

Ist das Trägheitsmoment der Rolle um ihre Achse J, dann ist die kinetische Energie der rotierenden Rolle.

```
> E[rot] = 1/2*J*omega^2;
```

$$E_{rot} = \frac{1}{2} J \omega^2 \tag{16}$$

```
> subs((15),(16));
```

$$E_{rot} = \frac{1}{2} \frac{J v^2}{r^2} \tag{17}$$

Die Energiebilanz mit der Rotationsenergie:
Auch die Rolle ruht zum Startzeitpunkt. Also ist auch die Energie der Rotation gleich 0 beim Startzeitpunkt. Die Summe der Energien bleibt 0.

```
> 0 = E[rot]+E[kin]+E[pot];
```

$$0 = E_{rot} + E_{kin} + E_{pot} \tag{18}$$

```
> subs((6),(7),(17),(18));
```

$$0 = \frac{1}{2} \frac{J v^2}{r^2} + \frac{1}{2} \left(M_1 + M_2 + M_3\right) v^2 + \left(M_1 - M_2 - M_3\right) g x \tag{19}$$

Vereinfachen mit $M_1 = M_2$.

```
> subs(M[2]=M[1],(19));
```

$$0 = \frac{1}{2} \frac{J v^2}{r^2} + \frac{1}{2} \left(2 M_1 + M_3\right) v^2 - g x M_3 \tag{20}$$

Auflösen nach v.

```
> solve((20),v)[1]: v=simplify(sqrt(%**2));
```

$$v = \sqrt{2} \sqrt{\frac{g x M_3 r^2}{2 r^2 M_1 + r^2 M_3 + J}} \tag{21}$$

Vergleich mit der Formel für den freien Fall liefert hier die Beschleunigung.

```
> a = simplify((rhs((21))^2)/(2*x)) assuming x>0;
```

$$a = \frac{g M_3 r^2}{2 r^2 M_1 + r^2 M_3 + J} \tag{22}$$

Das Trägheitsmoment eines Zylinders mit Masse M_4 und Radius r um seine Symmetrieachse aus der Formelsammlung:

```
> J = 1/2*M[4]*r^2;
```

$$J = \frac{1}{2} M_4 r^2 \tag{23}$$

Einsetzen in die Formel **(22)** für die Beschleunigung.

```
> simplify(subs((23),(22)));
```

$$a = \frac{2\,g\,M_3}{4\,M_1 + 2\,M_3 + M_4} \tag{24}$$

Auflösen nach der Masse M_4 der Rolle.

```
> isolate((24),M[4]);
```

$$M_4 = \frac{2\,M_3\,g}{a} - 4\,M_1 - 2\,M_3 \tag{25}$$

Gegeben ist

```
> a=0.4*Unit(m/s^2);
```

$$a = 0.40000 \left[\!\left[\frac{m}{s^2} \right]\!\right] \tag{26}$$

Den Zahlenwert ausrechnen.

```
> simplify(subs((26),parameter,(25)));
```

$$M_4 = 0.70333 \,[\![kg]\!] \tag{27}$$

```
>
```