```
> restart;
> Digits := 30: interface( displayprecision = 8 ):
```

Aufgabe

Gegeben ist die zurückzulegende Wegstrecke bei der eine bestimmte Endgeschwindigkeit (nicht gleich 0) erreicht werden soll. Es gibt zunächst eine positiv beschleunigte Bewegung mit einer gegebenen Anfangsgeschwindigkeit und Beschleunigung. Darauf folgt unmittelbar eine Abbremsbewegung mit gegebener negativer Beschleunigung und eben jener Endgeschwindigkeit.
1. Höchstgeschwindigkeit
2. Die benötigte Zeit bis die Höchstgeschwindigkeit erreicht wird.

Bearbeitung

Die Bewegung ist zusammengesetzt aus zwei Teilen.
1. Beschleunigen mit konstanter Beschleunigung.
2. Bremsen mit konstanter Bremsbeschleunigung.

Skizze des Geschwindigkeitsverlaufs (Geschwindigkeit v, Zeit t, Ort x):



Beide Teile sind gleichmäßig beschleunigte Bewegungen. Das Physikbuch [1] liefert für diesen Bewegungstyp fertige Formeln.

Die Beschleunigung a ist konstant.

```
> a = const;
```

$$a = const \quad (1)$$

Die Geschwindigkeit v ist eine lineare Funktion der Zeit t. Die Geschwindigkeit zum Startzeitpunkt wird v_0 genannt.

```
> v = a*t + v[0];
```

$$v = a\,t + v_0 \quad (2)$$

Der Ort x ist eine quadratische Funktion der Zeit t. Der Ort zum Startzeitpunkt wird x

$_0$ genannt.

```
> x = a/2*t^2 + v[0]*t + x[0];
```

$$x = \frac{1}{2}\, a\, t^2 + v_0\, t + x_0 \tag{3}$$

Aus diesen vorgegebenen Formeln kann die Bewegung beschrieben werden.
Geschwindigkeit v und Ort x werden zusammengesetzt aus den bekannten Formeln.

Der erste Teil dauert von t=0 bis t=T_1. Der zweite Teil dauert bis zur Zeit T_2.

Die Beschleunigung a ist im ersten Teil a_1 und im zweiten Teil a_2 .

```
> a = piecewise( t <= T[1], a[1], a[2] );
```

$$a = \begin{cases} a_1 & t \le T_1 \\ a_2 & \mathit{otherwise} \end{cases} \tag{4}$$

Die gegebene Anfangsgeschwindigkeit ist v_0 bei t = 0.

Die gesuchte Höchstgeschwindigkeit ist v_1 bei T_1.

Die gegebene Endgeschwindigkeit ist v_2 bei T_2.

Die Geschwindigkeit v folgt aus **(2)** für die beiden Teile.

```
> v = piecewise( t <= T[1], a[1]*t + v[0], a[2]*(t-T[1]) + v[1] )
  ;
```

$$v = \begin{cases} a_1\, t + v_0 & t \le T_1 \\ a_2\, (t - T_1) + v_1 & \mathit{otherwise} \end{cases} \tag{5}$$

Aus **(5)** folgt bei t = $T_{1:}$

```
> v[1] = simplify( subs(t=T[1],rhs((5))) );
```

$$v_1 = a_1\, T_1 + v_0 \tag{6}$$

Aus **(5)** folgt bei t = $T_{2:}$

```
> v[2] = simplify( subs(t=T[2],rhs((5)) ) ) assuming T[2]>T[1]:
  simplify(%,size);
```

$$v_2 = (T_2 - T_1)\, a_2 + v_1 \tag{7}$$

Der Anfangsort ist x = 0 bei t = 0.
Der Ort s_1 wird zum Zeitpunkt T_1 erreicht.

Der Ort s_2 beim Zeitpunkt T_2 ist die gegebene Streckenlänge.

Der Ort x folgt aus **(3)** für die beiden Teile.

```
> x = piecewise( t <= T[1], a[1]*t^2/2 + v[0]*t, a[2]*(t-T[1])
  ^2/2 + v[1]*(t-T[1]) + s[1] );
```

$$x = \begin{cases} \frac{1}{2}\, a_1\, t^2 + v_0\, t & t \le T_1 \\ \frac{a_2\, (t - T_1)^2}{2} + v_1\, (t - T_1) + s_1 & \mathit{otherwise} \end{cases} \tag{8}$$

Aus **(8)** folgt bei t = $T_{1:}$

```
> s[1] = simplify( subs(t=T[1],rhs((8))) );
```

(9)

$$s_1 = \frac{1}{2}\, a_1\, T_1^2 + v_0\, T_1 \tag{9}$$

Aus **(8)** folgt bei t = T_2:

```
> s[2] = simplify( subs(t=T[2],rhs((8)) ) ) assuming T[2]>T[1]:
  simplify(%,size);
```

$$s_2 = \frac{(T_1 - T_2)^2\, a_2}{2} + s_1 + v_1\, T_2 - v_1\, T_1 \tag{10}$$

Die Gleichungen **(6)**, **(7)**, **(9)** und **(10)** bilden ein Gleichungssystem mit den Unbekannten T_1, T_2, v_1 und s_1. Die anderen Größen in den Gleichungen sind bekannt. Dieses Gleichungssystem ist zu lösen.

Gleichung **(6)** nach der Zeit auflösen.

```
> isolate( (6), T[1] ): simplify(%,size);
```

$$T_1 = \frac{v_1 - v_0}{a_1} \tag{11}$$

Gleichung **(7)** nach der Zeitdifferenz (= die Zeit für den zweiten Teil der Bewegung) auflösen.

```
> isolate( (7), T[2]-T[1] ): simplify(%,size);
```

$$T_2 - T_1 = \frac{v_2 - v_1}{a_2} \tag{12}$$

Gleichungen **(11)** und **(12)** in Gleichungen **(9)** und **(10)** einsetzen. Damit werden die unbekannten Zeiten eliminiert.

```
> subs( (11), (9) );
```

$$s_1 = \frac{(v_1 - v_0)^2}{2\, a_1} + \frac{v_0\,(v_1 - v_0)}{a_1} \tag{13}$$

```
> algsubs( (12), (10) );
```

$$s_2 = \frac{(-v_2 + v_1)^2}{2\, a_2} - \frac{v_1\,(-v_2 + v_1)}{a_2} + s_1 \tag{14}$$

Gleichung **(13)** ind Gleichung **(14)** einsetzen. Damit wird die unbekannte Teilstrecke s_1 eliminiert.

```
> subs( (13), (14) );
```

$$s_2 = \frac{(-v_2 + v_1)^2}{2\, a_2} - \frac{v_1\,(-v_2 + v_1)}{a_2} + \frac{(v_1 - v_0)^2}{2\, a_1} + \frac{v_0\,(v_1 - v_0)}{a_1} \tag{15}$$

In dieser Gleichung ist nur v_1 eine Unbekannte.

Nach dem Auflösen der Klammern heben sich viele Summanden auf.

```
> expand( (15) );
```

$$s_2 = \frac{v_2^2}{2\, a_2} - \frac{v_1^2}{2\, a_2} + \frac{v_1^2}{2\, a_1} - \frac{v_0^2}{2\, a_1} \tag{16}$$

Auflösen.

```
> isolate( (16), v[1]^2 ); simplify(%,size);
```

$$v_1^2 = \frac{-s_2 + \frac{v_2^2}{2\,a_2} - \frac{v_0^2}{2\,a_1}}{\frac{1}{2\,a_2} - \frac{1}{2\,a_1}}$$

$$v_1^2 = \frac{\left(-2\,s_2\,a_2 + v_2^2\right)\,a_1 - v_0^2\,a_2}{a_1 - a_2} \quad \textbf{(17)}$$

Ist der Ausdruck auf der rechten Seite nicht-negativ, kann die Aufgabe gelöst werden.
Für die Aufgabenstellung ist nur die positve Wurzel sinnvoll.

```
> ( simplify(sqrt(lhs((17)))) assuming v[1] > 0 ) = sqrt(rhs((17)));
```

$$v_1 = \sqrt{\frac{\left(-2\,s_2\,a_2 + v_2^2\right)\,a_1 - v_0^2\,a_2}{a_1 - a_2}} \quad \textbf{(18)}$$

Das ist die gesuchte maximale Geschwindigkeit.

Die Zeit T_1 bis zum erreichen der Geschwindigkeit v_1 kann über **(11)** berechnet werden.
Die Gleichung **(18)** in **(11)** einsetzen ist unübersichtlicht. Direktes einsetzen der Zahlenwerte ist einfacher.

Zahlenbeispiel

Die gegebene Startgeschwindigkeit und Endgeschwindigkeit.

```
> v[0] = 0, v[2] = 30 *Unit(m/s);
```

$$v_0 = 0,\ v_2 = 30 \left[\!\left[\frac{m}{s} \right]\!\right] \quad \textbf{(19)}$$

Die Beschleunigung im ersten Teil und die Bremsbeschleunigung im zweiten Teil.

```
> a[1] = 5*Unit(m/s^2), a[2] = -3*Unit(m/s^2);
```

$$a_1 = 5 \left[\!\left[\frac{m}{s^2} \right]\!\right],\ a_2 = -3 \left[\!\left[\frac{m}{s^2} \right]\!\right] \quad \textbf{(20)}$$

Die Länge der Strecke.

```
> s[2] = 7500 * Unit(m);
```

$$s_2 = 7500 \left[\!\left[m \right]\!\right] \quad \textbf{(21)}$$

Die Zahlenwerte in **(18)** einsetzen und die Maximalgeschwindigkeit ausrechnen.

```
> subs( (19), (20), (21), (18) ): simplify(%): evalf(%);
```

$$v_1 = 169.37385 \left[\!\left[\frac{m}{s} \right]\!\right] \quad \textbf{(22)}$$

Einsetzen in **(11)** und die Zeit zum Erreichen der Maximalgeschwindigkeit v_1 ausrechnen.

```
> subs( (19), (20), (22), (11) ): simplify(%);
```

$$T_1 = 33.874769 \left[\!\left[s \right]\!\right] \quad \textbf{(23)}$$

Nach 33,9 Sekunden wird die Geschwindigkeit 169 m/s erreicht.

Gleichung **(12)** aufgelöst nach T2 liefert die Zeit bis zum Ende der Strecke.

```
> subs( (19), (20), (22), (23), isolate( (12), T[2] ) ): simplify(%);
```

$$T_2 = 80.332718 \left[\!\left[s \right]\!\right] \quad \textbf{(24)}$$

Zusammenstellen der Bewegungsdaten, Beschleunigung a(t), Geschwindigkeit v(t) und Ort x(t).

```
> subs( (23),(20), (4) );
```

$$a = \begin{cases} 5 \left[\!\left[\frac{m}{s^2} \right]\!\right] & t \le 33.874769 \, [\![s]\!] \\ -3 \left[\!\left[\frac{m}{s^2} \right]\!\right] & \textit{otherwise} \end{cases} \tag{25}$$

```
> subs( (19),(20),(22),(23),(5) );
```

$$v = \begin{cases} 5\, t \left[\!\left[\frac{m}{s^2} \right]\!\right] & t \le 33.874769 \, [\![s]\!] \\ -3\, (t - 33.874769 \, [\![s]\!]) \left[\!\left[\frac{m}{s^2} \right]\!\right] + 169.37385 \left[\!\left[\frac{m}{s} \right]\!\right] & \textit{otherwise} \end{cases} \tag{26}$$

```
> subs( (9),(19),(20),(22),(23),(8) );
```

$$x = \begin{cases} \frac{5\, t^2}{2} \left[\!\left[\frac{m}{s^2} \right]\!\right] \\ -\frac{3\, (t - 33.874769 \, [\![s]\!])^2}{2} \left[\!\left[\frac{m}{s^2} \right]\!\right] + 169.37385\, (t - 33.874769 \, [\![s]\!]) \left[\!\left[\frac{m}{s} \right]\!\right] + 2868.7500 \left[\!\left[\frac{m}{s^2} \right]\!\right] [\![s]\!]^2 \end{cases}$$

```
> unitFree := proc(t) subsindets(t, has_unit, proc (z) convert(z,
  unit_free) end proc) end proc:
> plot( unitFree(rhs((25))),unitFree(t=0..rhs((24))), caption =
  "Beschleunigung", labels=["t/s",typeset("a/",m/s^2)] );
```



Beschleunigung

```
> plot( unitFree(rhs((26))),unitFree(t=0..rhs((24))), caption =
  "Geschwindigkeit", labels=["t/s",typeset("v/",m/s)] );
```



Geschwindigkeit

```
> plot( unitFree(rhs((27))),unitFree(t=0..rhs((24))), caption =
  "Ort", labels=["t/s", "x/m"] );
```



Ort

Hilfsmittel

1. Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch
2. Maple 17, www.maplesoft.com