```
> restart:
> Digits := 25: interface( displayprecision=5 ):
```

Aufgabe

Gegeben ist Fachwerk aus drei starren Scheiben (Stäben). Das Fachwerk ist verschieblich. Die gegebene Skizze mit den wichtigsten Punkten grün eingezeichnet:



Punkt A - Auflager der Scheibe 1.
Punkt B - Endpunkt der Linearführung an der Scheibe 2
Punkt C - Gelenk zwischen Scheibe 2 und Scheibe 3.
Punkt D - Auflager der Scheibe 3.

Gesucht ist die Drehung in den beiden Auflagern A und D, wenn die Linearführung verlängert wird. z ist die Verschiebung in der Führung. Der Drehwinkel bei A ist α, der Drehwinkel bei D ist β. Der Winkel ist positiv, wenn gegen den Uhrzeigersinn (mathematisch positiv) gedreht wird.

Kartesisches Koordinatensystems wählen: Nullpunkt gleich Punkt A. X-Achse nach rechts, Y-Achse nach oben.

Schreibweise: Vektoren in Großbuchstaben, Skalare in Kleinbuchstaben. Die Punkte in der Ausgangslage bekommen den Index 0, die Punkte nach der Bewegung bekommen den Index 1.

Koordinaten der Punkt aus der Zeichnung ablesen. Alle Längen werden ohne Einheit aufgeschrieben. Die Längen sind immer in Längeneinheiten der Skizze gemeint.

```
> A[0]=<0,0>, B[0]=<2,3>, C[0]=<5,3>, D[0]=<7,0>; params := %:
```

$$A_0 = \begin{bmatrix} 0 \\ 0 \end{bmatrix}, B_0 = \begin{bmatrix} 2 \\ 3 \end{bmatrix}, C_0 = \begin{bmatrix} 5 \\ 3 \end{bmatrix}, \mathrm{D}_0 = \begin{bmatrix} 7 \\ 0 \end{bmatrix} \tag{1}$$

Die Verschiebung und die Drehungen sind in der gezeichneten Situation 0. Die Einheit aller Winkel ist Radiant.

```
> z[0]=0, alpha[0]=0, beta[0]=0; params2:=%:
```

$$z_0 = 0, \alpha_0 = 0, \beta_0 = 0 \tag{2}$$

Aus der analytischen Geometrie:

Die Verschiebung eines Punkts X in die Richtung R um die Strecke s. Der Richtungsvektor R muss nicht normiert sein, daher ist durch die Länge von R zu dividieren.

```
> Schieben := (X,R,s) -> X+s*R/abs(R): schieben(X,R,s)=Schieben(X,R,s);
```

$$schieben(X, R, s) = X + \frac{s\,R}{|R|} \tag{3}$$

Die Drehung eines Punkts X um den Punkt P um den Winkel α.

```
> Drehen := (X,P,alpha) -> Matrix([[cos(alpha),-sin(alpha)],[sin(alpha),cos(alpha)]]).(X-P) + P: drehen(X,P,alpha)=Drehen(X,P,alpha);
```

$$drehen(X, P, \alpha) = \begin{bmatrix} \cos(\alpha) & -\sin(\alpha) \\ \sin(\alpha) & \cos(\alpha) \end{bmatrix} \cdot (X - P) + P \tag{4}$$

Die Bewegung der Punkte beschreiben.

Die Punkte in der Ausgangslage werden mit dem Index 0 geschrieben. Die Punkte nach der Bewegung werden mit dem Index 1 geschrieben.

Der Punkt B wird durch das Linearlager um z verschoben. Die Richtung der Verschiebung ergibt sich aus der Linie zwischen B und A. Die gesamte Scheibe mit Lienarlager wird um das Lager A gedreht. Die beiden Bewegungen symbolisch notiert:

```
> B[1] = drehen( schieben(B[0],B[0]-A[0],z), A[0], alpha );
```

$$B_1 = drehen\left(schieben\left(B_0, B_0 - A_0, z\right), A_0, \alpha\right) \tag{5}$$

Der Punkt C ist wie der Punkt B ein Teil der starren Scheibe 2. Der Punkt C bewegt sich genau wie der Punkt B.

```
> C[1] = drehen( schieben(C[0],B[0]-A[0],z), A[0], alpha );
```

$$C_1 = drehen\left(schieben\left(C_0, B_0 - A_0, z\right), A_0, \alpha\right) \tag{6}$$

Der Punkt C ist auch ein Teil der starren Scheibe 3. Die Scheibe wird um das Lager D gedreht. Die Drehung symbolisch notiert:

```
> C[1] = drehen( C[0], D[0], beta );
```

$$C_1 = drehen\left(C_0, \mathrm{D}_0, \beta\right) \tag{7}$$

Der Punkt C muss durch die Bewegung **(6)** und durch die Bewegung **(7)** auf die gleiche Position gebracht werden. Die Bewegungen sind über diese Bedingung verknüpft.

```
> subs( (6), (7) );
```

$$drehen\left(schieben\left(C_0, B_0 - A_0, z\right), A_0, \alpha\right) = drehen\left(C_0, \mathrm{D}_0, \beta\right) \tag{8}$$

Umschreiben in die mathematischen Formel mit dem Zusammenhang **(3)** und **(4)**.

```
> subs( schieben=Schieben,drehen=Drehen, (8) ): evalf(%): sort(%);
```

$$A_0 + \begin{bmatrix} \cos(\alpha) & -\sin(\alpha) \\ \sin(\alpha) & \cos(\alpha) \end{bmatrix} \cdot \left(-A_0 + C_0 + \frac{(-A_0 + B_0)\,z}{|A_0 - B_0|}\right) = \mathrm{D}_0 + \begin{bmatrix} \cos(\beta) & -\sin(\beta) \\ \sin(\beta) & \cos(\beta) \end{bmatrix} \cdot \left(C_0 - \mathrm{D}_0\right) \tag{9}$$

Diese Formel gibt den Zusammenhang zwischen der Verschiebung und z und den beiden Drehungen um α und β wieder. Es kann mit den Symbolen für die Punkte weiter gerechnet werden. Die Formeln werden dann lang. Übersichtlicher werden die Formen, wenn die Koordinaten der Punkte eingesetzt werden. Differenzen der Vektoren ausrechnen

```
> subs( params,abs=rcurry(LinearAlgebra[VectorNorm],2), (9) ): eval
  (%): evalf(%);
```

$$\begin{bmatrix} \cos(\alpha)\ (5.+0.55470\,z)-\sin(\alpha)\ (3.+0.83205\,z) \\ \sin(\alpha)\ (5.+0.55470\,z)+\cos(\alpha)\ (3.+0.83205\,z) \end{bmatrix} = \begin{bmatrix} 7.-2.\cos(\beta)-3.\sin(\beta) \\ -2.\sin(\beta)+3.\cos(\beta) \end{bmatrix} \tag{10}$$

Die Verschiebung z ist vorgegeben. Die Drehwinkel sind abhängig von der Verschiebung z. Also sind die Drehwinkel Funktionen von z.

```
> subs( alpha=alpha(z), beta=beta(z), (10) );
```

$$\begin{bmatrix} \cos(\alpha(z))\ (5.+0.55470\,z)-\sin(\alpha(z))\ (3.+0.83205\,z) \\ \sin(\alpha(z))\ (5.+0.55470\,z)+\cos(\alpha(z))\ (3.+0.83205\,z) \end{bmatrix} = \begin{bmatrix} 7.-2.\cos(\beta(z))-3.\sin(\beta(z)) \\ -2.\sin(\beta(z))+3.\cos(\beta(z)) \end{bmatrix} \tag{11}$$

Die Vektoren sind genau dann gleich, wenn jede Komponente gleich ist. Die eine Gleichung mit den Vektoren in zwei getrennte Gleichungen zerlegen.

```
> lhs((11))[1]=rhs((11))[1];
```

$$\cos(\alpha(z))\ (5.+0.55470\,z)-\sin(\alpha(z))\ (3.+0.83205\,z)=7.-2.\cos(\beta(z))-3.\sin(\beta(z)) \tag{12}$$

```
> lhs((11))[2]=rhs((11))[2];
```

$$\sin(\alpha(z))\ (5.+0.55470\,z)+\cos(\alpha(z))\ (3.+0.83205\,z)=-2.\sin(\beta(z))+3.\cos(\beta(z)) \tag{13}$$

Gesucht ist nach dem Einfluss einer infinitesimalen Verschiebung dz auf die beiden Winkel. Dieser Einfluss ist die Ableitung dα/dz bzw. dβ/dz. Also die beiden Gleichung **(12)** und **(13)** nach z differenzieren.

```
> diff((12),z);
```

$$-\sin(\alpha(z))\ \frac{d}{dz}\alpha(z)\ (5.+0.55470\,z)+0.55470\cos(\alpha(z))-\cos(\alpha(z))\ \frac{d}{dz}\alpha(z)\ (3.+0.83205\,z)-0.83205\sin(\alpha(z))=2.\sin(\beta(z))\ \frac{d}{dz}\beta(z)-3.\cos(\beta(z))\ \frac{d}{dz}\beta(z) \tag{14}$$

```
> diff((13),z);
```

$$\cos(\alpha(z))\ \frac{d}{dz}\alpha(z)\ (5.+0.55470\,z)+0.55470\sin(\alpha(z))-\sin(\alpha(z))\ \frac{d}{dz}\alpha(z)\ (3.+0.83205\,z)+0.83205\cos(\alpha(z))=-2.\cos(\beta(z))\ \frac{d}{dz}\beta(z)-3.\sin(\beta(z))\ \frac{d}{dz}\beta(z) \tag{15}$$

Neben den gesuchten Ableitungen stehen noch Winkel α, β und Verschiebung z in den Gleichungen. Gesucht wird der Einfluss dα/dz und dβ/dz in der skizzierten Anfangsstellung, also bei $\alpha=\alpha_0$, $\beta=\beta_0$ und $z=z_0$.

```
> subs(diff(alpha(z),z)=da,diff(beta(z),z)=db,alpha(z)=alpha[0],
  beta(z)=beta[0],z=z[0],da=diff(alpha(z),z),db=diff(beta(z),z),
  (14));
```

$$-\sin(\alpha_0)\ \frac{d}{dz}\alpha(z)\ (5.+0.55470\,z_0)+0.55470\cos(\alpha_0)-\cos(\alpha_0)\ \frac{d}{dz}\alpha(z)\ (3.+0.83205\,z_0)-0.83205\sin(\alpha_0)=2.\sin(\beta_0)\ \frac{d}{dz}\beta(z)-3.\cos(\beta_0)\ \frac{d}{dz}\beta(z) \tag{16}$$

```
> subs(diff(alpha(z),z)=da,diff(beta(z),z)=db,alpha(z)=alpha[0],
  beta(z)=beta[0],z=z[0],da=diff(alpha(z),z),db=diff(beta(z),z),
  (15));
```

$$\cos(\alpha_0)\left(\frac{d}{dz}\alpha(z)\right)(5.+0.55470\,z_0)+0.55470\sin(\alpha_0)-\sin(\alpha_0)\left(\frac{d}{dz}\alpha(z)\right)(3.+0.83205\,z_0)+0.83205\cos(\alpha_0)=-2.\cos(\beta_0)\left(\frac{d}{dz}\beta(z)\right)-3.\sin(\beta_0)\left(\frac{d}{dz}\beta(z)\right) \quad (17)$$

Diese Startwert sind alle 0, siehe **(2)**. Die Werte in die beiden Gleichungen eingesetzt und weiter ausgerechnet.

```
> subs(params2,(16)): evalf(%);
```

$$0.55470-3.\,\frac{d}{dz}\alpha(z)=-3.\,\frac{d}{dz}\beta(z) \quad (18)$$

```
> subs(params2,(17)): evalf(%);
```

$$5.\,\frac{d}{dz}\alpha(z)+0.83205=-2.\,\frac{d}{dz}\beta(z) \quad (19)$$

Die beiden Gleichungen **(18)** und **(19)** bilden ein lineares Gleichungssystem für die gesuchten Ableitungen. Das Gleichungssystem gelöst:

```
> solve( [(18),(19)],{diff(alpha(z),z),diff(beta(z),z)} )[];
```

$$\frac{d}{dz}\alpha(z)=-0.066036,\ \frac{d}{dz}\beta(z)=-0.25094 \quad (20)$$

Das negative Vorzeichen bedeutet, dass sich die Scheibe nach rechts dreht, wenn das Linearlager nach oben geschoben wird.
Beispiel: Wird um dz=1 nach oben geschoben, so dreht sich die Scheibe 1 um 0,066 Radiant nach rechts und die Scheibe 2 dreht sich um 0,251 Radiant nach rechts.

Erweiterung

Die Skizze kann nur für eine Situation gemacht werden, hier für z=0. Mit der aufgestellten Gleichung für die Bewegung des Tragwerks **(11)** bzw. **(12)**+**(13)** kann die Bewegung für z>0 untersucht werden.

Für ein gegebenes z bilden die beiden Gleichungen **(12)** und **(13)** ein Gleichungssystem für die Drehwinkel α und β. Die Lösungen können mit Maple berechent und dargestellt werden:

```
> zValues := [seq(i/20.0,i=0..100)]:
> alphaValues := [0.0]:
> betaValues := [0.0]:
> for i from 2 to 101 do
     ret := fsolve(subs(alpha(z)=alpha,beta(z)=beta,z=zValues[i],{
  (12),(13)}),{alpha=alphaValues[i-1],beta=betaValues[i-1]});
     alphaValues:=[op(alphaValues),subs(ret,alpha)];
     betaValues:=[op(betaValues),subs(ret,beta)];
  end do:
> aPlot := plot( zValues, alphaValues, colour=green, legend=typeset
  (alpha(z)) ):
> bPlot := plot( zValues, betaValues, legend=typeset(beta(z)) ):
> plots[display]( aPlot, bPlot, labels=["z Verschiebung
  [Längeneinheit]","Winkel [Radiant]"], axes=boxed, legendstyle=
```

```
[location=right], labeldirections=["horizontal","vertical"] );
```



Die Ableitungen dα/dz und dβ/dz können mit dem Gleichungssystem **(14)** und **(15)** für jede Kombination (z,α,β) berechnet werden. Die Lösungen:

```
> daValues := []:
> dbValues := []:
> for i from 1 to 101 do
     ret := solve(subs(diff(alpha(z),z)=da,diff(beta(z),z)=db,alpha
  (z)=alphaValues[i],beta(z)=betaValues[i],z=zValues[i],{(14),(15)}),
  {da,db});
     daValues:=[op(daValues),subs(ret,da)];
     dbValues:=[op(dbValues),subs(ret,db)];
  end do:
> aPlot := plot( zValues, daValues, colour=green, legend=typeset
  (diff(alpha(z),z)) ):
> bPlot := plot( zValues, dbValues, legend=typeset(diff(beta(z),z))
  ):
> plots[display]( aPlot, bPlot, labels=["z Verschiebung
  [Längeneinheit]","Einfluss [Radiant/Längeneinheit]"], axes=boxed,
  legendstyle=[location=right], labeldirections=["horizontal",
  "vertical"], view=[0..5,0..-2.5] );
```



0
-0.5
-1
-1.5
-2
-2.5
Einfluss [Radiant/Längeneinheit]
0
1
2
3
4
5
z Verschiebung [Längeneinheit]
$\frac{d}{dz}\,\alpha(z)$
$\frac{d}{dz}\,\beta(z)$