```
> restart;
> Digits := trunc( evalhf(Digits) ):
  interface( displayprecision=7):
```

Aufgabe

Eine Masse (M=5kg) kann sich frei von Reibung an einer Stange auf und ab bewegen. Zum Zeitpunkt t=0 ist Y_0=1m und die Geschwindigkeit V_0=-1m/s (Masse bewegt sich senkrecht nach unten).
Fall A: Die Feder ist am Seil, mit einer Federkonstanten D=300N/m. Auf Y=0 beträgt die Federkraft 300N.
Fall B: Am Seil wird mit einer konstanten Kraft F gezogen. F ist so groß, dass sich etwa die gleiche Amplitude wie im Fall A ergibt.
Die Bewegungsgleichung soll numerisch gelöst werden.



Das Seil ist über eine Umlenkrolle geführt. Die Umlenkrolle hat den Abstand L=1m von der Stange.
Die Parameter:

```
> M=5*Unit(kg),L=1*Unit(m),D=300*Unit(N/m),Y[0]=1*Unit(m),V[0]=
  -1*Unit(m/s),F[0]=300*Unit(N);
```

$$M=5\ [\![kg]\!],\ L=[\![m]\!],\ \mathrm{D}=300\ \left[\!\!\left[\frac{kg}{s^2}\right]\!\!\right],\ Y_0=[\![m]\!],\ V_0=-\left[\!\!\left[\frac{m}{s}\right]\!\!\right],\ F_0=300\ [\![N]\!] \quad \textbf{(1)}$$

Fall A

Vereinfachen, indem alle Effekte, die in der Aufgabe nicht explizit genannt, vernachlässigt werden:
- Masse des Seils vernachlässigen.
- Masse der Feder vernachlässigen.
- Größe der Masse m vernachlässigen; einen Massenpunkt betrachten.
- Masse und Radius der Umlenkrolle vernachlässigen; das Seil als Gerade von einem festen Punkt (Mitte der Umlenkrolle) zum Massenpunkt betrachten.
- Keine Reibungseffekte (Luftwiderstand, innere Reibung in der Feder, Reibung an der Rolle).
- Dehnung des Seils vernachlässigen; konstante Seillänge annehmen.
- Für die Feder das Hookesche Gesetz unabhängig von der Dehnung annehmen.

Aufstellen der Bewegungsgleichung über Betrachtung der Kräfte.
Auf die Masse wirkt das Gewicht. Das Gewicht wirkt gegen die y-Richtung nach unten. Die y-

Komponente des Gewichts ist

```
> F[g,y] = -M*g;
```

$$F_{g,y} = -M g \tag{2}$$

Dabei ist g die Fallbeschleunigung.

```
> g = 9.81*Unit(m/s^2);
```

$$g = 9.81 \left[\left[\frac{m}{s^2}\right]\right] \tag{3}$$

An der Masse zieht das Seil mit der Kraft F_s.

Die y-Komponente dieser Kraft kann aus der Skizze abgelesen werden. Seil, gestichelte Linie und oberer Teil der Stange bilden ein rechtwinkliges Dreieck. Dieses Dreieck ist ähnlich dem Kräftedreieck.

```
> F[s,y] = - F[s]*y/sqrt(L^2+y^2);
```

$$F_{s,y} = -\frac{F_s y}{\sqrt{L^2+y^2}} \tag{4}$$

Weil der Körper reibungslos entlang der Stange gleitet, wird jede Kraftkomponente senkrecht zur Stange durch eine entgegengesetzte Reaktionskraft der Stange ausgeglichen. (Zwangsbedingung, Zwangskräfte)
Also kann von der Kraft des Seils auf die Masse nur die Komponente parallel zur Stange betrachtet werden. Die beiden von außen wirkenden Kraftkomponenten wurden aufgestellt.
d'Alembertsches Prinzip mit der Trägheitskraft aus der Beschleunigung a der Masse m:

```
> F[g,y] + F[s,y] = M*a;
```

$$F_{g,y} + F_{s,y} = M a \tag{5}$$

```
> subs( (2),(4), (5) );
```

$$-M g - \frac{F_s y}{\sqrt{L^2+y^2}} = M a \tag{6}$$

Die Kraft F_s mit der das Seil an der Masse zieht ist gleich der Kraft der Feder, weil idealisiert Seil und Rolle nur die Kraft umlenken.
Die Ausdehnung der Feder wird bestimmt durch die Strecke um die das Seil gezogen wird.
Nullpunktwahl für die gezogene Strecke: die Position der Umlenkrolle.
Ist die Masse auf Position y wird das Seil um die Strecke l gezogen.

```
> l = sqrt( L^2 + y^2 );
```

$$l = \sqrt{L^2+y^2} \tag{7}$$

Die Kraft der Feder ist proportional zu diese Länge (Hookesches Gesetz).

```
> F[s] = D*l + N;
```

$$F_s = \mathrm{D}\, l + N \tag{8}$$

```
> subs( (7), (8) );
```

$$F_s = \mathrm{D} \sqrt{L^2+y^2} + N \tag{9}$$

Dabei ist D die Federkonstante und N die Kraft bei l=0. Gegeben ist die Kraft F_0 bei y=0.

Die Umrechung:

```
> subs( y=0, F[s]=F[0], (9) ): simplify(%) assuming L>0;
```

$$F_0 = \mathrm{D}\, L + N \tag{10}$$

```
> isolate((10),N);
```

$$N = F_0 - \mathrm{D}\, L \tag{11}$$

Alle Größen auf der rechten Seite sind gegeben. Damit ist die Größe N ebenfalls eine bekannte Konstante.
Die Zahlenwerte **(1)** der Aufgabe einsetzen liefert den einfachen Wert:

```
> eval((11),{(1)}): simplify(%);
```

$$N = 0 \tag{12}$$

Federkraft **(9)** und Kräftegleichgewicht **(6)** einsetzen.

```
> subs( (9), (6) ); expand(%); %/M: expand(%);
```

$$-M g - \frac{\left(\mathrm{D}\sqrt{L^2+y^2} + N\right) y}{\sqrt{L^2+y^2}} = M a$$

$$-M g - y\,\mathrm{D} - \frac{y N}{\sqrt{L^2+y^2}} = M a$$

$$-g - \frac{y\,\mathrm{D}}{M} - \frac{y N}{M\sqrt{L^2+y^2}} = a \tag{13}$$

Die Beschleunigung a ist die die zweite Ableitung des Orts y nach der Zeit.

```
> subs( a=diff(y(t),t,t), subs(y=y(t),(13)) );
```

$$-g - \frac{y(t)\,\mathrm{D}}{M} - \frac{y(t)\,N}{M\sqrt{L^2+y(t)^2}} = \ddot{y}(t) \tag{14}$$

Damit ist die Differenzialgleichung für den Ort als Funktion der Zeit y(t) aufgestellt.
Eine Möglichkeit diese Gleichung analytisch zu lösen hat Maple nicht gefunden.
Doch bei den Zahlenwerte der Aufgabe führen auf **(12)** und damit auf

```
> subs( (12), (14) );
```

$$-g - \frac{y(t)\,\mathrm{D}}{M} = \ddot{y}(t) \tag{15}$$

Der Schwierige Ausdruck mit der Wurzelfunktion fällt aufgrund der gewählten Parameter bei dieser Aufgabe weg.
Diese spezielle Gleichung ist die Differenzialgleichung für eine harmonische Schwingung. Die Lösung ist

```
> dsolve((15)): subs(_C1=K[1],_C2=K[2],%): sort(%);
```

$$y(t) = K_1 \cos\left(\frac{\sqrt{\mathrm{D}}\,t}{\sqrt{M}}\right) + K_2 \sin\left(\frac{\sqrt{\mathrm{D}}\,t}{\sqrt{M}}\right) - \frac{M g}{\mathrm{D}} \tag{16}$$

K_1 und K_2 sind Konstanten, die aus den Anfangswerten bestimmt werden.
Die Schwingung erfolgt um den y-Wert:

```
> subs(sin=0,cos=0,rhs((16))): %=subs((1),(3),%): simplify(%);
```

$$-\frac{M g}{\mathrm{D}} = -0.1635000 \ [\![m]\!] \tag{17}$$

Die Kreisfrequenz der Schwingung ist

```
> omega = sqrt(D/M); lhs(%)=evalf(eval(rhs(%),{(1)})): simplify(%)
  ;
```

$$\omega = \sqrt{\frac{\mathrm{D}}{M}}$$

$$\omega = 7.745967 \left[\!\!\left[\frac{1}{s} \right]\!\!\right] \tag{18}$$

Die Periodendauer der Schwingung ist

```
> omega*T=2*Pi: isolate(%,T); evalf(eval(%,(18))): simplify(%);
```

$$T=\frac{2\,\pi}{\omega}$$

$$T=0.8111557\ [\![s]\!] \tag{19}$$

Die gegebene Anfangsposition Y_0 liefert die Konstante K_1.

```
> y(0)=Y[0];
```

$$y(0)=Y_0 \tag{20}$$

```
> subs( t=0, (16) ); subs((20),%); isolate(%,K[1]);
```

$$y(0)=K_1\cos(0)+K_2\sin(0)-\frac{M\,g}{\mathrm{D}}$$

$$Y_0=K_1-\frac{M\,g}{\mathrm{D}}$$

$$K_1=Y_0+\frac{M\,g}{\mathrm{D}} \tag{21}$$

Die Zahlenwerte einsetzen.

```
> eval((21),{(1),(3)}): simplify(%);
```

$$K_1=1.163500\ [\![m]\!] \tag{22}$$

Die gegebene Anfangsgeschwindigkeit V_0 liefert die Konstante K_2.

```
> v(t) = diff(y(t),t);
```

$$v(t)=\dot{y}(t) \tag{23}$$

```
> subs((16),(23)); simplify(%): expand(%);
```

$$v(t)=\frac{\partial}{\partial t}\left(K_1\cos\left(\frac{\sqrt{\mathrm{D}}\ t}{\sqrt{M}}\right)+K_2\sin\left(\frac{\sqrt{\mathrm{D}}\ t}{\sqrt{M}}\right)-\frac{M\,g}{\mathrm{D}}\right)$$

$$v(t)=-\frac{K_1\sin\left(\frac{\sqrt{\mathrm{D}}\ t}{\sqrt{M}}\right)\sqrt{\mathrm{D}}}{\sqrt{M}}+\frac{K_2\cos\left(\frac{\sqrt{\mathrm{D}}\ t}{\sqrt{M}}\right)\sqrt{\mathrm{D}}}{\sqrt{M}} \tag{24}$$

```
> v(0)=V[0];
```

$$v(0)=V_0 \tag{25}$$

```
> subs(t=0,(24)); subs((25),%); isolate(%,K[2]);
```

$$v(0)=-\frac{K_1\sin(0)\sqrt{\mathrm{D}}}{\sqrt{M}}+\frac{K_2\cos(0)\sqrt{\mathrm{D}}}{\sqrt{M}}$$

$$V_0=\frac{K_2\sqrt{\mathrm{D}}}{\sqrt{M}}$$

$$K_2=\frac{V_0\sqrt{M}}{\sqrt{\mathrm{D}}} \tag{26}$$

Die Zahlenwerte einsetzen.

```
> evalf(eval((26),{(1)})): simplify(%);
```

$$K_2=-0.1290994\ [\![m]\!] \tag{27}$$

Alle Zahlenwert zusammentragen für die vollständig ausgerechnete Funktion.

```
> subs( sqrt(D)*t/sqrt(M)=omega*t, (16) );
```

$$y(t) = K_1 \cos(\omega t) + K_2 \sin(\omega t) - \frac{M g}{D} \tag{28}$$

```
> subs( (17),(18),(22),(27), (28) ):
  subs(y(t)=y*Unit(m),t=t*Unit(s),%)/Unit(m):
  simplify(%);
```

$$y = 1.163500 \cos(7.745967\, t) - 0.1290994 \sin(7.745967\, t) - 0.1635000 \tag{29}$$

Die Schwingung:

```
> plot(rhs((29)),t=0..simplify(2.5*rhs((19))/Unit(s)), labels=
  ["t/s","y/m"] );
```



Fall B

Alle Überlegungen bis zur Gleichung **(6)** können aus dem Fall A übernommen werden:

```
> (6);
```

$$-M g - \frac{F_s y}{\sqrt{L^2 + y^2}} = M a \tag{30}$$

Hier ist die Kraft F_s konstant. Der Zahlenwert ist nicht gegeben, er muss später ermittelt werden.

```
> F[s] = F;
```

$$F_s = F \quad \textbf{(31)}$$

```
> subs( (31), (30) );
```

$$-M g - \frac{F y}{\sqrt{L^2 + y^2}} = M a \quad \textbf{(32)}$$

Die Beschleunigung a ist die zweite Ableitung nach der Zeit t.

```
> subs( y=y(t), a=diff(y(t),t,t), (32) ); %/M: expand(%);
```

$$-M g - \frac{F y(t)}{\sqrt{L^2 + y(t)^2}} = \ddot{y}(t)\, M$$

$$-g - \frac{F y(t)}{M \sqrt{L^2 + y(t)^2}} = \ddot{y}(t) \quad \textbf{(33)}$$

Diese ist die zu lösenden Bewegungsgleichung im Fall B.

Zur Vorbereitung der allgemeinen Lösung:
Ein Sonderfall der Bewegung: die Ruhelage, a=0 in Gleichung **(32)** einsetzen.

```
> subs(a=0,(32));
```

$$-M g - \frac{F y}{\sqrt{L^2 + y^2}} = 0 \quad \textbf{(34)}$$

Das ist eine quadratische Gleichung für y. Die beiden Lösungen:

```
> solve((34),{y});
```

$$\left\{y = -\frac{M g L}{\sqrt{-g^2 M^2 + F^2}}\right\}, \left\{y = \frac{M g L}{\sqrt{-g^2 M^2 + F^2}}\right\} \quad \textbf{(35)}$$

Die Lösung y>0 verlangt eine negative Kraft F. Weil die Kraft durch ein Seil übertragen wird, ist das bei dieser Aufgabe nicht möglich. Es bleibt die Lösung mit y<0.

```
> (35)[1][1];
```

$$y = -\frac{M g L}{\sqrt{-g^2 M^2 + F^2}} \quad \textbf{(36)}$$

Eine Ruhelage gibt es nur, wenn F > Mg ist. Das ist ebenfalls eine Bedingung für die Schwingung, weil die Schwingung um die Ruhelage herum ausgeführt wird.

Weil die Gleichung **(33)** nicht analytisch lösbar ist, eine Vereinfachung: Linearisierung für $y \ll L$.
Die Taylorentwicklung der Wurzelfunktion:

```
> sqrt(L^2+y^2): % = taylor( %, y, 2 ): simplify(%) assuming L>0;
```

$$\sqrt{L^2 + y^2} = L + \mathrm{O}(y^2) \quad \textbf{(37)}$$

Diese Näherung in die **(33)** eingesetzt:

```
> subs(y(t)^2=0,(33)): simplify(%) assuming L>0: expand(%);
```

$$-g - \frac{F y(t)}{M L} = \ddot{y}(t) \quad \textbf{(38)}$$

Die Linearisierung führt auf die Bewegungsgleichung einer harmonischen Schwingung. Die Lösung ist bekannt.

```
> dsolve((38)): subs(_C1=K[1],_C2=K[2],%): sort(%);
```

$$y(t) = -\frac{L M g}{F} + K_1 \cos\left(\frac{\sqrt{F}\, t}{\sqrt{L}\,\sqrt{M}}\right) + K_2 \sin\left(\frac{\sqrt{F}\, t}{\sqrt{L}\,\sqrt{M}}\right) \quad \textbf{(39)}$$

Der konstante Summand in **(39)** zeigt im Vergleich mit der exakten Ruhelage **(36)**: Diese Näherung gilt für eine Kraft F $\gg$ Mg. Mit dieser Kraft ist die Ruhelage y 0 und bei einer kleinen Auslenkung gilt bei der Schwingung y $\ll$L.
Aus den Winkelfunktionen kann die Kreisfrequenz der Schwingung abgelesen werden:

```
> omega = sqrt(F/(L*M));
```

$$\omega = \sqrt{\frac{F}{M L}} \tag{40}$$

Für diese Aufgabe ist die Linearisierung nicht brauchbar, weil die Anfangsposition Y_s = 1m = L gegeben ist.

Die Gleichung **(33)** muss also numerisch gelöst werden.
Ein Zahlenwert für die Kraft F fehlt. Setzen (Ausprobiert, bis die Amplitude im Plot zum Teil A passt.):

```
> F = 460*Unit(N);
```

$$F = 460 \, [\![N]\!] \tag{41}$$

Alle Zahlenwerte in **(33)** einsetzen.
Die Gleichung von Einheiten befreien. y soll in Meter und die Zeit t in Sekunden eingesetzt werden.

```
> subs( (1),(3),(41),(33)):
  subsindets(%, has_unit, proc (z) options operator, arrow;
  convert(z, unit_free) end proc);
```

$$-9.81 - \frac{92\, y(t)}{\sqrt{1 + y(t)^2}} = \ddot{y}(t) \tag{42}$$

Das ist eine Differenzialgleichung zweiter Ordnung. Für eine numerische Lösung wird die Gleichung umgewandelt in ein System von zwei Differenzialgleichungen erster Ordnung.
Dazu die Geschwindigkeit einführen.

```
> v(t) = diff(y(t),t);
```

$$v(t) = \dot{y}(t) \tag{43}$$

```
> subs( isolate(diff((43),t),diff(y(t),t,t)), (42) );
```

$$-9.81 - \frac{92\, y(t)}{\sqrt{1 + y(t)^2}} = \dot{v}(t) \tag{44}$$

Die beiden Gleichungen **(43)** und **(44)** bilden ein Differenzialgleichungssystem erster Ordnung für Ort y(t) und Geschwindigkeit v(t).
Die Anfangswerte sind gegeben

```
{ y(0) = Y[0], v(0) = V[0] };
```

$$\{v(0) = V_0, y(0) = Y_0\} \tag{45}$$

```
> subs( (1), (45) ): map(z->lhs(z)=convert(rhs(z),unit_free),%);
```

$$\{v(0) = -1, y(0) = 1\} \tag{46}$$

Das Programm Maple berechnet eine numerische Lösung über den Befehl dsolve. Die Standardlösungsmethode ist Runge-Kutta-Fehlberg der Ordnung 4-5. Das Programm liefert:

```
> f := dsolve( {(43),(44)} union (46), numeric, range=0..2 ):
> plots[odeplot](f,[t,y(t)]);
```



Hilfsmittel

1. Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch
2. Quarteroni, Sacco, Saleri: Numerische Mathematik 2, Springer Verlag
3. Scherf: Modellbildung und Simulation dynamischer Systeme, Oldenbourg Wissenschaftsverlag
4. Maple 17, www.maplesoft.com