```
> restart;
> Digits := 30: interface( displayprecision=5 ):
> with(LinearAlgebra):
```

Aufgabe

Gesucht ist die Gleichung einer Ellipse. Gegeben sind vier Punkte [x-Koordinate, y-Koordinate im kartesischen Koordinatensystem] auf der Ellipse. (Punkte gegen den Uhrzeigersinn sortiert.)

```
> P = [ <1.73,-1.00>,<0.50,0.87>,<-1.73,1.00>,<-0.50,-0.87>];
  Pk := rhs(%):
```

$$P = \left[\begin{bmatrix} 1.73 \\ -1.00 \end{bmatrix}, \begin{bmatrix} 0.50 \\ 0.87 \end{bmatrix}, \begin{bmatrix} -1.73 \\ 1.00 \end{bmatrix}, \begin{bmatrix} -0.50 \\ -0.87 \end{bmatrix} \right] \tag{1}$$

Lösungsweg

Die Koordinaten der Punkte sind auffällig: Die Koordinaten der Punktpaare P_1, P_2 und P_3, P_4 unterschieden sich nur durch Vorzeichen. Im Diagramm wird die besondere Eigenschaft der Punkte sichtbar, wenn auch die Verbindungslinien P_i-Ursprung eingezeichnet werden.

```
> p := map( t -> plot( [[0,0],[t[1],t[2]]], color=blue ), Pk ):
  p := [ op(p), plot( map(t->[t[1],t[2]],Pk), style=point, symbol=
  asterisk, symbolsize=16, scaling=constrained ) ]:
> plots[display]( p );
```



Aus der Zeichnung folgt die Vermutung: Die 4 Verbindungslinien P_i-Ursprung schließen immer genau ein Vielfaches von 90° ein. Dieses muss noch nachgerechnet werden.
Die Winkel über das Skalarprodukt berechnen.

```
> cos(alpha[i,i+1])=P[i].P[i+1]/(abs(P[i])*abs(P[i+1]));
```

$$\cos(\alpha_{i,\,i+1}) = \frac{P_i \cdot P_{i+1}}{|P_i|\,|P_{i+1}|} \tag{2}$$

```
> for i from 1 to 4 do cos(alpha[i,1+(i mod 4)])=Pk[i].Pk[1+(i mod
  4)]/(Norm(Pk[i],2)*Norm(Pk[1+(i mod 4)],2)); end do;
```

$$\cos(\alpha_{1,\,2}) = -0.0024936$$

$$\cos(\alpha_{2,\,3}) = 0.0024936$$

$$\cos(\alpha_{3,\,4}) = -0.0024936$$

$$\cos(\alpha_{4,\,1}) = 0.0024936 \tag{3}$$

Die Winkel sind alle nahe 90°. Die Koordinaten sind auf 3 Stellen gegeben, daher sind die Winkel nicht exakt 90°. Die Längen der Verbindungslinien sind paarweise gleich, weil nur die Vorzeichen anders sind.
Also sind die betrachteten Verbindungslinien P_i-Ursprung Achsen der Ellipse. Die gefundenen Achsen

müssen nicht die Hauptachsen sein. Die Aufgabe ist nicht eindeutig lösbar. Es gibt eine Schar von Ellipsen, die durch die 4 Punkte gegeben sind.

(a) Ansatz: Die betrachteten Verbindungslinien P_i-Ursprung sind die Hauptachsen der Ellipse.

Damit wird die kleinste Ellipse berechnet, die durch die gegebenen Punkte führt.

Eine Ellipse kann als transformierer Einheitskreis betrachtet werden. In diesem Fall kann die Koordinatentransformation fast direkt abgelesen werden.

Die Basisvektoren des Koordinatensystems´ sind die Halbachsen der Ellipse.

```
> e´[x] = Pk[1];
```

$$e'_x = \begin{bmatrix} 1.73 \\ -1.00 \end{bmatrix} \tag{4}$$

```
> e´[y] = Pk[2];
```

$$e'_y = \begin{bmatrix} 0.50 \\ 0.87 \end{bmatrix} \tag{5}$$

Aus dem Koordinatensystem´ in das Koordinatensystem transformieren.

```
> <x,y> = e´[x] * x´ + e´[y] * y´;
```

$$\begin{bmatrix} x \\ y \end{bmatrix} = e'_x x' + e'_y y' \tag{6}$$

```
> <x,y> = <Pk[1]|Pk[2]> . <x´,y´>;
```

$$\begin{bmatrix} x \\ y \end{bmatrix} = \begin{bmatrix} 1.73\, x' + 0.50\, y' \\ -x' + 0.87\, y' \end{bmatrix} \tag{7}$$

Die Transformationsmatrix

```
> T = <Pk[1]|Pk[2]>;
```

$$T = \begin{bmatrix} 1.73 & 0.50 \\ -1.00 & 0.87 \end{bmatrix} \tag{8}$$

Die Inverse der Transformationsmatrix

```
> T´ = <Pk[1]|Pk[2]>^(-1);
```

$$T' = \begin{bmatrix} 0.43389 & -0.24936 \\ 0.49873 & 0.86280 \end{bmatrix} \tag{9}$$

Aus dem Koordinatensystem ins Koordinatensystem´ transformieren.

```
> <x´,y´> = T´. <x,y>;
```

$$\begin{bmatrix} x' \\ y' \end{bmatrix} = T' \cdot \begin{bmatrix} x \\ y \end{bmatrix} \tag{10}$$

```
> subs( (9), (10) ); simplify(%);
```

$$\begin{bmatrix} x' \\ y' \end{bmatrix} = \begin{bmatrix} 0.43389 & -0.24936 \\ 0.49873 & 0.86280 \end{bmatrix} \cdot \begin{bmatrix} x \\ y \end{bmatrix}$$

(11)

$$\begin{bmatrix} x' \\ y' \end{bmatrix} = \begin{bmatrix} 0.43389\,x - 0.24936\,y \\ 0.49873\,x + 0.86280\,y \end{bmatrix} \quad \textbf{(11)}$$

Der Einheitskreis im Koordinatensystem´

```
> x´^2 + y´^2 = 1;
```

$$x'^2 + y'^2 = 1 \quad \textbf{(12)}$$

ist die gesuchte Ellipse im Koordinatensystem.

```
> subs( lhs((11)) [1]=rhs((11)) [1], lhs((11)) [2]=rhs((11)) [2], (12) );
```

$$(0.43389\,x - 0.24936\,y)^2 + (0.49873\,x + 0.86280\,y)^2 = 1 \quad \textbf{(13)}$$

Zur Kontrolle die berechnete Ellipse **(13)** und die gegebenen Punkten P_i plotten.

```
> pi := plots[implicitplot]( (13), x=-1.9..+1.9,y=-1.5..+1.5, color=
  green, gridrefine=1 ):
> plots[display](p,pi);
```



(b) Ansatz: Eine Ellipse mit Parameter für die Länge und Richtung der Hauptachsen.

Die impliziete Gleichung der Ellipse

```
> a*(cos(alpha)*x+sin(alpha)*y)^2 + b*(-sin(alpha)*x+cos(alpha)*y)
```

```
^2 = 1;
```

$$a\,(\cos(\alpha)\,x+\sin(\alpha)\,y)^2+b\,(-\sin(\alpha)\,x+\cos(\alpha)\,y)^2=1 \quad \mathbf{(14)}$$

Die für eine Ellipse konstanten a, b, α sind aus den gegebenen Punkten zu berechnen.
Die vier Punkte in die Gleichung **(14)** einsetzen.

P_1:

```
> subs( x=Pk[1][1], y=Pk[1][2], (14) );
```

$$a\,(1.73\cos(\alpha)-\sin(\alpha))^2+b\,(-1.73\sin(\alpha)-\cos(\alpha))^2=1 \quad \mathbf{(15)}$$

P_2:

```
> subs( x=Pk[2][1], y=Pk[2][2], (14) );
```

$$a\,(0.50\cos(\alpha)+0.87\sin(\alpha))^2+b\,(-0.50\sin(\alpha)+0.87\cos(\alpha))^2=1 \quad \mathbf{(16)}$$

P_3:

```
> subs( x=Pk[3][1], y=Pk[3][2], (14) );
```

$$a\,(-1.73\cos(\alpha)+\sin(\alpha))^2+b\,(1.73\sin(\alpha)+\cos(\alpha))^2=1 \quad \mathbf{(17)}$$

P_4:

```
> subs( x=Pk[4][1], y=Pk[4][2], (14) );
```

$$a\,(-0.50\cos(\alpha)-0.87\sin(\alpha))^2+b\,(0.50\sin(\alpha)-0.87\cos(\alpha))^2=1 \quad \mathbf{(18)}$$

Das Gleichungssystem hat eine Lösungsschar. Die Lösungen mit α als Parameter aufgeschrieben.

```
> solve( [(15),(16),(17),(18)], {a,b} );
```

$$\Big\{a = -\big(0.49873\,(2431.\cos(\alpha)^2+43300.\cos(\alpha)\sin(\alpha)+27429.\sin(\alpha)^2)\big)\Big/\big(-34700.\cos(\alpha)^3\sin(\alpha)+10051.\cos(\alpha)^4-34700.\cos(\alpha)\sin(\alpha)^3-10051.\sin(\alpha)^4\big),\ b = \big(0.49873\,(27429.\cos(\alpha)^2-43300.\cos(\alpha)\sin(\alpha)+2431.\sin(\alpha)^2)\big)\Big/\big(-34700.\cos(\alpha)^3\sin(\alpha)+10051.\cos(\alpha)^4-34700.\cos(\alpha)\sin(\alpha)^3-10051.\sin(\alpha)^4\big)\Big\} \quad \mathbf{(19)}$$

Eine Lösung zum Winkel α=0,8 rad ist

```
> alpha=0.8: evalf( subs(%,(19) ) ) union {%};
```

$$\{a=1.0445,\ b=0.20009,\ \alpha=0.8\} \quad \mathbf{(20)}$$

Diese Lösung in die Gleichung **(14)** der Ellipse eingesetzt.

```
> subs( (20), (14) ); simplify(%);
```

$$1.0445\,(\cos(0.8)\,x+\sin(0.8)\,y)^2+0.20009\,(-\sin(0.8)\,x+\cos(0.8)\,y)^2=1$$

$$0.60997\,x^2+0.84405\,x\,y+0.63462\,y^2=1 \quad \mathbf{(21)}$$

Zur Kontrolle der Plot der berechneten Ellipse **(21)** und der gegebenen Punkte P_i.

```
> pi := plots[implicitplot]( (21), x=-1.8..+1.8,y=-1.8..+1.8, color=
```

```
  green, gridrefine=1 ):
> plots[display](p,pi);
```