```
> restart;
> interface( imaginaryunit='junit' ):
```

Aufgabe

Ein schubsteifer Balken der Länge L ruht auf zwei Lagern und hat eine konstante Streckenlast q.



- Die Skizze in der Aufgabenstellung ist mit der Festlegung eines Koordinatensystems in der Mitte des Balkens ergänzt.
- Die Biegesteifigkeit EI des Balkens ist konstant.
- Die Aufgabe ist mit der linearen Theorie schubsteifer Balken zubearbeiten.

Wie lange müssen bei gegebener Länge L die Abschnitte L_1 und L_2 sein, damit die maximale Durchbiegung möglichst klein wird?

Abkürzungen

Die Längen L_s, L_2, L und die Positionen L/2, L_2/2 werden in der Rechnung oft verwendet.

Definition: S für die halbe Länge.

```
> S = L/2;
```

$$S = \frac{L}{2} \tag{1}$$

Definition: r für den Anteil der Länge zwischen den Lagern.

```
> L[2] = r*L;
```

$$L_2 = r\,L \tag{2}$$

Bestimmung der Lagerreaktionen

Die Auflager wirken mit der Kraft F_1 und F_2 auf den Balken. Beide Auflager können nur in Z-Richtung wirken.

Der Balken ruht. Die Summe der angreifenden Kräfte ist 0 und die Summe der angreifenden Drehmomente ist 0.

Alle Kräfte haben nur eine Z-Komponente. Z-Komponente der Kräftegleichung.

Die Streckenlast q hat ein negatives Vorzeichen, weil die Last in minus Z-Richtung wirkt.

```
> -q*2*S + F[1] + F[2] = 0;
```

$$-2\,q\,S + F_1 + F_2 = 0 \tag{3}$$

Drehmomentgleichung, Drehmomente bezogen auf Nullpunkt des Koordinatensystems.

```
> eval( int(-q*zeta,zeta=-S..S) + F[2] * (-r*S) + F[1] * (r*S) =
  0, 1 );
```

$$\int_{-S}^{S} -q\,\zeta\,\mathrm{d}\zeta - F_2\,r\,S + F_1\,r\,S = 0 \tag{4}$$

```
> (4);
```

$$-F_2\,r\,S + F_1\,r\,S = 0 \tag{5}$$

Die Gleichungen **(3)** und **(5)** nach den gesuchten Lagerkräften auflösen.

```
> solve( {(3),(5)}, {F[1],F[2]} )[];
```

$$F_1 = q\,S, F_2 = q\,S \tag{6}$$

Die Geometrie ist symmetrisch, die Strecklast ist symmetrisch. Daher sind die beiden Reaktionskräfte durch die Auflager gleich.

Bestimmung der Biegelinie

- Die beiden Lager nehmen keine Momente auf. Die Biegemomente werden nur durch die Biegung des Balkens ausgeglichen.
- Das rechte Lager ist ein Gleitlager. Daher entsteht keine Normalkraft im Balken.
- Die Streckenlast ist so klein, dass die Biegung mit der linearen Theorie berechnet werden kann.

Skizze des freigeschnittenen Balkens:



Durch die Auflager entstehen zwei Querkräfte, die an Punkten angreifen. Daher wird der Momentenverlauf und die Biegelinie auf drei Intervalle geteilt. (Alternativ zur stückweisen Beschreibung der Schnittmomente können Föppl-Klammern verwendet werden.)

Intervall 2 von -rS bis +rS.
Momentenverlauf M_2 auf dem Intervall 2.

```
> M[2](x) = eval( -int(-q*(zeta+S), zeta=-S..x ) -(x+r*S)*F[1], 1 );
```

$$M_2(x) = -\int_{-S}^{x} -q\,(\zeta + S)\,\mathrm{d}\zeta - (x + r\,S)\,F_1 \tag{7}$$

```
> (7); simplify(%): simplify(%,size);
```

$$M_2(x) = \frac{q\,(x^2 - S^2)}{2} + q\,S\,(x + S) - (x + r\,S)\,F_1$$

$$M_2(x) = \frac{(x + S)^2\,q}{2} - (x + r\,S)\,F_1 \tag{8}$$

Die Reaktionskraft F_1 aus Gleichung **(6)** einsetzen.

```
> subs( (6), (8) );
```

$$M_2(x) = \frac{(x+S)^2 q}{2} - (x + r S) q S \tag{9}$$

Die Differenzialgleichung für die Biegelinie.

```
> diff(w[2](x),x,x) = - M[2](x)/EI;
```

$$w_2''(x) = -\frac{M_2(x)}{EI} \tag{10}$$

Den Momentenverlauf aus **(23)** einsetzen.

```
> subs( (9), (10) ): simplify(%);
```

$$w_2''(x) = \frac{q\left(-x^2 - S^2 + 2 r S^2\right)}{2\,EI} \tag{11}$$

Integrieren der Differenzialgleichung.

```
> dsolve((11),w[2](x)): subs(_C1=A[2],_C2=B[2],%): simplify(%):
  collect(%,x);
```

$$w_2(x) = -\frac{q x^4}{24\,EI} + \frac{\left(-6 q S^2 + 12 q r S^2\right) x^2}{24\,EI} + A_2 x + B_2 \tag{12}$$

Die Lager geben die Position des Balkens an den Rändern des Interfalls 2 vor. Die Verschiebung w_2 muss an den Stellen der Lagern gleich 0 sein.

```
> Eval( w[2](x) = 0, x=r*S );
```

$$\left(w_2(x) = 0\right)\Big|_{x = r S} \tag{13}$$

```
> Eval( w[2](x) = 0, x=-r*S );
```

$$\left(w_2(x) = 0\right)\Big|_{x = -r S} \tag{14}$$

Aus den Bedingungen **(13)** und **(14)** können die Konstanten A_2 und B_2 berechnet werden.
$w_2(x)$ aus **(12)** in Gleichungen **(13)** und **(14)** einsetzen.

```
> value( subs( (12), (13) ) );
```

$$-\frac{q r^4 S^4}{24\,EI} + \frac{\left(-6 q S^2 + 12 q r S^2\right) r^2 S^2}{24\,EI} + A_2 r S + B_2 = 0 \tag{15}$$

```
> value( subs( (12), (14) ) );
```

$$-\frac{q r^4 S^4}{24\,EI} + \frac{\left(-6 q S^2 + 12 q r S^2\right) r^2 S^2}{24\,EI} - A_2 r S + B_2 = 0 \tag{16}$$

Die Gleichungen **(15)** und **(16)** bilden ein lineares Gleichungssystem für A_2 und B_2.
Das Gleichungssystem lösen.

```
> solve( {(15),(16)}, {A[2],B[2]} )[];
```

$$A_2 = 0, B_2 = \frac{q S^4 r^2 \left(r^2 + 6 - 12 r\right)}{24\,EI} \tag{17}$$

Die gefundenen Konstanten in die Biegelinie **(12)** einsetzen.

```
> subs( (17), (12) ): simplify(%,size): collect(%,x);
```

$$w_2(x) = -\frac{q x^4}{24\,EI} + \frac{\left(-\left(r^2 + 6 - 12 r\right) S^2 q + r^2 S^2 q\right) x^2}{24\,EI} + \frac{q S^4 r^2 \left(r^2 + 6 - 12 r\right)}{24\,EI} \tag{18}$$

Kontrolle: Die Symmetrie der Geometrie und der Kräfte verlangt die gleiche Symmetrie in der

Biegelinie

```
> w[2](x) = w[2](-x);
```

$$w_2(x) = w_2(-x) \tag{19}$$

```
> subs( (18), subs(x=-x,(18)), (19) );
```

$$-\frac{q\,x^4}{24\,EI} + \frac{\left(-\left(r^2+6-12\,r\right)S^2\,q + r^2\,S^2\,q\right)x^2}{24\,EI} + \frac{q\,S^4\,r^2\left(r^2+6-12\,r\right)}{24\,EI} = -\frac{q\,x^4}{24\,EI} + \frac{\left(-\left(r^2+6-12\,r\right)S^2\,q + r^2\,S^2\,q\right)x^2}{24\,EI} + \frac{q\,S^4\,r^2\left(r^2+6-12\,r\right)}{24\,EI} \tag{20}$$

```
> simplify(rhs((20))-lhs((20))=0);
```

$$0 = 0 \tag{21}$$

Intervall 1 von -S bis -rS.
Momentenverlauf M_1 auf dem Intervall 1.

```
> M[1](x) = eval( -int(-q*(zeta+S), zeta=-S..x ), 1 );
```

$$M_1(x) = -\int_{-S}^{x} -q\,(\zeta+S)\,\mathrm{d}\zeta \tag{22}$$

```
> (22); simplify(%,size);
```

$$M_1(x) = \frac{q\left(x^2-S^2\right)}{2} + q\,S\,(x+S)$$

$$M_1(x) = \frac{(x+S)^2\,q}{2} \tag{23}$$

Die Differenzialgleichung für die Biegelinie. w ist die Verschiebung in Z-Richtung.

```
> diff(w[1](x),x,x) = - M[1](x)/EI;
```

$$w_1''(x) = -\frac{M_1(x)}{EI} \tag{24}$$

Den Momentenverlauf aus **(23)** einsetzen.

```
> subs( (23), (24) );
```

$$w_1''(x) = -\frac{(x+S)^2\,q}{2\,EI} \tag{25}$$

Integrieren der Differenzialgleichung. Beim Integrieren entstehen die Integrationskonstanten A und B.

```
> dsolve((25),w[1](x)): subs(_C1=A[1],_C2=B[1],%);
```

$$w_1(x) = -\frac{(x+S)^4\,q}{24\,EI} + A_1\,x + B_1 \tag{26}$$

Die Integrationskonstanten werden aus den Übergangsbedingungen zum Intervall 2 berechnet.
Der Balken hat keine Lücke.

```
> Eval(w[1](x) = w[2](x), x=-r*S );
```

$$\left.\left(w_1(x) = w_2(x)\right)\right|_{x=-r\,S} \tag{27}$$

Der Balken hat keinen Knick.

```
> Eval(diff(w[1](x),x) = diff(w[2](x),x), x=-r*S );
```

$$\left.\left(w_1'(x) = w_2'(x)\right)\right|_{x=-r\,S} \tag{28}$$

Die Funktionen w_2 und w_1 aus **(18)** und **(26)** einsetzen.

```
> value( subs( (18), (26), (27) ) ): simplify(%,size);
```

$$-\frac{S^4 (r-1)^4 q}{24\, EI} - A_1 r S + B_1 = 0 \qquad \textbf{(29)}$$

```
> value( subs( (18), (26), (28) ) ): simplify(%,size);
```

$$\frac{S^3 (r-1)^3 q}{6\, EI} + A_1 = \frac{S^3 q\, r\, (r^2 + 3 - 6\, r)}{6\, EI} \qquad \textbf{(30)}$$

Gleichung **(30)** kann nach A_1 aufgelöst werden. B_1 folgt danach aus Gleichung **(29)**.

```
> solve( {(29),(30)}, {A[1],B[1]} )[];
```

$$A_1 = -\frac{S^3 q\, (3\, r^2 - 1)}{6\, EI},\ B_1 = \frac{S^4 q\, (r^4 - 16\, r^3 + 6\, r^2 + 1)}{24\, EI} \qquad \textbf{(31)}$$

Die gefundenen Konstanten in die Biegelinie **(26)** einsetzen.

```
> subs( (31), (26) ): simplify(%,size): collect(%,x);
```

$$w_1(x) = -\frac{q\, x^4}{24\, EI} - \frac{q\, S\, x^3}{6\, EI} - \frac{q\, S^2 x^2}{4\, EI} - \frac{S^3 q\, r^2 x}{2\, EI} + \frac{q\, r^2 (r^2 - 16\, r + 6)\, S^4}{24\, EI} \qquad \textbf{(32)}$$

Intervall 3 von +rS bis +S.
Momentenverlauf M_3 auf dem Intervall 3.

```
> M[3](x) = eval( -int(-q*(zeta+S), zeta=-S..x ) - (x+r*S)*F[1] -
  (x-r*S)*F[2], 1 );
```

$$M_3(x) = -\int_{-S}^{x} -q\, (\zeta + S)\, \mathrm{d}\zeta - (x + r\, S)\, F_1 - (x - r\, S)\, F_2 \qquad \textbf{(33)}$$

```
> (33);
```

$$M_3(x) = \frac{q\, (x^2 - S^2)}{2} + q\, S\, (x + S) - (x + r\, S)\, F_1 - (x - r\, S)\, F_2 \qquad \textbf{(34)}$$

Die Reaktionskräfte aus Gleichung **(6)** einsetzen.

```
> subs( (6), (34) ); simplify(%,size);
```

$$M_3(x) = \frac{q\, (x^2 - S^2)}{2} + q\, S\, (x + S) - (x + r\, S)\, q\, S - (x - r\, S)\, q\, S$$

$$M_3(x) = \frac{q\, (S - x)^2}{2} \qquad \textbf{(35)}$$

Die Differenzialgleichung für die Biegelinie.

```
> diff(w[3](x),x,x) = - M[3](x)/EI;
```

$$w_3''(x) = -\frac{M_3(x)}{EI} \qquad \textbf{(36)}$$

Den Momentenverlauf aus **(35)** einsetzen.

```
> subs( (35), (36) );
```

$$w_3''(x) = -\frac{q\, (S - x)^2}{2\, EI} \qquad \textbf{(37)}$$

Integrieren der Differenzialgleichung.
Beim Integrieren entstehen die Integrationskonstanten A und B.

```
> dsolve((37),w[3](x)): subs(_C1=A[3],_C2=B[3],%);
```

(38)

$$w_3(x) = -\frac{q\,(S-x)^4}{24\,EI} + A_3\,x + B_3 \tag{38}$$

Die Integrationskonstanten werden aus den Übergangsbedingungen zum Intervall 2 berechnet.
Der Balken hat keine Lücke.

```
> Eval(w[3](x) = w[2](x), x=r*S );
```

$$\left(w_3(x) = w_2(x)\right)\Big|_{x=r\,S} \tag{39}$$

Der Balken hat keinen Knick.

```
> Eval(diff(w[3](x),x) = diff(w[2](x),x), x=r*S );
```

$$\left(w_3{}'(x) = w_2{}'(x)\right)\Big|_{x=r\,S} \tag{40}$$

Die Funktionen w_2 und w_3 aus **(18)** und **(38)** einsetzen.

```
> value( subs( (18), (38), (39) ) ): simplify(%,size);
```

$$-\frac{S^4\,(r-1)^4\,q}{24\,EI} + A_3\,r\,S + B_3 = 0 \tag{41}$$

```
> value( subs( (18), (38), (40) ) ): simplify(%,size);
```

$$-\frac{S^3\,(r-1)^3\,q}{6\,EI} + A_3 = -\frac{S^3\,q\,r\,(r^2+3-6\,r)}{6\,EI} \tag{42}$$

Gleichung **(42)** kann nach A_3 aufgelöst werden. B_3 folgt danach aus Gleichung **(41)**.

```
> solve( {(41),(42)}, {A[3],B[3]} )[];
```

$$A_3 = \frac{S^3\,q\,(3\,r^2-1)}{6\,EI},\ B_3 = \frac{S^4\,q\,(r^4-16\,r^3+6\,r^2+1)}{24\,EI} \tag{43}$$

Die gefundenen Konstanten in die Biegelinie **(38)** einsetzen.

```
> subs( (43), (38) ): simplify(%,size): collect(%,x);
```

$$w_3(x) = -\frac{q\,x^4}{24\,EI} + \frac{q\,S\,x^3}{6\,EI} - \frac{q\,S^2\,x^2}{4\,EI} + \frac{S^3\,q\,r^2\,x}{2\,EI} + \frac{q\,r^2\,(r^2-16\,r+6)\,S^4}{24\,EI} \tag{44}$$

Kontrolle: Die Symmetrie der Geometrie und der Kräfte verlangt die gleiche Symmetrie in der Biegelinie

```
> w[1](x) = w[3](-x);
```

$$w_1(x) = w_3(-x) \tag{45}$$

```
> subs( (32), subs(x=-x,(44)), (45) );
```

$$-\frac{q\,x^4}{24\,EI} - \frac{q\,S\,x^3}{6\,EI} - \frac{q\,S^2\,x^2}{4\,EI} - \frac{S^3\,q\,r^2\,x}{2\,EI} + \frac{q\,r^2\,(r^2-16\,r+6)\,S^4}{24\,EI} = -\frac{q\,x^4}{24\,EI} - \frac{q\,S\,x^3}{6\,EI} - \frac{q\,S^2\,x^2}{4\,EI} - \frac{S^3\,q\,r^2\,x}{2\,EI} + \frac{q\,r^2\,(r^2-16\,r+6)\,S^4}{24\,EI} \tag{46}$$

```
> simplify(rhs((46))-lhs((46))=0);
```

$$0 = 0 \tag{47}$$

Veranschaulichung der Biegelinie

Plotten der berechneten Biegelinie.
Zahlenwerte für die Parameter S (bzw. L), q und EI sind in der Aufgabe nicht gegeben. Für das Plotten werden die Funktionen $w_i(x)$ entdimensioniert. Die Koordinate x entlang des Balkens wird

ersetzt durch eine relative Koordinate ξ.

```
> xi = x/S;
```

$$\xi = \frac{x}{S} \tag{48}$$

Die Verschiebung w wird ersetzt durch eine relative Verschiebung ω.

```
> omega = w/(q*S^4/EI);
```

$$\omega = \frac{w\,EI}{q\,S^4} \tag{49}$$

Aus der Biegelinie $w_1(x)$ in Gleichung **(32)** wird die entdimensionierte Biegelinie $\omega_1(\xi)$ durch einsetzen von **(48)** und **(49)**.

```
> subs( subs(omega=omega[1](xi),subs(w=w[1](xi*S)),rhs((49))=lhs((49))),
        subs( isolate((48),x), (32) )*coeff(rhs((49)),w)):
  simplify(%,size):
  collect(%,xi);
```

$$\omega_1(\xi) = -\frac{\xi^4}{24} - \frac{\xi^3}{6} - \frac{\xi^2}{4} - \frac{r^2\,\xi}{2} + \frac{r^2\,(r^2 - 16\,r + 6)}{24} \tag{50}$$

Aus der Biegelinie $w_2(x)$ in Gleichung **(18)** wird die entdimensionierte Biegelinie $\omega_2(\xi)$ durch einsetzen von **(48)** und **(49)**.

```
> subs( subs(omega=omega[2](xi),subs(w=w[2](xi*S)),rhs((49))=lhs((49))),
        subs( isolate((48),x), (18) )*coeff(rhs((49)),w)):
  simplify(%,size):
  collect(%,xi);
```

$$\omega_2(\xi) = -\frac{\xi^4}{24} + \left(-\frac{1}{4} + \frac{r}{2}\right)\xi^2 + \frac{r^2\,(r^2 + 6 - 12\,r)}{24} \tag{51}$$

Aus der Biegelinie $w_3(x)$ in Gleichung **(44)** wird die entdimensionierte Biegelinie $\omega_3(\xi)$ durch einsetzen von **(48)** und **(49)**.

```
> subs( subs(omega=omega[3](xi),subs(w=w[3](xi*S)),rhs((49))=lhs((49))),
        subs( isolate((48),x), (44) )*coeff(rhs((49)),w)):
  simplify(%,size):
  collect(%,xi);
```

$$\omega_3(\xi) = -\frac{\xi^4}{24} + \frac{\xi^3}{6} - \frac{\xi^2}{4} + \frac{r^2\,\xi}{2} + \frac{r^2\,(r^2 - 16\,r + 6)}{24} \tag{52}$$

Die Biegelinie ω(ξ) mit Parameter r zusammensetzen.

```
> omega(chi,r) = piecewise( xi<-r, omega[1](xi), -r<=xi and xi<=
  r, omega[2](xi), omega[3](xi) );
```

$$\omega(\chi, r) = \begin{cases} \omega_1(\xi) & \xi < -r \\ \omega_2(\xi) & -r \le \xi \le r \\ \omega_3(\xi) & \mathit{otherwise} \end{cases} \tag{53}$$

```
> Whelp := eval( subs( (50), (51), (52), rhs((53)) ) ):
  W := (Xi,R) -> evalf( subs(xi=Xi,r=R,Whelp ) ):
```

Plot der Biegelinie für verschiedene Werte des Parameters r. (Anteil der Balkenlänge zwischen den Auflagern.)
Bei großem r, also großem Abstand der Auflager ist die maximale Verschiebung in der Mitte. Bei kleinem r, also kleinem Abstand der Auflager und damit langen Kragarmen ist die maximale Verschiebung an den Balkenenden.
Dem Plot nach ist die maximale Durchbiegung in der Nähe von r = 0,6 am kleinsten.

```
> plot( [W(xi,0.4),W(xi,0.5),W(xi,0.6),W(xi,0.7)], xi=-1..+1,
  gridlines=true, labels=['x','omega'], legend=['r=0.4','r=0.5',
  'r=0.6','r=0.7'] );
```



Optimieren der Lagerpositionen

Die maximale Verschiebung bestimmen: Extremstellen der Funktion w(x) auf dem Intervall [-S ... + S] suchen. Die Schreibarbeit wird reduziert, wenn die entdimensionierte Funktion $\omega(\xi)$ verwendet wird.

In den Plots sind die lokalen Extremstellen bei ξ=-1, ξ=0 und ξ=1 sichtbar. Dennoch zur Vollständigkeit die Extremstellen berechnen.

Extremstellen im Intervall 1 suchen. Die Ableitung der entdimensionierten Biegelinie im Intervall 1, **(50)**.

```
> diff( (50), xi );
```

$$\frac{d}{d\xi}\,\omega_1(\xi) = -\frac{1}{6}\,\xi^3 - \frac{1}{2}\,\xi^2 - \frac{1}{2}\,\xi - \frac{1}{2}\,r^2 \qquad \textbf{(54)}$$

Nullstellen der Ableitung sind Extremstellen.

```
> lhs((54)) = 0;
```

$$\frac{d}{d\xi}\,\omega_1(\xi) = 0 \qquad \textbf{(55)}$$

```
> subs( (55), (54) );
```

$$0 = -\frac{1}{6}\,\xi^3 - \frac{1}{2}\,\xi^2 - \frac{1}{2}\,\xi - \frac{1}{2}\,r^2 \qquad \textbf{(56)}$$

Die Gleichung hat genau eine Lösung.

```
> isolate( (56), xi );
```

$$\xi = \left(-3\,r^2 + 1\right)^{1/3} - 1 \qquad \textbf{(57)}$$

Nur ein Wert im Intervall [-1..-r] ist ein Kandidat für eine Extremstelle, weil die Funktion ω_1 nur für das Intervall 1 des Balkens gilt.
Die Lösung **(57)** als Funktion von r plotten.

```
> plot( [-r,rhs((57))], r=0..1, gridlines=true );
```



Es gibt einen kleinen Bereich um r=0,56 in dem eine Extremstelle innerhalb des Intervalls 1 liegt.
Die linke Grenze des Intervalls wird durch die Bedingung $\xi \leqslant$ -r gegeben.

```
> -r = rhs((57));
```

$$-r = \left(-3\,r^2 + 1\right)^{1/3} - 1 \qquad \textbf{(58)}$$

```
> solve( {(58),r>0}, {r} )[]; evalf(%);
```

$$r = 3 - \sqrt{6}$$
$$r = 0.550510257 \qquad \textbf{(59)}$$

Die rechte Grenze ist durch die Bedingung -1 $\leqslant$ gegeben.

```
> -1 = rhs((57));
```

$$-1 = \left(-3\,r^2 + 1\right)^{1/3} - 1 \qquad \textbf{(60)}$$

```
> solve( {(60),r>0}, {r} )[]; evalf(%);
```

$$r = \frac{\sqrt{3}}{3}$$
$$r = 0.5773502693 \qquad \textbf{(61)}$$

Die Biegelinien für diesen Parameterbereich plotten.
Die Extremstellen liegen im Intervall 1 links von den Auflagern. Durch die Symmetrie liegen im Intervall 2 rechtes von den Auflagern Extremstellen.

Sichtbar ist, dass die Extremstellen der Biegelinie keine Punkte maximler Verschiebung sind.

```
> plot( [W(xi,0.550),W(xi,0.560),W(xi,0.577)], xi=-1..+1,
  gridlines=true, labels=['x','omega'], legend=['r=0.550','r=
  0.560','r=0.577'] );
```



Extremstellen im Intervall 2 suchen. Die Ableitung der entdimensionierten Biegelinie im Intervall 1, **(51)**.

```
> diff( (51), xi );
```

$$\frac{d}{d\xi}\,\omega_2(\xi) = -\frac{\xi^3}{6} + 2\left(-\frac{1}{4} + \frac{r}{2}\right)\xi \qquad \textbf{(62)}$$

Nullstellen der Ableitung sind Extremstellen.

```
> lhs((62)) = 0;
```

$$\frac{d}{d\xi}\,\omega_2(\xi) = 0 \qquad \textbf{(63)}$$

```
> subs( (63), (62) );
```

$$0 = -\frac{\xi^3}{6} + 2\left(-\frac{1}{4} + \frac{r}{2}\right)\xi \qquad \textbf{(64)}$$

Die Gleichung hat genau eine Lösung.

```
> isolate( (64), xi );
```

$$\xi = 0 \qquad \textbf{(65)}$$

Die relative Verschiebung an dieser Stelle folgt aus einsetzen in die Gleichung **(51)** der Biegelinie.

```
> subs( (65), (51) ); collect(%,r);
```

$$\omega_2(0) = \frac{r^2\,(r^2 + 6 - 12\,r)}{24}$$

$$\omega_2(0) = \frac{1}{24}\,r^4 - \frac{1}{2}\,r^3 + \frac{1}{4}\,r^2 \qquad \textbf{(66)}$$

Im Intervall 2 müssen keine Extremstellen gesucht werden, weil die Biegelinie symmetrisch ist.

Als Kandiaten für die maximale Verschiebung gibt es noch die Enden des Balkens. Die Stelle ξ=-1 zu betrachten is ausreichend, weil die Biegelinie symmetrisch ist.
Verschiebung an der Stelle ξ=-1 durch einsetzen in die Biegelinie **(50)** berechnen.

```
> subs( xi=-1, (50) ); collect(%,r);
```

$$\omega_1(-1) = -\frac{1}{8} + \frac{r^2}{2} + \frac{r^2\,(r^2 - 16\,r + 6)}{24}$$

$$\omega_1(-1) = -\frac{1}{8} + \frac{3}{4}\,r^2 + \frac{1}{24}\,r^4 - \frac{2}{3}\,r^3 \tag{67}$$

Plotten der Kandidaten für die maximale Verschiebung.

```
> plot( [rhs((66)),rhs((67))], r=0..1, legend=[lhs((66)),lhs((67))], gridlines=true );
```



Der Plot zeigt, dass im Parameterbereich r = [0,4 .. 0,6] die maximale Verschiebung möglichst klein wird.
Diesen Bereich größer darstellen. Den Betrag der Verschiebung plotten, damit die beiden Kurven besser vergleichbar sind.

```
> plot( [abs(rhs((66))),abs(rhs((67)))], r=0.4..0.6, legend=[abs(lhs((66))),abs(lhs((67)))], gridlines=true );
```



Sind die Auflager eng zusammen (= kleines r) dann ist die maixmale Verschiebung am Ende des Balkens. Sind die Auflager weit auseinander (= großes r), dann ist die maximale Verschiebung in

der Mitte.
Die maximale Verschiebung ist minimal beim Übergang, bei dem r sind die Verschiebungen am Ende des Balkens gleich der Verschiebung in der Mitte.

Die Bestimmungsgleichung für das gesuchte r.

```
> lhs((66)) = lhs((67));
```

$$\omega_2(0) = \omega_1(-1) \qquad (68)$$

Die Funktionen einsetzen.

```
> subs( (66), (67), (68) ); %-lhs(%);
```

$$\frac{1}{24} r^4 - \frac{1}{2} r^3 + \frac{1}{4} r^2 = -\frac{1}{8} + \frac{3}{4} r^2 + \frac{1}{24} r^4 - \frac{2}{3} r^3$$

$$0 = -\frac{1}{8} + \frac{1}{2} r^2 - \frac{1}{6} r^3 \qquad (69)$$

Eine kubische Gleichung ist zu lösen. Die analytischen Lösungen sind sehr unhandlich. Daher gebe ich nur die numerischen Lösungen an.

```
> fsolve( (69), {r} );
```

$$\{r = -0.4652268748\}, \{r = 0.5537017978\}, \{r = 2.911525077\} \qquad (70)$$

Der Parameter r muss im Bereich 0..1 liegen. Es bliebt nur eine Lösung.

```
> fsolve( (69), {r}, 0..1 )[];
```

$$r = 0.5537017978 \qquad (71)$$

Die gesuchten Längen sind damit

```
> subs( (71), (2) );
```

$$L_2 = 0.5537017978\, L \qquad (72)$$

```
> subs( (72), L[1] = ( L - L[2] )/2 );
```

$$L_1 = 0.2231491011\, L \qquad (73)$$

Ein Plot der entdimensionierten Biegelinie beim gefunden Parameter.
Zu sehen ist die gleiche Verschiebung an den Enden und in der Mitte des Balkens.

```
> plot( W(xi,rhs((71))), xi=-1..+1, gridlines=true, labels=['x', 'omega'] );
```



Hilfsmittel
- Balke: Einführung in die Technische Mechanik, Springer Verlag
- Schreiber: Technische Mechanik 1, Skript Uni Kassel, http://www.ifm.maschinenbau.uni-kassel.de/~lsch/lsch_skripte.html
- Maple 15, http://www.maplesoft.com/