```
> restart;
```

Bewegung einer Kugel im vertikalen Wasserstrom

Eine Kugel mit Masse m, Radius r ist in einem Wasserstrom. Der Wasserstrom ist senkrecht nach oben gerichtet. Die Kugel hat nur eine senkrechte Startgeschwindigkeit. Die Bewegung der Kugel erfolgt nur senkrecht, die Bewegung kann eindimensional beschrieben werden.



Die Geschwindigkeit v der Wasserströmung ist gegeben. Die Strömungsgeschwindigkeit ist an jedem Punkt un zu jedem Zeitpunkt der Strömung gleich.
Zu berechnen ist die Geschwindigkeit u(t) der Kugel.
Die Startgeschwindigkeit der Kugel ist u(0) = u_0. Die Startposition der Kugel ist x(0)=0.

Auf die Kugel wirkt das Gewicht der Kugel. Die Fallbeschleunigung g wirkt in -X Richtung.

```
> F[G]=-m*g;
```

$$F_G = -m\, g \tag{1}$$

Auf die Kugel wirkt die Auftriebskraft in X Richtung. (Archimedisches Prinzip). Darin ist ρ die Dichte des Wassers

```
> F[A]=V*rho*g;
```

$$F_A = V\, \rho\, g \tag{2}$$

Das Volumen V der Kugel mit Radius r.

```
> V=4*Pi*r^3/3;
```

$$V = \frac{4\, \pi\, r^3}{3} \tag{3}$$

Annahme: die Umströmung ist laminar.

Auf die Kugel wirkt die Wasserströmung durch Reibung. Die Kraft ist abhängig von der Geschwindigkeitsdifferenz v-u. Bei kleiner Geschwindigkeitsdifferenz ist die Umströmung der Kugel laminar, bei höherer Geschwindigkeit wird die Strömung turbulent.
Die Kraft auf die Kugel folgt aus dem Stokesschen Reibungsgesetz. Darin ist η die dynamische Viskosität des Wassers.

```
> F[R] = 6*Pi*eta*r*(v-u(t));
```

$$F_R = 6\,\pi\,\eta\,r\,(v - u(t)) \tag{4}$$

Die Summe der Kräfte auf die Kugel steht im Gleichgewicht zur Trägheitskraft. (Newton, d'Alembert)

```
> m*diff(u(t),t) = F[G]+F[A]+F[R];
```

$$m\,\dot{u}(t) = F_G + F_A + F_R \tag{5}$$

Einsetzen der Kräfte.

```
> subs((1),(2),(4),(5));
```

$$m\,\dot{u}(t) = -m\,g + V\,\rho\,g + 6\,\pi\,\eta\,r\,(v - u(t)) \tag{6}$$

Die stationäre Lösung der Differenzialgleichung **(6)** bestimmen.

```
> diff(u(t),t) = 0;
```

$$\dot{u}(t) = 0 \tag{7}$$

```
> subs((7),(6));
```

$$0 = -m\,g + V\,\rho\,g + 6\,\pi\,\eta\,r\,(v - u(t)) \tag{8}$$

Diese Gleichung liefert die Grenzgeschwindigkeit u_s. Hat die Kugel diese Grenzgeschwindigkeit erreicht, bleibt die Geschwindigkeit der Kugel konstant u_s.

```
> subs(u(t)=u[s],(8));
```

$$0 = -m\,g + V\,\rho\,g + 6\,\pi\,\eta\,r\,\left(v - u_s\right) \tag{9}$$

Auflösen nach der gesuchten Geschwindigkeit u_s.

```
> isolate((9),u[s]);
```

$$u_s = \frac{-m\,g + V\,\rho\,g}{6\,\pi\,\eta\,r} + v \tag{10}$$

```
> simplify((10),size): sort(%,[v,V]);
```

$$u_s = v + \frac{g\,(\rho\,V - m)}{6\,\pi\,\eta\,r} \tag{11}$$

Ist die Dichte der Kugel größer als die Dichte ρ des Wassers, dann ist die Grenzgeschwindigkeit kleiner als die Strömungsgeschwindigkeit v.

Die Differenzialgleichung **(6)** ist inhomogen. Der homogene Anteil:

```
> lhs((6))=rhs((6))-(-m*g+V*rho*g+6*Pi*eta*r*v); simplify(%);
```

$$m\,\dot{u}(t) = 6\,\pi\,\eta\,r\,(v - u(t)) - 6\,v\,\pi\,\eta\,r$$

$$m\,\dot{u}(t) = -6\,\pi\,\eta\,r\,u(t) \tag{12}$$

Die allgemeine Lösung einer Differenzialgleichung der Form $\frac{dx}{dt} = const.\cdot x(t)$ ist bekannt, die Exponentialfunktion.

```
> dsolve((12)): subs(_C1=K,%);
```

$$u(t) = K\,e^{-\frac{6\pi\eta r t}{m}} \tag{13}$$

Darin ist K die Integrationskonstante.
Eine spezielle Lösung der Differenzialgleichung **(6)** ist bereits berechnet, die stationäre Lösung **(11)**.
Die allgemeine Lösung der Differenzialgleichung **(6)** ist die Summe dieser beiden Teillösungen.

```
> lhs((13))=rhs((13))+lhs((11));
```

$$u(t) = K\,e^{-\frac{6\pi\eta r t}{m}} + u_s \tag{14}$$

Zum Startzeitpunkt t=0 ist die Geschwindigkeit u_0.

```
> u(0)=u[0];
```

$$u(0) = u_0 \tag{15}$$

Daraus wird die Konstante K bestimmt.

```
> subs(t=0,(14));
```

$$u(0) = K\,e^0 + u_s \tag{16}$$

```
> subs((15),(16)): simplify(%);
```

$$u_0 = K + u_s \tag{17}$$

```
> isolate((17),K);
```

$$K = u_0 - u_s \tag{18}$$

In die Lösung **(14)** einsetzen.

```
> subs((18),(14));
```

$$u(t) = (u_0 - u_s)\,e^{-\frac{6\pi\eta r t}{m}} + u_s \tag{19}$$

Die Zeitkonstante im Exponenten bestimmt, wie schnell die Grenzgeschwindigkeit erreicht wird.

```
> ln((rhs((19))-u[s])/(u[0]-u[s])): simplify(%) assuming real: Tau=-
  t/%;
```

$$\mathrm{T} = \frac{m}{6\pi\eta r} \tag{20}$$

```
> algsubs(1/rhs((20))=1/lhs((20)),(19));
```

$$u(t) = (u_0 - u_s)\,e^{-\frac{t}{\mathrm{T}}} + u_s \tag{21}$$

Der Ort der Kugel x folgt aus Integration der Geschwindigkeit u(t).

```
> x(t)=int(u(t),t);
```

$$x(t) = \int u(t)\,dt \tag{22}$$

```
> subs((21),(22)); simplify(%,size): lhs(%)=rhs(%)+K;
```

$$x(t) = \int \left((u_0 - u_s)\,e^{-\frac{t}{\mathrm{T}}} + u_s \right) dt$$

$$x(t) = -(u_0 - u_s)\,\mathrm{T}\,e^{-\frac{t}{\mathrm{T}}} + u_s\,t + K \tag{23}$$

Der Buchstabe K wird als Integrationskonstante wiederverwendet.

Die Kugel ist zum Startzeitpunkt t=0 am Ort 0. Aus der Startbedingung die Konstante K bestimmen.

```
> x(0) = 0;
```

$$x(0) = 0 \tag{24}$$

```
> subs(t=0,(23));
```

$$x(0) = -(u_0 - u_s)\,\mathrm{T}\,e^0 + K \tag{25}$$

```
> subs((24),(25));
```

$$0 = -(u_0 - u_s)\,\mathrm{T} + K \tag{26}$$

```
> isolate((26),K);
```

$$K = (u_0 - u_s)\,\mathrm{T} \tag{27}$$

Einsetzen in **(23)**.

```
> subs((27),(23)); lhs(%)=simplify( rhs(%)-u[s]*t, size )+u[s]*t;
```

$$x(t) = -(u_0 - u_s)\,\mathrm{T}\,e^{-\frac{t}{\mathrm{T}}} + u_s\,t + (u_0 - u_s)\,\mathrm{T}$$

$$x(t) = -(u_0 - u_s)\,\mathrm{T}\left(e^{-\frac{t}{\mathrm{T}}} - 1\right) + u_s\,t \tag{28}$$

Das ist die Bewegung der Kugel.

Annahme: Die Umströmung der Kugel ist turbulent.

Bei größerer Differenzgeschwindigkeit kann die Strömung um die Kugel turbulent werden. Einen Hinweis auf die Strömungsart liefert die Reynolds Zahl ausgerechnet mit der Differenzgeschwindigkeit. Die Abhängigkeit der Reibungskraft von der Geschwindigkeit bei einer turbulenten Umströmung ist

```
> F[R] = c*rho*Pi*r^2*(v-u(t))^2/2;
```

$$F_R = \frac{c\,\rho\,\pi\,r^2\,(v - u(t))^2}{2} \tag{29}$$

Diese Gleichung gilt nur für $v-u(t) > 0$, weil das Vorzeichen von F_R immer positiv ist. Anders formuliert: Die Gleichung gilt nur, wenn die Strömungsgeschwindigkeit v größer ist als die Geschwindigkeit u(t) der Kugel.

Der Widerstandsbeiwert c für die Kugel ist 0,45 bei $\mathrm{Re} < 2 \cdot 10^5$ und 0,13 bei $\mathrm{Re} = 10^6$.

Die Trägheitskraft der Masse steht im Gleichgewicht zu den angreifenden Kräften.

```
> m*diff(u(t),t) = F[G]+F[A]+F[R];
```

$$m\,\dot{u}(t) = F_G + F_A + F_R \tag{30}$$

```
> subs((1),(2),(29),(30));
```

$$m\,\dot{u}(t) = -m\,g + V\,\rho\,g + \frac{c\,\rho\,\pi\,r^2\,(v - u(t))^2}{2} \tag{31}$$

Das ist eine Differenzialgleichung für die Geschwindigkeit u(t) der Kugel.

Die Bewegung auf eine stationäre Geschwindigkeit untersuchen. Also auf eine Situation mit Beschleunigung gleich Null.

```
> diff(u(t),t)=0;
```

$$\dot{u}(t) = 0 \tag{32}$$

```
> subs((32),(31));
```

(33)

$$0 = -m\,g + V\,\rho\,g + \frac{c\,\rho\,\pi\,r^2\,(v - u(t))^2}{2} \quad \textbf{(33)}$$

Die Geschwindigkeit in diesem Fall ist konstant, die stationäre Geschwindigkeit u_s.

```
> subs(u(t)=u[s],(33));
```

$$0 = -m\,g + V\,\rho\,g + \frac{c\,\rho\,\pi\,r^2\,(v - u_s)^2}{2} \quad \textbf{(34)}$$

Die quadratische Gleichung hat zwei Lösungen.

```
> solve((34),{u[s]}): expand({%}): subsindets(%,radical,x->simplify
  (x)): map(x->subs(A=m-rho*V,subs(rho*V-m=-A,x)),%)[];
```

$$\left\{u_s = v - \frac{\sqrt{2}\,\sqrt{c\,\rho\,g\,(-\rho\,V + m)}}{c\,\rho\,\sqrt{\pi}\,r}\right\}, \left\{u_s = v + \frac{\sqrt{2}\,\sqrt{c\,\rho\,g\,(-\rho\,V + m)}}{c\,\rho\,\sqrt{\pi}\,r}\right\} \quad \textbf{(35)}$$

Nur die erste Lösung erfüllt die Bedingung $v - u(t) > 0$.

```
> (35)[1][];
```

$$u_s = v - \frac{\sqrt{2}\,\sqrt{c\,\rho\,g\,(-\rho\,V + m)}}{c\,\rho\,\sqrt{\pi}\,r} \quad \textbf{(36)}$$

Die Wurzel ist nur dann definiert (im Reellen), wenn $m > \rho\,V$, also das Gewicht der Kugel größer ist als die Auftriebskraft.
Umgekehrt formuliert: Ist die Auftriebskraft größer als das Gewicht der Kugel, dann gibt es keinen stationären Zustand unter der Bedinung $v - u(t) > 0$.
Diese Einschränkung ist plausibel, was die Betrachtung eines Spezialfalls zeigt. Bei einer Startgeschwindigkeit $u(0) = 0$ und $\rho\,V > m$ und $v > 0$ wird die Kugel durch die Reibungskraft und die Auftriebskraft gegen das Gewicht beschleunigt. Wird von der Kugel die Strömungsgeschwindigkeit erreicht, $u(t) = v$, dann ist die Reibungskraft Null, aber die Auftriebskraft beschleunigt die Kugel gegen das Gewicht weiter. Die Geschwindigkeit der Kugel wird größer als die Strömungsgeschwindigkeit. Damit wird die Bedingung $u(t) - v > 0$ verletzt. Innerhalb des Gültigkeitsbereichs der Differenzialgleichung **(31)** wird kein stationärer Zustand erreicht.

Weiter mit der Lösung der Differenzialgleichung.

Die Geschwindigkeitsdifferenz substituieren.

```
> w(t)=v-u(t);
```

$$w(t) = v - u(t) \quad \textbf{(37)}$$

```
> diff((37),t);
```

$$\dot{w}(t) = -\dot{u}(t) \quad \textbf{(38)}$$

```
> isolate((38),diff(u(t),t));
```

$$\dot{u}(t) = -\dot{w}(t) \quad \textbf{(39)}$$

```
> subs(rhs((37))=lhs((37)),(39),(31));
```

$$-m\,\dot{w}(t) = -m\,g + V\,\rho\,g + \frac{c\,\rho\,\pi\,r^2\,w(t)^2}{2} \quad \textbf{(40)}$$

Die Lösung dieser Differenzialgleichung ist

```
> dsolve((40)): subs(_C1=K,%): factor(%);
```

(41)

$$w(t) = -\frac{\tan\left(\frac{r\,(t+K)\,\sqrt{2}\,\sqrt{c\,\rho\,\pi\,g\,(\rho\,V-m)}}{2\,m}\right)\sqrt{2}\,\sqrt{c\,\rho\,\pi\,g\,(\rho\,V-m)}}{c\,\rho\,\pi\,r} \tag{41}$$

Darin ist K die Integrationskonstante.
Jetzt, wie bereits bei der Grenzgeschwindigkeit u_s, den Fall $m > \rho\,V$ betrachten. Bei der Schreibweise der Gleichung **(41)** treten zwei Wurzel mit imaginärem Wert auf. Doch $w(t)$ ist wieder reell.
Es gilt $\tan(I\,x) = I\tanh(x)$ mit der imaginären Einheit I.

```
> subs(sqrt(c*rho*Pi*g*(rho*V-m))=I*sqrt(c*rho*Pi*g*(m-rho*V)),(41))
  ;
```

$$w(t) = \frac{-\mathrm{I}\tan\left(\frac{\frac{\mathrm{I}}{2}\,r\,(t+K)\,\sqrt{2}\,\sqrt{c\,\rho\,\pi\,g\,(-\rho\,V+m)}}{m}\right)\sqrt{2}\,\sqrt{c\,\rho\,\pi\,g\,(-\rho\,V+m)}}{c\,\rho\,\pi\,r} \tag{42}$$

```
> simplify((42)): subs(A=m-rho*V,subs(rho*V-m=-A,%));
```

$$w(t) = \frac{\tanh\left(\frac{r\,(t+K)\,\sqrt{2}\,\sqrt{\pi}\,\sqrt{c\,\rho\,g\,(-\rho\,V+m)}}{2\,m}\right)\sqrt{2}\,\sqrt{c\,\rho\,g\,(-\rho\,V+m)}}{\sqrt{\pi}\,c\,\rho\,r} \tag{43}$$

In dieser Darstellung sind auch die Werte der Wurzel reell.
Die auftretenden Wurzel sind aus der Grenzgeschwindigkeit **(36)** bekannt.

```
> (v-(36))*denom(v-rhs((36)))/sqrt(2): subs(rhs(%)=lhs(%),(43));
```

$$w(t) = \tanh\left(\frac{r^2\,(t+K)\,\pi\,c\,\rho\,(v-u_s)}{2\,m}\right)(v-u_s) \tag{44}$$

Die Substitution **(37)** rückgängig machen.

```
> subs((37),(44));
```

$$v-u(t) = \tanh\left(\frac{r^2\,(t+K)\,\pi\,c\,\rho\,(v-u_s)}{2\,m}\right)(v-u_s) \tag{45}$$

```
> isolate((45),u(t));
```

$$u(t) = -\tanh\left(\frac{r^2\,(t+K)\,\pi\,c\,\rho\,(v-u_s)}{2\,m}\right)(v-u_s)+v \tag{46}$$

Die Zeitkonstante kann zusammengefasst werden

```
> Tau = (2*m)/(c*Pi*r^2*rho*(v-u[s]));
```

$$\mathrm{T} = \frac{2\,m}{c\,\pi\,r^2\,\rho\,(v-u_s)} \tag{47}$$

```
> subs((t+K)/rhs((47))=(t+K)/lhs((47)),(46));
```

$$u(t) = -\tanh\left(\frac{t+K}{\mathrm{T}}\right)(v-u_s)+v \tag{48}$$

Hilfsmittel:
- Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch
- Maple 14