```
> restart;
> Digits:=30: interface( displayprecision=5 ):
```

Aufgabe

Die Masse eines beladenen Aufzugkorbes beträgt m_1 = 900kg, die des Gegengewichts m_2 = 700kg. Das Seil hat die Masse m_3 = 100kg. Das Trägheitsmoment der Seiltrommel wird vernachlässigt. Die Reibungskraft an Führungen und Rollen beträgt F_r = 398N. Die Gewichtskraft des Seils soll nicht zur Beschleunigung beitragen.

a) Welches Antriebsmoment ist an der Trommel (d=40cm) erforderlich, um den Aufzug nach T = 2s eine Steiggeschwindigkeit von v_s = 2,4m/s zu erteilen?

b) Mit welcher Beschleunigung beginnt der Aufzug zu sinken, wenn kein Antrieb und keine Bremse betätigt werden?

Die Rechnung soll mit der d'Alembertsche Trägheitskraft erfolgen.

```
> parameters := m[1]=900*Unit(kg),m[2]=700*Unit(kg),m[3]=100*Unit(kg),F[r]=398*Unit(N),d=40*Unit(cm),T=2*Unit(s),v[s]=2.4*Unit(m/s):
```

Skizze



Rechnung

Die Aufgabe ist eine Variante der Atwoodschen Fallmaschine.

An der Rolle greifen zwei Gewichte parallel zur y-Achse an. Lauf Aufgabe wird das Gewicht der Masse m_3 vernachlässigt.

```
> F[s,1] = -m[1]*g;
```

$$F_{s,\,1} = -m_1\, g \tag{1}$$

```
> F[s,2] = -m[2]*g;
```

$$F_{s,\,2} = -m_2\, g \tag{2}$$

Dabe ist g die Fallbeschlenigung an der Erdoberfläche.

```
> g = evalf(ScientificConstants[Constant]
  (standard_acceleration_of_gravity,units));
```

$$g = 9.80665 \left[\!\left[\frac{m}{s^2} \right]\!\right] \tag{3}$$

Die beiden Gewichte führen zu Drehmomenten (parallel zur z-Achse) um die Achse der Rolle. Die Summe dieser Momente:

```
> M[s] = -d/2*F[s,1] + d/2*F[s,2];
```

$$M_s = -\frac{1}{2}\, d\, F_{s,\,1} + \frac{1}{2}\, d\, F_{s,\,2} \tag{4}$$

Die Massen werden in der Zeit T auf die Geschwindigkeit v_s beschleunigt.

Annahme: Die Beschleunigung a ist konstant.

```
> a=v[s]/T;
```

$$a = \frac{v_s}{T} \tag{5}$$

Es wirken Trägheitskräfte der Massen entgegen. Die Trägheit aller drei Massen wird berücksichtigt. Die Trägemasse des Seils wird auf der rechten Seite zusammengezogen. (Das formal korrekt aufzuschreiben ist aufwendig, weil viele Stücke des Seils sich in unterschiedliche Richtungen bewegen.)

```
> F[t,1] = -a*m[1];
```

$$F_{t,\,1} = -a\, m_1 \tag{6}$$

```
> F[t,2] = a*m[2];
```

$$F_{t,\,2} = a\, m_2 \tag{7}$$

```
> F[t,3] = a*m[3];
```

$$F_{t,\,3} = a\, m_3 \tag{8}$$

Die Trägheitskräfte führen ebenfalls zu Drehmomenten um die Achse der Rolle. Die Summe dieser Momente:

```
> M[t] = -d/2*F[t,1] + d/2*F[t,2] + d/2*F[t,3];
```

$$M_t = -\frac{1}{2}\, d\, F_{t,\,1} + \frac{1}{2}\, d\, F_{t,\,2} + \frac{1}{2}\, d\, F_{t,\,3} \tag{9}$$

Die Reibung F_r wirkt der Bewegung entgegen. Die Seile reiben. An der Rolle greift die Reibung daher auch im Abstand d/2 an. Das Drehmoment aus der Reibung:

```
> M[r] = d/2*F[r];
```

$$M_r = \frac{1}{2}\, d\, F_r \tag{10}$$

Alle Momente und das Drehmoment des Antriebs M_A addieren sich zu 0. (d'Alembert)

```
> M[A]+M[r]+M[t]+M[s]=0;
```

$$M_A + M_r + M_t + M_s = 0 \tag{11}$$

Die Drehmomente einsetzen.

```
> subs( (4),(9),(10), (11) );
```

(12)

$$M_A + \frac{1}{2}\, d\, F_r - \frac{1}{2}\, d\, F_{t,1} + \frac{1}{2}\, d\, F_{t,2} + \frac{1}{2}\, d\, F_{t,3} - \frac{1}{2}\, d\, F_{s,1} + \frac{1}{2}\, d\, F_{s,2} = 0 \tag{12}$$

Die Kräfte einsetzen.

```
> subs( (1),(2),(6),(7),(8), (12) );
```

$$M_A + \frac{1}{2}\, d\, F_r + \frac{1}{2}\, d\, a\, m_1 + \frac{1}{2}\, d\, a\, m_2 + \frac{1}{2}\, d\, a\, m_3 + \frac{1}{2}\, d\, m_1\, g - \frac{1}{2}\, d\, m_2\, g = 0 \tag{13}$$

Nach dem gesuchten Antriebsmoment auflösen.

```
> isolate( (13), M[A] ); simplify(%,size);
```

$$M_A = -\frac{1}{2}\, d\, F_r - \frac{1}{2}\, d\, a\, m_1 - \frac{1}{2}\, d\, a\, m_2 - \frac{1}{2}\, d\, a\, m_3 - \frac{1}{2}\, d\, m_1\, g + \frac{1}{2}\, d\, m_2\, g$$

$$M_A = -\frac{1}{2}\, d\, \left((m_1 + m_2 + m_3)\, a + m_1\, g - m_2\, g + F_r \right) \tag{14}$$

Die Beschleunigung a aus **(5)** einsetzen.

```
> subs( (5), (14) );
```

$$M_A = -\frac{1}{2}\, d\, \left(\frac{(m_1 + m_2 + m_3)\, v_s}{T} + m_1\, g - m_2\, g + F_r \right) \tag{15}$$

Die Werte aus der Aufgabe und den Parameter **(3)** einsetzen und ausrechnen.

```
> subs( parameters,(3),(15) ): simplify(%): lhs(%)=convert(rhs(%),
  units,'N*m');
```

$$M_A = -879.86600\ [\![N\, m]\!] \tag{16}$$

Das Vorzeichen ist negativ, weil der Antrieb ein Moment im Uhrzeigersinn liefert.

Im Teil (b) der Aufgabe ist der Antrieb ausgeschaltet und keine Bremse aktiv.
Die Gleichung **(11)** mit $M_A = 0$ gilt.

```
> subs( M[A]=0, (11) );
```

$$M_r + M_t + M_s = 0 \tag{17}$$

Die Drehmomente einsetzen.

```
> subs( (4),(9),(10), (17) );
```

$$\frac{1}{2}\, d\, F_r - \frac{1}{2}\, d\, F_{t,1} + \frac{1}{2}\, d\, F_{t,2} + \frac{1}{2}\, d\, F_{t,3} - \frac{1}{2}\, d\, F_{s,1} + \frac{1}{2}\, d\, F_{s,2} = 0 \tag{18}$$

Die Kräfte einsetzen.

```
> subs( (1),(2),(6),(7),(8), (18) );
```

$$\frac{1}{2}\, d\, F_r + \frac{1}{2}\, d\, a\, m_1 + \frac{1}{2}\, d\, a\, m_2 + \frac{1}{2}\, d\, a\, m_3 + \frac{1}{2}\, d\, m_1\, g - \frac{1}{2}\, d\, m_2\, g = 0 \tag{19}$$

Die Beschleunigung a ist nicht gleich dem Werte aus Gleichung **(5)**, sondern die gesuchte Größe. Also nach der Beschleungung a auflösen.

```
> isolate( (19),a ); simplify(%,size);
```

$$a = \frac{-\frac{1}{2}\, d\, F_r - \frac{1}{2}\, d\, m_1\, g + \frac{1}{2}\, d\, m_2\, g}{\frac{1}{2}\, d\, m_1 + \frac{1}{2}\, d\, m_2 + \frac{1}{2}\, d\, m_3}$$

$$a = \frac{(-m_1 + m_2)\, g - F_r}{m_1 + m_2 + m_3} \tag{20}$$

Die Werte aus der Aufgabe und den Parameter **(3)** einsetzen und ausrechnen.

```
> subs( parameters,(3),(20) ): simplify(%);
```

$$a = -1.38784 \left[\left[\frac{m}{s^2} \right]\right] \tag{21}$$

Die Beschleunigung ist negativ, weil die Masse m_s nach unten fällt.