

```
> restart;
> Digits:=30: interface( DisplayPrecision=12, imaginaryunit=j );
```

Aufgabe 1

Eine ideale Punktschallquelle erzeugt im Abstand r_1 einen Schallintensitätspegel L_1 .

Wie hoch ist der Pegel L_2 im Abstand r_2 ?

Welche elektrische Leistung P nimmt die Quelle auf, wenn sie mit einem Wirkungsgrad Schall X erzeugt.

```
> r[1] = 10*Unit(m), r[2] = 70*Unit(m), L[1] = 80*Unit(dB), X = 1/100;
```

$$r_1 = 10 \text{ [m]}, r_2 = 70 \text{ [m]}, L_1 = 80 \text{ [dB]}, X = \frac{1}{100} \quad (1)$$

Bearbeitung

Aus der Formelsammlung [1]: Schallpegeldifferenz bei einer punktförmigen Schallquelle. Schallintensitätspegel L_i im Abstand r_i zur Quelle. \lg ist der Logarithmus zur Basis 10.

```
> L[1]-L[2] = 20*Unit(dB)*lg(r[2]/r[1]);
```

$$L_1 - L_2 = 20 \lg\left(\frac{r_2}{r_1}\right) \text{ [dB]} \quad (2)$$

Auflösen nach gesuchtem Pegel L_2 .

```
> isolate((2),L[2]): sort(%);
```

$$L_2 = -20 \lg\left(\frac{r_2}{r_1}\right) \text{ [dB]} + L_1 \quad (3)$$

Alle Größen auf der rechten Seite sind bekannt. Die Werte einsetzen und ausrechnen.

```
> subs(lg=log10,(1),(3)): evalf(%);
```

$$L_2 = 63.0980391997 \text{ [dB]} \quad (4)$$

Der Schallintensitätspegel im Abstand $r_2 = 70\text{m}$ beträgt 63 dB.

Aus der Formelsammlung [1]: Schallintensität I im Abstand r einer punktförmigen Schallquelle mit Ausgangsleistung P .

```
> I = P/(4*Pi*r^2);
```

$$I = \frac{P}{4 \pi r^2} \quad (5)$$

(= gleichmäßige Verteilung der Schalleistung auf die Kugelschale)

Aus der Formelsammlung [1]: Umrechnung von Schallintensität I auf den Schallintensitätspegel L .

```
> L = 10 * lg( I/I[0] ) *Unit(dB);
```

$$L = 10 \lg\left(\frac{I}{I_0}\right) \text{ [dB]} \quad (6)$$

Die Bezugsintensität ist dabei

```
> I[0] = 1e-12 *Unit(W/m^2): lhs(%) = convert( rhs(%), units, W/m^2 );
```

$$I_0 = 1 \cdot 10^{-12} \left[\frac{W}{m^2} \right] \quad (7)$$

Der Wirkungsgrad X verknüpft die aufgenommene elektrische Leistung P_{el} mit der abgegebenen Schalleistung P .

```
> P = X * P[el];
```

$$P = X P_{el} \quad (8)$$

Gleichung (8) in Gleichung (5) einsetzen.

```
> subs( (8), (5) );
```

$$I = \frac{X P_{el}}{4 \pi r^2} \quad (9)$$

Auflösen nach der gesuchten elektrischen Leistung.

```
> isolate( (9), P[el] );
```

$$P_{el} = \frac{4 I \pi r^2}{X} \quad (10)$$

Die Schallintensität ist noch unbekannt. (6) auflösen nach der Schallintensität.

```
> (6)/(10*Unit(dB)) :  
10^rhs(%)=10^lhs(%) :  
applyrule(10^lg(x::algebraic)=x, lhs(%) )=rhs(%) :  
solve(%, {I}) [1];
```

$$I = 10^{\frac{L}{10} \llbracket dB \rrbracket} I_0 \quad (11)$$

Einsetzen in (10).

```
> subs( (11), (10) );
```

$$P_{el} = \frac{4 \cdot 10^{\frac{L}{10} \llbracket dB \rrbracket} I_0 \pi r^2}{X} \quad (12)$$

In der Aufgabe ist L_1 anstelle von L und r_1 anstelle von r gegeben.

```
> subs( L=L[1], r=r[1], (12) );
```

$$P_{el} = \frac{4 \cdot 10^{\frac{L_1}{10} \llbracket dB \rrbracket} I_0 \pi r_1^2}{X} \quad (13)$$

Auf der rechten Seite sind alle Größen bekannt. Einsetzen und ausrechnen.

```
> subs( (1), (7), (13) ): simplify(%);
```

$$P_{el} = 12.5663706144 \llbracket W \rrbracket \quad (14)$$

Die Punktquelle nimmt eine elektrische Leistung von 13 W auf.

Aufgabe 2

2 Überlandleitungen verlaufen parallel im Abstand s . Jede Leitung erzeugt für sich ein Summton, dessen Schallintensität I_1 in einer Entfernung r_1 senkrecht zur Leitung beträgt.

Bestimmen Sie Gesamtschallintensitätspegel L und Gesamtschallintensität I in der Mitte zwischen den Leitungen.

```
> I[1] = 2.5e-6*Unit(W/m^2), s = 1000*Unit(m), r[1] = 100*Unit(m);
```

$$I_1 = 2.5 \cdot 10^{-6} \left[\frac{\text{kg}}{\text{s}^3} \right], s = 1000 \text{ [m]}, r_1 = 100 \text{ [m]} \quad (15)$$

Bearbeitung

Vereinfachungen:

- Die Überlandleitung wird als gerade linienförmige Schallquelle behandelt.
- Bei der Überlagerung der beiden Summtöne tritt keine Interferenz oder Schwebung auf; die Schallintensitäten werden addiert.

Aus der Formelsammlung [1]: Die Schallintensität I einer linienförmigen Schallquelle im Abstand r zur Quelle bei einer Linienleistung P_l der Quelle.

> I = P[1]/(2*Pi*r) ;

$$I = \frac{P_l}{2 \pi r} \quad (16)$$

(= gleichmäßige Verteilung der Schallleistung auf die Zylinderschale)

Diese Formel ausgeschrieben für die Intensität I_1 im Abstand r_1 und die Intensität I_2 in der Mitte zwischen den beiden Leitungen = Abstand $s/2$.

> subs(I=I[1], r=r[1], (16));

$$I_1 = \frac{P_l}{2 \pi r_1} \quad (17)$$

> subs(I=I[2], r=s/2, (16));

$$I_2 = \frac{P_l}{\pi s} \quad (18)$$

Mit Gleichung (17) die Unbekannte Linienleistung P_l aus Gleichung (18) eliminieren.

> subs(isolate((17),P[1]), (18));

$$I_2 = \frac{2 I_1 r_1}{s} \quad (19)$$

Gesamtschallintensität I_g der beiden Leitungen berechnen durch Addieren der einzelnen Intensitäten (siehe Vereinfachung).

> I[g] = 2*I[2];

$$I_g = 2 I_2 \quad (20)$$

> subs((19), (20));

$$I_g = \frac{4 I_1 r_1}{s} \quad (21)$$

Alle Größen auf der rechten Seite sind bekannt. Einsetzen und ausrechnen.

> subs((15), (21)): lhs(%) = convert(rhs(%), units, W/m^2);

$$I_g = 1.000000000000 \cdot 10^{-6} \left[\frac{\text{W}}{\text{m}^2} \right] \quad (22)$$

Die Gesamtschallintensität in der Mitte zwischen den beiden Leitungen beträgt 10^{-6} W/m^2 .

Die Umrechnung von Schallintensität in Schallintensitätspegel nach Gleichung (6).

> subs(I[0]=R, I=I[g], L=L[g], R=I[0], (6));

$$L_g = 10 \lg \left(\frac{I_g}{I_0} \right) \quad [dB] \quad (23)$$

Einsetzen der berechneten Schallintensität aus (21).

> subs ((21), (23));

$$L_g = 10 \lg \left(\frac{4 I_1 r_1}{s I_0} \right) \quad [dB] \quad (24)$$

Alle Größen auf der rechten Seite sind bekannt. Einsetzen und ausrechnen.

> subs (lg=log10, (15), (7), (24)): evalf(%);

$$L_g = 60.0000000000 \quad [dB] \quad (25)$$

Der Gesamtschallintensitätspegel beträgt 60 dB.

▼ Hilfsmittel

[1] Hering, Martin, Stohrer: Physikalisch-Technisches Taschenbuch, VDI Verlag

[2] Maple 14