```
> restart;
```

Gegeben ist die Lage des Punkts P in Polarkoordinaten.
Die Radius-Koordinate des Punkts P als Funktion der Winkel-Koordinate:
(Ich folge hier der Konvention rad = Radiant = 1 für die Einheit des Winkels. Daher taucht rad nicht in den Formeln auf.)

```
> r = A * theta;
```

$$r = A\,\theta \tag{1}$$

mit der Konstanten

```
> A = 0.1*Unit(m);
```

$$A = 0.1\,[\![m]\!] \tag{2}$$

Die Winkel-Koordinate als Funktion als Funktion der Zeit:

```
> theta = B * t;
```

$$\theta = B\,t \tag{3}$$

mit der Konstanten

```
> B = 2*Unit(1/s);
```

$$B = 2\left[\!\!\left[\frac{1}{s}\right]\!\!\right] \tag{4}$$

(3) in **(1)** eingesetzt liefert die Radius-Koordinate als Funktion der Zeit.

```
> subs((3),(1)): simplify(%);
```

$$r = A\,B\,t \tag{5}$$

Damit ist die Lage des Punkts P in Polarkoordinaten als Funktion der Zeit t aufgeschrieben.
Für Geschwindigkeit und Beschleunigung müssen Ableitungen berechnet werden. Das ist in kartesischen Koordinaten besonders einfach. Also Umwandeln der Koordinaten. Die allgemeine Beziehung ist:

```
> x(t) = r*cos(theta);
```

$$x(t) = r\cos(\theta) \tag{6}$$

```
> y(t) = r*sin(theta);
```

$$y(t) = r\sin(\theta) \tag{7}$$

Die Polarkoordinaten des Punkts P aus **(3)** und **(5)** einsetzen.

```
> subs((3),(5),(6));
```

$$x(t) = A\,B\,t\cos(B\,t) \tag{8}$$

```
> subs((3),(5),(7));
```

$$y(t) = A\,B\,t\sin(B\,t) \tag{9}$$

Damit sind die kartesischen Koordinaten des Punkts als Funktionen der Zeit bestimmt.
Die Ableitungen ergeben die Komponenten der Geschwindigkeit..

```
> v[x] = diff(x(t),t);
```

$$v_x = \frac{\mathrm{d}}{\mathrm{d}t}\,x(t) \tag{10}$$

```
> v[y] = diff(y(t),t);
```

$$v_y = \frac{\mathrm{d}}{\mathrm{d}t}\,y(t) \tag{11}$$

Einsetzen der Koordinatenfunktionen **(8)** und **(9)** für den Punkt P.

```
> subs((8),(10)): simplify(%);
```

$$v_x = -A\,B\,\left(-\cos(B\,t) + B\,t\sin(B\,t)\right) \tag{12}$$

```
> subs((9),(11)): simplify(%);
```

(13)

$$v_y = A\,B\,(\sin(B\,t) + B\,t\cos(B\,t)) \tag{13}$$

Der Betrag der Geschwindigkeit.

```
> v = sqrt( v[x]^2 + v[y]^2 );
```

$$v = \sqrt{v_x^2 + v_y^2} \tag{14}$$

```
> subs((12),(13),(14));
```

$$v = \sqrt{A^2 B^2 (-\cos(B\,t) + B\,t\sin(B\,t))^2 + A^2 B^2 (\sin(B\,t) + B\,t\cos(B\,t))^2} \tag{15}$$

```
> simplify((15)) assuming A>0,B>0;
```

$$v = A\,B\sqrt{1 + B^2 t^2} \tag{16}$$

Die Ableitung der Geschwindigkeit nach der Zeit ergibt die Beschleunigung.

```
> a[x] = diff(v[x](t),t);
```

$$a_x = \frac{\mathrm{d}}{\mathrm{d}t}\, v_x(t) \tag{17}$$

```
> a[y] = diff(v[y](t),t);
```

$$a_y = \frac{\mathrm{d}}{\mathrm{d}t}\, v_y(t) \tag{18}$$

Einsetzen der Koordinatenfunktionen **(12)** und **(13)** für die Geschwindigkeit des Punkts P.

```
> subs(subs(v[x]=v[x](t),(12)),(17)): simplify(%);
```

$$a_x = -A\,B^2 (2\sin(B\,t) + B\,t\cos(B\,t)) \tag{19}$$

```
> subs(subs(v[y]=v[y](t),(13)),(18)): simplify(%);
```

$$a_y = -A\,B^2 (-2\cos(B\,t) + B\,t\sin(B\,t)) \tag{20}$$

Der Betrag der Beschleunigung

```
> a=sqrt( a[x]^2 + a[y]^2 );
```

$$a = \sqrt{a_x^2 + a_y^2} \tag{21}$$

```
> subs((19),(20),(21));
```

$$a = \sqrt{A^2 B^4 (2\sin(B\,t) + B\,t\cos(B\,t))^2 + A^2 B^4 (-2\cos(B\,t) + B\,t\sin(B\,t))^2} \tag{22}$$

```
> simplify((22)) assuming A>0, B>0;
```

$$a = A\,B^2\sqrt{4 + B^2 t^2} \tag{23}$$

Die Zahlenwerte sollen berechnet werden für

```
> t=1.0*Unit(s);
```

$$t = 1.0\ [\![s]\!] \tag{24}$$

Einsetzen der Zahlenwerte ergibt

```
> eval((16),{(2),(4),(24)}): simplify(%);
```

$$v = 0.4472135954 \left[\!\left[\frac{m}{s}\right]\!\right] \tag{25}$$

```
> eval((23),{(2),(4),(24)}): simplify(%);
```

$$a = 1.131370850 \left[\!\left[\frac{m}{s^2}\right]\!\right] \tag{26}$$

Die Geschwindigkeit beträgt 0,45 m/s und die Beschleunigung beträgt 1,1 m/s² zum Zeitpunkt t=1s.

Alternativ kann die gesamte Aufgabe in Polarkoordinaten gerechnet werden.
Die Formeln dazu aus den Buch "Technische Mechanik für Ingenieure" von Müller und Ferber.
Die Komponenten der Geschwindigkeit:

```
> v[r] = diff(r(t),t);
```

$$v_r = \frac{d}{dt} r(t) \tag{27}$$

```
> v[theta] = r(t) * diff(theta(t),t);
```

$$v_\theta = r(t) \left(\frac{d}{dt} \theta(t)\right) \tag{28}$$

Einsetzen der Koordinatenfunktionen **(3)** und **(5)** für den Punkt P:

```
> subs(subs(r=r(t),(5)),(27)): simplify(%);
```

$$v_r = A\,B \tag{29}$$

```
> subs(subs(r=r(t),(5)),subs(theta=theta(t),(3)),(28)): simplify(%);
```

$$v_\theta = A\,B^2\,t \tag{30}$$

Betrag der Geschwindigkeit

```
> v=sqrt(v[r]^2+v[theta]^2);
```

$$v = \sqrt{v_r^2 + v_\theta^2} \tag{31}$$

```
> subs((29),(30),(31));
```

$$v = \sqrt{A^2\,B^2 + A^2\,B^4\,t^2} \tag{32}$$

```
> simplify((32)) assuming A>0,B>0;
```

$$v = A\,B\,\sqrt{1 + B^2\,t^2} \tag{33}$$

Die Komponenten der Beschleunigung:

```
> a[r] = diff(r(t),t,t) - r(t)*diff(theta(t),t)^2;
```

$$a_r = \frac{d^2}{dt^2} r(t) - r(t) \left(\frac{d}{dt} \theta(t)\right)^2 \tag{34}$$

```
> a[theta] = r(t)*diff(theta(t),t,t) + 2*diff(r(t),t)*diff(theta
(t),t);
```

$$a_\theta = r(t) \left(\frac{d^2}{dt^2} \theta(t)\right) + 2 \left(\frac{d}{dt} r(t)\right) \left(\frac{d}{dt} \theta(t)\right) \tag{35}$$

Einsetzen der beiden Koordinaten **(3)** und **(5)** für den Punkt P.

```
> subs(subs(r=r(t),(5)),subs(theta=theta(t),(3)),(34)): simplify(%);
```

$$a_r = -A\,B^3\,t \tag{36}$$

```
> subs(subs(r=r(t),(5)),subs(theta=theta(t),(3)),(35)): simplify(%);
```

$$a_\theta = 2\,A\,B^2 \tag{37}$$

Betrag der Beschleunigung:

```
> a=sqrt(a[r]^2+a[theta]^2);
```

$$a = \sqrt{a_r^2 + a_\theta^2} \tag{38}$$

```
> subs((36),(37),(38));
```

$$a = \sqrt{A^2\,B^6\,t^2 + 4\,A^2\,B^4} \tag{39}$$

```
> simplify((39)) assuming A>0,B>0;
```

$$a = A\,B^2\,\sqrt{4 + B^2\,t^2} \qquad \textbf{(40)}$$

Die Ergebnisse für den Betrag von Geschwindigkeit und Betrag der Beschleunigung sind in beiden Rechenwegen gleich.
Die Berechung der Zahlenwerte **(25)** und **(26)** ist in den Poloarkoordinaten gleich.