Aufgabe

Ein offener Wasserbehälter aus Stahl mit einer Masse von 25kg ist mit 50 liter Wasser von 15°C gefüllt (Stahlbehälter hat die gleiche Temperatur.) Im Behälter befindet sich eine elektrische Heizung mit einer Leistung von 5 kW.
Barometerstand pb = 0,1 MPa.

a) nach welcher Zeit sind 20 Kg Wasser verdampft, wenn vom Behälter an die Umgebungsluft ein Wärmestrom von 1000 kJ/h abgegeben wird?

b) Der Wassergehalt wird so isoliert, dass keine Wärmeverluste auftreten. Nach welcher Zeit ist das gesamte Wasser verdampft (Isolierung soll sich nicht mit aufheizen)

```
> restart;
> Digits:=30: interface( displayprecision=5 ):
> dT := `&Delta;T`:
> parameters := []:
> addParameter := proc(arg)
      global parameters;
      parameters := [ op(parameters), arg ];
  end proc:
```

Rechnung Teil a.

Stahlmasse, die erwärmt werden muss.

```
> m[S] = 25*Unit(kg); addParameter(%):
```

$$m_S = 25 \ [[kg]] \tag{1}$$

Wasservolumen, das erwärmt werden muss.

```
> V[W] = 50*Unit(liter); addParameter(%):
```

$$V_W = 50 \ [[L]] \tag{2}$$

Spezifische Wärmekapazität von Stahl aus "Taschenbuch der Physik" von Stöcker.

```
> c[S] = 0.51*Unit(kJ/(kg*K)): lhs(%)=convert(rhs(%),units,kJ/(kg*
  K) ); simplify(%): addParameter(%):
```

$$c_S = 0.51000 \left[\left[\frac{kJ}{kg\,K}\right]\right] \tag{3}$$

Spezifische Wärmekapazität von Wasser.

```
> c[W] = 4.187*Unit(kJ/(kg*K) ): lhs(%)=convert(rhs(%),units,kJ/
   (kg*K) ); simplify(%): addParameter(%):
```

$$c_W = 4.1870 \left[\left[\frac{kJ}{kg\,K}\right]\right] \tag{4}$$

Dichte von Wasser. (Bei 20°C, die Temperaturabhängigkeit aller Stoffeigenschaften wird hier vernachlässigt.)

```
> rho[W] = 0.998*Unit(kg/liter): lhs(%)=convert(rhs(%),units,
  kg/liter ); simplify(%): addParameter(%):
```

$$\rho_W = 0.99800 \left[\left[\frac{kg}{L}\right]\right] \tag{5}$$

Bis zum Siedepunkt 100°C muss Stahbehälter und Wasser um ΔT erwärmt werden.

```
> dT = (100-15)*Unit(K); addParameter(%):
```

$$\Delta T = 85 \ [[K]] \tag{6}$$

Wärmestrom an die Umgebung.

```
> P[U] = 1000.0*Unit(kJ/h): lhs(%)=convert(rhs(%),units,kJ/h );
  simplify(%); addParameter(%):
```

$$P_U = 1000.0 \left[\!\left[\frac{kJ}{h} \right]\!\right]$$

$$P_U = 277.78 \; [\![W]\!] \tag{7}$$

Wärmestrom aus der Heizung.

```
> P[H] = 5*Unit(kW); addParameter(simplify(%)):
```

$$P_H = 5 \; [\![kW]\!] \tag{8}$$

Die Temperaturänderung wird in der Zeit t_1 erfolgen. Energiebilanz für diesen Zeitraum:

```
> (m[S]*c[S] + V[W]*rho[W]*c[W])*dT + P[U]*t[1] = P[H]*t[1];
```

$$\left(m_S c_S + V_W \rho_W c_W\right) \Delta T + P_U t_1 = P_H t_1 \tag{9}$$

Auflösen nach der Zeit t_1.

```
> isolate((9),t[1]): simplify(%): sort(%);
```

$$t_1 = \frac{\left(V_W c_W \rho_W + c_S m_S\right) \Delta T}{P_H - P_U} \tag{10}$$

Die Zahlenwerte ergeben.

```
> subs(parameters,(10)): simplify(%); lhs(%)=convert(rhs(%),units,
  min);
```

$$t_1 = 3990.3 \; [\![s]\!]$$

$$t_1 = 66.504 \; [\![\mathrm{min}]\!] \tag{11}$$

Verdampfen soll die Wassermasse

```
> m[V] = 20*Unit(kg);
```

$$m_V = 20 \; [\![kg]\!] \tag{12}$$

Spezifische Verdampfungsenthalpie von Wasser.

```
> h[V] = 2265*Unit(kJ/kg): lhs(%)=convert(rhs(%),units,kJ/kg );
  simplify(%): addParameter(%):
```

$$h_V = 2265 \left[\!\left[\frac{kJ}{kg} \right]\!\right] \tag{13}$$

Die Verdampfung wird in der Zeit t_2 erfolgen. Die Temperatur des Wassers und des Behälters bleibt dabei konstant. Die Energiebilanz für diesen Zeitraum:

```
> m[V]*h[V] + P[U]*t[2] = P[H]*t[2];
```

$$m_V h_V + P_U t_2 = P_H t_2 \tag{14}$$

Auflösen nach der Zeit t_2.

```
> isolate((14),t[2]): simplify(%): sort(%);
```

$$t_2 = \frac{h_V m_V}{P_H - P_U} \tag{15}$$

Die Zahlenwerte ergeben.

```
> subs(parameters,(12),(15)): simplify(%); lhs(%)=convert(rhs(%),
  units,min);
```

$$t_2 = 9592.9 \; [\![s]\!]$$

$$t_2 = 159.88 \; [\![\mathrm{min}]\!] \tag{16}$$

Die Zeit für den gesamten Vorgang.

```
> t = t[1]+t[2];
```

$$t = t_1 + t_2 \tag{17}$$

```
> subs((11),(16),(17));
```

$$t = 226.39 \; [\![\mathrm{min}]\!] \tag{18}$$

Rechnung Teil b.

Fast alle Überlegungen aus Teil a können übernommen werden.

Die Temperaturänderung wird in der Zeit t_1 erfolgen. Energiebilanz für diesen Zeitraum, es fehlt der Wärmestrom an die Umgebung:

```
> (m[S]*c[S] + V[W]*rho[W]*c[W])*dT = P[H]*t[1];
```

$$\left(V_W c_W \rho_W + c_S m_S\right) \Delta T = P_H t_1 \tag{19}$$

Auflösen nach der Zeit t_1.

```
> isolate((19),t[1]): simplify(%): sort(%);
```

$$t_1 = \frac{\left(V_W c_W \rho_W + c_S m_S\right) \Delta T}{P_H} \tag{20}$$

Die Zahlenwerte ergeben.

```
> subs(parameters,(20)): simplify(%); lhs(%)=convert(rhs(%),units,
  min);
```

$$t_1 = 3768.6 \; [\![s]\!]$$

$$t_1 = 62.810 \; [\![\mathrm{min}]\!] \tag{21}$$

Verdampfen soll die gesamte Wassermasse.

```
> m[V] = V[W]*rho[W];
```

$$m_V = V_W \rho_W \tag{22}$$

Die Verdampfung wird in der Zeit t_2 erfolgen. Die Temperatur des Wassers und des Behälters bleibt dabei konstant. Die Energiebilanz für diesen Zeitraum, auch hier fehlt der Wärmestrom an die Umgebung:

```
> m[V]*h[V] = P[H]*t[2]; subs((22),%);
```

$$m_V h_V = P_H t_2$$

$$V_W \rho_W h_V = P_H t_2 \tag{23}$$

Auflösen nach der Zeit t_2.

```
> isolate((23),t[2]): simplify(%): sort(%);
```

$$t_2 = \frac{V_W h_V \rho_W}{P_H} \tag{24}$$

Die Zahlenwerte ergeben.

```
> subs(parameters,(24)): simplify(%); lhs(%)=convert(rhs(%),units,
  min);
```

$$t_2 = 22605. \; [\![s]\!]$$

$$t_2 = 376.74 \; [\![\mathrm{min}]\!] \tag{25}$$

Die Zeit für den gesamten Vorgang.

```
> t = t[1]+t[2];
```

$$t = t_1 + t_2 \tag{26}$$

```
> subs((21),(25),(26));
```

$$t = 439.55 \, [\![\mathrm{min}]\!] \tag{27}$$