```
> restart;
> Digits := trunc( evalhf(Digits) ):
> interface( displayprecision = 5 ):
> dT := ΔT:
  dotQ := diff(Q(t),t):
> param := []:
  aParam := proc ( s )
      global param;
      param := [op(param),simplify(s)];
  end:
```

Aufgabe

In einem geschlossenen Gebäude befindet sich ein Stromerzeuger. Der Stromerzeuger liefert 65 kW Abwärme, die durch Ventilation ausgetragen werden. Um wie viel Kelvin steigt die Raumtemperatur, wenn der Ventilator für 10 / 15 / 20 Minuten abgestellt wird.

Außentemperatur beträgt 28°C.
Raumluftvolumen beträgt 500 m³.

Vorgegebenen Zeitpunkte

```
> [ 10, 15, 20 ] *~ Unit('min'); simplify(%);
```

$$[10\,[[\mathrm{min}]], 15\,[[\mathrm{min}]], 20\,[[\mathrm{min}]]]$$

$$[600\,[[s]], 900\,[[s]], 1200\,[[s]]] \quad (1)$$

Wärmestrom aus dem Generator

```
> P = 65*Unit('kW'); aParam(%):
```

$$P = 65\,[[kW]] \quad (2)$$

Außentemperatur

```
> T[amb] = convert( 28.0*Unit('Celsius'), temperature, 'K' );
  aParam(%):
```

$$T_{amb} = 301.15\,[[K]] \quad (3)$$

Luftvolumen

```
> V[L] = 500*Unit(m^3); aParam(%):
```

$$V_L = 500\,[[m^3]] \quad (4)$$

1. Abschätzung

Annahmen:
- Das Gebäude enthält nur Luft.
- Das Gebäude ist geschlossen. (Kein Luftaustausch mit der Umgebung.)
- Der Luftdruck im Gebäude ist konstant. (isobar; die Wände geben der Wämeausdehnung nach.)
- Die Gebäudewände isolieren perfekt. (adiabatisch; keine Wärmestrom in die Umgebung.)

spezifische isobare Wärmekapazität von trockener Luft [1]

```
> c[L] = convert( 1.005 * Unit(J/(g*K)),units,kJ/(kg*K) ); aParam
  (%):
```

$$c_L = 1.0050\,\left[\left[\frac{kJ}{kg\,K}\right]\right] \quad (5)$$

Dichte von trockener Luft (bei 273 K, 1013 hPa) [1]

```
> rho[L] = 1.2928*Unit(kg/m^3); aParam(%):
```

(6)

$$\rho_L = 1.2928 \left[\!\left[\frac{kg}{m^3} \right]\!\right] \quad \textbf{(6)}$$

Wärmekapazität der Luft

```
> C[L] = c[L] * rho[L] * V[L];
```

$$C_L = c_L \, \rho_L \, V_L \quad \textbf{(7)}$$

In die Luft abgegebene Wärme nach der Zeit t;

```
> Q = P*t;
```

$$Q = P\, t \quad \textbf{(8)}$$

Temperaturänderung ΔT der Luft durch Aufnahme der Wärme.

```
> Q = C[L] * dT;
```

$$Q = C_L \, \Delta T \quad \textbf{(9)}$$

```
> isolate((9),dT);
```

$$\Delta T = \frac{Q}{C_L} \quad \textbf{(10)}$$

Die Formeln **(7)** und **(8)** einsetzen.

```
> subs((7),(8),(10));
```

$$\Delta T = \frac{P\, t}{c_L \, \rho_L \, V_L} \quad \textbf{(11)}$$

Die Temperaturänderung bei den gegebenen Zeitpunkten **(1)** ausrechnen.

```
> print( 'Zeit', 'Temperaturanstieg' );
  for temp in (1) do
    temp2 := t = temp;
    print( temp2, simplify( subs( param, temp2, (11) ) ) );
  end:
```

$$\mathit{Zeit}, \mathit{Temperaturanstieg}$$

$$t = 600 \, [\![s]\!], \Delta T = 60.034 \, [\![K]\!]$$

$$t = 900 \, [\![s]\!], \Delta T = 90.051 \, [\![K]\!]$$

$$t = 1200 \, [\![s]\!], \Delta T = 120.07 \, [\![K]\!] \quad \textbf{(12)}$$

2. Abschätzung

Annahmen aus 1. Abschätzung mit Änderungen:

- Das Gebäude enthält Luft und einen Generator.
- Luft und Generator haben die gleiche Temperatur. (Großer Wärmedurchgangskoeffizent zwischen Generator und Luft führt zu geringer Temperaturdifferenz.)

Abschätzen der Wärmekapazität des Generators:
Die Verlustleitung ist mit 65 kW gegeben. Bei einem Wirkungsgrad von 90% ist die Verlustleistung 10% der Eingangsleistung und die elektrische Leistung 90% der Eingangsleistung. Die elektrische Leistung des Generators sind rund 600 kW.
Ein Generator mit 600 kW hat eine Masse von rund 3 Tonnen. [2]
Ist in der Aufgabe mit "Stromerzeuger" ein Dieselgenerator gemeint, dann wird die elektrische Leistung kleiner sein, weil der Gesamtwirkungsgrad kleiner ist. Die Masse der gesamten Maschine bleibt bei rund 3 Tonnen. ([3], Ein 600 kW Dieselgenerator mit 5t.)
Der Generator besteht zu großen Teilen aus Eisen und Kupfer.

Für die Rechnung: 3 Tonnen Eisen.

```
> m[Fe] = 3*Unit('t'); simplify(%); aParam(%):
```

$$m_{Fe} = 3 \; [\![t]\!]$$

$$m_{Fe} = 3000 \; [\![kg]\!] \tag{13}$$

spezifische Wärmekapazität von Gusseisen [1]

```
> c[Fe] = convert( 0.532 * Unit(J/(g*K)),units,kJ/(kg*K) );
  aParam(%):
```

$$c_{Fe} = 0.53200 \left[\!\left[\frac{kJ}{kg\,K} \right]\!\right] \tag{14}$$

Die Rechenschritte entsprechen den Schritten aus der 1. Abschätzung.

Die Wärmekapazität von Luft und Generator

```
> C = C[L] + m[Fe]*c[Fe];
```

$$C = C_L + m_{Fe}\, c_{Fe} \tag{15}$$

```
> subs((7),(15));
```

$$C = c_L\, \rho_L\, V_L + m_{Fe}\, c_{Fe} \tag{16}$$

Temperaturänderung ΔT durch Aufnahme der Wärme.

```
> Q = C * dT;
```

$$Q = C\, \Delta T \tag{17}$$

```
> isolate((17),dT);
```

$$\Delta T = \frac{Q}{C} \tag{18}$$

Die Formeln **(16)** und **(8)** einsetzen.

```
> subs((16),(8),(18));
```

$$\Delta T = \frac{P\, t}{c_L\, \rho_L\, V_L + m_{Fe}\, c_{Fe}} \tag{19}$$

Die Temperaturänderung bei den gegebenen Zeitpunkten **(1)** ausrechnen.

```
> print( 'Zeit', 'Temperaturanstieg' );
  for temp in (1) do
    temp2 := t = temp;
    print( temp2, simplify( subs( param, temp2, (19) ) ) );
  end:
```

$$\mathit{Zeit},\ \mathit{Temperaturanstieg}$$

$$t = 600 \; [\![s]\!],\ \Delta T = 17.367 \; [\![K]\!]$$

$$t = 900 \; [\![s]\!],\ \Delta T = 26.051 \; [\![K]\!]$$

$$t = 1200 \; [\![s]\!],\ \Delta T = 34.734 \; [\![K]\!] \tag{20}$$

3. Abschätzung

Die Annahmen aus der 2. Abschätzung mit Änderungen:

- Die Wände des Gebäude übertragen Wärme.
- Die Wände haben keine Wärmekapazität.
- Zum Ausschaltzeitpunkt war die Temperatur im Gebäude gleich der Umgebungstemperatur.

Wärmedurchgangskoeffizient der Wände abschätzen:

Der Generator im Gebäude erzeugt viel Verlustleistung. Eine Wärmedämmung der Wände wären sinnlose Kosten. Daher die Annahme: Es ist eine Betonwand ohne Wärmedämmung. [1,3] gegeben einen Wärmedurchgangskoeffizient von rund 3,3 W/(m² K) an.

Fläche der Wände abschätzen:
Das Luftvolumen ist mit 500 m³ gegeben. Diese Volumen ist eine Abschätzung für das Volumen des Gebäudes verwenden. Die Höhe der Halle auf 5m schätzen. Eine Wandlänge mit 5m annehmen. Bleibt 20m für die andere Wandlänge.

Fläche aller Wände

```
> A = 2*( 5*5 + 20*5 )*Unit(m^2); aParam(%):
```

$$A = 250 \, [\![m^2]\!] \tag{21}$$

Wärmedurchgangskoeffizient der Wände

```
> U = convert( 3.3*Unit(W/(m^2*K)), units, W/(m^2*K) ); aParam(%)
  :
```

$$U = 3.3 \left[\!\left[\frac{W}{m^2 K}\right]\!\right] \tag{22}$$

Die Raumtemperatur (Temperatur von Luft und vom Generator) als Funktion der Zeit ansehen T(t). Der Wärmestrom durch die Wand nach außen, auch eine Funktion der Zeit.

```
> dotQ = A*U*(T(t)-T[amb]);
```

$$\dot{Q}(t) = A\,U\left(T(t) - T_{amb}\right) \tag{23}$$

Die Verlustleistung P erwärmt zum Teil das Innere mit Wärmekapazität C und strömt zum Teil durch die Wände nach außen.

```
> P = C*diff(T(t),t) + dotQ;
```

$$P = C\,\dot{T}(t) + \dot{Q}(t) \tag{24}$$

```
> subs( (23), (24) );
```

$$P = C\,\dot{T}(t) + A\,U\left(T(t) - T_{amb}\right) \tag{25}$$

Das ist eine Differenzialgleichung für die Temperatur T.
Zuerst den Grenzwert bestimmen, die maximale Temperatur bei der die Erwärmung zum Stillstand kommt.

```
> diff(T(t),t)=0;
```

$$\dot{T}(t) = 0 \tag{26}$$

```
> subs((26),T(t)=T[max],(25));
```

$$P = A\,U\left(T_{\max} - T_{amb}\right) \tag{27}$$

```
> isolate((27),T[max]);
```

$$T_{\max} = \frac{P}{A\,U} + T_{amb} \tag{28}$$

Die Zahlenwerte einsetzen.

```
> simplify( subs( param, (28) ) );
```

$$T_{\max} = 379.94 \, [\![K]\!] \tag{29}$$

```
> lhs((29))=convert( rhs((29)),'temperature','Celsius' );
```

$$T_{\max} = 106.79 \, [\![degC]\!] \tag{30}$$

Zurück zur Differenzialgleichung **(25)**.

```
> isolate( (25), C*diff(T(t),t) ); expand(%/(A*U));
```

$$C\,\dot{T}(t) = P - A\,U\left(T(t) - T_{amb}\right)$$

(31)

$$\frac{C\,\dot{T}(t)}{A\,U} = \frac{P}{A\,U} - T(t) + T_{amb} \quad \textbf{(31)}$$

Auf der rechten Seite können Terme zur Grenztemperatur zusammengefasst werden.

```
> algsubs(rhs((28))=lhs((28)),(31));
```

$$\frac{C\,\dot{T}(t)}{A\,U} = T_{\max} - T(t) \quad \textbf{(32)}$$

Die Zeitkonstante ist ablesbar.

```
> tau = C/(A*U);
```

$$\tau = \frac{C}{A\,U} \quad \textbf{(33)}$$

Die Wärmekapazität aus **(16)** einsetzen.

```
> subs( (16), (33) );
```

$$\tau = \frac{c_L\,\rho_L\,V_L + m_{Fe}\,c_{Fe}}{A\,U} \quad \textbf{(34)}$$

Die Zahlenwerte einsetzen.

```
> simplify( subs(param,(34) ) );
```

$$\tau = 2722.0\ [\![s]\!] \quad \textbf{(35)}$$

```
> algsubs( rhs((33))=lhs((33)), (32) ): expand( % );
```

Die Differenzialgleichung mit der Zeitkonstante aufschreiben.

$$\dot{T}(t)\,\tau = T_{\max} - T(t) \quad \textbf{(36)}$$

Die Lösung dieser Differenzialgleichung ist bekannt.
Der Abstand zwischen der Starttemperatur T_{amb} und der Endtemperatur T_{max} nimmt mit $e^{-t/\tau}$ ab.

```
> subs( Tmax=T[max], dsolve( {subs(T[max]=Tmax,(36)),T(0)=T[amb]}
  ) );
```

$$T(t) = T_{\max} + e^{-\frac{t}{\tau}}\left(T_{amb} - T_{\max}\right) \quad \textbf{(37)}$$

Die Temperatur T(t) und Temperaturänderung ΔT bei den gegebenen Zeitpunkten **(1)** ausrechnen.

```
> temp3 := subs((29),(35),(37)):
  for temp in (1) do
    temp2 := t = temp;
    temp2 := simplify( subs( param, temp2, temp3 ) );
    print( temp2 );
    print( subs(param,dT = rhs(temp2)-T[amb]) );
  end:
```

$$T(600\ [\![s]\!]) = 316.74\ [\![K]\!]$$
$$\Delta T = 15.586\ [\![K]\!]$$
$$T(900\ [\![s]\!]) = 323.33\ [\![K]\!]$$
$$\Delta T = 22.182\ [\![K]\!]$$
$$T(1200\ [\![s]\!]) = 329.24\ [\![K]\!]$$
$$\Delta T = 28.089\ [\![K]\!] \quad \textbf{(38)}$$

Hilfsmittel

[1] Stöcker: Taschenbuch der Physik, Verlag Harri Deutsch

[2] Ergebnis einer Websuche, http://www.weier-energie.de/index.php?id=92&L=1
[3] Ergebnis einer Websuche, http://www.china-dieselgenerator.com/blog/?p=662
[4] Wikipedia Wärmeübergangskoeffizient, http://de.wikipedia.org/wiki/Wärmeübergangskoeffizient
[ ] Maple, www.maplesoft.com